

# The TUTOR Language

Bruce Arne Sherwood

To my Mother, who saw me through!

Love,

Bruce

FUND of PLATO PROY  
DAVIS

# The TUTOR Language

# The TUTOR Language

Bruce Arne Sherwood  
*Computer-based Education Research Laboratory  
and Department of Physics  
University of Illinois  
Urbana, Illinois*

CONTROL DATA  
EDUCATION COMPANY

 a service of  
CONTROL DATA CORPORATION



© 1977 by Bruce Arne Sherwood.

All rights reserved. No part of this material may be reproduced by any means without permission in writing from the publisher and the author.

ISBN: 0-918852-00-5

Library of Congress Catalog Card Number: 77-77589

Printed in the United States of America

## Preface

The PLATO IV computer-based education system was developed in the Computer-based Education Research Laboratory (CERL) of the University of Illinois, Urbana. PLATO IV is the result of 15 years of research and development effort led by Donald Bitzer, director of CERL. The University of Illinois system presently links 950 graphical-display terminals to a large Control Data Corporation computer in Urbana. Some of these terminals are located as far away as San Diego and Washington, D.C. Additional PLATO systems with their own complements of terminals are located elsewhere in the United States. Students are individually tutored at terminals by interacting with PLATO lesson materials created by teachers. There are over 4,000 hours of PLATO lessons already available. These lessons span a wide range of subject areas and are used by students in elementary schools, community colleges, military training bases, universities, and commercial training programs. Authors of lesson materials are teachers who use the TUTOR language to tell PLATO how to interact with students on an individual basis. This book explains the TUTOR language in detail and is intended to help authors write quality lesson materials.

In 1967, Paul Tenczar (then a graduate student in zoology) concluded that existing methods of creating computer-based lesson material on the earlier PLATO III system were unnecessarily difficult. As a result he originated the TUTOR language. There followed a rapid increase in the number of authors and in the number and degree of sophistication of

## Preface

the lessons they wrote. This active author community in turn spurred the continual development and refinement of TUTOR by requesting additional needed features. In 1970, CERL began implementing the PLATO IV system, which afforded a rare opportunity to take stock of the evolution of TUTOR up to that point and make a fresh start. Many useful simplifications were made, and many important features were added. The growth of PLATO IV into a continental network brought together an ever-wider spectrum of authors through the rich interpersonal communications facilities available on PLATO, and the suggestions and criticisms from these authors contributed to the present form of the TUTOR language. Also of great importance has been the large number of students who have used PLATO lessons, and whose experiences have influenced the development of TUTOR to meet their needs. The TUTOR language described in this book is, therefore, based on heavy use-testing.

In the earliest phase Paul Tenczar and Richard Blomme were mainly responsible for TUTOR development. Since then, many people have been involved, some as full-time CERL staff members and some as high school, undergraduate, or graduate students. It is impossible to adequately acknowledge the various contributions, and difficult even to list all of those who have played a major role, but an attempt should be made. Paul Tenczar is head of TUTOR development. Full-time people have included David Andersen, Richard Blomme, John Carstedt (Control Data), Ruth Chabay, Christopher Fugitt, Don Lee, Robert Rader, Donald Shirer, Michael Walker, and this author. They have been assisted by James Parry and Masako Secrest, and by Doug Brown, David Frankel, Sherwin Gooch, David Kopf, Kim Mast, Phil Mast, Marshall Midden, Louis Steinberg, Larry White, and David Woolley. William Golden has also provided useful advice.

All of these people have been involved mainly with “software”, the programming of the PLATO computer in such a way as to permit authors and students to write and use computer-based lessons. Of equal importance to the technical success of PLATO are the CERL scientists, engineers, and technicians who invented, designed, and implemented the unique terminals and telecommunications devices (“hardware”) which form the PLATO educational network. CERL personnel who have been heavily involved in hardware development include Donald Bitzer, Jack Stifle, Fred Ebeling, Michael Johnson, Roger Johnson, Frank Propst, Dominic Skaperdas, Gene Slottow, and Paul Tucker.

The latter part of Chapter 1 is adapted from a PLATO III document, “The TUTOR Manual”, by R. A. Avner and P. Tenczar.

I thank Elaine Avner and Jeanne Weiner for editorial assistance, Sheila Knisley for typing, and Stanley Smith for photographic work. I appreciate the encouragement William Golden gave me to finish the task.

# Contents

1	Introduction	<b>1</b>
	How to Use This Book	1
	Sample PLATO Lessons	3
	The PLATO Keyboard	8
	Basic Aspects of TUTOR	13
2	More on Creating Displays	<b>23</b>
	Coarse Grid and Fine Grid	23
	The <code>-box-</code> , <code>-vector-</code> , and <code>-circle-</code> Commands	25
	Large-size Writing: <code>-size-</code> and <code>-rotate-</code>	26
	Animations (Moving Displays): <code>-erase-</code> and <code>-pause-</code>	28
	<code>-pause-</code> , <code>-time-</code> , and <code>-catchup-</code>	30
	The <code>-mode-</code> Command	33
	Automated Display Generation	35
3	Building Your Own Tools: The <code>-do-</code> Command	<b>39</b>
4	Doing Calculations in TUTOR	<b>43</b>
	Giving Names to Variables: <code>-define-</code>	47
	Repeated Operations: The Iterative <code>-do-</code>	49
	Showing the Value of a Variable	51
	Passing Arguments to Subroutines	53

**Contents**

5	Sequencing of Units Within a Lesson	<b>59</b>
	Summary of Sequencing Commands	69
	The -helpop- Command: “Help on Page”	72
	The -imain- Command	73
6	Conditional Commands	<b>77</b>
	Logical Expressions	80
	The Conditional -write- Command (-writec-)	82
	The Conditional -calc- Commands: -calcc- and -calcs-	84
	The Conditional -mode- Command	85
	The -goto- Command	85
	The Conditional Iterative -do-	90
	The -if- and -else- Commands	91
7	Judging Student Responses	<b>95</b>
	Student Specification of Numerical Parameters	101
	Student Specification of Non-Numerical Parameters	104
	Difference Between Numeric and Alphabetic Information	105
	More On -answer- and -wrong- (Including -list- and -specs-)	106
	Building Dialogs With -concept- and -vocabs-	111
	Numbering Vocabulary Words	117
	The -judge- Command	118
	Finding Key Words: The -match- and -storen- Commands	123
	Numerical and Algebraic Judging: -ansv- and -wrongv-	126
	Handling Scientific Units: -ansu-, -wrongu-, and -storeu-	133
	The -exact- and -exactc- Commands	136
	The -answerc- Command: A Language Drill	137
	Summary	139
8	More About Judging	<b>141</b>
	Stages in Processing the -arrow- Command	141
	Repeated Execution of -join-	142

Judging Commands Terminate Regular State	144
The <code>-goto-</code> is a Regular Command	146
Interactions of <code>-arrow-</code> with <code>-size-</code> , <code>-rotate-</code> , <code>-long-</code> , <code>-jkey-</code> , and <code>-copy-</code>	149
Applications of <code>-jkey-</code> and <code>-ans-</code>	151
Modifying the Response: <code>-bump-</code> and <code>-put-</code>	156
Manipulating Character Strings	159
Catching Every Key: <code>-pause-</code> , <code>-keytype-</code> , and <code>-group-</code>	164
Touching the Screen	168
Summary	169
<b>9 Additional Display Features</b>	<b>171</b>
More on the <code>-write-</code> Command	171
Extensions to the Basic Character Set	175
The “initial entry unit” ( <code>ieu</code> )	177
Smooth Animations Using Special Characters	178
Creating a New Character Set	179
Micro Tables	181
The Graphing Commands: Plotting Graphs with Scaling and Labeling	182
Summary of Line-drawing Commands: <code>-draw-</code> , <code>-gdraw-</code> , <code>-rdraw-</code>	185
The <code>-window-</code> Command	190
More on Erasing: The <code>-eraseu-</code> Command	192
Keeping Things on the Screen: “inhibit erase”	196
Interaction of “inhibit erase” with <code>-restart-</code>	199
The <code>-char-</code> and <code>-plot-</code> Commands	199
The <code>-dot-</code> Command	200
<b>10 Additional Calculation Topics</b>	<b>201</b>
Defining Your Own Functions	202
Arrays	204
Segmented Variables	207
Branching Within a Unit: <code>-branch-</code> and <code>-doto-</code>	212
Array Operations	214
<i>Integer Variables and Bit Manipulation</i>	217
Byte Manipulation	229

## Contents

	Vertical Segments	230
	Alphanumeric to Numeric: The -compute- Command	231
	The -find- Command	235
	The -exit- Command	236
11	Manipulating Data Bases	<b>237</b>
	The -common- Command	237
	The Swapping Process	240
	Common Variables and the Swapping Process	243
	The -storage- Command	246
	Using -datasets-	248
	Sorting Lists	248
12	Miscellany	<b>249</b>
	Other Terminal Capabilities	249
	Student Response Data	251
	Additional Tools for Teaching Foreign Languages	252
	Routers and -jumpout-	254
	Instructor Mode	255
	Special “terms”	255
	<b>APPENDICES</b>	<b>257</b>
	Appendix A. Where to Get Further Information	258
	Appendix B. List of TUTOR Commands Additional TUTOR Commands Not Discussed in This Book	259
	Appendix C. List of Built-in -Calc- Functions	261
	System Variables	262





# Introduction

1

## How to Use This Book

This book describes in detail the TUTOR language, which is used by teachers to create lesson materials on the PLATO IV computer-based education system. Teachers use the TUTOR language to express to the PLATO computer how PLATO should interact with individual students. Students and teachers interact with PLATO through terminals each of which includes a plasma display panel screen and a typewriter keyboard, as shown. Using TUTOR, an author of a computer-based lesson can tell PLATO how to display text, line drawings, and animations on the student's screen. The author can ask PLATO to calculate for the student, to offer the student various sequencing options, and to analyze student responses.

The TUTOR language was originally created and developed for educational purposes. However, educational interactions are probably the most subtle and difficult of all the interactions a person might have with the author of materials presented through a computer. It is now clear that other kinds of interactions are also handled well by means of TUTOR, including recreation and communication. Nevertheless, for concreteness this book will concentrate on the instructional applications of TUTOR.

## The TUTOR Language

It is hoped that you have already studied the textbook "Introduction to TUTOR" by J. Chesquiere, C. Davis, and C. Thompson, and the associated PLATO lessons. These materials are designed to teach you not only basic aspects of TUTOR but also how to create and test your own lessons on the PLATO system. The present book, "The TUTOR Language," does not attempt to describe the latter aspects, such as how to insert or delete parts of your lesson and how to try out your new lesson. It does cover all aspects of the TUTOR *language*: that is, *what* statements to give PLATO but not *how* to type these statements into a permanent PLATO lesson space. By studying this book you could, in principle, write down on paper a lesson expressed in the TUTOR language, but when you go to a PLATO terminal to type in your new lesson, you may not know what buttons to push to get started. Also, this book discusses TUTOR in more detail than does "Introduction to TUTOR," which makes "The TUTOR Language" less appropriate for your initial study.

It is also hoped that as you study this book you will try things out at a PLATO terminal. TUTOR is designed for interactive use, in which case an author writes a short segment of a lesson, tries it, and revises it on the basis of the trial. Normally, the sequence write, try, revise, and try again takes only a few minutes at a PLATO terminal. It is far better to create a lesson this way than to write out a complete lesson on paper, only to find upon testing that the overall structure is inappropriate.

It is also helpful to try the sample lesson fragments discussed in this book. It is literally impossible to describe fully in this book how the examples would appear on a PLATO terminal. The PLATO medium is far richer than the book medium. One striking example is the PLATO facility for making animations such as a car driving across the screen. As another example, you must experience it directly to appreciate how easy it is at a PLATO terminal to draw a picture on the screen (by moving a cursor and marking points), then let PLATO *automatically* create the corresponding TUTOR language statements which would produce that picture. PLATO actually writes a lesson segment for you!

This book is written in an informal style. Sometimes, when the context is appropriate, topics are introduced in a different chapter than would be required by strict adherence to a formal classification scheme. In these cases, the feature is at least mentioned in the other chapter, and the index at the end of the book provides an extensive cross-linkage. The order of presentation, emphasis, examples, and counter-examples are all based on extensive experience with the kinds of questions working authors tend to ask about TUTOR.

If you are a fairly new TUTOR author, read this book lightly to get acquainted with the many features TUTOR offers. Plan to return to the book from time to time as your own authoring activities lead you to seek detailed information and suggestions. Your initial light reading should help orient you to finding appropriate sections for later intensive study. After you feel you know TUTOR inside and out, read this book carefully one last time, looking particularly for links among diverse aspects of the language. This last reading will mean much more to you than the first!

If you are already an experienced TUTOR author, read this book carefully with two goals in mind: to spot features unused in your past work but of potential benefit, and to acquire a more detailed understanding of the structural aspects of the language, with particular emphasis on judging.

The remainder of this introductory chapter contains some interesting examples of existing PLATO lessons, a description of the PLATO keyboard including the use of the special function keys, and a review of the most basic aspects of TUTOR.

## Sample PLATO Lessons

Figures 1-1 through 1-6 on the following pages give several examples of interesting PLATO lessons. All were written in the TUTOR language. They have been chosen to give you some idea of the broad range of TUTOR language capabilities. Each example is illustrated with a photograph of the student's screen at a significant or representative point in the lesson. (See the note at the bottom of pg. 7.)

The PLATO terminal's display screen consists of a plasma display panel which contains 512 horizontal electrodes and 512 vertical electrodes mounted on two flat plates of glass between which is neon gas. Any or all of the quarter-million (512×512) intersections of the horizontal and vertical electrodes can be made to glow as a small orange dot. (The word "plasma" is the scientific name for an ionized gas; the orange glow is emitted by ionized neon gas.) As can be seen in the sample photographs, the PLATO terminal can draw lines and circles on the plasma panel as well as display text using various alphabets. Both drawings and text are actually made up of many dots. TUTOR has many display features for writing or erasing text and drawings on the plasma panel.

## The TUTOR Language

Type your question about the unknown and then press NEXT.

When you have identified the compound press BACK.

» does it dissolve in H<sub>2</sub>O

It is slightly soluble in water.

SCORE = 1

For tables of data press DATA. To review press LAB.

For help press HELP.

Fig. 1-1. Dialog in which a chemistry student attempts to identify an unknown compound by asking experimental questions. (Stanley Smith)




Fig. 1-2. Game of mathematical strategy in which two grade-school children compete in constructing advantageous mathematical expressions from random numbers appearing on the spinners. (Bonnie Anderson)




Fig. 1-3. Tutorial on vectors in which the student walks a boy and girl around the screen and measures their vector displacements. (Bruce Sherwood)




Fig. 1-4. Russian sentence drill. The markings under the student's translation of the second sentence indicate incorrect words and misspellings. (Constance Curtin)

## The TUTOR Language

Add parts to the cell below to synthesize the protein chain...  
Leucine--Aspartic acid--Glutamic acid

<u>GACTTAGTC</u> <u>CTGATCAG</u>	$\Delta$ -Leucine o-Aspartic acid $\alpha$ -Glutamic acid
<u>CUGAUCAG</u>	<u>GAC</u> <u>UUA</u> <u>GUC</u>

What would you like to add?

(HELP, DATA, LAB, BACK)

Fig. 1-5a

DNA  
GACTTAGTC

CUGAUCAG  
m-RNA

OH, OH! You don't have all the parts!  
In particular, ribosomes are missing

Fig. 1-5b

Graphical illustration of the biochemical steps involved in protein synthesis. The student introduces appropriate DNA, RNA, etc., into an initially empty cell, then watches the synthesis proceed. Here the synthesis breaks down for lack of a crucial part. (Paul Tenczar)




Fig. 1-6. Using graphics to teach Esperanto without using English. Here the stars have been circled to emphasize the student's mistake in counting. (Judith Sherwood)

These are actual photographs of the plasma panel. The display shows orange text and drawings on a black background, but the pictures are shown here as black on white for ease of reproduction. The plasma panel size is 22 cm. square (8.5 in. square).

## The TUTOR Language




Fig. 1-7

### The PLATO Keyboard

Every PLATO terminal has a keyboard like the one pictured above. The keyboard has a number of special features which are closely related to certain aspects of the TUTOR language, such as the HELP key which allows students to access optional sections of a lesson written in TUTOR.

The central white keys include letters, the numbers 0 through 9 along the top row, and punctuation marks. Note that the *numbers* 0 and 1 are different from the *letters* o and l. The zero has a slash through it to distinguish it unmistakably from the letter o. Except for these distinctions, the white keys are the same as the keys on a standard typewriter. Capital letters are typed by pressing either of the SHIFT keys while striking a letter key. Some keys show two different characters, such as the keys in the upper row, e.g., depressing a SHIFT key while striking a "4" produces a "\$".\*

Eight of the letter keys (d, e, w, q, a, z, x, and c all clustered around the s key) have arrows marked on them pointing in the eight compass directions. Typing "e" with a SHIFT key depressed normally produces a capital "E" on the screen, not a northeast arrow. The directional arrows are shown because these keys are sometimes used to control the motion of a cursor or pointer on the screen. In this context, the student presses an un-shifted "e" and the lesson interprets this as a command to move a cursor northeast on the screen, rather than a command to display an "e"

\*Since this book deals with technical entities, which are set off by quotation marks, it is necessary to violate certain rules of punctuation.



on the screen. Such redefinitions of what a key should do in a particular context provide enormous flexibility. Another interesting example is the use of the keyboard to type Russian text in the Cyrillic alphabet.

Spaces (blank characters) are produced by striking the long "space bar" at the bottom of the keyboard. Holding down a shift key while hitting the space bar produces a backspace. An example of the backspace's use is in underlining. The underlined word "cat" is produced by typing "c", "a", "t", backspace, backspace, backspace, underline, underline, underline (underline is shift-6, not to be confused with the minus sign or dash). Typing "T", backspace, "H", will superimpose the two letters, making a "H." The backspace is used for superimposing characters, whereas the ERASE key (just to the right of the letter p) is used to correct typing errors.

A few black keys on the left side of the keyboard are mainly associated with mathematical operations: they include plus, minus (also used as a dash), times and divide ( $\div$  is equivalent to the slash /). The  $\Leftarrow$  is used in TUTOR calculations to assign values to variables. The TAB key is most often used by authors writing lessons rather than by students studying lessons. The TAB key's function is similar to the tabulate function on standard typewriters, e.g., pressing TAB once is equivalent to hitting the space bar as many times as is necessary to reach a preset column on the screen. Shift-TAB, called CR for "carriage return", to continue the typewriter analogy, moves typing down one line and to the left margin. Shift-plus produces a  $\Sigma$  (which means summation in mathematical notation) and shift-minus produces a  $\Delta$  (which means difference in mathematical notation).

The black keys at the right of the keyboard are called "function" keys because they carry out various functions rather than displaying a character on the screen. By far the most important function key is NEXT. The cardinal rule for studying PLATO lessons is "When in doubt, press NEXT." Pressing NEXT causes the next logical thing to happen, such as proceeding on to a new display, asking for a response to be judged, erasing an entire incorrect response, etc. The second most important function key is ERASE, which is used to correct typing errors. Each press of ERASE erases one character from the screen. Pressing shift-ERASE (abbreviated as ERASE1) erases an entire word rather than a single character. Note the difference from the backspace (shift-space) which does not erase and is used for superimposing characters.

The EDIT key is also used for correcting typing. Suppose you have typed "the quik brown fox" when you notice the missing "c" in "quick". You could press ERASE1 twice to erase "fox" and "brown", use ERASE to get rid of the "k", then retype "ck brown fox". The EDIT key makes

## The TUTOR Language

such retyping unnecessary. Instead of hitting ERASE1, you press EDIT which makes the entire sentence disappear. Press EDIT again, and the entire first word "the" appears. Press EDIT again and you see "the quik" on the screen. Use ERASE to change this to "the quick". Now hit EDIT twice to bring in the words "brown" and "fox". The final result is "the quick brown fox". This takes longer to describe here in words, but pressing the EDIT key a few times is much easier and faster than doing all the retyping that would otherwise be necessary. The EDIT1 key (shift-EDIT) brings back the entire remaining portion of a sentence. For example, after inserting the "c" to make "the quick", you could hit EDIT1 once to bring back "brown fox". You should type some sentences at a PLATO terminal and study the effects produced by EDIT and EDIT1.

The COPY key is closely related to the EDIT key and is used mainly by authors. While EDIT and EDIT1 cycle through words you have just typed, COPY and COPY1 bring in words from a pre-defined "copy" sentence. These keys are used heavily when changing or inserting portions of a lesson.

The display "a<sup>2</sup>b" can be made by hitting "a", then SUPER, then "2", then "b". SUPER makes a non-locking movement higher on the screen for typing superscripts. Notice that SUPER is struck and released, not held down while typing the superscript. Striking shift-SUPER makes a locking movement, so that the sequence "a", shift-SUPER, "2", "b" will produce "a<sup>2</sup>b". The SUB key is similar to SUPER. For example, the display "H<sub>2</sub>O" is made by typing "H", SUB, "2", "O". A locking subscript results from shift-SUB, which is also what is used to get down from a locking superscript. Similarly, shift-SUPER will move up from a locking subscript.

There are 34 additional characters not shown on the keyboard which are accessible through the MICRO key. For example, striking and releasing the MICRO key followed by hitting "p" produces a "π". The sequence MICRO-a produces an α. Typing "e", MICRO, "q", produces "è", whereas typing "E", MICRO, "q", produces "É". Note the "auto-backspacing" which not only backspaces to superimpose the accent mark but also places the accent mark higher on a capital letter. Six MICRO options involve autobacking:  $\grave{}$ (q),  $\acute{}$ (e),  $\ddot{}$ (u),  $\hat{}$ (x),  $\tilde{}$ (n), and  $\_$ (c). The last accent mark (MICRO-c) is used for creating cedillas (ç and Ç) and does not involve a different height for capitals. It is easy to remember these keys because of natural associations. The  $\grave{}$  and  $\acute{}$  accent marks are on the q and e keys which have the ↖ and ↗ arrows marked on them. The umlaut  $\ddot{}$  usually appears on a "u" (German ü). The circumflex  $\hat{}$  is on the x key. The tilde  $\tilde{}$  usually appears on an "n" (Spanish ñ).

The Greek letters  $\alpha$ ,  $\beta$ ,  $\delta$ ,  $\theta$ ,  $\lambda$ ,  $\mu$ ,  $\pi$ ,  $\rho$ ,  $\sigma$ , and  $\omega$  are produced by typing MICRO followed by a, b, d, t, l, m, p, r, s, or w. Here is a complete list:

key	MICRO-key	key	MICRO-key
		$\emptyset$	$\triangleleft$
		1	$\triangleright$ ("embed" symbols)
a	$\alpha$ (alpha)	< (shift- $\emptyset$ )	$\leq$ (less than or equal)
b	$\beta$ (beta)	> (shift-1)	$\geq$ (greater than or equal)
d	$\delta$ (delta)	[ (shift-2)	{ (braces)
t	$\theta$ (theta)	] (shift-3)	}
l	$\lambda$ (lambda)	\$ (shift-4)	# (pound sign)
m	$\mu$ (mu)	5	@ (each)
p	$\pi$ (pi)	6	$\triangleright$ (arrow)
r	$\rho$ (rho)	=	$\neq$ (not equal)
s	$\sigma$ (sigma)	) (shift=)	$\equiv$ (identity)
w	$\omega$ (omega)	;	$\sim$ (approximate)
q	` (grave)	o	$^\circ$ (degree sign)
e	˘ (acute)		(vertical line)
c	˘ (cedilla)	D	$\rightarrow$ (east)
u	¨ (umlaut)	W	$\uparrow$ (north)
n	˜ (tilde)	A	$\leftarrow$ (west)
x	ˆ (circumflex)	X	$\downarrow$ (south)
C	© (copyright)	,	$\updownarrow$ (special)
Q	(leftward writing)	+	& (ampersand)
R	(rightward writing)	/	\ (backwards slash)
CR (shift- TAB)	(special TAB for leftward writing)	x (shift-x)	o (matrix multiply) x (vector product)

These are the standard MICRO definitions. You can change these by setting up your own micro table. This is discussed in Chapter 9.

The standard character set includes all the characters we have seen so far, including the Greek letters and other characters accessible through the MICRO key. The shifted MICRO key, called FONT, lets you shift from this standard set of characters to another set of up to 126 special characters which you can design.

These special characters might be the Cyrillic, Arabic, or Hebrew alphabet, or they can be pieces of pictures, such as the characters  $\curvearrowright$ ,  $\curvearrowleft$ , and  $\curvearrowright$  which form a car when displayed side by side:  $\curvearrowright\curvearrowleft\curvearrowright$ . Unlike MICRO which only affects the next keypress, FONT locks you in the

**The TUTOR Language**

alternate “font” or character-set. You press FONT again to return to the standard font. The creation of new character sets is described in Chapter 9.

If the author activates it, the ANS key can be used by the student to get the correct answer to a question. This is discussed in Chapter 7. The shifted ANS key, TERM, when pressed causes the question “what term?” to appear at the bottom of the screen. At this point, you can type any one of various keywords in order to move to a different part of the lesson. The use of TERM is discussed in Chapter 5.

If you set up an optional help sequence, the student can press the HELP key to enter the sequence. The student can then press BACK (or BACK1) to go back to where he or she was when originally requesting help, or will be brought back to the original point upon completion of the help sequence. You could also specify a different help sequence accessible by pressing HELP1 (shift-HELP). The six keypresses HELP, HELP1, LAB, LAB1, DATA, and DATA1 can, if activated by the author, allow the student a choice of six different help sequences. You can also activate NEXT, NEXT1, BACK and BACK1, but these simply let the student move around in the lesson without remembering or returning to the original place. In other words, these four keys do not initiate help sequences. Usually, BACK is reserved for review sequences or similar situations where you want to back up.

The STOP key throws out output destined for the terminal. A useful example is the case of skimming through pages of text in an on-line catalog or collection of notes. If you decide you want to skip immediately to the next page, you might press STOP in order to avoid the wait required to finish plotting the present page.

The STOP1 or shift-STOP key plays a crucial role in PLATO usage. You press STOP1 to leave a lesson you are studying. When a student is ready to leave the terminal he or she presses STOP1, which performs a “sign-out” function. Among other things, the sign-out procedure brings the student’s permanent status record up to date so that days later he or she can sign-in and resume working at the same point in the lesson. When an author presses STOP1 to leave a lesson that he or she is testing, the author is taken back to a point in the PLATO system where he or she can make changes in the lesson before trying it again.

The key next to HELP, with the square (□) on it, is similar to the EDIT key, but retrieves one character at a time, instead of a whole word. It is particularly useful when used in association with the EDIT key. The shifted square key is presently used as the ACCESS key, as described in Chapter 9.

## Basic Aspects of TUTOR

In their simplest form, lessons administered by the PLATO interactive educational system consist of a repeating sequence: a display on the student's screen followed by the student's response to this display. The display information may consist of sentences, line drawings, graphs, animations (moving displays)—nearly anything of a pictorial nature, and in any combination. The student responds to this display by pressing a single key (e.g., the HELP or NEXT key), by pointing at a particular area of the screen, by typing a word, sentence, or mathematical expression, or even by making a geometrical construction. Lesson authors provide enough details about the possible student responses so that PLATO can maintain a dialog with the student. The sequence of a display followed by a response is the basic building block of a lesson and is called a "unit" in the TUTOR language. This "display-response" terminology is convenient but is not intended to imply that the student is in a subservient position. Often what we will conventionally call the student "response," is a question or a command issued to PLATO to respond with a display of some kind.

An author constructs a lesson by writing one unit at a time. For each unit, the author uses the TUTOR language to specify: (1) the display that will appear on the student's screen; (2) how PLATO is to handle student responses to this display; and (3) how the current unit connects to other units.

A statement written in the TUTOR language appears as follows:

write	How are you today?
⏟	⏟
command	tag

The first part of the statement (-write-) is called the "command," while the remainder (How are you today?) is called the "tag." Command names mnemonically represent PLATO functions. The tag gives additional specifications on how the function is to be carried out. In this case, the tag specifies what text is to be written on the screen.

The following is an entire unit written in TUTOR. Figure 1-8 shows what a student would see on the screen while working on the unit.

## The TUTOR Language

unit geometry  
at 1812  
write What is this figure?  
draw 510;1510;1540;510  
arrow 2015  
answer <it,is,a> (right,rt) triangle  
write Exactly right!  
wrong <it,is,a> square  
write Count the sides!




Fig. 1-8

We will discuss each statement of this unit in detail.

unit geometry

The `-unit-` statement initiates each unit. The tag (`geometry`) will become useful later when units are connected together to form a lesson. Each unit must have a name. No two units in a lesson may have the same name.

at 1812

The `-at-` statement specifies at what position on the screen a display will occur. The tag "1812" means that we will display something on the 18th line in the 12th character position. The top line of the screen is line 1 and the bottom line is line 32. There are 64 character positions going from 01 at the left edge of the screen to 64 at the right. Thus, 101 refers to line 1, character position 01 (the upper left corner of the screen), while 3264 refers to line 32, character position 64 (the lower right corner of the screen). Note that "0" means the number zero, as distinct from the letter "O".




Fig. 1-9. Illustration of "at 1812"

write What is this figure?

The `-write-` statement causes the text contained in the tag to be displayed on the student's screen. The writing starts at line 18, character position 12, as specified by the preceding `-at-` statement.

draw 510;1510;1540;510

The `-draw-` statement specifies a straight-line figure to be displayed on the screen. In this particular case a series of straight lines will be drawn starting at location 510 (line 5, character position 10), going vertically downward to location 1510, then to the right to location 1540, and finally back to the starting point, 510. This produces a right triangle on the student's screen.

arrow 2015

## The TUTOR Language

The `-arrow-` statement acts as a boundary-line that separates preceding display statements from the following response-handling statements. Thus, what precedes the `-arrow-` command produces the screen display which remains visible while the student works on the question. Statements after the `-arrow-` command are used in handling student responses to the display. In addition, the `-arrow-` statement notifies TUTOR that a student response is required at this point in the lesson. The tag of the `-arrow-` statement locates the student response on the screen. An arrowhead is shown on the screen at this place to indicate to the student that a response is desired and to tell him or her where the response will appear. In this case the arrowhead will appear on line 20, character position 15. The student's typing will start at 2017, leaving a space between the arrowhead and the student's first letter.

```

answer <it,is,a> (right,rt) triangle
.
.
.
.
wrong <it,is,a>square

```

The `-answer-` and `-wrong-` statements are used to evaluate the student's response. The special brackets `<` and `>` enclose optional words, while the parentheses enclose important words which are to be considered synonyms. Thus any of the following student responses would match the `-answer-` statement: "a right triangle", "it is a rt triangle", "it triangle", etc.

If the response matches the tag of the `-answer-` statement, TUTOR writes "ok" after the student's response. For a match to a `-wrong-` statement, "no" is written. An "ok" judgment allows the student to proceed to the next unit, whereas a "no" judgment requires the student to erase and try again. Any response not foreseen by `-answer-` or `-wrong-` statements is judged "no".

Having matched the student's response, TUTOR proceeds to execute any display statements following the matched `-answer-` or `-wrong-` statement. Thus, student responses of "a right triangle" and "square" will trigger appropriate replies. In the absence of specific `-at-` statements, TUTOR will display these replies three lines below the student's response on the screen.

Special help is provided to the student if his answer is partially



correct. Here is what happens if the student responds with “a lovely tringle, right?”:

> a lovely tringle, right?  
 xxxxxxxxΔ===== ←

TUTOR automatically marks up the student’s response to give detailed information on what is wrong with the response. The word “lovely” does not belong here and is marked with xxxxxxxx, the Δ shows where a word is missing, the word “tringle” is misspelled and is underlined, and the word “right” is out of order, as is indicated by the small arrow.

Statements can be added to the current example unit which will greatly improve it. Consider the following:

	unit	geometry
	at	1812
	write	What is this figure?
	draw	510;1510;1540;510
	arrow	2015
	specs	okcap
	answer	<it,is,a> (right,rt) triangle
	write	Exactly right!
Handling additional responses	answer	<it,is,a> three*sided (right,rt) polygon
	write	Yes, or a right triangle.
	wrong	<it,is,a> triangle
	at	1605
	write	Please be more specific. It has a special angle.
	draw	1410;1412;1512
	wrong	<it,is,a> square
	write	Count the sides!






Fig. 1-10.

As you can see, any number of -answer- and -wrong- statements can be added to the response-handling section of the unit. Time and effort spent by an author in providing for student responses other than the common answer can greatly increase the ability to carry on a personal dialog with each student. Figure 1-10 shows what the student will see if he responds with “a triangle”. The construction “three\*sided” is called a “phrase”. A phrase is a set of words to be considered together for purposes of spelling and word order. As another example, a question about Columbus’s flagship might involve the phrase “Santa\*María”.

The TUTOR Language


The `-specs-` statement is introduced here. It is used to give optional specifications on how the student's response is to be handled. In this case the tag, "okcap", specifies that any capitalization in the student's response is optional. Without this specification, TUTOR would consider "Right Triangle" to be misspelled. There are many convenient options available in a `-specs-` statement. For example, "specs okextra,noorder" specifies that extra words not mentioned explicitly in following `-answer-` and `-wrong-` statements are all right, and that the student's word order need not be the same as the word order of the `-answer-` and `-wrong-` statements to achieve a match. Such options can be used to greatly broaden the range of responses which can be handled properly.

Lessons could be written using only the commands already discussed. Explanatory units could be written using only display commands. Tutorial units could be interspersed to test a student's understanding of the lesson material. Thus a single linear chain of units could form a lesson. However, mastery of a few more TUTOR commands opens up a wealth of "branching" or sequencing possibilities. Branching, the technique of allowing alternate paths through a lesson, is one of the keys to personal dialog with each student. The example unit will, therefore, be expanded to include `-next-`, `-nextnow-`, `-back-`, and `-help-` commands:

	unit	geometry
	next	moregeom
	help	thelp1
	back	intro
	at	1812
	write	What is this figure?
	draw	510;1510;1540;510
	arrow	2015
	specs	okcap
	answer	<it,is,a> (right,rt) triangle
	write	Exactly right!
	wrong	<it,is,a> triangle
	at	1605
	write	Please be more specific.
		It has a special angle.
	draw	1410;1412;1512
	wrong	<it,is,a> square
	nextnow	treview

The tag of the `-next-` statement following the `-unit-` command gives

the name of the next unit the student will see upon the successful completion of unit "geometry". The -next- statement is necessary because the next unit for a student in a highly-branching lesson sequence may not be the unit following in the lesson. For example, a diagram of the lesson flow involving unit "geometry" might be:



In moving from one unit to another the screen normally is automatically erased to make room for the displays produced by the following unit.

The -help- statement refers to a help unit which the student may reach through the use of the HELP key. Help units are constructed in the same manner as unit "geometry". However, the last (or only) unit in a help sequence is terminated by an -end- command. Upon completing the last help unit, the student is returned to the "base" unit, the unit from which the student branched (in this case unit "geometry"). The student

## The TUTOR Language

need not complete the entire help sequence. He may press BACK or shift-BACK to return to the base unit from any point in the help sequence. Help units for unit "geometry" could appear as follows:

```

.
.
.
* These units are help units for "geometry".
unit  thelp1
at    1828
write The figure has three sides.
draw  510;1510;1540;510
*
unit  thelp2
at    1828
write It also has three angles.
draw  510;1510;1540;510
*
unit  thelp3
at    1828
write Note the right angle.
draw  510;1510;1540;510
end

```



Any statement which begins with an asterisk (\*) has no effect on the operation of the lesson and may be used anywhere to insert comments which describe the units. A comment statement between units improves readability by guiding the eye to the unit subdivisions of the lesson.


The -back- statement permits the student to move to a different unit by pressing the BACK key. Because of its name, it is customary to associate a review sequence with the BACK key. If a student is in a non-help unit that does not contain a -back- statement, the BACK key does nothing. In a help-sequence unit that has no -back- statement, the BACK key returns the student to the original base unit.

If the student calls the figure "a square", he or she will see this response judged "no" and get the reply "Count the sides!" The -nextnow- statement is used to force the student through additional material. It locks the keyboard so that only the NEXT key has any effect. In particular, the student cannot erase his or her response. When the student presses NEXT, he or she will be sent to unit "review". Upon

completion of one or more units of review about triangles, the author might return the student to unit "geometry". Thus, this student's lesson flow might consist of:

- 1) a discussion of geometric figures;
- 2) a question about a right triangle;
- 3) an error causing -nextnow- to lock the keyboard;
- 4) further study of triangles; and finally
- 5) a return to the right triangle.

Consider now the problem of using unit "geometry" for a second student response. Additional display information is needed to ask the student a second question, and another -arrow- command is needed plus a second set of response-handling statements. The unit could appear as follows:

unit	geometry	
next	moregeom	
back	intro	
help	thelp1	
at	1812	
write	What is this figure?	
draw	510;1510;1540;510	
arrow	2015	
.	} Response-handling statements for first arrow.	
.		
.		
.		
 endarrow		
at	2512	
write	How many degrees in a right angle?	
help	angles	
arrow	2815	
.	} Response-handling statements for second arrow	
.		
.		

The -endarrow- command delimits the response-handling statements associated with the first -arrow-. Only when the first -arrow- is satisfied by an "ok" judgment will TUTOR proceed past the -endarrow- command to present the second question. The statement "help angles" overrides the earlier statement "help thelp1". If the student presses the HELP

## The TUTOR Language

key while working on the second -arrow- he or she will reach unit "angles" rather than unit "thelp1".

The second question could have been given in a separate unit rather than following an -endarrow- command. The major difference is that the entire screen is normally erased as the student proceeds to a new unit, whereas here the second question was merely added to the existing screen display. Even if there is only one -arrow- command in a unit, -endarrow- can be useful, for it can be followed by display or other statements to be performed only after the -arrow- is satisfied. This is particularly convenient if there are several -answer- commands corresponding to several different classes of acceptable responses.

Fourteen TUTOR commands have been illustrated in this chapter. This repertoire is adequate to begin lesson writing. If you have access to a PLATO terminal, it would be useful at this point to try out the ideas discussed so far.

## More on Creating Displays

## 2

Particular attention should be paid to the question of how to display text and line drawings to the student. Good or poor displays of material in a lesson can make the difference between a successful or unsuccessful lesson. Imaginative use of graphics, including animations (moving displays), will capture the attention of the student and transmit your message much more efficiently than would mere text. You have already seen how to write text and draw figures by using the `-at-`, `-write-`, and `-draw-` commands. This chapter will discuss how to achieve finer control over screen positions, how to draw circles and circular arcs, how to display large-size text and write at an angle, and how to erase portions of the screen. The ability to erase a portion of the screen makes it possible to create animated displays.

### Coarse Grid and Fine Grid

It is convenient to specify a line number and character position for displaying text. We have seen that the TUTOR statement `"at 1812"` instructs PLATO to display information starting on the 18th line at the 12th character position. Line 1 is at the top of the screen and line 32 is at the bottom. Each line has room for 64 characters, with character position 01 at the left and character position 64 at the right. This numbering scheme is called the "coarse grid" or "gross grid".

The TUTOR Language

Sometimes it is necessary to position text or draw a figure with finer control than is permitted by the coarse grid. The PLATO screen consists of a grid of 512 by 512 dots, and the position of any of these quarter-million dots can be specified by giving two numbers—the number of dots from the left edge of the screen (often called “x”) and the number of dots up from the bottom of the screen (often called “y”):




Fig. 2-1.

The position shown would be referred to as the “fine grid” location “384,128” in an -at- or -draw- statement. This position is equivalent to the coarse grid location 2449 (line 24, character position 49). As an example, consider the following unit:

```

unit double
at 384,128
write DOUBLE WRITING
at 385,129
write DOUBLE WRITING
    
```

This unit would write “DOUBLE WRITING” twice, displaced horizontally and vertically by one dot, which looks like this:



# DOUBLE WRITING

Fig. 2-2.

(Greatly enlarged.)

The `-draw-` command permits mixing the two numbering schemes:

```
draw 1215;1225;120,240;1855
```

This means “draw a straight line from 1215 to 1225, draw a second straight line from there to (120,240), then draw a third straight line from there to 1855”. Note that each point, whether expressed in coarse grid or fine grid, must be set off by a semicolon.

The `-box-`, `-vector-`, and `-circle-` Commands




Fig. 2-3.

Figure 2-3 illustrates how rectangular boxes are often drawn as part of a display. Although such boxes can be drawn using the `-draw-` command, it is even more convenient to use a `-box-` command, since you merely give two corners of the box. For example:

## The TUTOR Language

```
box 1215;1835
```

is exactly equivalent to:

```
draw 1215;1235;1835;1815;1215
```

Fine grid coordinates can also be used for the corners of the box. The sides of the box can be made thicker for additional emphasis. For example, “box 1215;1835;2” will draw a box with sides two dots thick.

Another frequently drawn object is a “vector”—a line with an arrowhead used to point out something on the screen. The statement “vector 512;920” will draw a line from 512 to 920, with an arrowhead added at the 920 end. Fine grid coordinates can be used. The size of the arrowhead in relation to the line can be controlled by adding another number. For example, “vector 512;120;6” will show an arrowhead about half as large as normal. Making the arrowhead size *negative* draws an “open” rather than a “closed” arrowhead.

Circles are drawn by specifying a center with an `-at-` command, then using a `-circle-` command to specify the radius (as a number of dots):

```
at      1215
circle 50
circle 75
```

This will draw two circles whose radii are 50 dots and 75 dots long, centered at screen location 1215. Notice that the screen position is restored to the center of the circle after drawing a complete circle.

A portion or arc of a circle can be drawn by specifying starting and ending angles, as in “circle 125,0,45”, which will draw a 45-degree circular arc, starting at 0 degrees (0 is “east” or “horizontally to the right”; and 360 degrees is again “east”). After drawing an arc, the screen position is left at the end of the arc rather than at the center of the circle.

The `-circleb-` command is just like `-circle-`, but it draws a broken or dashed circle or circular arc.

The basic line-drawing commands (`-draw-`, `-box-`, `-vector-`, and `-circle-`) are used together to build complicated drawings.

### Large-size Writing: `-size-` and `-rotate-`

It is possible to display text in larger than normal size, and even write at an angle. This is particularly useful in showing an eye-catching title on a page. Here is a sample display with the corresponding TUTOR statements. The “\$\$” permits a comment to appear after a tag.




Fig. 2-4.

```

unit   title
size   9.5   $$ text 9.5 times normal size
rotate 45    $$ text rotated 45 degrees
at     2519
write  Latin
size   0     $$ return to normal writing
rotate 0
at     3125
write  Lesson on Verbs

```

For technical reasons the large-size writing comes on the screen much more slowly than does normal text, but the speed is adequate for short titles. Use “size 0” to return to normal writing. Normal writing is unaffected by -rotate-. However, you may use “size 1” if you wish to rotate standard size text. Size 1 writing appears at the same slow speed as larger writing (about 6 characters per second, or 60 words per minute). Only size 0 writing is rapid (180 characters per second, or 1800 words per minute).

**BE SURE TO RETURN TO SIZE 0!!** If you forget to place a “size 0” statement after the completion of the special writing, all of your text will be written slowly (and possibly rotated). It is also good practice to say “rotate 0”, so that the next time you use “size” the rotation will be through 0 degrees unless stated otherwise.

You can magnify the width of the characters differently from the height. For example, “size 2,5” will make the characters twice as wide and five times as high as they are normally.

Because “sized” writing is slow, it should be used only for special effects. It should be avoided on pages which are seen repeatedly, such as a table of contents for a lesson, because the student will be irritated by the enforced wait. In such cases, it is better to achieve emphasis by other, faster techniques, such as drawing a box around a heading written in “size 0”.

The TUTOR Language

Animations (Moving Displays): -erase- and -pause-

An animated display can be created by repetitively displaying some text, pausing, erasing the text and rewriting it in a new position on the screen. Here is a unit which will show two balloons floating upwards. The unit is split in order to show the changes in the display. (See Figures 2-5a through 2-5c.)

	unit	balloons	
	at	3020	
	write	Watch the balloons go up!	
	at	250,100	
	write	00	\$\$ use 00 for balloons
☞	pause	1.5	\$\$ suspend processing for 1.5 seconds
	at	250,100	
☞	erase	2	\$\$ erase two characters
	at	250,150	\$\$ reposition 50 dots higher
	write	00	
	pause	1.5	




Fig. 2-5a.




Fig. 2-5b.

```

at      250,150
erase   2
at      250,200
write   00
pause   1.5
.
.
.

```




Fig. 2-5c.

The statement “erase 2” selectively erases two character positions without disturbing the rest of the screen. In particular, the text “Watch the balloons go up!” will stay on the screen.

There are other forms of the -erase- command. The statement “erase 12,3” will selectively erase a block of 12 character positions on three consecutive coarse grid lines. The statement “erase” with no tag will erase the entire screen instantaneously. The same full-screen erase normally takes place automatically upon moving to a new main unit.

#### -pause-, -time-, and -catchup-

The -pause- statement with a tag in seconds suspends processing for the specified amount of time. If the tag is omitted, TUTOR waits for the

## The TUTOR Language

student to strike a key, any key, rather than wait a specified amount of time. This form is particularly suitable in more complicated situations where the student may want to study each step before proceeding. Here is an example:




Fig. 2-6a.





Fig. 2-6b.

unit     discuss  
at        520  
write    There are several kinds  
          of -erase- commands  
          for selective and full  
          erasing of the screen.

 pause  
at        1520  
write    "erase 5" will erase  
          5 spaces. "erase 25,4"  
          will erase 25 spaces  
          on 4 lines.

pause  
at        2520  
write    An -erase- command  
          with a blank tag  
          will erase the whole  
          screen.




Fig. 2-6c.

Each time the student presses a key to move past the -pause- command, more text is added to the screen. This prevents the student from feeling overwhelmed by too much text all at once. Each new paragraph is added only when the student signals by pressing a key that he or she wants to go on. On the other hand, this structure leaves the earlier paragraphs on the screen so that the student can look back to review. If the -pause- commands were replaced by -unit- commands, each paragraph would reside in a separate main unit. When the student presses NEXT to move on to the next main unit, the screen is completely erased to make room for the next display. This would accomplish the objective of letting the student control the rate of presentation of new material but would not leave the earlier paragraphs on the screen for review and comparison.

It is inadvisable in this application to use "pause 15" rather than "pause", since the student would have no control over the presentation rate. Any time delay you choose will be too fast for some students and too slow for others. A timed -pause- is mainly useful for animations. Sometimes it is appropriate to move on after a long time if the student hasn't pressed a key. This can be achieved with a -time- command:

```
.
time 30
pause
.
```

## The TUTOR Language

The “time 30” statement will “press the timeup key” after 30 seconds, so that if the student does not press a key, TUTOR will. The student can move on sooner by pressing a key before then. However, this is not possible if you use “pause 30”.

To summarize, there are three types of -pause- situations:

- 1) pause n pause n seconds whether  
keys are pressed or not
- 2) pause wait for any key
- 3) time n } wait for any key or n seconds  
pause }

Occasionally, you might want to send several seconds worth of output to the student’s screen, then pause two seconds, then add something else. If you write several seconds of display including text and drawings which take several seconds to paint on the screen, followed by:

```
pause 2  
write More text. . .
```

you will not get the desired effect because TUTOR will add “More text . . .” right after the initial material headed toward the terminal (since the “pause 2” ends before the initial display is finished). The student will see no gap between the first and second parts of the display. The problem can be solved with a -catchup- command:

```
.  
. .  
. .  
catchup  
pause 2  
write More text. . .
```

The -catchup- command tells TUTOR to let the terminal “catch up” on its work up to that point before continuing. *Then* you pause an additional two seconds, and you get the desired effect.



### The -mode- Command

The -erase- command may be used to erase blocks of character positions or the whole screen. However, something else is needed for selectively erasing line drawings created with -draw- and -circle- statements. The PLATO terminal can be placed in an erasure mode in which the terminal interprets all display instructions as requests to erase rather than to light up the corresponding screen dots. This is done with the -mode- command:

```

unit    modes
at      2517
write   Selective erase of a figure
draw    1210;2010;2050;1210  $$ triangle
pause   $$ wait for a key
mode    erase
☞ draw  1210;2010;2050      $$ part of the triangle
mode    write
at      510
write   One line left.
    
```




Fig. 2-7a.





Fig. 2-7b.

The TUTOR Language

The "write" mode is the normal display mode. Be sure to specify "mode write" when you are through with "mode erase", or all further writing in that unit will be invisible!

In the standard mode ("write") it is possible to superimpose or overstrike text with another -write- statement. If, however, a "mode rewrite" statement is executed, the second -write- statement will erase the previous text as it writes the new text, and there will be no superposition. Compare these sequences in write and rewrite modes:



In the rewrite case the second -write- statement wipes out the 3-character area as it writes the new information. Each character area is 8 dots wide by 16 dots high. This determines the number of rows and columns in coarse grid. In the coarse grid,  $(512/8)=64$  characters fit across the screen, and  $(512/16)=32$  lines of characters fill the screen vertically.

The statement "erase 2" is actually equivalent to:

```

mode  rewrite
write (two spaces)
mode  (previous mode)
                
```

Writing spaces (blank characters) in rewrite mode wipes out an entire character area.

The balloon animation in Figures 2-5a through 2-5c could have been written:

```

.
.
.
at    250,100
write 00
pause 1.5
mode  erase
at    250,100
                
```

```

write 00      $$ instead of "erase 2"
mode write
.
.
.

```

This form would be different from the form using "erase 2" if there were other screen dots lit in this area. The form which uses "erase 2" completely erases two character positions while "write 00" in the erase mode erases only the dots that make up the letters "00" without disturbing neighboring dots.

### Automated Display Generation

It should be mentioned that an author working at a PLATO terminal can use a moving cursor to design a display involving text, line figures, circles and arcs. The PLATO system then automatically creates corresponding TUTOR statements which would produce that display. The author can alter these statements, convert them back into a display, and add to or alter the resulting display. This facility makes it unnecessary in most cases to worry about the details of screen positions. Here is an example of such operations:




Fig. 2-8. Move the cursor (the "+") to draw the road and to mark the ends of the tree trunk.




Fig. 2-9. Draw the tree trunk.

The TUTOR Language




Fig. 2-10. Specify a circle for the top of the tree. Draw the house. Place text of various kinds on the screen. (The car uses special characters.)




Fig. 2-11. PLATO automatically generates TUTOR statements corresponding to the desired display.




Fig. 2-12. Recall the display and add a flag to the house.

```

unit    display
draw    1812;1852;skip;1844;1544
at      344,288
circle  16
draw    1837;1637;1535;1633;1831
at      184,225
write   (R)
at      2821
write   Automated display-making
size    3
rotate  -38
at      2521
write   Look!
size    8
rotate  8
draw    1535;1335;1333;256,296;272,296

```

Fig. 2-13. PLATO appends a -draw-state-ment corresponding to the flag.

The TUTOR Language




Fig. 2-14. Final result. The illustrations in this book were created by these techniques. The screen displays were photographed.

# Manipulating Data Bases

11


In this chapter we will discuss the tools available in TUTOR for creating and using “data bases” (small or large blocks of data such as test scores, population statistics, map coordinates, etc.). In the process of discussing these tools we will also learn more about the internal workings of the PLATO system.

## The -common- Command

The “student variables” v1 through v150 are associated with the individual student. It is possible to use “common variables” which are common to *all* those students studying a particular lesson. These common variables can be used to send messages from one student to another, to hold a bank of data used by all the students, to accumulate statistics on student use of the lesson, to contain test items in a compact, standardized form, etc.

As a first example of the use of the -common- command, let’s count the number of students who have entered our lesson. We will also count how many of these students are female:

```


common 2 $$ two common variables
define total=vc1,females=vc2

```

(Continued on the next page.)

## The TUTOR Language

```

*
unit      ask
calc      total←total+1
at        1215
write     Are you a female?
arrow     1415
answer    yes
calc      females←females+1
answer    no
no
write     Yes or no, please!
endarrow
at        1615
write     There are <s,total> students, of whom
          <s,females> are female.

```

The `-common-` command tells TUTOR to set up two common variables, `vc1` and `vc2`, which we have defined as “total” and “females”. These common variables are automatically initialized to zero before the first student enters this lesson. The first student increments “total” to one (“`calc total←total+1`”) and may also increment “females”. The second student to enter the lesson causes “total” to increase to two and may also change “females”. Each student is shown the present values of “total” and “females”, which depend on what other students are doing. We must use common variables `vc1` and `vc2` rather than the student variables `v1` and `v2` because the student variables cannot be directly affected by actions of other students. Another way to see this is to point out that when there are five students in this lesson, they share a single `vc1` and a single `vc2`, whereas they each have their own `v1` and their own `v2`: there are five `v1`'s and five `v2`'s but only one `vc1` and `vc2`.

Integer common variables are `nc1`, `nc2`, etc., and indexed common variables are written as `vc(index)` or `nc(index)`.

The statement “`common 2`” tells TUTOR to associate a two-word set of common variables with this lesson. For reference purposes, it is good style to place the `-common-` command near the beginning of the lesson. There can be only one `-common-` statement in a lesson. Like `-define-`, `-vocab-`, and `-list-`, the `-common-` command is not executed for each student. Rather, when TUTOR is preparing the lesson for the first student who has requested it, a set of common variables is associated with the lesson and all these common variables are initialized to zero. Additional students entering the lesson merely share the common variables previously set up.

Suppose a class of fourteen students uses our lesson from 10 a.m. to 11 a.m. The fourteenth student comes at 10:05 and gets a message on the



screen saying "There are 14 students, of whom 8 are female". As long as the lesson is in active use, each new student who enters the lesson increases "total" (vc1). However, when all the students leave at 11:00, the lesson is no longer in active use and will eventually be removed from active status to make room for other lessons. When another class comes at 3:00 p.m., the lesson is not in active use and TUTOR must respond to the first student's request for the lesson by preparing the lesson for active use. In the preparation process the statement "common 2" tells TUTOR to set up two common variables and initialize them to zero. The first student to enter the lesson at 3:00 is told "There are 1 students, of whom 1 are female". She is *not* told "There are 15 students, of whom 9 are female", despite the fact that the previous student (at 10:05 that morning) had been told there were 14 students, 8 female. The "common 2" statement will cause the common variables to be zeroed every time the lesson is prepared for active use.

The type of common which is set up by the statement "common 2" is called a "temporary common". It lasts only as long as the lesson is in active use, and its contents are initialized to zero whenever the lesson is moved from inactive to active status. Temporary common can be used for such things as telling the students how many students are present, what their names are, and whether a student at another terminal who has finished a particular section of the lesson is willing to help a student who is having difficulties. Messages can be sent from one student to another through a temporary common by storing the message in the common area with an identifying number, so that the appropriate student can pick up the message and see it with a -showa-. The lesson simply checks occasionally for the presence of a message.

When a student signs out you usually want to change the temporary common in some way. For example, if you are keeping a count of the number of students presently using the lesson, you increase the count by one when a student signs in and you decrease the count by one when the student leaves. The -finish- command lets you define a unit to be executed when the student presses shift-STOP to sign out:

```
finish decrease
.
.
unit decrease
calc count←count-1
```

In this case unit "decrease" will be done each time a student signs out. Normally the -finish- command should be put in the "ieu". As with -imain-, the pointer set by the -finish- command is not cleared at each new main unit. A later -finish- command overrides an earlier one, and

## The TUTOR Language

“finish q” or a blank -finish- statement will clear the pointer. Like all unit pointer commands, -finish- can be conditional. Only a limited amount of processing is permitted in a -finish- unit to insure that the student can sign out promptly.

We can keep a permanent, on-going count of students who enter the lesson by using a “permanent common”. Instead of writing “common 2”, we write “common italian,counts,2”, where “italian” is the name of a permanent lesson storage space and “counts” is the name of a common block stored there. This is the same format used for character sets (the -charset- command) and micro tables (the -micro- command). When the common block is first set up in the lesson space, its variables are initialized to zero. Let’s suppose that the fourteen students who come in at 10:00 a.m. are the very first students ever to use our lesson. The statement “common italian,counts,2” will cause TUTOR to fetch this (zeroed) common block from permanent storage. As before, the fourteenth student arrives at 10:05 and is told “There are 14 students, of whom 8 are female”. At 11:00 a.m. these students leave and our lesson is no longer in active use. At some point, room is needed for other active lessons (and commons), at which point our permanent common, with its numerical contents of 14 (students) and 8 (females) *is sent back to permanent storage*. At 3:00 p.m. the first student (a female) of the afternoon class causes TUTOR to prepare the lesson and retrieve the permanent common from permanent storage *without* initializing the common variables to zero. The result is that she gets the message “There are 15 students, of whom 9 are female”. (There is an -initial- command which can be used to define a unit to be executed when the first student references the common. This makes it possible to perform initializations on a permanent common.)

The key feature of permanent common is that it is retrieved from storage when needed and *returned* in its altered state to permanent storage when the associated lesson is no longer active. In our case, we could enter the lesson months after its initial use and see the total number of students who have entered the lesson during those months. Other uses of permanent common include the storage of data bases accessed by the students, such as census data in a sociology course or cumulative statistical data on student performance in the course.

### The Swapping Process

Before discussing additional applications of common variables, it is useful to describe the “swapping” process by which a single computer can appear to interact with hundreds of students simultaneously. The

computer actually handles students one at a time but processes one student and shifts to another so rapidly that the students seem to be serviced simultaneously. In order to process a student, the student's lesson and individual status (including the variables v1 through v150) must be brought into the "central memory" of the computer. After a few thousandths of a second of processing, the student's modified status is transferred out of the central memory (to be used again at a later time) and another student's lesson and status are transferred into central memory. This process of transferring back and forth is called "swapping," and the large storage area where the lessons and status banks are held is called the "swapping memory." The swapping memory must be large enough to hold all the status banks and lessons which are in *active* use; that is, in use by students presently working at terminals. It is not necessary for the swapping memory to also hold the many lessons not presently in use nor the status banks for the many students not using the computer at that time. These inactive lessons and status banks are kept in a still larger "permanent storage" area. (See Fig. 11-1.)




Fig. 11-1.

## The TUTOR Language

When a student sits down at a terminal and identifies herself as “Jane Jones” registered in “french2a”, her status bank is fetched from permanent storage to see what lesson she was working on and where in the lesson she left off last time. If the lesson is already in the swapping memory (due to active use by other students), Jane Jones is simply connected up to that lesson, and, as she works through the lesson, her lesson and her changing status bank will be continually swapped to central memory. If, on the other hand, the required lesson is not presently in active use, it must be moved from permanent storage to the swapping memory. (This involves a translation of the TUTOR statements into a form which the computer can process later at high speed.) This fetching of the inactive lesson from permanent storage to prepare an active version in the swapping memory will typically be done once in a half-hour or more often as the student moves from one lesson to another. In contrast, the swapping of the active lesson to central memory happens every few seconds as the student interacts with the lesson. Therefore, the swapping transfer rate must be very high (whereas a low transfer rate between permanent storage and the swapping memory is adequate).

When Jane Jones leaves for the day, her status bank is transferred from the swapping memory to permanent storage. This makes it possible for her to come back the next day and restart where she left off.

The question arises as to why there are three different memories: central memory, swapping memory, and permanent storage. For example, why not keep everything in the central memory where students can be processed? It turns out that central memory is extremely expensive, but permanent storage in the form of rotating magnetic disks is very cheap. Why not do swapping directly between permanent storage and central memory? The rate at which lessons can be fetched from permanent storage is much too slow to keep the computer busy: the computer would handle only a small number of students because a lot of time would be wasted waiting for one student to be swapped for another. If the cost of the computer were shared by a small number of students, the cost would be prohibitively high. In order to boost the productivity of the computer, a special swapping memory is used which permits rapid swapping. This minimizes unproductive waiting time and raises the number of students that can be handled. The swapping memory is cheaper than central memory but considerably more expensive than permanent storage.

There is, therefore, a hierarchy of memories forced on us by economic and technological constraints. The expensive, small central memory is the place where actual processing occurs, and there is never more than one student in the central memory. Material is swapped back and forth to a large medium-cost swapping memory whose most important feature is a very high transfer rate to central memory. Permanent storage is an even larger and cheaper medium for holding the entire set of

lessons and student status banks. It has a low transfer rate to the swapping memory.

### Common Variables and the Swapping Process

Now it is possible to describe more precisely the effect of a -common- statement in a lesson. Just as an individual student's lesson and status bank (including the student variables v1 through v150) are swapped between central memory and the swapping memory, so a set of common variables associated with the lesson is swapped between central memory and the swapping memory. There is in central memory an array of 1500 variables, called v1 through v1500, into and out of which a set of common variables is swapped. As long as the -common- statement specifies a set of no more than 1500 common variables, this set will automatically swap into and out of the central memory array v1 to v1500. (See Fig. 11-2.) (There is a -comload- command which can be used to specify which portions of a common to swap if the common contains more than the 1500 variables which will fit into central memory.) All 1500 variables in the central memory array are set to zero before bringing a lesson, status bank, and common into central memory, so that any of these variables not loaded by the common will be zero.




Fig. 11-2.

## The TUTOR Language

Note that the student status banks and commons are swapped *in and out* of central memory in order to retain any changes made during the processing in central memory. On the other hand, lessons are brought into central memory but are *not* sent back since no changes are made to the lesson. (A lesson only has to be copied into but not out of central memory.) The separation of the modifiable status banks and commons from the unchanging lessons makes it possible for a single copy of a lesson to serve many students.

It is dangerous to use vc-variables without a -common- statement or to use vc-variables outside the range loaded by the common (e.g., referring to vc3 when there is a “common 2” statement in the lesson). For example, consider this sequence in a lesson which has no -common-statement:

```

.
.
.
calc   vc735←18.34
pause  2
show   vc735
.
.
.

```

This will show 0, *not* 18.34. The “pause 2” statement causes this student’s material to be swapped out to the swapping memory for two seconds while many other students are processed. When the student is swapped back into central memory, all the vc-variables are zeroed. As a matter of fact, vc735 may temporarily take on many different values during those two seconds as different students are processed. On the other hand, a “common 800” would insure that vc1 through vc800 would be saved in the swapping memory and restored after two seconds, so that the “18.34” stored in vc735 would again be available to be shown (unless it had been changed by a student using the same common who was processed during the two-second wait). Similarly, because the student variables v1 through v150 are part of the swapped student status bank, the sequence:

```

.
.
.
calc   v126←3.72
pause  2

```

```
show v126
```

```
.
.
.
```

will correctly show “3.72”. The contents of the student variables cannot get lost in the swapping process because these variables are saved in the swapping memory and restored to central memory the next time this student is processed.

The fact that common variables are shared by all students studying the lesson is extremely useful but can cause difficulties if you are not careful. Suppose you want to add up the square roots of the absolute values of `vc101` through `vc1000`:

```
.
.
.
calc total←0
doto 8sum,index←101,1000
      total←total+[abs(vc(index))].5
8sum
show total
.
.
.
```

This iterative calculation will take longer than one “time-slice” (the computing time TUTOR gives you before interrupting your processing to service other students). You are swapped out and will be swapped back into central memory later to continue the computation. It might take several time-slices to complete the computation, and in between your time-slices other students are processed. This time-slicing mechanism insures that no one student can monopolize the computer and deny service to others. Suppose two students, Jack and Jill, are studying this lesson and sharing its common. Suppose that Jack has reached the part of the lesson that contains the `-doto-` shown above. If, at the same time, Jill runs through calculations that modify `vc101` through `vc1000`, her modifications will be made during the interruptions in Jack’s processing. The total that Jack calculates will, therefore, be based on changing values and will *not* be the total at a particular instant. Jack calculates a partial total, Jill makes some changes, Jack continues to do more calculations in the `-doto-`, then Jill makes further changes, etc. At the end Jack has a peculiar total made up of partial totals made at different times. Even more drastic things will happen if “total” is itself a common variable: Jill might do “total←0” right in the middle of Jack’s summation!

## The TUTOR Language

If it is necessary to get an accurate total at a specific instant, it is necessary to lock out Jill and other students from modifying common until the totaling is complete. This is done by doing a “reserve common” statement before starting Jack’s calculation and a “release common” statement after the calculation is complete. The `-reserve-` command checks to make sure no other student has reserved the common, and then reserves the common. The system variable “zreturn” is set to `-1` if the `-reserve-` was able to get control of the common. Otherwise “zreturn” is set to the station number of the student who had previously reserved the common. Normally, if you can’t get the common, you loop waiting for the other person to do a “release common”:

```

      .
      .
      .
      8again
      reserve common
      branch zreturn,x,8again

```

Notice that you must reserve and release common for Jill as well as for Jack (doing it for one but not the other will not prevent the other from looking at or changing the common).

Don’t forget the “release common” for a student, or other students will get hung up waiting for the common to be available. When a student who has reserved a common signs out of the lesson, TUTOR automatically releases the common.

Note that a lock is certainly needed if different students are *storing* information into the same area of common. There is often no problem with having different students *reading* information out of the same area of common and no problem when storing information in *different* areas of common. Logical conflicts are most serious when *modifying* the *same* part of common. However, even in this case there are usually no problems. In the example of counting the number of students in the lesson, we simply execute “`vc=vc+1`”, which cannot cause any problems since all of the modifications are completed in one simple step. (Note, however, that a very complicated `-calc-` statement, particularly one involving multi-element array operations, may take more than one time-slice to be performed.)

### The `-storage-` Command

In certain applications 150 individual student variables are not sufficient, even when using segmented variables. It is possible to set up



extra storage of up to 1500 variables to give a total of 1650 variables that are *individual*, not shared in a common. A “storage 350” statement will cause a storage block of 350 variables to be set up in the swapping memory for *each* student who enters the lesson. Like `-common-`, the `-storage-` command is not “executed” (it is rather an instruction to TUTOR to set up storage when the student enters the lesson). Like temporary common, the storage variables are zeroed when the storage is set up.

A `-transfr-` command can be used to move common or storage variables from swapping memory into the student variables or into the “vc” area. Usually, however, common is loaded automatically into the “vc” area. If the common is larger than 1500 variables, a `-comload-` command must be used to specify which part of this large common is to be swapped into and out of which section of `vc1` through `vc1500`. In the case of `-storage-`, there is no automatic swapping. Instead, a `-stoload-` command is used to specify what parts of the storage are to be moved into what area of the “vc” variables. Here is a typical example:

```
common 1000
storage 75
stoload vc1001,1,75
```

The common will be automatically swapped in and out of `vc1` through `vc1000`. The 75 storage variables will be swapped in and out of `vc1001` through `vc1075`. It is good form to define all these matters:

```
define comlong=1000, stlong=75
      stbegin=vc(comlong+1)
      (etc.)
common comlong
storage stlong
stoload stbegin,1,stlong
.
.
calc stbegin<=37.4
```

While `-common-` and `-storage-` are “non-executable” commands, `-comload-` and `-stoload-` are executable, so that swapping specifications can be changed during the lesson.

The student’s current variables `v1` through `v150` are saved with other restart information when he or she signs out. Therefore, when the student signs in the next day, these variables will have the values they had when the student left. Storage variables are *not* saved, however. All storage variables are initialized to zero when the storage block is set up upon

## The TUTOR Language

entry into the lesson, as with temporary common. If it is necessary to file away more than the standard 150 student variables, you could split up a common into different pieces for individual students. For example, if you need to save 200 extra variables for no more than 20 students, you could split up a 4000-variable common into 20 pieces each containing 200 variables. An alternative is to use "dataset" operations, which permit you to directly control the transfer of blocks of individual data between the permanent storage (magnetic disks) and the swapping memory.

### Using "datasets"

A PLATO "dataset" is a file of records kept in the permanent (magnetic disk) storage. You can write some data out to the 5th record of the dataset, then get it back months later simply by reading back the 5th record of that dataset. Each record is made up of many words, and the record word size is specified at the time the dataset is created. (Currently the minimum record size is 64 words.) One record might, for example, hold exam scores for a particular student.

In order to perform operations on a dataset, you first must execute a `-dataset-` command to tell PLATO which of your datasets you are going to be working on at the moment. You can then execute any number of `-dataout-` commands to send data out to the dataset, and any number of `-datain-` commands to read such information back. You can use a `-reserve-` command to reserve specific records, similar to using a "reserve common". You must use a `-release-` command to permit others again to manipulate those records. (For details, see the PLATO on-line "aids".)

### Sorting Lists

When manipulating a data base it is often necessary to sort a list of items into alphabetic or numeric order. The `-sort-` (numeric) and `-sorta-` (alphabetic) commands will transform a disordered list into a sorted list. These commands will also sort an associated list of items at the same time. For example, you might have student names in one part of a common, and corresponding grades in another part of the common. You could use a `-sorta-` command to place the names in alphabetical order, and at the same time you could have the `-sorta-` command similarly re-order the grades to correspond with the altered order of the students. (See the PLATO on-line "aids" for details.)

## Additional Calculation Topics

10

Before discussing additional TUTOR calculational capabilities, let's review briefly those aspects which have been covered so far:

- 1) Expressions follow the rules of high school algebra. Multiplication takes precedence over division, which takes precedence over addition and subtraction. Superscripts may be used to raise numbers to powers. The symbol  $\pi$  may be used to mean 3.14159. . . . The degree sign ( $^{\circ}$ ) may be used to convert between degrees and radians.
- 2) There are 150 student variables, v1 through v150, which may be named with the `-define-` command. These variables can be set or altered by assignment ( $\Leftarrow$ ) and by `-store-`, `-storen-`, or `-storea-` commands. If a "define student" set of definitions is provided, the student may use variable names in his or her responses.
- 3) Logical expressions are composed using the operators `=`, `≠`, `>`, `<`, `≥`, `≤`, `$and$`, `$or$`, and the "not" function. Logical expressions have the value true (-1) or false (0).
- 4) There are several available system variables such as "where", "wherey", "anscnt", "jcount", "spell", etc. Available system functions include  $\sin(x)$ ,  $\sqrt{x}$ , etc. A full list of system variables and functions is given in Appendix C.
- 5) The `-show-` command (and its relatives `-showt-`, `-showz-`, `-showe-`, and `-showo-`) will display the numerical value of an

## The TUTOR Language

expression. The `-showa-` command will display stored alphanumeric information. These commands may be embedded within `-write-` and `-writec-` statements.

- 6) The `-calcc-` and `-calcs-` commands make it easy to perform (conditionally) one of a list of calculations or assignments.
- 7) The `-randu-` command with one argument picks a fraction between 0 and 1. With two arguments, it picks an integer between 1 and the limit specified. There is a set of commands associated with permutations: `-setperm-`, `-randp-`, `-remove-`, and `-modperm-`.
- 8) The iterative form of the `-do-` command facilitates repetitive operations.

Now let's look at additional TUTOR calculational capabilities.

## Defining Your Own Functions

While many important functions such as  $\ln(x)$  and  $\log(x)$  are built-in to the TUTOR language, it is frequently convenient to define your own functions. To take a simple example, suppose you define a cotangent function:

```
define cotan(a)=cos(a)/sin(a)
```

Then, later in your lesson you can write:

```
calc r<=cotan(3x+y-5)
```

and TUTOR will treat this as though you had written:

```
calc r<=[cos(3x+y-5)/sin(3x+y-5)]
```

Such use of functions not only saves typing but improves readability.

**CAUTION:** In defining a function, the *arguments* must not be already defined. For example, the following definition will be rejected by TUTOR (with a suitable error message):

```
define x=v1
      cube(x)=x3
```

This must be rewritten as:

```
define x=v1
      cube(dummy)=dummy3
```

or anything similar. A function definition may involve previously defined quantities on the *right* side of the “=” sign, however. You might have:

```
define x=v1
      new(c)=c4+2x
```

In this case you might have a -calc- that looks like:

```
calc x<-15.7
     y<-3new(8)
```

and this would be equivalent to:

```
calc x<-15.7
     y<-3[(8)4+2x]
```

Sometimes it is convenient to define “functions” that have *no* arguments:

```
define r=v1
      quad=r2-100
      r3=r1/3
      root=sqrt(r)
      prod=r3×root
      trans=(r<-prod)
```

Note that “prod” depends on two previous definitions, each of which (in turn) depends on the definition of “r”. There is no limit on how deep you can go in definition levels. The unusual definition of “trans” permits you to write an unusual -calc- (where the assignment is implicit in the definition of “trans”):

```
calc trans
```

Essentially anything is a legal definition. The only rule is that the definition make sense when enclosed in parentheses (since a defined name when encountered in an expression is replaced by its meaning and surrounded by parentheses). This means that you cannot define “minus=-” because (-), a minus sign enclosed in parentheses, is not permitted in an expression. On the other hand, “minus=-1” is all right because (-1) is meaningful.

A function may have up to six arguments. Here is a function of two arguments:

The TUTOR Language

```
define modulo(N,base)=N-[base×int(N/base)]
```

This means that modulo (17,5) in an expression will have the value 2; the “int” or “integral part” function throws away the fractional part of 17/5, leaving 3, so that we have  $(17-5 \times 3) = (17-15) = 2$ . This modulo function, therefore, gives you what is left over in division of “N” by “base”.

Here are a couple of other examples of multi-argument function definitions:

```
define big(a,b)=-[a×(a≥b)+b×(b>a)]
      small(a,b)=-[a×(a≤b)+b×(b<a)]
```

The minus sign appears because logical true is represented by -1. If you have “big(x+y,z)” in an expression, with  $(x+y)=7$  and  $z=3$ , this expands to:

$$-[7 \times (7 \geq 3) + 3 \times (3 > 7)]$$

which reduces to  $-[7 \times (-1) + 3 \times (0)]$  which is 7. So our “big” function picks out the larger of two arguments.

### Arrays

It is often important to be able to deal with arrays of data such as a list of exam scores, the number of Americans in each 5-year age group together with their corresponding mortality and fertility rates, a list of which pieces are where on a chess board, or the present positions of each of several molecules in the simulation of the motion of a gas.

Suppose we have somehow entered the exam scores for twenty students into variables v31, v32, v33 . . . up to v50. Here is a unit which will let you see the score of the 5th or 13th or Nth student:

```
unit      see
back     index
at       1215
write    Which student number?
         (Press BACK when done.)
arrow    1518
store    N
wrongv   10.5,9.5 $$ range 1 to 20
write    The score of the <s,N>th student is <s,v(30+N)>.
```



(The -wrongv- rather than -ansv- makes it easy to ask another question.)  
 The new element here is the “indexed variable”:

$$v(30+N)$$

which means “evaluate  $30+N$ , round to the nearest integer, and choose the corresponding variable”. For example, if  $N$  is 9,  $v(30+N)$  is  $v(39)$  or  $v39$ . If  $N$  is 13.7,  $v(30+N)$  means  $v44$ .

We might list and total all the scores:

```

.
.
calc  total←0          $$ initialization step
do    showem,N←1,20
at    3035
write The average score is <s,total/20>.
*
unit  showem
at    835+100N
show  v(30+N)
calc  total←total+v(30+N)
    
```

As usual, it is preferable to define a name for this data, such as:

```
define scores(i)=v(30+i)
```

in which case we would write our last unit as:

```

unit  showem
at    835+100N
show  scores(N)
calc  total←total+scores(N)
    
```

Due to the special meaning attached to “ $v(\text{expression})$ ” you must exercise some care in using a variable named “ $v$ ”, in that you must write “ $v \times (a+3b)$ ” and not “ $v(a+3b)$ ” if you mean multiplication. We will see later that the same restriction applies to the names “ $n$ ”, “ $vc$ ”, and “ $nc$ ”. This restriction does not apply to students entering algebraic responses, where “ $v(a+3b)$ ” is taken to mean “ $v \times (a+3b)$ ”. Students can use indexed variables only if they are named (as in “scores” in the above example). Such definitions must, of course, be in the “define student” set.

Suppose you have three sets of exam scores for the twenty students. This might conveniently be thought of as a  $3 \times 20$  (“two-dimensional”)

The TUTOR Language

array. Suppose we put the first twenty scores in v31 through v50, the second set in v51 through v70, and the third set in v71 through v90. It might be convenient to redefine your array in the following manner:

```
define scores(a,b)=v(10+20a+b)
```

Then, if you want the 2nd test score for the 13th student, you just refer to scores (2,13) which is equivalent to v(10+40+13) or v(63). If you wanted to display all the scores you might use "nested" -do- statements:

```
.
.
do column,i<=1,3
*
unit column
do rows,j<=1,20
*
unit rows
at 820+10i+100j
show scores(i,j)
```

Unit "column" is done three times and for each of these iterations, unit "rows" is performed twenty times.

There is an alternative way to define our array:

```
define i=v1,j=v2
scores=v(10+20i+j)
```

Then our unit "rows" would look like:

```
unit rows
at 820+10i+100j
show scores
```

The indices specifying which test is for which student are implicit. This form is particularly useful when you have large subroutines where "i" and "j" are fixed and it would be tiresome to type over and over again "scores(i,j)". Just set "i" and "j", then -do- the subroutine.

It is frequently necessary to initialize an entire array to zero. One way to do this is with -do- statements:

```
unit clear
do clear2,i<=1,3
*
```



```

unit clear2
do clear3,j<=1,20
*
unit clear3
calc scores(i,j)<=0

```

A simpler way to accomplish the same task is to say:

```
zero scores(1,1),60
```

You simply give the starting location (the first of the 60 variables) and the number of variables to be cleared to zero. As another example, you can clear all of your variables by saying:

```
zero v1,150
```

Not only is the `-zero-` command simpler to use, but TUTOR can carry out the operation several hundred times faster! TUTOR keeps a block of its own variables, each of which always contains zero. When you ask for 150 variables to be cleared, TUTOR does a rapid block transfer of 150 of its zeroed variables into your specified area. This ultra-high-speed block transfer capability can be used in other ways. For example:

```
transfr v10;v85;25
```

performs a block transfer of the 25 variables starting with v10 to the 25 variables starting with v85. In this way you can move an entire array from one place to another with one `-transfr-` command, and at speeds hundreds of times faster than are possible by other means.

## Segmented Variables

Storing three scores for each of your twenty students required the use of 60 variables, out of an available 150. We're running out of room! You can save space by defining "segmented" variables which make it easy to keep several numbers in each student variable. For example, you can write a definition of the form:

```
define segment,score=v31,7
```

This identifies "score" as an array which starts at v31 and consists of segments holding positive integers (whole numbers) smaller than 2<sup>7</sup> (which is 128). It turns out that each student variable will hold 8 such

## The TUTOR Language

segments, so “score(8)” is the last segment in v31, while “score(9)” is the first segment in v32. Since “score(60)” is the fourth segment in v38, we need only eight variables to hold all sixty scores. You can use “score(expr)” in calculations. The expression “expr” will be rounded to the nearest integer and the appropriate segment referenced. As a simple example:

```
calc score(23)←score(3)+5
```

will get the third segment, add 5 to it, and store the result in the twenty-third segment.

If we define a segmented one-dimensional array “score”, we can define a two-dimensional array as before:

```
define segment,score=v31,7
      scores(a,b)=score(20a-20+b)
```

With these definitions, “scores(1,1)” means “score (20-20+1)” or “score(1)”, which is the first segment in v31. As before, “scores” could use implicit indices:

```
define i=v1,j=v2
      scores=score(20i-20+j)
```

In this case you use “scores” rather than “scores(expr1,expr2)” in calculations. NOTE: At the present writing, the commands -zero- and -transfr- cannot be used with segmented variables because these commands refer to entire variables. You could, however, zero all of the scores by saying “zero v31,8” which sets v31 through v38 to zero, which has the effect of zeroing all the segments contained in those eight variables. You can make such manipulations more readable by defining your segmented array this way:

```
define start=v31
      segment,score=start,7
```

Then you can write “zero start,8” rather than “zero v31,8”. Similar remarks apply to the -transfr- command.

It is possible to store integers (whole numbers) that can be negative as well as positive:

```
define segment,temp=v5,7,signed
```

The addition of the word “signed” (or the abbreviation “s”) permits you to hold in “temp(i)” any integer from -63 to +63. The range 2<sup>7</sup> (128) has been cut essentially in half to accommodate negative as well as positive values. The following table summarizes the unsigned and signed ranges of integers permissible for various segment size specifications up to 30 (sizes up to 59 are allowed, though beyond 30 there is only one segment per variable).

Segment size	n	2 <sup>n</sup>	unsigned range	signed range	No. of segments per variable
1		2	0 to 1	—	60
2		4	0 to 3	-1 to +1	30
3		8	0 to 7	-3 to +3	20
4		16	0 to 15	-7 to +7	15
5		32	0 to 31	-15 to +15	12
6		64	0 to 63	-31 to +31	10
7		128	0 to 127	-63 to +63	8
8		256	0 to 255	-127 to +127	7
9		512	0 to 511	-255 to +255	6
10		1 024	0 to 1 023	-511 to +511	6
11		2 048	0 to 2 047	-1 023 to +1 023	5
12		4 096	0 to 4 095	-2 047 to +2 047	5
13		8 192	0 to 8 191	-4 095 to +4 095	4
14		16 384	0 to 16 383	-8 191 to +8 191	4
15		32 768	0 to 32 767	-16 383 to +16 383	4
16		65 536	0 to 65 535	-32 767 to +32 767	3
17		131 072	0 to 131 071	-65 535 to +65 535	3
18		262 144	0 to 262 143	-131 071 to +131 071	3
19		524 288	0 to 524 287	-262 143 to +262 143	3
20		1 048 576	0 to 1 048 575	-524 287 to +524 287	3
21		2 097 152	0 to 2 097 151	-1 048 575 to +1 048 575	2
22		4 194 304	0 to 4 194 303	-2 097 151 to +2 097 151	2
23		8 388 608	0 to 8 388 607	-4 194 303 to +4 194 303	2
24		16 777 216	0 to 16 777 215	-8 388 607 to +8 388 607	2
25		33 554 432	0 to 33 554 431	-16 777 215 to +16 777 215	2
26		67 108 864	0 to 67 108 863	-33 554 431 to +33 554 431	2
27		134 217 728	0 to 134 217 727	-67 108 863 to +67 108 863	2
28		268 435 456	0 to 268 435 455	-134 217 727 to +134 217 727	2
29		536 870 912	0 to 536 870 911	-268 435 455 to +268 435 455	2
30		1 073 741 824	0 to 1 073 741 823	-536 870 911 to +536 870 911	2

Table 10-1.

## The TUTOR Language

As an example of the use of this table, suppose you are dealing with integers in the range from  $-1200$  to  $+1800$ . You would need a segment size of 12 (signed), which gives a range from  $-2047$  to  $+2047$ . There would be 5 segments in each variable. Your `-define-` might look like:

```
define segment,dates=v140,12,signed
```

It is not necessary to understand the rationale behind this table in order to be able to use segments effectively. Explanations of the underlying “binary” or “base 2” number system and the associated concept of a “bit” are discussed later in an optional section of this chapter.

Segments are frequently used to set “flags” or markers in a lesson. For example, you might like to keep track of the topics the student has completed or which questions in a drill have been attempted. A segment size of just one is sufficient for such things, with the segment first initialized to zero, then set to one when the topic or question has been covered. The definition might look like this:

```
define flags=v2
      segment,flag=flags,1
```

In the first unit, (not the “initial entry unit”) use the statement “zero flags” to clear all sixty segments in `v2`. If you use up to 120 markers you would use “zero flags,2” to clear two variables, each containing 60 segments. When the student completes the fourth topic you use “calc flag(4) $\leftarrow$ 1” to set the fourth flag. You can retrieve this information at any time to display to the student which topics he or she has completed. Note that the `-restart-` command can be used to restart the student somewhere after the first unit (where the flags would otherwise be cleared), so that you can remind the student of which sections he or she completed during previous sessions.

Although only whole numbers can be kept in segments, it is possible to use the space-saving features of segments even when dealing with fractional numbers. Suppose you have prices of items which (in dollars and cents) involve fractions such as \$37.65 (37 dollars plus 65 hundredths of a dollar). Assume that \$50 is the highest price for an item. Simply express the prices in *cents*, with the top price then being 5000 cents. Using the table, we see that a segment size of 13 will hold positive integers up to 8191, so we say:

```
define price=v1 $$ in dollars and cents
      segment,cents=v2,13
      put(i)=[cents(i) $\leftarrow$ 100price]
      get(i)=[price $\leftarrow$ cents(i)/100]
```

A sequence using these definitions might look like:

```

calc  price←28.37
.
.
.
calc  put(16)      $$ equivalent to "cents(16)←100price"
.
.
.
show  get(16)      $$ equivalent to "price←cents(16)/100"

```

The final -show- will put "28.37" on the screen, even though between the "put" and "get", the number was the integer "2837". Notice the unusual "calc put(16)" which has an assignment (←) implicit in the definition of "put". Also notice that the variable "price" is changed as a side-effect of "get". If this is not desired, we could define "get(i)=cents(i)/100".

As another example of the use of segments with fractional numbers, suppose you have automobile trip mileages up to 1000 miles which you want to store to the nearest tenth-mile (such as 243.8 miles). In this case you must multiply by 10 when storing into a segment and divide by 10 when retrieving the information. You would use a segment size of 14, since your biggest number is 10000. It should be pointed out that rounding to the nearest integer occurs when storing a non-integer value into a segment:

```

calc  miles←539.47
      seg(2)←10 miles  $$ 5394.7 becomes 5395
      miles←seg(2)/10  $$ 5395/10 or 539.5

```

So, by going into and out of the segment, the "539.47" has turned into "539.5".

Aside from the restriction to integers, calculations with segmented variables have one further disadvantage: they are much slower than calculations with whole variables. This is due to the extra manipulations the computer must perform in computing which variable contains the Nth segment, and extracting or inserting the appropriate segment. Segments save space at the expense of time. In many cases this does not matter, but you should avoid doing a lot of segment calculations in a heavily-computational repetitive loop, such as an iterative -do- which is done ten thousand times. (There are other kinds of segments, "vertical" segments, which are handled much faster but these have quite different space requirements than regular segmented variables.)

## The TUTOR Language

## Branching Within a Unit: -branch- and -doto-

All of the branching or sequencing commands discussed so far referred to -unit-s (or -entry-s). It is often convenient to be able to branch *within* a unit, which is possible with the -branch- command:

```

unit      somethin
branch   count←4,5,x,8after
at       1215
write    "count" is equal to 4
5
do       countit
8after   count←15

```

The tag of the -branch- command is like the tag of a -goto-, except that unit names are replaced by "statement labels." These labels appear at the beginning of statements and must start with a *number* (0 through 9) to distinguish them from commands, which start with letters. A statement beginning with a label need not have any tag (as in the line above labeled "5"), but it can have a tag like that of a -calc-, as in the last statement above ("8after count←15"). In fact, a labeled statement *is* essentially a -calc- statement. As with -goto-, "x" in a -branch- means "fall through" to the next statement.

It is not permissible in a unit to label two statements with the same label (nor can you have two units with the same name in a lesson). On the other hand, since -branch- operates only *within* a unit and cannot refer to labels in other units, it is all right to use the same label in different units. (Similarly, you can use the same *unit* name in different *lessons*.) Note that -entry- is similar to -unit-, so -branch- cannot be used to branch to a label if an -entry- command intervenes.

It is often convenient to use -branch- rather than -goto-. In addition, -branch- requires less computer processing than -goto-, so that heavily computational iterations are better done with -branch- where possible. Generally speaking, about the only time you must consider the computational efficiency of one TUTOR technique compared with another is when you do a large number of iterations of some process. Unless you are making many passes through the same statements, merely write your TUTOR statements in what seems to be the simplest and most readable manner. It is a mistake to spend time worrying about questions of efficiency if the student will make only one pass through the statements.

Just as -branch- is a fast -goto- within a unit, there is a fast -doto- (analogous to the iterative -do-) for use within a unit:

```


      doto 8end,i←first,last,incr
      calc a←b×sin(5i°)
      at 100,200+2a-i
      write T
      8end
      circle 100
    
```

The tag of the `-doto-` is similar to an iterative `-do-`, but instead of naming a unit to be done repeatedly you name a statement label. For each iteration TUTOR executes statements from the `-doto-` down to the named statement label. After the last iteration is performed, TUTOR proceeds to the statement which follows the `-doto-` label (`-circle-` in the above example).


Just as it is possible to have nested `-do-` iterations, it is also possible to have nested `-doto-`s. Here is a comparison of `-do-` and `-doto-` for displaying a two-dimensional array:

-do-	-doto-
do column,i←1,3	doto 4,i←1,3
unit column	doto 4,j←1,20
do rows,j←1,20	at 820+10i+100j
unit rows	show scores(i,j)
at 820+10i+100j	4
show scores(i,j)	


This nested `-doto-` example has the structure:




Other possible structures include the following:



The TUTOR Language



Note that in each case the “inner” -doto-s are nested within the “outer” -doto-s. Here is a counter-example of a structure which is *not* permissible:



When do you use -doto- instead of an iterative -do-? Use -doto- whenever the contents of the loop are very short, because the “overhead” associated with each -doto- iteration is much less than the “overhead” associated with each -do- iteration. This is due to the extra manipulation involved in getting to the “done” unit. If the contents of the loop are long, the overhead becomes insignificant, and either -do- or -doto- can be used, whichever you prefer or whichever is more readable.

Array Operations

You have seen how to operate on individual elements of an array by using indexed variables. It is also possible to define an array in such a way as to permit operating on the array *as a whole*. Here are two sets of statements, one using true “arrays” and the other using indexed variables, with both routines calculating the sum of sixty scores (three scores for each of twenty students):

TRUE ARRAY	INDEXED VARIABLE
define total=v1	define total=v1,i=v2,j=v3
array,scores(3,20)=v31	scores(a,b)=v(10+20a+b)



```

calc   total←Sum(scores)           calc   total←0
doto   4,i←1,3                      doto   4,j←1,20
doto   4,j←1,20                    calc   total←total+scores(i,j)
calc   4                              4

```

The calculation using indexed variables involves initializing “total” to zero, then using nested -doto-s to add in each element of “scores”. The true array calculation is much simpler, involving a single -calc- statement!

The statement “define array,scores(3,20)=v31” tells TUTOR to put scores (1,1) in v31, scores (1,2) in v32, scores (1,3) in v33, etc., with scores (2,1) in v51, scores (2,2) in v52, etc. Moreover, this “array” definition permits you to work with the whole array, and there are various array functions such as “Sum” to help you. The expression “Sum(scores)” means “add up all the numbers in all the elements of the array”. Similarly, the statement “scores←scores+1” will cause *all* sixty array elements to be increased by one.

Such whole-array operations are not possible with indexed variables, because (with indexed variables) TUTOR does not know how many elements make up the whole array. On the other hand, the complexities of handling true arrays limits their size to 255 elements at present and to only two “dimensions” (that is, you can’t say “define array,points(2,5,4)=v1”, which would define a three-dimensional array). So, ordinary indexed variables do have their uses, particularly when manipulating large databases (as discussed in the next chapter). While the most useful feature of true arrays is the ability to deal with all elements at once, you can also refer to individual elements, such as scores(2,15), just as you would with indexed variables.

Suppose we define two arrays, A and B, both ten variables long:

```

define array,A(10)=v141
        array,B(10)=v131

```

The following calculations involving these arrays will have the specified results:

#### CALCULATION

#### RESULT

A←2B

Each element of A is assigned the value of two times the corresponding element of B: A(1)←2B(1), A(2)←2B(2), etc.

(Continued on next page.)

The TUTOR Language

CALCULATION (continued)	RESULT (continued)
A←25	Each element of A is set to 25.
A←1/A	Each element of A is replaced by its reciprocal.
A←A+B	Corresponding elements of A and B are added together, and the sum replaces the element of A: A(1)←A(1)+B(1), A(2)←A(2)+B(2), etc.
A←3.4cos(B)	A(1)←3.4cos(B(1)), etc.
A←B <sup>2</sup>	Presently not allowed: use A←BxB instead.
A←A \$and\$ B	Each element of A is replaced by -1 or 0, depending on a logical "and" of the corresponding elements of A and B (which should of course contain logical values, -1 and 0, to begin with).

There are a couple of special operators unique to array manipulations: A ° B gives the standard "matrix multiplication", with row-by-column multiplication and summation, and AxB gives the standard "vector product" or "cross product". If A and B are one-dimensional arrays, the matrix multiplication A ° B yields a single number, known in mathematics as the "dot product". The symbol ° is typed by means of MICRO-x, and "x" is typed by MICRO-shift-x.

There are some useful functions:

Sum(A)	Adds up all the elements of A
Prod(A)	The product of all the elements: A(1) x A(2) . . . . x A(10)
Min(A)	Picks out the smallest value
Max(A)	Picks out the largest value
And(A)	A(1)\$and\$A(2)\$and\$A(3) . . . . . \$and\$A(10)
Or(A)	A(1)\$or\$A(2)\$or\$A(3) . . . . . \$or\$A(10)
Rev(A)	Reverses the order of the elements
Transp(A)	Produces the transposed array: A(i,j)←A(j,i)

Combinations of the various operations and functions can be used to your advantage. For example, a common statistical calculation involves the square root of the sum of the squares of all array elements. This can be easily obtained from sqrt(Sum(AxA)), or from sqrt(A ° A) if A is a one-dimensional array.

Arrays can be filled with a `-set-` command and displayed with a `-showt-` command:

```

define  array,C(2,3)=v16
set     C<-100,200,300
        400,500,600
at      1215
showt   2C,5  $$ 5 figures

```

} will display

```

        200 400 600
        800 1000 1200

```

The `-set-` command fills elements in order. For example:  $C(1,1)$ ,  $C(1,2)$ ,  $C(1,3)$ ,  $C(2,1)$ ,  $C(2,2)$ ,  $C(2,3)$ . The `-showt-` (“show tabular”) command shows the numbers appropriately on the screen. You can also use `-showe-`, `-showo-`, and `-showa-` (but not `-show-` or `-showz-` at present).

It is often convenient for the array elements to be offset, so that the first element is not numbered “one”. For example, you might want an array of the world population from 1900 to 1970. In this case, simply say “`define array,popul(1900;1970)=v1`”, which assigns  $\text{popul}(1900)$  to  $v1$  and  $\text{popul}(1970)$  to  $v71$ . Note the semicolon in the `-define-`. A two-dimensional array with offsets is written “`define array,D(-3,0;5,8)=v1`”, where  $D(-3,0)$  is in  $v1$ ,  $D(-3,1)$  is in  $v2$ , etc. The last element of this array is  $D(5,8)$ .

## Integer Variables and Bit Manipulation

This section goes much more deeply into the way a computer represents numbers and character strings. You might start off by skimming this section to see whether you will need to study it in detail. You will need this material only if you pack several pieces of data in one variable or if you want to use `-calc-` operations on character strings.

A variable such as  $v150$  can hold a number as big as  $10^{322}$  (the number 1 followed by 322 zeros) or a non-zero number as small as  $10^{-293}$  (a 1 in the 293rd position after the decimal point). These huge or tiny numbers may be positive or negative, from  $\pm 10^{-293}$  up to  $\pm 10^{322}$ . Any number held in  $v150$  is recorded as sixty tiny “bits” of information. For example, whether the number is positive or negative is one bit of information, and whether the magnitude is  $10^{+200}$  or  $10^{-200}$  is another bit of information. The remaining 58 bits of information are used to specify precisely the number held in  $v150$ .

What is a bit? A bit is the smallest possible piece of information and represents a two-way (binary) choice such as yes or no, or true or false, or

## The TUTOR Language

up or down (anything with two possibilities). A number is positive or negative and these two possibilities can be represented by one bit of information. Numbers themselves can be represented by bits corresponding to yes or no. Let us see how any number from zero to seven can be represented by three bits corresponding to the yes or no answers to just three questions. Suppose a friend is thinking of a number between zero and seven and you are to determine it by asking the fewest possible questions to be answered yes or no. Suppose the friend's number is 6:

- a) Is it as big as 4?            Yes.
- b) Is it as big as 4+2?        Yes.
- c) Is it as big as 4+2+1?    No.

From this you correctly conclude that the number is 6. You determined that the number was made up of a 4, a 2, and no 1. You might also say that the number can be represented by the sequence "yes,yes,no"!

As another example, try to guess a number between zero and 63 chosen by the friend. Suppose it is 37:

- a) Is it as big as 32?            Yes.
- b) Is it as big as 32+16?        No.
- c) Is it as big as 32+8?         No.
- c) Is it as big as 32+4?         Yes.
- d) Is it as big as 32+4+2?      No.
- e) Is it as big as 32+4+1?      Yes.

So the number is 37, or perhaps "yes,no,no,yes,no,yes". Try this questioning strategy on any number from zero to 63 and you will find that six questions are always sufficient to determine the number. The strategy depends on cutting the unknown range in two each time (a so-called "binary chop").

Conversely, any number between zero and 63 can be represented by a sequence of yes and no answers to six such questions. What number is represented by the sequence

yes,yes,no,yes,no,yes?

This number must be built up of a 32, a 16, no 8, a 4, no 2, and a 1.  $32+16+4+1$  is 53, so the sequence represents the number 53.

Because a yes or no answer is the smallest bit of information we can extract from our friend, we say any number between zero (six nos) and 63 (six yeses) can be represented by six bits. If on the other hand we know the number is between zero and seven, three bits are sufficient to describe

the number fully. Similarly, numbers up to 15 ( $2^4-1$ ) can be expressed with four bits, and numbers up to 31 ( $2^5-1$ ) with five bits. Each new power of two requires another bit because it requires another yes/no question to be asked.

This method of representing numbers as a sequence of bits, each bit corresponding to a yes or no, is called "binary notation" and is the method normally used by computers. Whether a computer bit represents yes or no is typically specified by a tiny electronic switch being on or off, or by a tiny piece of iron being magnetized up or down. A TUTOR variable contains *sixty* bits of yes/no information and could therefore be used to hold a *positive integer* as big as ( $2^{60}-1$ ), which is approximately  $10^{18}$ , or 1 followed by 18 zeros. What do we do about *negative integers*? Instead of using all sixty bits we could give up one bit to represent whether the number is positive or negative (again, a two-way or binary bit of information) and just use 59 bits for the magnitude of the number. In this way we could represent positive or negative *integers* up to  $\pm(2^{59}-1)$ , which is approximately plus or minus one-half of  $10^{18}$ .

But what do we do about bigger numbers, or numbers such as 3.782 which are not integers? The scheme used on the CONTROL DATA® PLATO computer is analogous to the scientific notation used to express large numbers. For example,  $6.02 \times 10^{23}$  is a much more compact form than 602 followed by 21 zeros, and it consists of two essential pieces: the number 6.02 and the *exponent* or *power* of ten (23). Instead of using 59 bits for the number, we use only 48 bits and use 11 bits for the exponent. Of these 11 bits, one is used to say whether the exponent is positive or negative (the difference between  $10^{+6}$ , a million, and  $10^{-6}$ , one-millionth). The remaining ten bits are used to represent exponents as big as one thousand ( $2^{10}-1$  is 1023, to be precise). The exponent is actually a power of two rather than ten, as though our scientific notation for the number 40 were written as  $5 \times 2^3$  instead of  $4 \times 10^1$ . That is, instead of expressing the number 40 as  $4 \times 10^1$ , we express it as  $5 \times 2^3$ , putting the 5 in our 48-bit number and the 3 in the 11-bit exponent storage place. In this way we split up the 60 bits as:

- 1 bit for positive or negative number
- 1 bit for positive or negative exponent
- 10 bits for the power of two
- 48 bits for the number

The 48-bit number will hold an integer as big as ( $2^{48}-1$ ), which is about  $2.5 \times 10^{14}$ . If we wish to represent the number 1/4, the variable will have a number of  $2^{47}$  and an exponent of -49:

$$2^{47} \times 2^{-49} = 2^{-2} = 1/4$$

## The TUTOR Language

That is, the 48-bit number will hold a large integer,  $2^{47}$ , and the exponent or power of 2, will be  $-49$ . The complicated format just described is that used by the PLATO computer when we calculate with variables v1 through v150. It automatically takes care of an enormous range of numbers by separating each number into a 48-bit number and a power of two. This format is called "fractional" or "floating-point" format because non-integral values can be expressed and the position of the decimal point floats automatically right or left as operations are performed on the variable.

Sometimes this format is not suitable, particularly when dealing with strings of characters. The `-storea-` and `-pack-` commands place ten alphanumeric characters into each variable or "word" (a computer variable is often called a "word" because it can contain several characters). We simply split up the sixty bits of the word into ten characters of six bits each, six bits being sufficient to specify one of 64 possible characters, from character number zero to character number 63 ( $2^6 - 1$ ). In this scheme character number 1 corresponds to an "a", number 2 to a "b", number 26 to a "z", number 27 to a "0", number 28 to a "1", etc. A capital D requires *two* 6-bit character slots including one for a "shift" character (which happens to be number 56) and one for a lower-case "d" (number 4). The `-showa-` command takes such strings of 6-bit character codes and displays the corresponding letters, numbers, or punctuation marks on the student's screen.




Fig. 10-1.

Nonsensical things happen when a `-showa-` command is used to display a word which contains a floating-point number. The two sign bits (for the number and for the exponent) and the first four bits of the exponent make up the first 6-bit character code. The last six bits of the exponent are taken as specifying the second 6-bit code. Then the remaining 48 bits are taken as specifying eight 6-bit character codes. Small wonder that using a `-showa-` on anything other than character strings usually puts gibberish on the screen. On the other hand, using a `-show-` with a character string gives nonsense: the floating-point exponent is made up out of pieces of the first and second 6-bit character codes, the 48-bit number comes from the last eight character codes, and whether the number and the exponent are positive or negative is determined by the first two bits of the first character code. (See Fig. 10-1)

So far we have kept numerical manipulations (`-calc-`, `-store-`, `-show-`) completely separate from character string manipulations (`-storea-`, `-showa-`). The reasons should now be clear. It is sometimes advantageous, however, to be able to use the power of `-calc-` in manipulating character strings and similar sequences of bits. For such manipulations we would like to notify TUTOR *not* to pack numbers into a variable in the useful but complicated floating-point format. This is done by referring to "integer variables":

n1,n2,n3-----n149,n150

The integer variable `n17` is the same storage place as `v17`, but its internal format will be different. If we say `"calc v17←6"`, TUTOR will put into variable number 17 the number 6, expressed as  $6 \times 2^{45}$  with an exponent of  $-45$ , so that the complete number is  $6 \times 2^{45} \times 2^{-45}$ , or 6. If on the other hand we say `"calc n17←6"`, TUTOR will just put the number 6 into variable number 17. (See Fig. 10-2.) Since the number 6 requires only three bits to specify it, variable 17 will have its first 57 bits unused (unlike the situation when we refer to the 17th variable as `v17`, in which case both the exponent and the magnitude portions of the variable contain information).




Fig. 10-2.

## The TUTOR Language

Consider the following sequence:

```

calc    n17←6
.
.
.
at      1223
showa   n17,10

```

This will cause an “f” (the 6th letter in the alphabet) to appear on the screen at location 1223. The first 9 character codes in n17 are zero, and these zero or “null” codes have no effect on the screen or screen positioning. Indeed, a “showa n17,9” would display nothing since the “6” is in the *tenth* character slot. If we use “show n17”, we will only see a “6” on the screen. The integer format of n17 alerts -show- not to expect a floating-point format.

If we say “calc n23←5.7”, variable n23 will be assigned the value 6. *Rounding* is performed in assigning values to integer variables. If truncation is desired, use the “int” function: “n23←int(5.7)” will assign the integer part (5) to n23. Indexed integer variables are written as “n(index)” in analogy with “v(index)”.

The -showa- and -storea- commands may be used with either v-variables or n-variables. These commands simply interpret any v- or n-variable as a character string. This is the reason why we were able to use -showa- and -storea- without discussing integer variables.

It is possible to shift the bits around inside an integer variable. In particular, a “circular left shift”, abbreviated as “\$cls\$”, will move bits to the left, with a wrap-around to the right end of the variable. For example:

```

calc    n17←6 $cls$ 54
.
.
.
at      1223
showa   n17,1 $$ show one character

```

will display an “f” even though the -showa- will display only the first character, because the “6” has been shifted left 54 bit positions (9 six-bit character positions). A circular left shift of 54 may also be thought of as a right circular shift of 6 because of the wrap-around nature of the circular shift.

We have been using “n17” as an example, but we should actually be writing “inum” or some such name, where we have used a -define- to



specify that "inum=n17". For the remainder of this chapter we revert, therefore, to the custom of referring to variables (v or n) by name rather than number. Also, if we want the character code corresponding to the letter "f" we should use "f" rather than 6. For example:

```
calc inum←"f" $cls$ 54
```

is equivalent to but much more readable than:

```
calc n17←6 $cls$ 54.
```

The quotation marks can be used to specify strings of characters. For example:

```
calc inum←"cat"
```

will put these numbers in inum:

null	null	null	null	null	null	null	c	a	t
0	0	0	0	0	0	0	3	1	20

Fig. 10-3.

A "showa inum,10" will display "cat". Notice, particularly, that using quotes in a -calc- to define a character string puts the string at the *right* ("right adjusted"), whereas the -storea- and -pack- commands produce *left*-adjusted character strings. It is possible to create left-adjusted character strings by using *single* quote marks: inum←'cat' will place the "cat" in the first three character positions rather than the last three.

Let us now return to our early example of the number 37 expressed as the sequence of six bits "yes,no,no,yes,no,yes". If we let 1 stand for "yes", and 0 for "no", we might write this sequence as:

100101

which stands for:

$$(1 \times 32) + (0 \times 16) + (0 \times 8) + (1 \times 4) + (0 \times 2) + (1 \times 1) = 32 + 0 + 0 + 4 + 0 + 1 = 37$$

or even more suggestively:

The TUTOR Language

$$(1 \times 2^5) + (0 \times 2^4) + (0 \times 2^3) + (1 \times 2^2) + (0 \times 2^1) + (1 \times 2^0) = 32 + 0 + 0 + 4 + 0 + 1 = 37$$

(Note that  $2^0$  equals 1.) Writing the sequence in this way is analogous to writing 524 as:

$$(5 \times 10^2) + (2 \times 10^1) + (4 \times 10^0) = 500 + 20 + 4 = 524$$

In other words, when we write 524 we imply a “place notation” in base 10 such that each digit is associated with a power of 10:  $5 \times 10^2$ ,  $2 \times 10^1$ ,  $4 \times 10^0$ . Similarly, rewriting our yes and no sequences as 1 and 0 sequences, we find that the string of ones and zeros turns out to be the place notation in base 2 for the number being represented.

Here are some examples. ( $1001_2$  means 1001 in base 2.)

$$\begin{aligned} 1001_2 &= 2^3 + 2^0 = 8 + 1 = 9 \\ 1100_2 &= 2^3 + 2^2 = 8 + 4 = 12 \\ 110101_2 &= 2^5 + 2^4 + 2^2 + 2^0 = 32 + 16 + 4 + 1 = 53 \\ 1000001_2 &= 2^6 + 2^0 = 64 + 1 = 65 \end{aligned}$$

This base 2 (or “binary”) notation can be used to represent any pattern of bits in an integer variable, and with some practice you can mentally convert back and forth between base 10 and base 2. This becomes important if you perform certain kinds of bit manipulations.

An important property of binary representations is that shifting left or right is equivalent to multiplying or dividing. Consider these examples:

$$\begin{aligned} & \leftarrow \text{shift left 2 places} \\ 9 \text{ \$cls\$ } 2 = 1001_2 \text{ \$cls\$ } 2 = 100100_2 = 36 \\ & \text{(left shift 2 is like multiplying by } 2^2 \text{ or 4)} \end{aligned}$$

$$\begin{aligned} & \leftarrow \text{shift left 3 places} \\ 9 \text{ \$cls\$ } 3 = 1001000_2 = 72 \\ & \text{(left shift 3 like multiplying by } 2^3 \text{ or 8)} \end{aligned}$$

So, a *left* shift of N bit positions is equivalent to *multiplying* by  $2^N$ . A *right* shift of N bit positions is equivalent to *division* by  $2^N$  (assuming no bits wrap around to the left end in a  $60-N$ ). There exists an “arithmetic right shift”,  $\$ars\$, which is not circular but simply throws away any bits that fall off the right end of the word:$

$$9 \text{ sars} \$ 3 = 1001_2 \text{ sars} \$ 3 = \overset{\text{thrown away}}{\cancel{10}0}1 = 1.$$

This corresponds to a division by  $2^3$ , with truncation ( $9/2^3 = 9/8$  which truncates to 1).

A major use of the 60 bits held in an integer variable is to pack into one word many pieces of information. For example, you might have 60 “flags” set up or down (1 or 0) to indicate 60 yes or no conditions, perhaps corresponding to whether each of 60 drill items has been answered correctly or not. Or you might keep fifteen 4-bit counters in one word: each 4-bit counter could count from zero to as high as 15 ( $2^4-1$ ) to keep track of how well the student did on each of fifteen problems. Ten bits is sufficient to specify integers as large as 1023: you could store six 10-bit baseball batting averages in one word, with suitable normalizations. Suppose a batting average is .324. Multiply by a thousand to make it an integer (324) and store this integer in one of the 10-bit slots. When you withdraw this integer, divide it by a thousand to rescale it to a fraction (.324). When we discussed arrays we had exam scores ranging from zero to 100. The next larger power of two is 128 ( $2^7$ ), so we need only 7 bits for each integer exam score. Eight such 7-bit quantities could be stored in one 60-bit word.

How do you extract a piece of information packed in a word? As an example, suppose you want three bits located in the 19th of twenty 3-bit slots of variable “spack”:

$$\text{inum} \leftarrow (\text{spack} \text{ sars} \$ 3) \text{ sars} \$ 7$$





Fig. 10-4.

The TUTOR Language

The number 7 is  $111_2$  (base 2:  $4+2+1$ ), so it is a 3-bit quantity with all three bits “set” or “on” (non-zero). The \$mask\$ operation pulls out the corresponding part of the other word, the 3-bit piece we are interested in. In an expression ( $x$  \$mask\$  $y$ ), the result will have bits set (1) only in those bit positions where both  $x$  and  $y$  have bits set. In those bit positions where either  $x$  or  $y$  have bits which are “reset” or “off” ( $\emptyset$ ), the \$mask\$ operation produces a  $\emptyset$ . We could also have used a “segment” definition to split up the word into 3-bit segments.

A 4-bit mask would be 15 ( $1111_2$ ) and a 5-bit mask would be 31 ( $11111_2$ ). (Again, “segment” definitions of 4 or 5 bits could be used.) You might even need a mask such as  $11\emptyset111_2$  (or 55) which will extract bits located in the five bit positions where  $11\emptyset111_2$  has bits set. There should be a simpler way of writing down numbers corresponding to particular bit patterns. Certainly, reading the number 55 does not immediately conjure up the bit pattern  $11\emptyset111_2$ !

A compact way of expressing patterns of bits depends on whether or not each set of three bits can represent a number from  $\emptyset$  to 7:



Just as each digit in a decimal number (base 10) runs from  $\emptyset$  to 9, so do the individual numerals run from  $\emptyset$  to 7 in an octal number (base 8). Octal numbers are useful only because they represent a compact way of expressing bit patterns. With practice, you should be able to convert between octal and base 2 instantaneously, and between base 8 and base 10 somewhat slower! See the table below.

base 10	base 8		base 2
0	0		0 or 000
1	1		1 or 001
2	2		10 or 010
3	3		11 or 011
4	4		100
5	5		101
6	6		110
7	7		111

These should be memorized

base 10 (continued)	base 8 (continued)	base 2 (continued)
8	10	1000
9	11	1001
10	12	1010
11	13	1011
12	14	1100
13	15	1101

The conversion between base 8 and base 2 is a matter of memorizing the first eight patterns, after which translating  $1101011011101_2$  to octal is simply a matter of drawing some dividers every three bits:

$$\begin{array}{c|c|c|c|c} 1 & 101 & 011 & 011 & 101 \\ \hline 1 & 5 & 3 & 3 & 5 \end{array} = 15335_8$$

What is  $15335_8$  in base 10?

$$\begin{array}{c|c|c|c|c} 8^4 & 8^3 & 8^2 & 8^1 & 8^0 \\ \hline 4096 & 512 & 64 & 8 & 1 \end{array}$$

$$1 \quad 5 \quad 3 \quad 3 \quad 5 = 1 \times 4096 + 5 \times 512 + 3 \times 64 + 3 \times 8 + 5 = 5853_{10}$$

How about the octal version of the number 79? The biggest power of 8 in 79 is  $8^2$  (64), and 79 is 15 more than 64. In turn, 15 is  $1 \times 8^1 + 7 \times 8^0$ , so:

$$79_{10} = 1 \times 64 + 1 \times 8 + 7 \times 1 = 1 \times 8^2 + 1 \times 8^1 + 7 \times 8^0 = 117_8$$

Luckily, in bit manipulations the conversions between base 2 and base 8 are more important than the harder conversions between base 8 and base 10.

To express an octal number in TUTOR, use an initial letter "o":

`x $mask$ o37`

will extract the right-most 5 bits from x, because  $o37 = 37_8 = 011111_2$ , which has 5 bits set. Naturally, a number starting with the letter "o" must not contain 8's or 9's.

You can display an octal number with a `-showo-` command (show octal):

`showo 39`

The TUTOR Language

will display “0000000000000000000047” on the screen ( $39_{10}=47_8$ ). The default format is twenty (3-bit) octads, corresponding to a whole 60-bit word:

showo 39,4

will display “0047”, showing just four octads.

Now that we have discussed the octal notation, it is possible to point out what happens to negative numbers:

showo -39

will display “77777777777777777730”. A negative number is the “complement” of the positive number (binary 1’s are changed to 0’s and binary 0’s are changed to 1’s). In octal, the complement of 0 is 7 ( $000_2 \rightarrow 111_2 = 7_8$ ), and the complement of 7 is 0. In the example shown, octal  $47_8$  is  $100111_2$ , whose complement is  $011000_2$ , or  $30_8$ . Notice that the *left*-most bit (the “sign” bit) of a negative number is always set. In order for a negative number to stay negative upon performing an “arithmetic right shift”, all the left-most bits are set. So,

o40000000000000000003242 \$ars\$ 6

yields:

o7740000000000000000032.

Only the sign bit was set among the left-most bits before the shift (o40 is  $100000_2$ ), but after the shift the first seven bits are all set. The “circular left shift”, \$cls\$, does not do anything special with the sign bit.

It is interesting to see the bits set for floating-point numbers:

```

.
.
.
.
calc   v1←3
at     1215
write  pos=<o,v1>   $$ o for -showo-
       neg=<o,-v1>

```

will make this display:

```

pos = 17216000000000000000
neg = 6057177777777777777

```

Note that the negative number is the complement of the positive. The 48-bit magnitude (6000000000000000) represents a huge integer ( $6 \times 2^{45}$ ). The eleven bits between the sign bit and the 48-bit magnitude give the power of two ( $-46$ ) by which the magnitude is to be scaled ( $3 = 6 \times 2^{45} \times 2^{-46} = 6 \times 2^{-1} = 3$ ). A bias of  $2000_8$  is added to the correct exponent ( $-46$ , or  $-56_8$ ) to give an eleven-bit exponent of  $1721_8$ . Exponents less than  $2000_8$  represent negative powers and exponents greater than  $2000_8$  represent positive powers.

We have encountered octal numbers (e.g.,  $o327$ ) which can be shifted left ( $\$cls\$$ ) and right ( $\$ars\$$ ) and complemented (by making them negative). Pieces can be extracted with a  $\$mask\$$  operation. Additional bit operations are  $\$union\$$ ,  $\$diff\$$ , and “bitcnt”. The “bitcnt” function gives the number of bits set in a word:  $bitcnt(o25)$  is 3, because  $o25$  is  $010101_2$ , which has 3 bits set;  $bitcnt(-o25)$  is 57, since the complement will have only 3 of 60 bits *not* set; and  $bitcnt(0)$  is 0. Like  $\$mask\$$ ,  $\$union\$$  and  $\$diff\$$  operate on the individual bit positions, with all 60 done at once:

- x  $\$mask\$$  y produces a 1 only where *both* x and y have 1's.
- x  $\$union\$$  y produces a 1 where *either* x or y or *both* have 1's.
- x  $\$diff\$$  y produces a 1 only where x and y *differ*.

Note that  $\$union\$$  might be called “merge”, since 1's will appear in every bit position where either x or y have bits set. The  $\$diff\$$  operation might also be referred to as an “exclusive” union, since it will merge bits except for those places where *both* x and y have bits set.

While  $\$mask\$$  can be used to extract a piece of information from a word, a  $\$mask\$$  that includes all *but* that piece followed by a  $\$union\$$  can be used to insert a new piece of information.

These bit operations can be used with arrays. For example, if A, B, and C are true arrays, the statement “ $C \leftarrow A \$diff\$ B$ ” will replace each element of C by the bit difference of the corresponding elements of A and B.

## Byte Manipulation

The most common use of bit manipulations is for packing and unpacking “bytes” consisting of several bits from words each of which contain several bytes. This can lead to major savings in space. If an exam score lies always between 0 and 100, only seven bits are required to hold each score, since  $(2^7 - 1)$  is 127. Another way to see this is to write the largest 7-bit quantity:  $1111111_2 = 177_8 = 1 \times 8^2 + 7 \times 8^1 + 7 \times 8^0 = 64 + 56 + 7 = 127$ . This is one less than  $200_8$ , which requires an eighth bit. We can fit

## The TUTOR Language

eight 7-bit bytes into each 60-bit word. Happily, TUTOR will do the bookkeeping, as we saw earlier:

```
define segment,scores=n31,7
```

This definition makes it possible to work with this “segmented” array as though it were an ordinary array:

```
calc ss<-scores(3)
      scores(17)<-83
      etc.
```

These refer to the 3rd and 17th bytes. The first eight 7-bit bytes reside in n31, with the last 4 bits unused. The next eight bytes are in n32, etc. The 17th byte is the first 7-bit byte in n33.

Just as it is possible to give up one bit of a 60-bit word in order to have negative as well as positive numbers, so it is possible to have both positive and negative numbers stored in a segment array:

```
define segment,temp=v52,8,signed
.
.
.
.
calc temp(23)<-95
```

With 8-bit bytes we can have numbers in the range of  $\pm 127$ . The word “signed” may be abbreviated by “s”.

Now that you understand the bit structure of a variable, you should be able to understand the table (Table 10-1) provided earlier of segment ranges and the number of segments per variable. Look at the table now and see whether you can check the entries in the table.

## Vertical Segments

We might call the segments discussed so far “horizontal” segments (the segments move horizontally across each word). It is possible to define “vertical” segments (each of which occupies only part of a word): successive segments are found in the same position in successive words, rather than in different positions within the same word. As an example, “define segmentv,left=n51,1,30” defines vertical segments each occupying the left half of words n51, n52, n53, etc. Each segment



starts in bit position 1 of each word, and each segment is 30 bits long. The right halves of the words could be specified with “define segmentv, right=n5,31,30”, whose elements begin in the 31st bit position and are 30 bits wide. An “s” can be added to denote signed segments, as with horizontal segments.

Aside from the intrinsic usefulness of this kind of segmenting of words, the simpler structure permits TUTOR to process vertical segments much faster than horizontal segments, and only slightly slower than normal whole-word variables.

You can save space with true arrays by putting the elements in vertical segments. The -define- statement looks like “define arraysegv, A(10)=n5,3,12,s”. This example defines a ten-element array, with A(1) represented by a 12-bit signed segment starting in bit position 3 of n5. It is not yet possible to define a true array in horizontal segments.

### Alphanumeric to Numeric: The -compute- Command

The -store- command analyzes the judging copy of the student’s response character string and produces a numerical result. This is actually a two-step process. First, the character string is “compiled” into basic computer instructions and then these machine instructions are “executed” to produce the numerical result. During the compilation process the “define student” definitions and the built-in function definitions (sin, cos, arctan, etc.) are used to recognize the meaning of names appearing in the character string. Numbers expressed as alphanumeric digits are converted to true numerical quantities. For example, the character string 49 becomes a number by a surprisingly indirect process. The character code for “4” is 31 since “z” is 26, “0” is 27, etc. The character code for “9” is 36. The number expressed by typing 49 is obtained from the formula:

$$\begin{aligned} &10(31-27)+(36-27) \text{ or } 10("4"-“0”) + (“9”-“0”) \\ &10(4)+(9) \\ &40+9 \\ &49 \end{aligned}$$

For these and similar reasons, the compilation process is ten to a hundred times slower than the execution process. Therefore, TUTOR attempts to compile the student’s response only once, while the resulting machine instructions may be used many times.


## The TUTOR Language

The first `-store-`, `-ansv-`, `-wrongv-`, `-storeu-`, `-ansu-`, or `-wrongu-` command encountered during judging triggers compilation. All these commands following the first one simply reuse the compiled machine instructions. If a `-bump-` or `-put-` makes any changes in the judging copy, a following `-store-` or related command will have to recompile. Similarly, a “judge rejudge” will force recompilation by any of these commands. Note that re-execution is *always* performed even if recompilation isn’t, because the student might refer to defined variables whose values have been altered.

While `-store-` will compile and execute from the judging copy, the *regular* `-compute-` command will compile and execute from *any* stored character string:

```
compute result,string,#characters,pointer
```

For example:



After compilation, the “pointer to machine instructions” contains the location of the machine instructions in a special `-compute-` storage area. You must zero the pointer at first to force compilation. TUTOR will then set the pointer appropriately, so that re-executions of the `-compute-` command can simply re-execute the saved machine instructions. Here is a unit which permits the student to plot functions of interest to him or her.

```

define      student
            x=v1
define      ours,student
            result=v2,string=v3,point=v35
origin      100,250
bounds      0,-200,300,200
scalex      10
scaley      2
*
unit        graph
next        graph

```




Fig. 10-5.

back	graph	
axes		\$\$ display the axes
labelx	1	
labeledy	0.2	
at	3105	
write	Type a function of x:	
arrow	where+2	
storea	string,jcount	
ok		
calc	x←point←0	
compute	result,string,jcount,point	
goto	formok,x,badform	
gat	0,result	\$\$ draw from here
doto	8plot,x←.1,10,.1	
compute	result,string,jcount,point	
goto	formok,x,badform	
gdraw	;x,result	
8plot		
*		
unit	badform	
at	3207	
writec	formok, . . .	\$\$ tell what's wrong
judge	wrong	

Different functions can be superimposed by changing the response instead of pressing NEXT or BACK. The first -compute- in this unit calculates the value of the student's function for x equal to zero. The -gat-command positions us at location (0, result) so that the first -gdraw- will draw a line starting at that point. The system variable "formok" has the value -1, if compilation and execution succeed; 0 if compilation succeeds but execution fails (due to such errors as trying to take the square root of a negative number); and various positive integral values for various compilation errors (missing parentheses, unrecognized variable names, etc.).

Note that predefined functions can be more easily plotted with a -funct- command. For example, the student could specify a value for "n", and you could plot a polynomial simply by using "funct x^n,x←0,10,.1". But, you must use -compute- if the student is permitted to try arbitrary functions of his or her own choosing.

As another example, the PLATO lesson "grafit" (written by this author) permits the student to write up to fifteen statements in the grafit

## The TUTOR Language

language and execute his or her program to produce graphical output (as seen in Fig. 10-6):




Fig. 10-6.


This student's program calculates the motion of a mass oscillating on the end of a non-standard spring. The two curves are the superposition of running the program twice with different values of the parameters. The heart of this lesson is a loop through a -compute- command with string, character count, and point all being indexed variables. The index is the line number, from 1 to 15. Each student response is analyzed using a -match- command looking for keywords such as "goto". Then the rest of the response is filed away with a -storea- into the string storage area corresponding to that line number. The 15 pointer variables are zeroed in the "ieu" (initial entry unit) to insure that when the student returns to a PLATO terminal after several days TUTOR won't be confused over whether the strings have been recently compiled or not. Also, whenever the student changes one of his or her statements, the corresponding pointer is zeroed in order to force recompilation of the altered character string. The student can press DATA to initialize parameters, LAB to specify what variable to plot against what variable, and HELP for a description of the grafit language. The student define set defines all 26 letters as variables the student can use.

Note that even though *s*, *i*, and *n* have been defined in the student define set, the student can use the “sin” function. The reason that the student’s “sin” is not interpreted as  $s \times i \times n$  is that TUTOR looks for the longest possible name in a string of characters typed by the student. One difference between the handling of student expressions and author expressions is that students cannot reference system variables such as “where”, “anscnt”, or “data” (the numerical value of the DATA key). If you want the student to be able to use “where”, define it in the student define set as “where=where”. While authors are discouraged from using primitive names such as *v47* (except in a -define- statement), students are not permitted to use primitives at all. This is done to protect the author’s internal information. Similarly, students cannot use the assignment symbol ( $\Leftarrow$ ), except in a -compute-, unless there is a “specs okassign”.

It should be mentioned that while -compute- converts alphanumeric information into a numerical result, there is an -itoa- command that can be used to convert an integer to an alphanumeric character string. Most often, however, the -pack- command with embedded -show- commands will be used to convert non-integer as well as integer values to the corresponding character strings.

## The -find- Command

The -search- command discussed in Chapter 8 is character-string oriented and will locate ‘dog’ even across variable or word boundaries: the “d” might be at the end of one word and the “og” at the beginning of the next word. The -find- command, in contrast, is word oriented. It will find which word contains a certain number or character string:



If *n1* contains 372, *n125* will return the value 0; if *n2* is the first word which contains 372, *n125* will be 1; etc. If none of the 50 words contains 372, *n125* will be set to -1. Notice that in -search- the return is 1, not 0, if the string is found immediately. This is due to the fact that in character strings we start numbering with character number 1. On the other hand, here the first word is  $n(1+0)$ .

Do not use *v*-variables in the first two arguments of -find- because -find- makes its comparisons by integer operations. The first argument can be a character string such as ‘dog’ or ‘dog’. You can look at every 3rd word by specifying an optional increment:

## The TUTOR Language

```
find "cat",n1,50,n125,3
                                }
                                optional
```

This will look for "cat" in n1, n4, n7, etc., and n125 would be returned 0, or 3, or 6, etc. Negative increments can be used to search backwards from the end of the list.

You can also specify that a "masked equality search" be made:

```
find "cat",n1,50,n125,1,o777700
                                }
                                mask
                                not optional
```

In this case, n125 will be zero if [(n1 \$diff\$ "cat") \$mask\$ o777700] is zero. The mask specifies that only a part of the word will be examined. The increment must be specified, even if it is one, to avoid ambiguity.

There is a `-findall-` command which will produce a list of all of the locations where something was found, rather than producing locations one at a time.

### The `-exit-` Command

Suppose you are seven levels deep in `-do-`s. That is, you have encountered seven nested `-do-` statements on the way to the present unit. The statement `"exit 2"` will take you out two levels. The next statement to be executed is the statement which follows the sixth `-do-`. A blank `-exit-` command (blank tag) takes you immediately to the statement following the first `-do-`. (Such operations are occasionally useful.) Notice that encountering a unit command at the end of a done subroutine will cause an automatic `"exit 1"`. It is superfluous to put `"exit 1"` at the end of a unit, since this effect is automatic.

## Additional Display Features

9

### More on the -write- Command

It should be pointed out that the -at- command not only specifies a screen position for subsequent writing but also establishes a left margin for “carriage returns” (CR on the keyset), much like a typewriter. Upon completion of one line of text, the next line will start at the left margin set by the last -at- command. There are carriage returns implicit in “continued” write statements:

```
at    1215
write Now is the
      time for all
      good men to
      come home.
```

The “at 1215” establishes a left margin at the 15th character position so that each line will start there. This example will produce an aligned screen display similar to the appearance of the tags of this continued -write- statement.

The setting of a margin by -at- has an unusual side effect. Consider:

```
at    2163
write The cow jumped.
```

## The TUTOR Language

This will put the following display on the screen:

```
Th
e
co
w
ju
mp
ed
.
```

This unusual display is caused by the setting of the left margin at character position 63, just two characters shy of the right edge of the screen. When a `-write-` would go past the right edge of the screen, TUTOR performs a carriage return to drop down one line, starting at the left margin. An `-arrow-` also sets a left margin with respect to the student typing a long response which would pass the right edge of the screen. Further typing appears on the next lower line starting at the margin set by `-arrow-`.

Occasionally, it is useful to position something on the screen without setting a margin. This can be done with an `-atnm-` command (“at with *no margin*”). The statement “`atnm 1215`”, is equivalent to “`at 1215`”, but does not change the current margin setting.

It is important to understand that writing characters on the screen automatically advances the terminal’s current screen position. Suppose we have consecutive `-write-` statements:

```
at    712
write horses
write and cows
```

This sequence will display “horseand cows” all on line 7. The first `-write-` (“horses”) advances the terminal’s screen position from the 712 specified by the preceding `-at-` to  $712+6=718$  (there being 6 characters in the text “horses”). Without an explicit `-at-` to change this, the second `-write-` (“and cows”) starts at position 718. Note that:

```
at    712
write horses
      and cows
```

would give a different display:

```
horses
and cows
```




because the “continued” -write- statement implies carriage returns.

TUTOR keeps track of the current screen position in a system variable named “where”. For example:

```

at      712
write  horses
at      where+305  $$ “where” is 712+6=718 here
write  and cows
    
```

will produce the display:



The statement “write horses” leaves the screen position at  $712+6=718$ , and the system variable “where” therefore has the value 718. When you then say “at where+305” this is equivalent to saying “at 718+305” or “at 1023”.

There are many uses of this “where” system variable. Here is another example:

```

at      1215
write  What is your name?
arrow  where+3
    
```

This will appear as:

What is your name? > Sam

The arrow has been positioned 3 characters beyond the end of the -write-statement’s display.

The positioning information is useful with other display commands as well. Consider this:

## The TUTOR Language

```

at      815
write  Look at this!
draw   where;815

```

This will display underlined text:

Look at this!

This is due to the fact that upon completion of the -write- statement, “where” refers to the beginning of the next character position just after the exclamation point. We simply draw from there back to the starting point. This form of the -draw- statement is so common that a concise form is permitted. For example, “draw ;815” is equivalent to “draw where;815”. Either form will draw a line or figure starting at the current screen position. This is particularly useful in constructing a graph (by connecting the new point to the last point with a line). The point reached with a -draw- (or *any* display command) will be the new screen position and may be referred to through the system variable “where”, which is kept up to date automatically by TUTOR.

There are fine-grid system variables “wherex” and “wherey” which correspond exactly to the coarse-grid “where”. The position “where+305” is equivalent to “wherex+(5×8),wherey-(3×16)” because a character space is 8 dots wide and 16 dots high. The minus sign is present because, in coarse grid, line 4 is *below* line 3, whereas in fine grid dot 472 is *above* dot 471.

Superscripts and subscripts may be typed either in a locking or nonlocking mode. To type “10<sup>23</sup>” you can either: (a) press 1, press 0, press SUPER, press 2, press SUPER, press 3 (non-locking case); or (b) press 1, press 0, press shift-SUPER (that is, hold down the shift key while pressing SUPER), press 2, press 3. To get down from a locked superscript you type shift-SUB (locking subscript). Notice that in typing superscripts or subscripts the SUPER and SUB keys are pressed *and released* before typing the material to be moved up or down. You do *not* hold these keys down while typing, unlike the shift key used for making capital letters.

It is possible to overstrike characters to make combinations. The symbol “v” can be made by typing v, backspace, SUPER, minus sign. This will superimpose a raised minus sign above the v. The backspace is typed holding down the shift key while hitting the wide space bar at the bottom of the keyset. Similarly, “horse” can be typed by typing “horse” followed by five backspaces and five underline characters. Note that these superpositions of characters won’t work in “mode rewrite”, where a new character is written on the screen. In mode rewrite, the last example would show up as “\_\_\_\_\_”, the “horse” having been wiped out by the characters whose only visible dots are the low, horizontal bars.

## Extensions to the Basic Character Set

We've seen examples of lower-case and upper-case characters, numbers, punctuation marks, superscripts, and subscripts. What if you need special accent marks, or an unusual mathematical symbol, or the entire Cyrillic alphabet for writing Russian? It is important that you be able to write text on the screen using the special symbols of your particular subject area. In addition, it is possible to use special characters to display small, intricate figures whose display would be slow and cumbersome if done with `-draw-` commands.

The PLATO terminal has 126 built-in characters (including those used so far) and storage for 126 additional characters which can be different in every lesson. For example, Russian lessons fill this additional character storage space with the Cyrillic alphabet, whereas there is a genetics lesson which fills the storage area with fruitfly parts which permit displaying flies by writing appropriate characters at appropriate positions on the screen. We will learn how to access all 252 characters (126 which are built-in and 126 which can be varied).

The 126 built-in characters include many useful symbols which do not appear on the keyset (since there aren't enough keys). This is due to the fact that the keys on the right of the keyset are reserved for various important functions (`ERASE`, `BACK`, `STOP`, etc.). In order to access the "hidden" characters it is necessary to first strike the `ACCESS` key (presently the `shift-□` key) and then to strike a second key. Like `SUPER` and `SUB`, the `ACCESS` key is not held down but struck. You can press `ACCESS`, then "a" to get a Greek alpha; `ACCESS-b` for beta; `ACCESS-m` for mu; `ACCESS=` for  $\neq$ ; and also `ACCESS-<or>` for  $\leq$  and  $\geq$ . It is useful to try `ACCESS` followed by every key (or shifted key) at a terminal to find approximately 36 useful hidden characters. In most cases, there is a mnemonic connection between the key which follows the `ACCESS` key and the hidden character which results, such as  $\neq$  being `ACCESS=`. `ACCESS` followed by comma gives the symbol  $\updownarrow$  mentioned in the discussion of the `-writec-` command in Chapter 6. `ACCESS-Ø` and `ACCESS-1` give the symbols  $\langle$  and  $\rangle$  used for embedding `-show-` commands in `-write-` statements. (In the discussion of "micro tables" later in this chapter, we will see that the `MICRO` key is equivalent to the `ACCESS` key, under normal circumstances.)

You can get at the "alternate font" of 126 additional, modifiable characters by pressing the `FONT` key (the shifted `MICRO` key), then typing regular keys, which will produce characters from the alternate font. Which characters appear depends on what character set has been previously loaded into the terminal. The `FONT` key toggles you between the standard built-in font and the alternate font (you stay in the alternate

## The TUTOR Language

font until you strike FONT to return to the standard font). It is, therefore, not necessary to strike FONT for each symbol (unlike the way ACCESS works).

Here is an example of the use of a special character set:

```
at      912
write  Now LOADING CHARACTER SET.
       Please be patient - loading
       takes about 17 seconds.
charset charsets,russian
erase  $$ full-screen erase to remove message
unit   intro
at     905
write  The Russian word карандаш means pencil.
```

Fig. 9-1.

The `-charset-` statement sends to the terminal the character set specified in the tag (character set "charsets,russian" in this case). Character patterns are transmitted to the terminal at a rate of 7.5 character patterns per second, so a full 126-character set will take about 17 seconds to send. Precede the `-charset-` command with a `-write-` statement to explain this delay to the student, so that he or she will not think that something is wrong or broken! The full-screen `-erase-` will remove the message upon completion of the loading process. Once the character patterns have been

loaded into the terminal, it is possible to write Russian text on the student's screen at the same high speed as English, 180 characters per second, which corresponds to a reading speed of almost two thousand words per minute.

TUTOR keeps track of which character set has been loaded into the terminal and skips a `-charset-` statement if loading is not required. In the above example, TUTOR would rush right through the message, skipping the `-charset-` and erasing the screen. There would not be the 17-second delay which occurs if the Cyrillic characters have not been loaded.

The `-write-` statement in unit "intro" is created by:

1. typing "write The Russian word";
2. striking the FONT key to select the alternate font;
3. typing the keys k, a, r, a, n, d, a, w (which causes карандаш to appear)
4. striking the FONT key to toggle back to the standard font
5. typing " means pencil."

Each character in the alternate font is associated with a key on the keyset. For example, the creators of the "russian" character set chose to associate the Cyrillic "д" with the "d" key because of the phonetic similarity of these two letters. Similarly, the Cyrillic "р" and "н" sound like the "r" and "n" letters with whose keys they are associated. Just as accessing some of the 126 built-in characters requires the ACCESS key, so a full 126-character alternate font will also necessitate the use of the ACCESS key to reach some of the characters.

If the student is to respond at an `-arrow-` with a Russian response, he or she must hit the FONT key in order to do so. Usually it is preferable to precede the first judging command with the statement "force font", which essentially hits the FONT key for the student. The student merely uses the regular typing keys, but the typing appears in the alternate font. Some languages, including Arabic, Hebrew, and Persian, are written right-to-left instead of left-to-right. For these languages use a "force font,left" and the student's typing will automatically go leftwards from the `-arrow-` in the alternate font.

### The "initial entry unit" (ieu)

You may have noticed that the first few statements of the previous example (which write a message, load a character set, and then erase the screen) are not preceded by a `-unit-` statement. This is intentional.

## The TUTOR Language

TUTOR statements which precede the first `-unit-` statement (“unit intro” in this case) constitute an “initial entry unit” which is performed whenever a student enters the lesson. The “initial entry unit” (or “ieu”) is the logical place to put various kinds of initializations, such as a `-charset-` statement to load characters which will be used throughout the lesson. Although `-define-`, `-vocabs-`, and `-list-` statements are not actually *executed* (they are only instructions to TUTOR on how to interpret `-calc-`, `-concept-`, and `-answer-` statements in preparing a lesson for student use), they can also be placed in the “ieu” at the beginning of the lesson, for the sake of readability.

The importance of the “ieu” lies in the fact that it is performed no matter where the student starts within the lesson (even if the student does not start at the first unit statement). TUTOR is capable of keeping track of a student’s place within a lesson, so that a student who leaves without finishing a lesson is able to restart the next day where he or she left off. It is important, in the restarting process, to load the appropriate character set. The restart procedure can *not* be executed properly if the `-charset-` statement comes after the first `-unit-` statement (since the student will not go through the first part of the lesson again).

Suppose the student is to restart in unit “middle”, which looks like this:

```
unit middle
next mid2
```

The “ieu” is utilized in such a way that TUTOR acts as though the “ieu” were done at the beginning of the restart unit:

```
unit middle
(do "ieu")
next mid2
```

This pseudo-do is the reason for following the `-charset-` statement with a full-screen erase. We don’t want the “loading” message to interfere with the display to be created by unit “middle”.


### Smooth Animations Using Special Characters

The `-charset-` command is not limited to its use with foreign alphabets. Special characters are often used to create pictures:

```
at 1319
write This  uses special characters!
```

The car is composed of several adjacent characters. Because characters can be drawn very fast (180 per second), dramatic animations are possible:

```

mode  rewrite
do    drive,x←100,400
*
unit  drive
at    x,200
write 

```

The car advances one dot at a time. If the car characters are designed in such a way as to leave a vertical column of blank dots at the back of the car, the “rewrite” mode will insure that the advancing car simultaneously erases its old position. If two columns are left blank, the car could be advanced two dots at a time and still completely wipe out the previous car display. This type of animation can run as fast as twenty or thirty moves per second, which creates the illusion of a smoothly moving object.

For the built-in characters there is an expandable and rotatable (but slow) line-drawn form available through the use of `-size-` and `-rotate-`, but these commands have no effect on charset characters. If a larger or rotated car is needed, it can be constructed with `-draw-` and `-circle-` commands, built up out of additional special characters, or produced with “lineset” characters. A lineset is like a charset, but the characters are made up of lines instead of dots. If “size” is not zero, and a lineset is in effect, alternate-font text is displayed as line-drawn characters which can be expanded and rotated.

## Creating a New Character Set

Figure 9-2 on the following page demonstrates how a special character is designed at a PLATO terminal. The author moves the cursor on an  $8 \times 16$  grid to specify which dots are to be lit. The author can inspect “in the small” the appearance of the character he designs “in the large”. The letter shown at the top of the page is the key with which this character will be associated when typing in the alternate font, just as character “д” is associated with key “d” in “charset russian”. The character pattern is stored in such a way that the author can (at any later time) recall the pattern and modify it. A character set can contain up to 126 special characters or as few as one or two characters.

## The TUTOR Language




Fig. 9-2.

Figure 9-3 shows how an author can create several 8 × 16 characters at once to be used together or separately. This option is particularly helpful when designing character-mode pictures.




Fig. 9-3.



Your own character set will be stored in an electronic storage area assigned to you. Such storage areas are called “lesson spaces” because they mainly hold TUTOR statements which describe a lesson to be administered to students by PLATO. Your lesson space might be called “italian3” and it is by this name that you refer to the lesson space when you want to look at the TUTOR statements or change them. Within this lesson space you can also have one or more character sets, which you will have named. Suppose in lesson space “italian3” you have stored a character set named “rome”. In this case, the TUTOR statement used to transmit this character set to a terminal is:

```
charset italian3,rome
      ^           ^
      |           |
lesson space   character set
```

The same format holds for linesets.

## Micro Tables

It is sometimes desirable to associate a string of several characters with a single key. For example, the symbol  $\bar{v}$  may be produced by  $v$ , backspace, superscript, minus sign. It is possible to set up a “micro table” so that  $\bar{v}$  may be produced simply by hitting the MICRO key followed by hitting “v”. Similarly, the micro table might specify that MICRO-e should be equivalent to typing e, shift-SUPER, k, x, SUPER, 2, shift-SUB to make  $e^{kx^2}$ . The micro table makes possible a kind of shorthand which can be useful both to authors composing -write- statements and to students typing complicated responses.

Like character sets, micro tables reside in lesson spaces. If lesson space “italian3” contains a micro table named “dante”, these micros can be made available to students by the statement:

```
micro italian3,dante
```

As with -charset-, the -micro- statement should be placed in the “ieu” (initial entry unit).

Figure 9-4 on the following page shows how an author defines an item in a micro table, by associating a string of characters with a particular key. Later the effect of striking MICRO followed by this key is *identical* to typing this string of characters. With a “force micro” in effect, the student does not even have to press MICRO. This makes it easy to redefine the keyboard.

## The TUTOR Language




Fig. 9-4.

If you do not specify your own micro table, a standard one is provided that lets you use the MICRO key as though it were the ACCESS key. For example, MICRO-p gives ACCESS-p, which is  $\pi$ . This means you can (and should) mention only the MICRO key to students in your typing directions to them. It is not necessary to mention ACCESS. Note, however, that ACCESS-p must be used to make a  $\pi$  if you have your own micro table with a different definition for MICRO-p.

### The Graphing Commands: Plotting Graphs with Scaling and Labeling

You may often want to plot a horizontal or vertical bar graph or other kinds of graphs to display relationships. There exists a group of TUTOR commands which *collectively* make it very easy to produce such displays. In particular, scaling of your variables to screen coordinates is automatic, as is the numerical labeling of the axes, with tick marks along the axes. Figure 9-5 shows some examples.

Suppose you want a graph to occupy the lower half of the screen. The horizontal x-axis should run from zero to ten and the vertical y-axis from zero to two. Both axes should be labeled appropriately. These statements will make the display shown in Figure 9-6.




Fig. 9-5.

unit	setup	
gorigin	50,50	\$\$ x,y graph origin
axes	400,150	\$\$ lengths in dots
scalex	10	\$\$ maximum x
scaley	2	\$\$ maximum y
labelx	2,.5	\$\$ major mark every 2,
*		minor every .5
labely	.5	\$\$ major mark every .5
graph	6,1.5,A	\$\$ x=6, y=1.5
graph	8,.5,BC	\$\$ x=8, y=.5
hbar	3,1.5	\$\$ horizontal bar to
*		3,1.5
vbar	4.5,1	\$\$ vertical bar to
*		4.5,1
gdraw	2,.5;4,1.5;7,0	
gat	4,2	
write	Top	




Fig. 9-6.

After specifying -gorigin- and -axes- in terms of fine-grid screen coordinates, the -scalex- and -scaley- commands associate scale values with the end points of the axes. These scale values determine how (x,y) coordinate positions given in later statements will be scaled to screen coordinates.

The TUTOR Language

The `-labelx-` and `-labeley-` commands cause numerical labels and tick marks to appear. The statement "graph 6,1.5,A" plots an A at  $x=6$ ,  $y=1.5$  in scaled coordinates. The `-hbar-` and `-vbar-` commands draw horizontal and vertical bars to the specified scaled points. The `-gdraw-` command is like `-draw-`, except points are specified in terms of scaled quantities. The `-gat-` command is like `-at-` but uses scaled quantities.

Read the example over and try to identify in the picture what part of the display results from each statement. (Keep in mind that each number in the tags of these statements could have been a complicated mathematical expression.)

The `-markx-` and `-marky-` commands are similar to `-labelx-` and `-labeley-` but merely display tick marks without writing numerical labels. The `-axes-` command has an alternative form which allows for axes in the negative directions. (See Figure 9-7.)





Fig. 9-7.

Although the commands were originally designed to make it easy to draw graphs, the automatic scaling features make these commands useful in many situations. Note, in particular, that you can move complicated displays around on the screen merely by changing the `-gorigin-` statement.

Additional graphing commands include `-gvector-` for drawing a line with an arrowhead at one end, `-polar-` for polar coordinates, and `-lscalex-` and `-lscaley-` for logarithmic scales. The `-bounds-` command has the same

effect as `-axes-` in establishing lengths, but no axes are drawn on the screen (a later blank `-axes-` command will display the axes). The `-gbox-` command is used to draw rectangular boxes easily. The `-gcircle-` command draws circles or, if the `x-` and `y-`scales are different, `-gcircle-` will draw an ellipse.

Functions can be plotted very easily with the `-funct-` command. For example, `"funct 5sin(2w),w<=1,5,.02"` will plot the function `"5sin(2w)"` by evaluating this function for values of `w` running from 1 to 5 in steps of `.02`. Note the similarity to the form of the iterative `-do-` statement. If there was an earlier `"delta .02"` statement, we can leave off the increment and simply write `"funct 5sin(2w),w<=1,5"`. If, in addition, we want the function to be plotted all the way from the left edge of the established axes to the right edge, we simply write `"funct 5sin(2w),w"`.

### Summary of Line-drawing Commands: `-draw-`, `-gdraw-`, `-rdraw-`

Recall that the `-draw-` statement has the form:

```
draw point1;point2;point3;etc.
```

Each point in a `-draw-` statement may be coarse-grid (such as `"1215"`) or fine-grid (such as `"135,245"`). Each point specification is set off by a semicolon in order to avoid ambiguities when mixing coarse-grid and fine-grid points, as in `"draw 1525;1932;35,120;1525"` (the first two points are given in coarse-grid; the third, in fine-grid; and the last point in coarse-grid coordinates).

A discontinuous line drawing can be made with a single `-draw-` statement by using the word `"skip"`:

```
draw 1518;1538;skip;1738;1718
```

Using `"skip"` in a `-draw-` statement means `"skip to the next point without drawing a line."` This example is essentially equivalent to:

```
draw 1518;1538
draw 1738;1718
```

The only difference between these otherwise equivalent forms is related to the fact that the system variables `"where"`, `"wherex"`, and `"wherey"` are not brought up to date until the *completion* of the `-draw-` statement. The sequence:

## The TUTOR Language

```
at 1319 $$ affects "where"  
draw 1518;1538;skip;1738;where
```

is equivalent to:

```
at 1319  
draw 1518;1538  
draw 1738;1319
```

since *during* the -draw- statement "where" has the value 1319. On the other hand, the sequence:

```
at 1319  
draw 1518;1538  
draw 1738;where
```

is equivalent to:

```
at 1319  
draw 1518;1538  
draw 1738;1538
```

since upon completion of the first -draw- statement, the value of "where" is 1538. This difference between a single -draw- using "skip" and separate -draw- statements is sometimes useful in drawing figures relative to some point.

As mentioned earlier, starting with a semicolon implies a continued drawing from the present screen location. The sequence:

```
at 1319  
draw ;1542;1942
```

is equivalent to:

```
at 1319  
draw where;1542;1942
```

and is also equivalent to:

```
draw 1319;1542;1942
```

Sometimes you have more points for a -draw- than will fit on one line. A "continued" -draw- can be written, with the command blank on succeeding lines:

```
draw 1512;1542;skip;100,200;
      400,200;400,400;
      100,400;100,200
```

This will behave as though all the points had been listed on one line.

To summarize, the `-draw-` statement contains fine-grid or coarse-grid points separated by semicolons, “skip” can be used for a discontinuous drawing, “where” and the fine-grid “wherex” and “wherey” are brought up to date upon *completion* of the `-draw-`, and starting the tag with a semicolon has the special meaning of continuing a drawing from the present screen position.

The `-gdraw-` command is like the `-draw-` command except that points are relative to the graphing coordinate system established by `-gorigin-`, `-axes-`, (or `-bounds-`), `-scalex-`, and `-scaley-` (or logarithmic scales set up by `-lscalex-` and `-lscaley-`). Of particular value are the “skip” option and starting with a semicolon (for continuing a drawing). The use of “where”, “wherex”, and “wherey” in a `-gdraw-` statement is normally not meaningful, since these system variables refer to the absolute screen coordinate system, not the graphing system. In the graphing coordinate system, there are only fine-grid, not coarse-grid points, so all points have the form “x,y”.

It is possible to use `-draw-` to draw something relative to the present screen position:

```
at      2215
draw wherex+25,wherey-75;wherex+200,wherey+150
```

(Remember that “wherex” and “wherey” do not change until the completion of the `-draw-` statement.) There is an `-rdraw-` command (“r” for “relative”) which makes such drawings simpler. The example just shown can be written:

```
rorigin 2215
rdraw 25,-75;200,150
```

Each point of an `-rdraw-` is taken to be relative to an origin established with an `-rorigin-` command.

The `-rdraw-` command is particularly useful for applications such as writing the same Chinese characters at different places on the screen. For each character, make a subroutine involving one or more `-rdraw-` statements. The characters can be positioned with `-rorigin-` statements:

## The TUTOR Language

```

rorigin 400,400
do chin1
rorigin 400,300
do chin2
etc.

```

Or you might include the `-rorigin-` statement in the character subroutines:

```

do chin1(400,400)
do chin2(400,300)

```

In this case each subroutine has a form like this:

```

unit chin1(a,b)
rorigin a,b
rdraw -75,30;75,30;etc.

```

Unlike `-draw-`, the `-rdraw-` command is affected by preceding `-size-` and `-rotate-` commands. Your Chinese characters can be enlarged and rotated:

```

size 3,5 $$ 3 times the width, 5 times the height
rotate 45 $$ rotated 45 degrees
do chin1(400,400)
do chin2(400,300)

```

(Another way to handle such things as Chinese characters is with `-lineset-`.) Figure 9-8 shows a design created with the following commands:

```

rorigin 250,250
do figure,a←0,360,15
*
unit figure
rotate a
rdraw -50,0;50,0;0,200;-50,0

```

The `-rotate-` command affects `-rdraw-` even with “size 0”, even though `-write-` is *not* rotated in size 0. (The `-write-` statement is unaffected in order to facilitate normal text operations.) As far as `-rdraw-` is concerned, size 0 is equivalent to size 1. As far as `-write-` is concerned, size 0 means “write text at 180 characters per second, unrotated”, whereas size 1 means “write line-drawn text at 6 characters per second, rotated”.






Fig. 9-8.

Note that `-rdraw-` and `-size-` are essentially reciprocal to `-gdraw-` and `-scalex-`. In the case of `-rdraw-`, a drawing gets bigger when `-size-` specifies a larger size. But, specifying a larger number in a `-scalex-` command implies that the same number of screen dots (given by `-axes-`) will now correspond to larger (scaled) numbers in a `-gdraw-`. This means that a *larger* `-scalex-` implies a *smaller* `-gdraw-` figure. Note that `-gorigin-` affects `-gdraw-` the same way that `-rorigin-` affects `-rdraw-`.

There is a complete set of “relative” commands for making displays relative to an origin specified by `-rorigin-`, and affected by `-size-` and `-rotate-`. Here is a summary:

“ABSOLUTE”	“RELATIVE” (-size-)	“GRAPHING” (-scalex-, -scaley-)
	<code>rorigin</code>	<code>gorigin</code>
<code>at</code>	<code>rat</code>	<code>gat</code>
<code>atnm</code>	<code>ratnm</code>	<code>gatnm</code>
<code>draw</code>	<code>rdraw</code>	<code>gdraw</code>
<code>box</code>	<code>rbox</code>	<code>gbox</code>
<code>vector</code>	<code>rvector</code>	<code>gvector</code>
<code>circle</code>	<code>rcircle</code>	<code>gcircle</code>

Note that `-rcircle-` will draw an ellipse if the x- and y-sizes are different (as in “size 1,4”, for example).

## The TUTOR Language

The “halfcirc” subroutine of Chapter 4 could be conveniently rewritten using relative commands:


	“ABSOLUTE”		“RELATIVE”
unit	halfcirc	unit	halfcirc
at	x,y	rorigin	x,y \$\$ sets rorigin and “rat 0,0”
circle	radius,0,180	rcircle	radius,0,180
draw	x-radius,y;x+radius,y	rdraw	-radius,y;radius,y

It is important to note that the relative specifications set by `-rorigin-`, `-size-`, and `-rotate-`, as well as the graphing specifications set by `-gorigin-`, `-bounds-`, `-scalex-` (or `-lscalex-`) and `-scaley-` (or `-lscaley-`) *carry over* from one main unit to another. If you would prefer to have these parameters set to some standard values at the beginning of each main unit, simply do the initializations in an `-imain-` unit. (Remember that the `-imain-` command allows you to specify a unit to be performed every time a new main unit is started.)

How do you decide which of the three sets of display commands to use? If you want to rotate a drawing, you must use relative commands, because the absolute and graphing commands are unaffected by the `-rotate-` command. If rotations are not involved, just use whichever commands seem most convenient at the moment. Absolute commands may be used quite often since they are the simplest and easiest to use. The graphing commands are certainly best for drawing graphs of functions, but they are also useful whenever it is convenient to think of your drawing in terms of numerical scale factors. Graphing commands are also needed if you use polar coordinates (invoked with the `-polar-` command). Sometimes you may use all three sets simultaneously. For example, in one of this author’s lessons, the most convenient way to produce the screen display was to give instructions at the bottom of the screen using absolute commands, draw figures scaled in centimeters using graphing commands, and superimpose a movable box on the (absolute) instructions by means of relative commands.

### The `-window-` Command

Sometimes it is useful to specify a “window” through which drawings are viewed. Parts of a figure extending outside the window are not drawn. A rectangular window is specified by giving the lower left and upper right corners of the desired window:



The corners could also be given in coarse-grid coordinates, as in "window 1524;1248".

Drawings constructed from the various `-draw-` commands and `-circle-` commands are affected by a preceding `-window-` command. Line-drawn text (size non-zero) produced by `-write-`, `-writec-`, `-show-`, etc., will also be windowed. Like `-size-` and `-rotate-`, windowing is not reset upon entering a new main unit. Be sure to use a blank `-window-` command (blank tag) to turn off windowing operations. It is quite common for an author to forget to turn off windowing and then wonder why some of the drawings aren't showing up! The correct structure is shown below. (See Figures 9-9 and 9-10.)

```

window  one corner;opposite corner
.
.
.
(windowed) display statements
.
.
window          $$ blank tag to turn off
    
```




Fig. 9-9.




Fig. 9-10

### More on Erasing: The -eraseu- Command

When a student's response is judged "no" or "wrong", he or she can correct the response by hitting ERASE or ERASE1 to erase a letter or word, or by hitting NEXT, EDIT, or EDIT1 to erase the entire response. If additional judging keys have been defined with a -jkey- command, these will act like NEXT and erase the response. If there is only one -arrow- command and no -endarrow-, these options are available even after an "ok" judgment (except that a NEXT key or another judging key takes the student to the next main unit rather than merely erasing the response). If there is a "force firsterase", the student need not clear an incorrect response by pressing NEXT before trying a different response. In this case, the first key of the new response will cause the old response to be erased.

If the student erases part or all of his or her response, the "ok" or "no" is erased. Moreover, the last response-contingent message to the student is erased, since it is no longer relevant. For example:

```

.
.
wrong  cat
write  The cat is
        not a canine.
.
.
.

```

The student types "cat" and presses NEXT:

```

> cat no

The cat is
not a canine.

```

Notice that there is a default -at- three lines below the response. Suppose the student now presses ERASE:

```

> ca

```

The “t”, the “no”, and the text of the -write- statement have all disappeared automatically. This is appropriate since the comment “The cat is not a canine” is no longer needed.

It is helpful to know that the method TUTOR uses for automatically erasing such text is by re-executing the *last* -write-, -writec-, or -show-statement in the erase mode. Suppose we change the lesson slightly:

```

.
.
.
wrong cat
write The cat is
      not a canine.
write Meow!
.
.

```

Now the sequence looks like this:

```

> cat no

The cat is
not a canine. Meow!

```

```

> ca

The cat is
not a canine.

```

## The TUTOR Language

Only the *last* -write- statement is removed, leaving “The cat is not a canine” on the screen. Notice that the normal automatic erasing can be prevented simply by adding an extra -write- statement. Even a blank -write- statement will do.

As another example, consider this:

```

.
.
.
wrongv 4
write   Number of apples=
show   apnum
.
.
.

```

Only the -show- will be erased, leaving “Number of apples=” on the screen. If this is not desirable, use an embedded -show-:

```

.
.
.
wrongv 4
write   Number of apples=<(s,apnum)>

```

Now the last -write- statement includes the showing of the number, and all the writing will be erased. It is important not to change “apnum” after the -write-. If you change its value from what it was when shown by the -write-, the re-execution in “mode erase” will turn off the wrong dots in the numerical part of the writing. Here is the type of sequence to be avoided:

```

.
.
.
wrongv 4
write   Number of apples=<(s,apnum)>
calc    apnum←apnum+25
.
.
.


```

The number will not be erased properly due to the change in “apnum”.

Similar problems can arise with the other `-show-` commands, including `-showa-`.

Sometimes the automatic erasing of the last text statement is insufficient. For example, if the reply to the student included a drawing produced with `-draw-`, or if there were several `-write-` statements, you would need some additional mechanism to remove the reply when the student presses ERASE. There is an `-eraseu-` command which you can use to specify a subroutine to be done when the student changes his or her response:

```

.
.
.
 eraseu  eblock
arrow   1215
.
.
.
unit    eblock
at      1512
erase   35,4
at      318
erase   42
.
.
.

```

Unit “`eblock`” will be done whenever the student changes a response. Only the *first* press of the ERASE key triggers the erase unit, since additional executions of the unit would be erasing nothing.

Another example involves an erase unit specific to a particular response:

```

.
.
.
wrong   3 dogs
do      woof
eraseu  remove
.
.
.

```

(Continued on the next page.)

## The TUTOR Language

```

unit    remove
mode    erase
do      woof
mode    write
eraseu
.
.
.

```

The statement “eraseu remove” defines unit “remove” as the unit to be done when the student presses ERASE (or NEXT, etc.). Unit “remove” in the example shown simply re-does unit “woof” in the erase mode, thus taking off the screen everything originally displayed by unit “woof”. The final blank -eraseu- clears the pointer so there is no longer an erase unit specified.

Notice the similarities between the -imain- and -eraseu- commands. Both specify units to be done under specific conditions.

### Keeping Things on the Screen: “inhibit erase”

Let us consider a modified version of the simple language drill discussed in Chapter 7.

```

unit    espo
next    espo
back    satisfy
at      512
write   Here is a simple drill
        on the first five
        Esperanto numbers.
        Press BACK when you
        feel satisfied with your
        understanding.
at      1812
write   Give the Esperanto for
randu   item,5
at      2015
writec  item-2,one,two,three,four,five
arrow   2113
answerc item-2;unu;du;tri;kvar;kvin

```

This version will greatly annoy the student after the first couple questions. Each time the student gets an “ok” and presses NEXT to move on



to the next unit, the screen is erased and the student suffers through the introductory paragraph being written again on the screen. It turns out to be very annoying to see the same text replotted this way.

This is a situation where most of the material on the screen is not changing and should not be replotted. Only the item and the student's typing need be erased to make room for a new item and a new response. One way to do this involves judging correct responses "wrong", as was done in the dialog using -concept- discussed in Chapter 7. You should use "specs nookno" to prevent the "no" from appearing, or you can use the regular -okword- and -noword- commands to change the standard TUTOR "ok" and "no". For example, use the statement "noword Fine!" to cause "Fine!" to appear for a correct response. You would need to do a "noword no" whenever the student answers incorrectly. With all responses judged "wrong" we stay at the -arrow- and do not move on to another main unit.

Another way to manage a screen on which little is changing involves "inhibit erase". This statement prevents the normal full-screen erase upon leaving the present main unit. The next main unit must also execute an "inhibit erase" if no erase is to be performed upon leaving the second unit. We can rewrite our drill using this feature:

```

unit      preespo
at        512
write     Here is a simple drill
          on the first five
          Esperanto numbers.
          Press BACK when you
          feel satisfied with your
          understanding.

at        1812
write     Give the Esperanto for
goto     espo1
*

unit      espo
at        2015
erase    5      $$ item area
at        2115
erase    15     $$ response area
entry    espo1
inhibit  erase  $$ leave instructions on screen
next     espo
back     satisfy

```



(Continued on next page.)

## The TUTOR Language

```

randu    item,5
at       2015
writec   item-2,one,two,three,four,five
arrow    2113
answerc  item-2;unu;du;tri;kvar;kvin

```

In unit “preespo” we display the instructions about the drill. We then go to “espo1”, where we “inhibit erase” and display the first item. After receiving an “ok”, the student moves on to the next main unit, “espo”. The screen is not erased since there was an “inhibit erase”. In unit “espo” we erase the area containing the displayed item, and we also erase the response area of the screen. We then fall through the -entry- command and display a new item. This process repeats continually, and only those parts of the screen which must be changed are erased.

It is important to place an explicit blank -erase- statement (“erase ”) at the beginning of unit “satisfy”. Since we have inhibited the normal full-screen erase, no erase will occur automatically when the student presses BACK to leave the drill. If unit “satisfy” does not explicitly erase the screen, the student will see a superposition of the drill display and the display produced by unit “satisfy”.

Similarly, if we specify a help unit, that unit should start with a full-screen erase. Upon completion of the help sequence, we should come back to unit “preespo” rather than “espo” in order to restore the screen display properly, like this:

```

.
.
.
entry  espo1
base   preespo  $$ to come back to preespo from help
help   esphelp
.
.
.

```

The -base- command puts us in a help sequence, with the base unit being “preespo”. When a base unit has already been specified, pressing HELP doesn’t change the base unit (in other words, there is only one “level” of help). When we reach an -end- command or press BACK, we will return to the base unit, which is preespo. Note that unit “satisfy” should have a blank base statement to insure that we are in a non-help sequence. Otherwise, pressing BACK in unit “satisfy” will bring us to the base unit “preespo” again.

## Interaction of "inhibit erase" with -restart-

There is a -restart- command which is used to specify in which unit a student should resume study upon returning to a PLATO terminal. For example, suppose the last -restart- statement encountered on Monday by student "Ann North" in course "lingvo" was "restart espo" in lesson "espnum". On Wednesday she returns to a PLATO terminal and identifies herself by name (Ann North) and course (lingvo). Her registration records will show that she is to be restarted in unit "espo" of lesson "espnum" and she will automatically be taken to that point. As discussed previously, the "ieu" (initial entry unit) will be done, which among other things permits character set loading.

Unfortunately, restarting at unit "espo" means that the basic drill instructions contained in unit "preespo" *will not appear* (see last example). This is basically an initialization problem. You should use -restart- commands in such a way as to restart students only at the *beginning* of a section of this kind. In this particular case, we should have had a "restart preespo" rather than "restart espo". This is analogous to our use of "base preespo" for returning from a help sequence. (The more common form of the -restart- is the blank -restart-, which means "restart in the present main unit." We would place a blank -restart- in unit "preespo".)

Aside from initialization questions related to TUTOR and the display screen, it should be pointed out that the *student* has comparable initialization problems. Since the student may be away for several days, it is often advisable to have your restart points only at the beginning of sections of the lesson. This way the student can ease back into the context, whereas restarting in the middle of a discussion may be quite confusing. In those lessons which include an index, the index unit may be the best restart point. On the other hand, you will want to arrange things to allow the student to restart in the middle of a section if that section is very long.

When a student restarts in a lesson, he or she starts at the unit specified by the last -restart- command. However, the student's saved variables, v1 through v150, have whatever values were current at the time he or she left the last PLATO class session. Therefore, some care is required to initialize appropriate variables in the restart unit.

## The -char- and -plot- Commands

In most cases, special characters are handled with a -charset- command and displayed with a -write- statement using the FONT key. Alternatively, -char- commands can be used to transmit character patterns

## The TUTOR Language

to the terminal. If a `-char-` command sends a pattern to character slot 35 of the terminal, that character can be displayed using the `-plot-` command: `"plot 35"`. The arguments of the `-char-` command can be computed expressions so that a character can be constructed algorithmically. Similarly, the `-plot-` command may have a mathematical expression for its tag in order to choose the Nth character. See Appendix A for sources of detailed information on the `-char-` command.

### The `-dot-` Command

The statement `"dot 125,375"` will plot a single dot at the specified location (`"dot 1817"` uses coarse grid). A sequence of `-dot-` commands can produce sixty dots per second on the plasma display panel. A `-draw-with one point` (`"draw 125,375"` or `"draw 1817"`) makes a single dot by drawing a minute line from this point to the same point (or itself) and, for technical reasons, will produce only twenty dots per second. (The commands `-rdot-` and `-gdot-` also exist.)

## More About Judging

## 8

The previous chapter described the array of major response-judging features of the TUTOR language. We can now discuss the judging process in more detail, after which we will see how to treat responses that don't quite fit the categories of the previous chapter.

### Stages in Processing the -arrow- Command

The following is a summary of the several stages of processing involved when there is an -arrow- command.

- Stage 1 The -arrow- command is executed. The arrow is displayed on the screen, and a marker is set to remember the unit and location within the unit of this -arrow- command. Regular processing continues until a judging command is encountered, at which point there is a wait while the student types a response.
- Stage 2 The student presses NEXT or otherwise completes his or her response. TUTOR uses its -arrow- marker to start judging at the statement following the -arrow- command. Only judging commands are executed; all regular commands are skipped. Execution of a -specs- command sets a -specs- marker to remember the unit and location within the unit of this -specs- command.

## The TUTOR Language


- Stage 3 Some judging command terminates judging and successive regular commands are executed until a judging command is encountered, which ends this regular processing, even if we are several levels deep in `-do-s`. There is *no* “undoing”. An `-arrow-` or `-endarrow-` will also halt this regular processing without permitting “undoing”. (If no judging command terminates the judging phase, the end of a unit with no more “undoing” to do; an `-endarrow-`; or another `-arrow-` will end Stage 3 and make a “no” judgment.)
- Stage 4 If the `-specs-` marker has been set, regular processing begins at the statement following the last `-specs-` command encountered. (The `-specs-` marker is cleared.) This processing terminates in the same way as the regular processing of Stage 3. If the judgment is not “ok,” the `-arrow-` is not satisfied. The student must erase part or all of the response and enter a different response, which initiates Stage 2 again.
- Stage 5 The *search* state is initiated if there is an “ok” judgment. TUTOR again uses the `-arrow-` marker to start processing at the statement following the `-arrow-` command, this time in a search for another `-arrow-`. Only `-join-s` are executed, all other commands (regular or judging) are skipped during this search state. If an `-arrow-` command is encountered, TUTOR begins Stage 1 for this additional `-arrow-`. If an `-endarrow-` command is encountered, the search state ends and regular commands are processed. If neither `-arrow-` nor `-endarrow-` is encountered, the student can press NEXT to go on to the next main unit, having satisfied all the `-arrow-s`.

This all sounds rather complicated, written out in this way, but in most practical cases this structure turns out to be quite natural and reasonable. It is, nevertheless, useful to look at some unusual cases to further clarify the various processing stages.

Repeated Execution of `-join-`

The following is an example of the repeated execution of a `-join-` in regular, judging, and search states (remember that `-join-` is similar to `-do-`):

unit	multy
calc	i<=0

	arrow	1514
	join	$i \leftarrow i+1, \text{ansdog}$
	endarrow	
	at	2514
	show	i
	*	
	unit	ansdog
	answer	dog
	write	Bowwow!

The conditional `-join-` has only one unit listed, so we will always join unit “ansdog” no matter what value the expression  $(i \leftarrow i+1)$  has. Upon first entering unit “multy”, we do the `-calc-`, the `-arrow-`, and the `-join-`, all in the regular state. This terminates at the `-answer-` command to await a student response. Note that  $i$  is now 1, due to the assignment  $(i \leftarrow i+1)$  contained in the conditional `-join-`. Suppose the student types “cat” and presses NEXT. TUTOR starts at the statement following the `-arrow-` and executes the `-join-` in the judging state (incrementing 1 to 2 in the process). No match is found for “cat”, so the student must give another response. Suppose the student now enters “dog”. TUTOR again starts judging just after the `-arrow-` and again executes the `-join-` (thus incrementing  $i$  to 3). This time there is a match to “answer dog” which changes the state from judging to regular. The “write Bowwow!” is executed, and the end of unit “ansdog” causes TUTOR to “undo” back into unit “multy”, where the `-endarrow-` signals the end of the statements associated with the `-arrow-`. Since we received an “ok” judgment, we are ready to search for any other `-arrow-s` that might be in unit “multy”. We return to the `-arrow-` one last time, this time in the search state. The `-join-` is executed to see whether there is an `-arrow-` command in unit “ansdog”, with the incidental result that  $i$  gets incremented to 4. No `-arrow-` is found in unit “ansdog” and we “undo” into the `-endarrow-` command, which changes us from search state to regular state. The `-at-` and `-show-` are executed and we get “4” on our screen, due to the quadruple execution of the `-join-`.

Aside from illustrating some consequences of the processing rules, this example should emphasize that using the assignment symbol ( $\leftarrow$ ) in a conditional `-join-` may have unexpected results. Note that `-join-` is the *only* command with these properties, due to the fact that it is the only command executed in regular, judging, and search states. It is important that `-join-` be universally executed in this way so that you can join judging commands in the judging state and even `-arrow-` commands in the search state, not just regular commands in the regular state.

## The TUTOR Language

## Judging Commands Terminate Regular State

The rule that a judging command terminates the processing of regular commands is an important and general rule. We have seen that this must be true upon first encountering an `-arrow-` (the first judging command after the `-arrow-` makes TUTOR wait for a student response, since that judging command needs a response to work on). Let's see another instance of the rule:

```

.
.
.
arrow  1518
answer dog
write  Bowwow
wrong  cat
write  Meow
wrong  horse
.
.
.

```

If the student says "dog", he or she gets a reply "Bowwow" and regular processing stops at the "wrong cat" because `-wrong-`, a judging command, terminates the regular state. Similarly, if the student response is "cat", the statement "write Meow" is the only regular statement which is executed. The judging commands delimit those regular commands associated with a match of a particular judging command. This delimiting effect is achieved because:


- 1) Regular commands are skipped in the judging state; and
- 2) The processing of regular commands ends whenever a judging command is encountered.

Now let's consider a slightly modified sequence:

```

.
.
.
arrow  1518
join   dogcat
write  Meow
wrong  horse
.
.
.

```






unit	dogcat
answer	dog
write	Bowwow
wrong	cat

Supposedly, the “join dogcat” will act as though the statements of unit “dogcat” were inserted where the -join- is, which should make this modified version equivalent to the earlier version. Indeed, the rule that a judging command terminates the processing of regular commands does make the two versions equivalent, as we will show. Remember, in this discussion, that -join- is the same as -do- except for the universal nature of -join-.

Suppose the student types “dog”. We start just after the -arrow-, in the judging state. The -join- is executed and we find a matching “answer dog” which ends judging and puts us in the regular state. The “write Bowwow” is executed. The statement “wrong cat” is encountered next. The judging command -wrong- stops the processing of regular commands and also prevents coming out of the joined unit. Even though we are one level deep in -join-s, TUTOR will *not* “unjoin” and the “write Meow” which follows the “join dogcat” will *not* be executed. What *will* happen is just what happens in the earlier version: we have an “ok” judgment which causes the search state to be initiated at the -arrow- (there was no -specs-). Thus, the two versions operate in identical manners because the -join- acts like a text insertion. Note that a response of “cat” will get a reply “Meow” because there is no judging command following the “wrong cat” (and a normal “undo” is performed at the end of unit “dogcat”).

This last example illustrates the importance of the rule “a judging command terminates the regular state.” It is this rule which insures that -join- (or -do-) will act like a text insertion.

In the discussion of the -goto- command in Chapter 6, we saw that a -goto- in a done unit destroys the strict text insertion character of the -do-. This is true in the present context as well. Suppose we insert a -goto- in unit “dogcat” (any -goto- will do, we’ll use a “goto q”):

unit	dogcat
answer	dog
write	Bowwow
 goto	q
wrong	cat

The student enters “dog” and we do unit “dogcat” where the match to “answer dog” flips us from the judging to the regular state. The regular


The TUTOR Language

commands `-write-` and `-goto-` are executed. (Note that `-goto-`, like `-do-`, is only regular whereas `-join-` is universal, being executed not only in regular but in judging and search states.) The execution of the `-goto-` prevents TUTOR from encountering the “wrong cat” which previously terminated the regular state. We have run out of things to do in unit “dogcat” and are one level deep in `-do-s`. TUTOR, therefore, “undoes” and executes the “write Meow” which follows the “join dogcat”! The student will see “BowwowMeow” on the screen. If, on the other hand, we replace the “join dogcat” with the statements contained in unit “dogcat” we would have:

```

.
.
.
arrow    1518
answer   dog
write    Bowwow
goto     q
wrong    cat
write    Meow
wrong    horse
.
.
.

```



and a response of “dog” would merely cause “Bowwow” to appear on the screen, *not* “BowwowMeow”.

We have again seen that a `-goto-` in a done unit can cause the `-join-` operation to behave differently from a text insertion. We get different effects depending on whether we `-join-` such a unit or put that unit’s statements in place of the `-join-` statement. You can avoid confusion by not using `-goto-` commands in “done” or “joined” units which contain `-arrow-` commands or judging commands.


The `-goto-` is a Regular Command

Since the `-goto-` command is a regular command, it is skipped in the judging and search states. Here is a sequence of commands which illustrates the fact that the `-goto-` is skipped in the *judging* state:

```


.
.
.

```

	arrow	1612
	goto	dogcat
	*	
	unit	dogcat
	answer	dog
	write	Bowwow
	wrong	cat

When the `-arrow-` is first encountered, an arrow is displayed on the screen at 1612. TUTOR continues in the regular state and executes the `-goto-`. The `-answer-` in unit “dogcat” ends this regular processing to await the student’s response. Suppose the student types “dog” and presses NEXT. TUTOR starts judging just after the `-arrow-`, skips the regular `-goto-` command, and finds no judging commands at all. The student’s response gets a default “no” judgment. The `-goto-` should be replaced by a `-join-` so that unit “dogcat” will be attached in the judging state.

Similarly, the following is an erroneous sequence which illustrates the fact that the `-goto-` command is skipped in the *search* state:

.		
.		
.		
	arrow	1612
	specs	bumpshift
	answer	dog
	goto	another
	wrong	cat
	*	
	unit	another
	arrow	2514
	answer	wolf

The student responds to the first `-arrow-` with “dog” and matches the “answer dog”, which switches the processing from the judging state to the regular state. The `-goto-` is executed, and in unit “another” we encounter an `-arrow-` command. This `-arrow-` command terminates the regular processing just as a judging command would. The `-specs-` marker was set, so we will now execute any regular commands following the `-specs-` command (there are none in this example). Since the student’s response was “ok”, the search state is now initiated. TUTOR starts at the “arrow 1612” looking for another `-arrow-` command. The `-specs-`, `-answer-`, `-goto-`, and `-wrong-` are skipped in the search state, and we come to the end of the unit without finding an `-arrow-`. Thus the `-goto-` did not

## The TUTOR Language

succeed in attaching a second -arrow-. If the -goto- is replaced by a -join-, the “wrong cat” will be associated with the *second* -arrow- (2514). This is due to the text insertion nature of the -join-, which interposes the statements of unit “another” between the “answer dog” and the “wrong cat”. One correct way to write this sequence is shown below:

```

.
.
.
arrow      1612
specs     bumpshift
answer    dog
wrong     cat
endarrow
goto      another    $$ or “do another”
*
unit      another
arrow     2514
answer    wolf

```

The -goto- or -do- placed after the -endarrow- will not cause any problems because the search state has been completed, and the -endarrow- flips us from the search state to the regular state.

Considerations of this kind suggest that some care must be exercised when using -join- or -do- to attach units containing -arrow- commands. To avoid unpredictable results follow these two rules:

- 1) A unit attached by -join- or -do- which contains one or more -arrow- commands *must* end with an -endarrow- command. This insures that the unit will end and “undo” in the regular state. (It is permissible to have regular commands following the -endarrow-.)
- 2) The attached unit containing one or more -arrow- commands *must not* contain any -goto- commands. (A -goto- can make TUTOR fail to see the -endarrow- or a judging command so that a premature “undo” occurs.)

If these two rules are followed, the -join- or -do- will act precisely as though you had inserted the statements of the attached unit where the -join- or -do- was. Here are examples of good and bad forms:

	<u>GOOD</u>		<u>BAD</u>
	unit      response		unit      response
...	answer    apple		answer    apple
join response	do        newton		goto      newton ( <u>Don't</u> use -goto- here)
...	wrong     pear		wrong     pear

GOOD (continued)  
 write Wrong fruit.  
 endarrow

BAD (continued)  
 write Wrong fruit.  
 (Do use -endarrow- here)

### Interactions of -arrow- with -size-, -rotate-, -long-, -jkey-, and -copy-

When an -arrow- command is performed, several things happen. An arrow character is displayed on the screen, cuing the student to enter a response. A note is made of the unit and location within that unit of the -arrow- command so that TUTOR can return to this marked spot when necessary. Even the trail of -do-s (and/or -join-s) which brought TUTOR to this -arrow- command is saved, so that each restart at the -arrow- will be at the appropriate level of -do- relative to the main unit. The current settings of -size- and -rotate- are saved, to be restored each time so that you can write a size-3 reply to a student's incorrect response without affecting the size of his or her corrected typing. In other words, response-contingent settings of -size- and -rotate- are temporary, whereas in other circumstances they are permanent until explicitly changed:

```

.
.
.
.
size      2
rotate    0
arrow     1718
answer    dog
size      4
rotate    30
write     Woof!
answer    wolf
endarrow
at        2218
write     This is in size 2, rotate 0.
.
.
.

```

The last writing appears in size 2, rotate 0 despite the size 4, rotate 30, that were contingent on the student's response, "dog." When the search state is initiated, the original size and rotate settings are restored.

## The TUTOR Language

Similarly, if “dog” had been judged wrong, the student’s revised typing would have been in size 2, not 4, because the original size and rotate are restored before waiting for the student’s revised input.

Executing an `-arrow-` command has other important initialization effects:

- 1) A default response limit of 150 characters is set. The student cannot enter a response longer than 150 characters (including “hidden” characters such as shift-codes and superscripts). This can be altered by following the `-arrow-` command with a `-long-` command to change this to as much as 300. If this is a “long 1,” judging will commence as soon as the student types one character. If more than 1 is specified, the student is prevented from entering more characters and must press NEXT to initiate judging, unless a “force long” statement has appeared in the unit.
- 2) A default specification of “judging keys” is set. In most cases, the NEXT key is solely responsible for starting the judging process. However, there are two other possible ways to begin judging: (1) hitting the limit with a “force long”; or (2) if there is a “long 1”, typing one character will begin judging. This can be altered by following the `-arrow-` command with a `-jkey-` command to specify *additional* judging keys (NEXT is *always* a judging key). One example is “jkey data,help” which would make the DATA and HELP keys equivalent to the NEXT key at this arrow.
- 3) A default specification is set to disable the COPY key. The `-arrow-` command can be followed with a `-copy-` command to specify a previously stored character string to be referenced with the COPY key. An example is “copy v51,v3”, where v51 is the start of the character string and v3 is the number of characters. This way of specifying a string of characters is the same as the scheme used with `-storea-` and `-showa-`.

Some explanation of the COPY and EDIT keys is required. The EDIT key is *always* available for the student to use in correcting his or her typing. Pressing the EDIT key the first time erases all typing, after which each press of the EDIT key brings back the typing one word at a time. This makes it easy to make corrections and insertions without a lot of retyping. Each press of the COPY key, on the other hand, brings in a word from the character string specified by the `-copy-` command, as opposed to bringing in the student’s own typed words with the EDIT

key. One example of the use of the COPY key is seen in the PLATO lesson editor. In this case, you as an author can use the COPY key in insert or replace mode to bring in portions of a preceding line without having to retype. The COPY key must be specifically activated by a -copy- command, but the EDIT key is always usable, unless you specify a -long- greater than the normal limit of 150. (To use the EDIT key on responses longer than 150 characters requires you to furnish an edit buffer through an -edit- command.)

The -long-, -jkey-, and -copy- commands all override default specifications set by the -arrow- command. They can be thought of as modifiers of the -arrow- command. If they are to have an effect on the student's first response, they not only must follow the -arrow- command but must *precede* any judging commands:

```

arrow 1518 $$ sets default values
{ jkey  help
  copy  cstring,ccount } These commands alter the default values.
{ long  15
  -specs- or -answer- or -store- or any other judging command

```

If -jkey-, -copy-, or -long- came *after* the first judging command, the -arrow- defaults would hold for the first response because the modifying command would not have been executed yet.

### Applications of -jkey- and -ans-

Use of the -jkey- command is well illustrated in the case of providing help to the student (through the HELP key) *without leaving the page*. (This is an alternative to the more commonly used -helpop- command described in Chapter 5.) If giving help requires an entire screen display, or a whole sequence of help units, it is best to use a -help- command to specify where to jump if the student presses HELP. The screen is then erased automatically to make room for the help page (unless the original base unit had an "inhibit erase" in it). On the other hand, sufficient help might consist merely of a brief comment or some additional line-drawings on the present page. A convenient way to provide such help without leaving the page is:

## The TUTOR Language

```

.
.
.
arrow    1815
jkey     help
answer   cat
no
write    Hint: it meows . . .

```

The statement “jkey help” makes the HELP key completely equivalent to the NEXT key. If the student presses HELP, judging is initiated, the student’s (blank) response does not match “cat”, and he or she gets “Hint: it meows . . .”. Without the -jkey- command, the HELP key would be ignored (which would be unfortunate). It is a very good idea to have the HELP key do *something* at all times so that the student can come to rely on help being available.

In this example, the student will get the same assistance whether he or she presses HELP or types “dog” followed by pressing NEXT. We could give different kinds of assistance in these two cases by changing the -write- statement to a -writec-:

```

.
.
.
arrow    1815
jkey     help
answer   cat
no
writec   key=help,Meow?,The answer is cat.

```

The system variable “key” always contains a number corresponding to the last key pressed by the student. In this case the last key will either be HELP or NEXT. If the student presses HELP, the logical expression “key=help” will be true (-1) and the student gets the reply “Meow?”. But, if the student presses NEXT, then the logical expression “key=help” is false (0) and the student gets “The answer is cat.” The lower-case word “help” is defined by TUTOR to mean (in a calculational expression) “the number corresponding to the HELP key.” Other similarly defined names include next, back, and help1 (for shift-HELP).

The following is another way of writing the same sequence:



```

arrow    1815
jkey     help
no       $$ terminate judging
judge    key=help,x,continue
write    Meow?
answer   cat
no
write    The answer is cat.

```

If key=help, we “fall through” the -judge- command and write “Meow?” If the key is not equal to help (that is, the student pressed NEXT), a “judge continue” is performed to return to the judging state. The “write Meow?” is skipped since -write- is a regular command. If the response does not match “cat”, the student will get the message “The answer is cat”. As usual, there are many ways in TUTOR to do the same thing! In a particular situation one scheme may be more appropriate than another.

There is an ANS key on the keyset which is often used to let students skip through material by just pressing ANS:

```

.
.
.
arrow    1817
jkey     ans
ok
judge    key=ans,x,continue
write    The answer is cat
answer   cat
.
.
.

```

Since the ANS key generates an ok judgment here, the student will move on immediately to the next arrow or unit without having to type the correct answer. This procedure could best be utilized when the student is in the review mode. That is, you might define “review=v1”, zero it initially, and set it to -1 only after the student has gone through the material once under his or her own power. With the following structure, the student will be able to use the ANS key only when reviewing the material:

## The TUTOR Language


```

.
.
.
arrow 1817
do review,jans,x
ok
judge key=ans,x,continue
.
.
.
unit jans
jkey ans

```

Another way to activate the ANS key for the student is to use the `-ans-` command with a blank tag.

```

.
.
.
arrow 2123
ans
 write The answer is cat.
.
.
.

```

In the above example, the single `-ans-` command is equivalent to the following:

```

jkey ans
ok
judge key=ans,x,continue

```

The `-ans-` command is a judging command and must be the *first* judging command after the `-arrow-`. When it is first encountered, it sets up ANS to be a judging key, and it is matched only if the ANS key is pressed. If the `-ans-` command is used only to provide a kind of help, but not to let the student pass on to the next item, put a “judge wrong” after the `-ans-` command.

In many places you may do specific things in response to the ANS and HELP keys. Elsewhere in the lesson it is appropriate merely to utilize these keys so that *something* will happen when they are pressed. Just put “jkey help,ans” after each such `-arrow-`. The student will then

get (at least) whatever reply you give him or her after the universal -no- that catches all unrecognized responses. Certainly, every -arrow- should provide some kind of feedback to unrecognized responses or the student will become perplexed. The “jkey help,ans” will further insure that a reasonable response to the student’s input is always forthcoming. Without this -jkey- statement, nothing would happen when the student presses ANS or HELP.

An additional procedure is advisable. Often a student will press NEXT an extra time, perhaps because he or she hadn’t noticed that a response was to be typed. This blank response, consisting only of a NEXT key, will probably get judged “no” at most arrows, which requires an additional NEXT (or ERASE) to clear the “no” judgment before typing a response. This can get confusing. In most cases it is best simply to ignore blank responses by means of the statement “inhibit blanks”, which can be put in the -imain- unit (see Chapter 5). This statement causes blank-NEXT inputs to be ignored, but other blank inputs such as HELP or ANS are not ignored.

Use a -join- to insert recurring statements after an -arrow-:

```

.
.
arrow  1917
join   anshelp
answer cat
.
.
.
unit   anshelp
inhibit blanks  $$ or the -inhibit- could be in an -imain- unit
jkey   ans,help

```

Placing “join anshelp” after each -arrow- will insure that extra NEXT keys are thrown out (while responses involving ANS or HELP keys, will fall through to whatever reply you give to unrecognized responses). Note that you must use -join-, not -do-, to attach unit “anshelp” if you add any judging commands to that unit.


Just as the -imain- command can be used to specify a unit to be done at the beginning of each new main unit, there is an -iarrow- command (“initialize arrow”) which can be used to specify a unit to be joined after every -arrow-. With the statement “iarrow anshelp”, it is unnecessary to write “join anshelp” after every -arrow- command. Unit “anshelp” will be joined automatically after every -arrow-.

## The TUTOR Language

## Modifying the Response: -bump- and -put-

It is possible to delete characters from the judging copy of the student's response by using the -bump- command:


```

arrow 1812
 bump as3 $$ delete all a's,s's, and 3's
answer rdvrk

```

This -answer- will be matched if the student types "33 aardvarks" because the -bump- command reduces the judging copy of the response to "rdvrk." The original response is not altered and can be recovered with a "judge rejudge". Also, the screen display is unaffected: the student still sees "33 aardvarks" on the screen just as he or she typed it. On the other hand, all judging commands following the -bump- are affected since they all operate on the judging copy (not on the original response). For example, a -storea- following the -bump- would give you "rdvrk". Here is another example:

```

define cfirst=v1,csecond=v2
      first=v11,second=v21
unit   conson
at     913
write  Type anything, and I'll
      remove the vowels:
arrow  1309
long   100    $$ from v11 to v21 is 100 characters
storea first,cfirst<=jcount
 bump aeiou
storea second,csecond<=jcount
ok
write  You typed <a,first,cfirst>.
      Remove vowels: <a,second,csecond>.
      You used <s,cfirst-csecond> vowels.

```

Note that "cfirst" is the number of characters (including hidden characters such as shift characters) in the original response, whereas "csecond" is the number of characters after the -bump- has removed the vowels. This is a true count since "jcount" always has an up-to-date character count of the judging copy, as influenced by -bump- and related opera-

tions. (You may recall that “specs bumpshift” also affects “jcount” by removing shift characters.) Suppose the student types “Apples taste funnier”. In this case, the student will get the reply:

You typed Apples taste funnier.  
Remove vowels: Ppls tst fnnr.  
You used 7 vowels.


The reason that the word “Apples” turns into “Ppls” with a capital “P” is that a capital “A” is really a shift character followed by a lower-case “a”. With the “a” bumped out, the shift character stands next to the “p”, making a capital “P”.

While the -bump- command will delete characters, the -put- command will change particular strings of characters:

```

.
.
.
arrow 1218
put cat=dog
put rat=mouse
storea first,jcount
ok
showa first,jcount

```



All occurrences of “cat” change into “dog”, and all occurrences of “rat” change into “mouse”. Suppose the student types “Scattered cats scratch rats”. The reply will be “Sdogtered dogs scmousech mouses”!

Both -bump- and -put- are judging commands. They operate on the student’s response. Like all judging commands, they stop processing when encountered during the processing of regular commands. The -put- command has a property similar to -store- in that it can terminate judging with a “no” judgment if it cannot handle the student’s response:

```

.
.
.
arrow 1218
put cat=enormous
write Too many cats!
ok
.
.
.

```

## The TUTOR Language

If the student has many “cats” in his or her response, the `-put-` may cause “jcount” to exceed the 150-character response limit. In this case, it changes to the regular state, and the student gets the message “Too many cats!” This regular `-write-` command normally is skipped, since we’re in the judging state.

The following is an equivalent form of `-put-` which is often easier to read:

```
put   cat=dog
putd  /cat/dog/
putd  ,cat,dog,
```

All three of these statements are equivalent. The `-putd-` (d for delimiter) takes the first character as the delimiter between the two character strings. Other examples of its use are:

```
putd  /=/equals/  $$ convert = sign
putd  / //        $$ remove all spaces
```

It is also possible to change variable character strings by using `-putv-` (v for variable):

```
putv   first,cfirst,second,csecond
      ↙           ↘
string and count  string and count
```

When you combine `-put-` and `-bump-` commands, you must be careful about how you arrange them. For example, the following sequence is nonsense:

```
bump a
put   cat=dog
```

With all a’s bumped the `-put-` will not find any cat’s. Similar remarks apply to sequences of `-put-` commands.

The `-bump-` command looks for single characters, so “bump B” will not merely bump capital B’s. All shift characters will be bumped as well as lower-case b’s. In other words, “bump B” is really “bump shift-b”. If you want to eliminate only capital B’s, use “putd /B//”. This will find occurrences of the string of characters “shift-b” and replace this string with a zero-length string, thus deleting the B.


The main purpose of `-bump-` and `-put-` is to make minor modifica-

tions to the student's response and convert it into a form which can be handled by standard judging commands. For example, the word-oriented judging commands (-answer-, -match-, -concept-, etc.) cannot find *pieces* of words. Suppose that for some reason you need to look for the fragment "elect", and you don't care whether this appears in the word "selection" or "electronics" or "electoral". Do this:

```

.
.
.
arrow    1723
specs    okextra
putd     /elect/ elect /
answer   elect
.
.
.

```



The -putd- is used here to put spaces before and after the string "elect" so that it stands out as a separate word. You could also use the values of "jcount" before and after executing the -putd- to determine whether "elect" was present. The number of times it appeared could also be determined from these values. The value of "jcount" will increase by two for each insertion of two extra spaces.

## Manipulating Character Strings

The judging commands -bump- and -put- operate on the judging copy of the student's response. It is sometimes useful to manipulate other strings of characters with -pack-, -move-, and -search-. These commands are regular commands, not judging commands. Like -showa-, they operate on stored character strings, not the judging copy of the student's response. These commands are mentioned here because they are often used in association with the analyzing of student responses. In particular, the judging command -storea- can be used to get the response character string. It can then be operated on with -move- and -search-. Finally, the altered character string can be loaded back into the judging copy with the judging command -loada- (load alphanumeric; the -loada- command is precisely the opposite of -storea-). Since this section deals with a rather esoteric topic, you might just skim through it now to get a rough idea of what character string manipulations look like. If you later find a need for such operations, you should study this section again.

The TUTOR Language

Here is an example of a -move- statement:

```
move v3,5,v52,21,8
```

This means “move 8 characters from the 5th character of the string that starts in v3 to the 21st character of the string that starts in v52.” The 21st through 28th characters of the v52 character string are replaced by the 5th through the 12th characters of the v3 character string. The v3 character string is unaffected. In other words, -move- has the form:

```
move string1,start1,string2,start2,#characters moved
```

If the number of characters to move is not specified, one character will be moved.

Here is an example of the use of -move-. Suppose the student types “ $x+4y = y-3$ ”, and we want to convert this into the form “ $x+4y-(y-3)$ ” before using -store- on it. Assume “str” has been defined:

```


.
.
.
arrow 1812          $$ x+4y=y-3
putd   .=.-(.      $$ x+4y-(y-3)
storea str,jcount
ok
move   ')',1,str,jcount+1  $$ x+4y-(y-3)
judge  continue        $$ to do judging -loada-
loada  str,jcount+1
store  result
ok
write  Subtracting the right side of
       your equation from the
       left side gives <(s,result)>.

```


In the -move- command the parenthesis within single quote marks, ')', means a character string one character long consisting of a right parenthesis. Similarly, 'dog' would denote a character string consisting of d,o, and g. Character strings up to ten characters in length may be described this way, using single quote marks. The -move- command shown above moves the first character of ')', which is just a right parenthesis, to the (jcount+1)th character position in “str”. This effectively appends a right parenthesis to the student’s character string (as modified by the -putd-). The -loada- command moves the final character string into the judging copy so that -store- can operate on it. Note carefully the switches from the judging state to the regular state and back again.



The `-search-` command is used to look for occurrences of specific character strings. It has the form:



Suppose we use `-storea-` to place the unaltered student response “`x+4y=y-3`” in “`str,jcount`”. Then use:



This `-search-` command will set the variable “`charnum`” to 5, since the equal sign is the 5th character in “`x+4y=y-3`”. If the search is unsuccessful, “`charnum`” is set to `-1`. As further illustration of `-move-` and `-search-`, let’s rewrite our earlier sequence without the `-putd-`:

```
arrow 1812
storea str,jcount
ok
search '=' ,1, str,jcount,1, charnum
* Now make room for the -( :
move str,charnum+1,str,charnum+2,jcount-charnum
*Next insert the -( :
move '-(',1,str,charnum,2      $$ move 2 characters
* Append the ) :
move ')',1,str,jcount+2
judge continue
loada str,jcount+2
store result
ok
```

## The TUTOR Language

The `-search-` finds the equal sign. The first `-move-` moves the latter part of the string to make room for the insertion of `'-'`. The second `-move-` makes the insertion which overwrites the characters (`=y`) which were there originally. The third `-move-` appends the `'`. Normally, the `-search-` would be followed by a `"goto charnum,noeq,x"` to take care of the case where the student did not use an equal sign, in which case `"charnum"` would be `-1`.

The single quote marks can be used to specify character strings up to ten characters long. Longer character strings can be placed in variables with a `-pack-` command:

```
pack v11,v3,abcdefghijklmnopqrstuvwxy
      ↙           ↘
string location  character count
```

This packs a character string 26 characters long into `v11` and following variables. The character count (26 in this case) is placed in `v3`. Since each variable holds ten characters, `v11` and `v12` will be full while `v13` will have the last six characters. The `-pack-` command might be considered analogous to `-storea-`, since both place character strings in variables. In the case of `-storea-`, the total character count can be gotten from the system-defined variable `"jcount"`. Here is another example:

```
pack v12,v1,H2SO4
...
showa v12,v1
```

This will display `"H2SO4"` on the screen. The character count in `v1` will be ten, including three shift codes and two subscripts. The character string `H2SO4` is actually composed of shift, h, subscript, 2, shift, s, shift, o, subscript, 4. The character count portion of a `-pack-` command can be left blank, as in `"pack v12,,dog"`, the result of which could be displayed later with the statement `"showa v12"`. It is possible to embed `"show"` commands in a `-pack-` statement:

```
pack string,count,There are $(s,total) left.
```

There is also a conditional form, `-packc-`, analogous to `-writec-`:

```
packc cond,string,count,dog,cat,horse,cow
      ↙           ↘           ↘           ↘
conditional  -1     0     1     ≥2
expression
```

There are other string-oriented commands. For example, `-clock-` will get the time, `-date-` gets today's date, `-name-` gets the (18-character) name the student is registered under, and `-course-` gets the course the student is registered in. These commands are used in the following illustration:

```
.
.
.
name    v1  $$ v1 and v2 for name
course  v3
clock   v4
date    v5
write   Hello! Your name is<a,v1,18>.
        You are registered in <a,v3>.
        The time is <a,v4>.
        The date is <a,v5>.
```

Suppose the student is registered as "sam nottingham" in a course "french4." It is 10:45:37 PM (22:45:37 on a 24-hour clock) on June 3, 1974. The student will receive this display:

```
Hello! Your name is sam nottingham.
You are registered in french4.
The time is 22.45.37.
The date is 06/03/74.
```

All of these commands, `-name-`, `-course-`, `-clock-`, and `-date-`, simply place the requested character string in the specified variable for use in a `-showa-`.

The `-clock-` command produces a *character string*. In addition, there is a system variable "clock" which may be used in calculational expressions. It holds the *number* of seconds of a daily clock to the nearest thousandth of a second, and is convenient for calculating the amount of time spent in a section of a lesson.

The `-date-` command also produces a *character string*. There is also a `-day-` command which produces a *number* corresponding to the number of days elapsed since January 1, 1973. This number of days and fraction of a day is accurate to one-tenth of a second.

The TUTOR judging commands offer a great deal of power. We have seen that the judging commands `-bump-` and `-put-` together with the regular string-oriented commands `-move-`, `-search-`, and `-pack-` can be used to change an otherwise intractable response into a form which can be handled with TUTOR judging commands. This is a useful scheme as

## The TUTOR Language

long as only minor modifications are required. However, if major modifications of the response are required in order to be able to use TUTOR judging facilities, it is usually simpler to “do your own judging.” That is, get the student’s response with a `-storea-` and then analyze it with string-oriented commands, together with the additional calculational machinery described in Chapter 9. You might not even want to use the built-in marker features of the `-arrow-` command, with the associated returns to the `-arrow-`, when there is a “no” judgment. In such circumstances you might write a subroutine to be used in place of `-arrow-` commands, which merely collects the student’s response:

```

unit      arrow(apos)
arrow     apos
storea   sstr,scnt←jcount
specs    nookno
ok
endarrow

```

Instead of writing “`arrow 1815`” with associated judging commands you would then write:

```

.
.
do arrow(1815)
calc,move,etc. to do your own judging
.
.

```

Naturally, this course of action is advisable only if you are trying to analyze responses which have a form very different from those classes of responses which can be handled well by TUTOR judging commands.

### Catching Every Key: `-pause-`, `-keytype-`, and `-group-`

Occasionally, it is useful to process individual keypresses without waiting for a NEXT key. We have already discussed such typical examples as moving a cursor and choosing a topic from an index. These examples used a “long 1” with an `-arrow-` in order to catch each keypress. There is another way to do this, involving the `-pause-` command which was introduced in Chapter 2 in connection with creating displays, particularly timed animations. As was pointed out in the discus-

sion of the `-jkey-` command in the present chapter, the system variable “key” contains a number corresponding to the most recent key pressed by the student. For example, if the student presses the letter “d”, the system variable “key” will have the numerical value 4 (since d is the 4th letter in the alphabet). Putting these notions together, we have the following kind of structure:

```
write  Press "d", please.
pause
writec key≠4,You didn't press d.,Good!
```

The blank `-pause-` statement (“blank” in the sense of having no tag) causes TUTOR to wait for the student to press a key. Any key will cause TUTOR to move past the `-pause-` to the next statement.

In the example shown, the `-pause-` is followed by a `-writec-` conditional on “key≠4”. This `-writec-` can be written in more readable form by replacing the “4” with a “d”:

```
writec  key≠"d",You didn't press d.,Good!
```

Enclosing the d with (double) quote marks is taken in calculational expressions to mean the number 4. Similarly, (`v3←“z”`) will assign the value 26 to v3. If the student presses 0 or 1, “key” will have the numerical value 27 or 28 respectively. That is, the 26 letters are followed by the numbers 0 through 9, then come various punctuation marks. If the student presses the plus key, “key” will have the numerical value “+”, which happens to be 37.

If the student presses a capital D, “key” will have the value 64+“d”, or 68. The shifted or upper case letters have “key” value 64 greater than the corresponding lower-case letters. Caution: some common keys such as parentheses have key numbers smaller than 64 despite requiring the shift key to type them. The most commonly used characters (lower-case letters, numbers, and common punctuation marks) have key numbers less than 64, independent of whether they are typed using the shift key. As for the function keys (NEXT, BACK, HELP1, etc.), we have seen (in connection with the `-jkey-` command) that the corresponding key numbers are given by next, back, help1, etc., as in:


```
goto  key=help1,yes,no
```

No quote marks are used for the function keys.

A more convenient way to determine which key has been pressed is to use a `-keytype-` command. Consider a cursor-moving procedure:

## The TUTOR Language

```

define    num=v5,x=v1,y=v2,dx=10,dy=10
unit      cursor
pause
 keytype  num,d,e,w,q,a,z,x,c
goto     num,cursor,x
.
.
calcs    num-1,y<=y,y+dy,y+dy,y+dy,y,y-dy,y-dy,y-dy

```

The `-keytype-` command searches through the listed keys (d, e, w, q, a, z, x, and c in this case) and, similar to the `-match-` command, sets “num” to -1 (if the key is not found in this list) or to 0, 1, 2, 3, etc. (if it is found). If the student presses d, “num” will be set to 0; if the student presses c, “num” will be 7; and if he or she presses D, “num” will be set to -1. The `-goto-` statement effectively causes all unlisted keys to be ignored.

Note that no quote marks are used in specifying keys in a `-keytype-` command. Capital letters and function keys may also be listed:

```
keytype  v3,a,A,b,B,next,data,timeup
```

While the `-keytype-` command is most often used in conjunction with a `-pause-` command, it can also be used in association with an `-arrow-` command or any time that you want to find out which key was pressed most recently. The function key `timeup` is one generated by TUTOR when a timing key is “pressed” as the result of an earlier `-time-` command or timed `-pause-` command (see Chapter 2).

Just as the `-list-` command can be used to specify a set of synonymous words and numbers for use in `-answer-` and `-match-`, so there is a `-group-` command available for specifying synonymous keys for use in a `-keytype-` command:

```

define    keynum=v23,algkey=v24
group     algebra,x,y,z
.
.
.
keytype   keynum,a,b,algebra,help
           ↓ ↓ ↓ ↓
           0 1  2  3

```

If the student presses any of the keys x, y, or z, the variable “keynum” will be assigned the value 2. An additional `-keytype-` command can be used to separate members of a group:

```

keytype  keynum,a,b,algebra,help
goto     keynum,none,ua,ub,alg,somehelp
.
.
unit     alg
keytype  algkey,x,y,z
.
.

```

Some particularly useful `-group-` definitions are built-in. Without specifying these definitions with your own `-group-` commands, you can (in a `-keytype-` command) refer to these groups in the following ways:

```

alpha    all 52 lower-case and upper-case letters
numeric  0 through 9
funct    function keys (next,help,etc.)

```

An example of the use of these built-in groups might be “`keytype v45,funct,a,b,c`”. You can also use previously defined or built-in groups to define new groups:

```

group  mine,a,b,c,help
group  ours,mine,d,e,f
group  all,A,B,C,ours,numeric,funct

```

It is important to note that if you use a `-pause-`, the key pressed will not cause the associated character to appear on the student’s screen. You are in *complete* control. You may write something on the screen or not, as you choose. Only if you use an `-arrow-` will the standard key display take place (with the associated ERASE and other standard typing features available). Similarly, if you press HELP, you will not automatically branch to a unit specified by a previous `-help-` command, because a blank `-pause-` gives you *every* key, function key or not.

There is a variant of the `-pause-` command which is usually more useful than the blank `-pause-`. You can define which keys are to be accepted, and all other keys will be ignored:

```

.
.
next    umore
help    discuss
data    tables

```

(Continued on the next page.)

## The TUTOR Language

```

.
.
pause keys=d,D,next,term,help,help1
.
.
.

```

Any key not listed here is completely ignored, as though the student had not pressed it. Of the function keys listed, the HELP key will take the student to unit “discuss”, since you have already specified what you want the HELP key to do. Note that this is not possible with a blank -pause- which catches *all* keys. Similarly, what the TERM key will do has been predefined (the student will be asked “what term?”). But the DATA key will be ignored since it is not listed in the -pause- statement, and the student cannot reach unit “tables” with the DATA key until he or she has passed the -pause-. Pressing d, D, NEXT, or HELP1 will take the student past the -pause-. The NEXT key is rather special here in that the preceding specification “next umore”, unlike “help discuss”, tells TUTOR what to do when the present main unit has been *completed*. Thus, pressing NEXT here takes us past the -pause- instead of branching us immediately to a different unit as HELP does.

You may prefer *not* to ignore the HELP key *nor* to use it to access unit “discuss”. In this case, the statement “help discuss” must *follow* the -pause- statement, or a “help q” must precede the -pause- in order to quit specifying a help unit.

## Touching the Screen

Most PLATO terminals have “touch panels” which make it possible for the student to respond by touching the screen. For example, a language drill might show the student pictures of various animals and ask the student to point to the dog. You need a way to tell at which part of the screen the student pointed. This is most easily done with -pause- and -keytype- statements, as in the following example:

```

pause keys=touch
keytype num,touch(1215),touch(100,200)

```

The first statement, using the built-in group “touch”, tells PLATO to expect a touch input. The -keytype- statement will set “num” to 0, if the student touches as close as possible to screen location “1215”; will set “num” to 1, if the student touches near location “100,200”; and will set



“num” to -1, if the student touches the screen elsewhere.

How close the student must be to location “1215” or location “100,200” depends on the resolution or fineness of the touch panel. Most touch panels cover the screen with a 16 by 16 grid of square touch areas. Each square is 32 dots by 32 dots in size, or 4 characters wide by 2 characters high. If the square touched by the student overlaps location “1215” or location “100,200”, TUTOR will consider that the student has pointed at that place.

You can define larger regions of the screen. For example:

```
keytype num,touch(1215;8,4),touch(100,200;64,32)
```

In this case, the `-keytype-` statement will set “num” to 0 if the student touches somewhere within a box whose lower left corner is at “1215”, whose width is 8 characters, and whose height is 4 characters. The variable “num” will be 1 if the student touches within a box whose lower left corner is at fine-grid location “100,200”, whose width is 64 dots, and whose height is 32 dots. The touch-panel square touched by the student must overlap one of your rectangles in order for TUTOR to consider that a rectangle has been touched.

You can abbreviate “touch” by “t” and write “t(1215)” instead of “touch(1215)”.

In addition to the `-pause-keytype` combination, you can also use a `-touch-` judging command with an `-arrow-`. See the PLATO on-line “aids” for details.

## Summary

In this chapter we have discussed, in some detail, the marker properties of the `-arrow-` command. The `-arrow-` command as we have seen serves as an anchor point which TUTOR clings to until the `-arrow-` is satisfied by an “ok” judgment (at which point a search is made for additional `-arrow-` commands). We looked at some cases involving the repeated execution of `-join-` in regular, judging, and search states, and of the *non*-execution of `-goto-` in the judging and search states. We have also looked at other side-effects of the `-arrow-` command, including initializations associated with `-size-`, `-rotate-`, `-long-`, `-jkey-`, and `-copy-`.

In addition, we have seen how the `-bump-` and `-put-` commands can be used to change a student’s response into a form more easily handled by the standard judging commands. This is particularly useful when only slight changes are necessary.

In Chapter 7 we saw how to store numeric and alphanumeric

## The TUTOR Language

responses for later processing (-store- and -storea-). These capabilities make it possible to “do your own judging” in those cases where the standard judging commands are not suitable. The basic TUTOR judging commands provide a great deal of power but cannot handle all possible situations. Fortunately, there is always the possibility of performing calculations on a stored student response, which means that TUTOR is open-ended in its judging power. The regular commands -search- and -move- can be used to manipulate stored character strings. (In Chapter 10 you will find discussions of “segments” and “bit manipulations” which permit you to use the -calc- command to perform additional operations on character strings.) We have also discussed how to handle input from the student by collecting each key with a -pause- command, then using -keytype- (aided by -group-) to make decisions on a key-by-key basis. We have learned, also, how to use similar techniques to determine where the student had touched the screen.

# Judging Student Responses

# 7

You now know quite a bit about how to express (in the TUTOR language) your instructions to PLATO on how to administer a lesson to a student. You may not have realized it, but in the process you have learned a great deal about the fundamental concepts of computer programming. You can calculate, produce complex displays, and construct rich branching structures. You have studied aspects of initialization problems, you have seen the importance of subroutines, and you have looked at some stylistic aspects of good programming practice such as defining variables, placing unit pointer commands at the head of main units, etc. With this solid background you are now ready for a detailed look at how to accept and judge student responses.

In Chapter 1 you saw a common type of judging situation in which you simply listed the anticipated responses after an -arrow- statement, together with the display or other actions to be performed depending on the particular response. Let us see how TUTOR actually processes these judging commands. We will consider a slightly different version of the "geometry" unit. Remember that in the -answer- and -wrong- statements, parentheses enclose synonyms, and angle brackets enclose ignorable words.

## The TUTOR Language

```

unit      geometry
draw      510;1510;1540;510
arrow     2015
at        1812
write     What is this figure?
answer    <it,is,a> (right,rt) triangle
write     Exactly right!
wrong     <it,is,a> square
write     Count the sides!

```




Fig. 7-1.

The order of the initial statements has been changed slightly. TUTOR starts executing this main unit by drawing the triangle. TUTOR next encounters the `-arrow-` command, places an arrowhead at position 2015, and *notes where this -arrow- command is* (the second command in unit “geometry”). TUTOR then executes the `-at-` and `-write-` to display the text: “What is this figure?”

Finally, TUTOR reaches the `-answer-` command. This “judging” command is *useless* at this time because the student *has not entered a response*. There is nothing more that can be done but wait for the student to type a response and enter it by pressing NEXT. We call commands which operate on the student’s response “judging” commands (such as `-answer-` and `-wrong-`). Other commands, such as `-draw-`, `-at-`, `-write-`, and `-calc-`, are called “regular” commands. We see that TUTOR must stop executing regular commands when a judging command is encountered. (This assumes the presence of an `-arrow-` command. An `-answer-` or other judging command without a preceding `-arrow-` is meaningless.)

When the student presses NEXT to enter his or her response, TUTOR looks at its notes and finds that the `-arrow-` was the second command in unit “geometry”. TUTOR starts looking just beyond there for judging commands to process the student’s response. It *skips* the regular commands `-at-` and `-write-` since these are not judging commands and are of no use at this point. It encounters the `-answer-` command and compares the student response with the specifications given in the tag of the `-answer-` command.

If there is not an adequate match, TUTOR goes to the next command looking for a judging command that might yield a match. In this case, the following command is a regular command (-write-) which is skipped. Next there is a -wrong- judging command, and if there is no match to the student's response, TUTOR keeps judging. At this point, there is a -write-regular command which is skipped.

Finally, we come to the end of the unit without finding a matching judging command and must give a "no" judgment to this response (and possibly mark up the response with underlining and X's if the response is fairly close to that specified by the -answer- command). (See Figure 7-1.) The process of starting immediately after the -arrow- in the "judging state" will be repeated each time the student tries again with a revised response.

If, on the other hand, the response adequately matches the -answer-statement, TUTOR has found a match and can terminate the execution of judging commands. It switches to processing regular commands with the result that the following "write Exactly right!" will be executed. (This regular command is skipped unless a match to the -answer- flips TUTOR out of the "judging state" into the "regular state".) Then TUTOR, *in the regular state*, comes to a judging command (-wrong-) which terminates the processing. TUTOR finishes up by placing an "ok" beside the student response. (Similarly, a match to the -wrong- would flip TUTOR to the regular state to execute the regular statement "write Count the sides!")

When the -arrow- is finally "satisfied" by an "ok" judgment, TUTOR returns one last time to the -arrow- and searches for any other -arrow- commands in the unit. In this search it skips both regular and judging commands. In our particular example no other -arrow- is found, so all arrows (one) in the unit have been satisfied. After the student has read our comment, he or she presses NEXT and proceeds to the next main unit.

It may seem wasteful to you that TUTOR keeps going back to the -arrow- only to skip over the regular commands preceding the first judging command. It turns out that skipping a command is an extremely fast procedure, and that keeping a single marker (the location of the -arrow- command within the unit) greatly simplifies the TUTOR machinery.

In the example, the replies "Exactly right!" or "Count the sides!" would be displayed at location 2317, three lines below the response on the screen. This standard positioning can, of course, be altered by an -at-statement. Here is another illustration:

unit	canine
at	2105
write	Name a canine:

(Continued on the next page.)

## The TUTOR Language

```

arrow    2308
answer   dog
write    A house pet.
answer   wolf
write    A wild one!
wrong    cat
write    A feline!

```

Suppose the student enters “wolf” as his response. TUTOR initiates the “judging state” just after the -arrow-. The first -answer- (dog) does not match, so TUTOR stays in the judging state and skips the “write A house pet.” There is a match to the following “answer wolf”, so judging terminates and the regular state starts. The “write A wild one!” is executed, not skipped. Next, TUTOR encounters a “wrong cat”, and since -wrong- is a judging command, this terminates the regular state. The student gets an “ok” judgment. TUTOR searches for another -arrow- but does not find one, so the student has successfully completed the unit. (Various units of this kind are illustrated with animated diagrams in the on-line “aids” available on PLATO.)

This method of processing judging and regular commands yields a readable programming structure, with judging commands delimiting the regular commands used to respond to the student. We have spent time discussing the details in order to simplify our later descriptions of the various types of judging commands used to match, modify, or store student responses.

It is important to point out that the -do- and -goto- commands are *regular* commands. They are, therefore, skipped over during the judging state and during the search state (looking for a possible additional -arrow- after an arrow has been satisfied). There is another command, -join-, which works much like -do- except that the -join- command is universally executed whether TUTOR is in the regular state, the judging state, or the search state. In particular, it is possible to -join- units containing judging commands, whereas a -goto- or -do- is incapable of accessing other units in the judging state (since these regular commands are skipped). Although the -do- command acts essentially like a -join-, it is, nevertheless, a regular command and is skipped during the judging and search states. Only the -join- command itself has the unique characteristic of being performed in all states (regular, judging, and search).

It is frequently useful to handle more than one response in a unit. Let’s ask “Who owned Mount Vernon?” and (after receiving a correct response) ask in what state it is located *but stay on the same page*:

```

unit     wash
at       812

```

```

write   Who lived at Mount Vernon?
arrow   1015
{
answer  <George,G> Washington
at      1120
write   Great!
wrong   Jefferson
at      1112
write   No, he lived at Monticello.
arrow   1715
{
at      1512
write   In what state is it located?
answer  (Va,Virginia)

```

If you say "Jefferson" the -wrong- is matched. Regular commands are executed until you run into the second -arrow-, which ends the range of the first -arrow-. In other words, when you are working on one -arrow-, the next -arrow- is a terminating marker. If you say "Washington", the student gets the "Great!" comment. Since the -arrow- is now satisfied, TUTOR starts at the first -arrow- searching for another -arrow-. In this search state, all commands other than -join- are skipped (-join- may be used to attach a unit that contains another -arrow-). A second -arrow- is encountered, which changes the search state into the regular state. The arrowhead is displayed on the screen and the location of this -arrow- within the unit is noted. The regular commands following this second -arrow- are processed to display the second question. The final -answer- command stops this processing to await the student's response.

There is another way to do this which is probably more readable:

```

unit    wash
next    wash
at      812
write   Who lived at Mount Vernon?
arrow   1015
{
answer  <George,G> Washington
at      1120
write   Great!
wrong   Jefferson
at      1112
write   No, he lived at Monticello.
endarrow
at      1512
write   In what state is it located?
arrow   1715
{
answer  (Va,Virginia)

```


## The TUTOR Language

The `-endarrow-` command defines the end of commands associated with the first `-arrow-`. Note that `-endarrow-` changes the search state to the regular state. One benefit of this form is that the second arrowhead appears on the screen *after* the text of the second question, which often seems more natural.

It may seem rather abrupt that the “Great!” and “In what state is it located?” both appear on the screen at the same time. It might be better to let the student digest the reply before presenting the second question. We might insert a `-pause-` (with the tag “keys=all”) just after the `-endarrow-`. Now TUTOR waits for you to press a key, (which signals that you want to go on) before presenting the next question.

The `-endarrow-` command is quite useful even in units which contain only one `-arrow-`:

```

.
.
.
arrow      1213
answer     dog
write      Bowwow!
answer     wolf
write      How!!
wrong      cat
write      Meow.
 endarrow
calc       y<=37+y
circle     100,250,250
    
```

The commands following the `-endarrow-` will be executed only after the `-arrow-` is satisfied, whether it be by the response “dog” or “wolf”. So this is a convenient way to finish up the unit.

While it is possible to `-join-` or `-do-` units which contain `-arrow-` commands, two seemingly arbitrary rules *must* be followed or you will get unpredictable results:

- 1) A unit attached by `-join-` or `-do-` which contains one or more `-arrow-` commands *must* end with an `-endarrow-` command (possibly followed by regular commands).
- 2) This attached unit must not contain any `-goto-` commands.

If you violate either of these rules, strange things will happen because TUTOR may “undo” from this unit several times (during judging, while processing regular commands, or in the search state).




If you follow these two rules, the `-join-` or `-do-` will act like a text-insertion device whereby your program will act as though you had inserted the attached unit where the `-join-` or `-do-` was. We will discuss these rules in more detail in Chapter 8.

### Student Specification of Numerical Parameters

The `-answer-` and `-wrong-` commands make it easy to specify a list of anticipated responses each of which (due to the specification of synonymous and optional words) can allow the student considerable latitude in the way he or she phrases his or her response. However, in some cases there can be no list of anticipated responses and a different technique must be used. For example, you might ask the student to specify a rocket's launch velocity and use his or her number to calculate and display the rocket's orbit. Or you might ask the student for his or her name for later use in personalized messages such as "Bill, you should look at Chapter 5." In such cases, all you can anticipate is that the response will be a number or a name, but you can't possibly list all possible numbers or names.

Here is an example of such a situation. We will provide the student with a desk calculator accessible on the DATA key. In the desk calculator mode the student can type complicated expressions (such as " $2+6^3$ ") and receive the evaluated result. (Students also have access to a similar calculator mode by typing "`TERM-calc`", which is a built-in PLATO feature.)

unit	mainline	
data	desk	
at	3020	
write	Press DATA for calculator	
.		
.		
.		
unit	desk	
next	desk	\$\$ for repeated use
at	1713	
write	Type an expression.	
	Press BACK when finished.	
arrow	1915	
 store	eval	\$\$ Be sure to define "eval".
ok		\$\$ Accept all responses.
write	The result is <{s,eval}>.	

## The TUTOR Language

The `-store-` command will evaluate the student's expression (e.g., "13sin30°") and store the result in "eval" (in this case, the number 6.5). The `-store-` command is a judging command because it operates on the student's response and can be executed only after the student initiates judging by pressing NEXT. The `-ok-` command is a universal `-answer-` which matches all responses, and unconditionally flips TUTOR from the judging state to the regular state. In this example, it accepts any response and enables the following `-write-` to display the evaluated result.

Note that a student need not use parentheses with functions. For example, `sqrt25`, `cos60°`, `arctan3` are all legal. However, such expressions are illegal in a `-calc-`. In a moment we'll see another way in which TUTOR is more tolerant of students than of authors.

What if the response cannot be evaluated, such as " $(-3)^{1/2}$ " or "19/" or "(3+5)))"? In this case, the student will get a "no" judgment. To see how this works, let's insert a `-write-` statement after the `-store-`:

```

.
.
.
store eval
write Cannot evaluate!
ok
.
.
.

```

Notice that this new `-write-` is normally skipped because the `-store-` leaves us in the judging state. But, if the student's expression cannot be evaluated, `-store-` makes a "no" judgment and *switches* us from the judging state to the regular state. TUTOR then executes the "write Cannot evaluate!", after which it encounters a judging command (`-ok-`) which stops the regular processing. Note that `-store-` terminates judging only on an error condition, whereas `-answer-` terminates judging only on a match, and `-ok-` *always* terminates judging.



You can tell the student precisely (in a `-writec-` statement) what is wrong with his or her expression by use of the *system* variable "formok". This variable is  $-1$  if the student's expression can be evaluated but takes one of several positive integral values for specific errors such as unbalanced parentheses, bad form, unrecognized variable name, etc. The variable "formok" is defined automatically to perform this function. (If you yourself define "formok=v3" you override the system definition and you won't get these features.) The particular values assumed by "formok" can be obtained through on-line documentation at a PLATO terminal.

You can also give the student some storage variables. Let's define a couple of variables for the student:

```
define student $$ special define set
bob=v30,cac=v31
```

Place these defines *ahead of everything else* in the lesson. Suppose you do a -calc- to assign bob←18 and cac←3. If the student types "2bob" he gets 36. Or he can type "bobcac" and get 54, whereas bobcac would be illegal in a -calc- where you would need bob×cac or bob(cac). Only names defined in the set of definitions labeled "student" may be used by the student in this way. Attempted use (by the student) of names in your other sets of defines will give a value of "formok" corresponding to "unrecognized variable name".

We have discussed a desk calculator, but clearly the store/ok combination will work in any situation where we let the student choose a number. Another good example is in an index of chapter numbers:

```
unit table
base
term index $$ or access by means of shift-DATA,
at 1218 $$ as in Chapter 5
write Choose a chapter:
      1) Introduction
      2) Nouns
      3) Pronouns
      4) Verbs
arrow 1822
 long 1 $$ get one digit; don't wait for NEXT
store chapter
 no
jump chapter,x,x,intro,unoun,pron,verb,x
write Pick a number between 1 and 4.
```

(As previously mentioned in Chapter 5, it would be better to execute the -base- command only after deciding to jump, so that the student could still use the BACK key to return to the original unit.) The -long- command following an -arrow- (but preceding any judging commands) sets a limit on the length of the student's response. The "long 1" is particularly useful here because the student need not press NEXT but has only to press the single number to begin the judging process. (For -long- of greater than 1 there must be an accompanying "force long" statement or else a NEXT key is required.) The -long- command must precede any

## The TUTOR Language

judging command since the “long” specification is needed *before* the student starts typing (whereas we proceed past the judging command only *after* the student enters a response). The -long- command may be thought of as a kind of modifier of the -arrow- command, in the sense that the -arrow- sets a default maximum response length which is overridden (or modified) by the following -long- statement.

The -no- in this index unit is similar to an -ok- command in that it unconditionally terminates judging, but the -no- command makes a “no” judgment. If “chapter” is a number from 1 to 4, the -jump- will take the student to his chosen chapter. (Since -jump- erases the screen the “no” will not be seen.) If, however, “chapter” is not in range, we fall through the -jump- to an error message, and there will be a “no” next to the response (and the student must try again).


## Student Specification of Non-Numerical Parameters

Now that we have seen how to let the student specify a number, let’s see how to ask the student to tell us his or her name or nickname to permit us to communicate by name:

```

unit      meet
at        1215
write     Hello, my name is Sam Connor.
           What's your name?
arrow     162Ø
long      8          $$ limit to 8 characters
storea   name       $$ define "name" earlier
ok
write     Pleased to meet you, <a,name>!

```



The -storea- command is a judging command which will store *alphabetic* information as distinguished from numeric information. The <a,name> is the embedded form of the statement “showa name” which will display alphabetic information. This unit will feed back to you any name you give it. Notice that you can’t enter a name of more than 8 characters because of the -long- command. TUTOR stores a capital letter as a “shift” character plus the lower-case letter, so if capitalized, the name must be shorter because a capital letter counts as *two* characters. (Insert a “force long” statement anywhere before the -storea- if you would like judging to start upon hitting the -long- limit, without having to press NEXT.)

A statement of the form “storea name,3” will store just the first three characters of the student’s response. You can get and keep a character count of the length of the student’s name, including “shift” characters, by referring to the system variable “jcount”, which is a count of the number of characters in the copy of the student response used for judging—hence the “j”. With these facts in mind, change the -storea- to:

```
storea name,(namlNg←jcount)
```

This will store the whole response and save the length. Be sure to define both “name” and “namlNg”, but do *not* define “jcount” or you will override TUTOR’s definition of its function. Also, to show the precise number of characters, change the embedded -showa- to:

```
<a,name,namlNg>
```

The reason for saving the present value of “jcount” in “namlNg” is that “jcount” will change at each -arrow- in the lesson, whereas throughout the lesson you will repeatedly use “showa name,namlNg” or <a, name,namlNg> to call the student by name. So, you want “namlNg” to keep the name length. Incidentally, a -showa- with only a single argument (such as “showa name”) will show ten characters, which is the number of characters (including shift characters) that will fit in one of your variables.

It is possible to store alphabetic information which is longer than ten characters. Change the “long 8” to “long 20”. Suppose you’ve defined “name=v24.” In this case, you must make sure that you are *not using v25*, and change your defines if necessary. The 20-character name will need both v24 *and* v25 since each variable can hold only ten characters. With these changes it is possible to enter a long name (e.g., Benjamin Franklin, which is 19 characters counting shift characters).

### Difference Between Numeric and Alphabetic Information

When we were studying the desk calculator unit, we defined a variable “bob=v30” for the student. Suppose the student responds with the word “bob”. If we use a numeric -store-, we will get the number presently contained in v30, which might be 529.3. If we use an alphabetic -storea-, we will get the string of characters “bob” which is simply a name and nothing more. Perhaps the distinction is most easily seen with an example, which you should write and try out at a PLATO terminal:

## The TUTOR Language

```

define student
    bob=v1
define ours,student          $$ include "student" set of defines
    name=v2,num=v3
unit test
calc bob←π                  $$ π means 3.14159 . . . . .
arrow 1815
store num
storea name
ok
write num=<s,num>
    name=<a,name,jcount>

```

Consider various responses. For example, “2bob” should give a *numeric*  $2\pi$  (6.2832) and an *alphabetic* “2bob”. Most often, we speak of “alphanumeric” information (letters and numbers) in the latter case. The response “3-4/5” yields a numeric 2.2 and an alphanumeric “3-4/5”.

In other words, a storea/showa combination feeds back *exactly* the alphanumeric text entered by the student. However, a -store- involves a numerical evaluation of the student’s response, and a later -show- converts this numerical result into appropriate characters to display on the screen (so that you can read the result). You might interchange the “num” and “name” arguments on the -store- and -storea- commands to see the unusual things that happen if you pair -store- with -showa- (instead of -show-) or if you pair -storea- with -show- (instead of -showa-).

To sum up, if you accept numeric information with a -store-, display it with a -show-. If you accept alphanumeric information with a -storea-, display it with a -showa-.

### More On -answer- and -wrong- (Including -list- and -specs-)

There are some additional features of -answer- (and -wrong-) which should be pointed out. First, -answer- will not only handle word or sentence responses but will also handle numbers:

```
answer 7 women <and> 5 men
```

This -answer- will be matched by a student response of the form “14/2 women and 3+2 men” because simple expressions such as 14/2 or 3+2 are evaluated by the -answer- command. Currently, the -answer- command will not handle very complicated numerical expressions.

(Later we will discuss the `-ansv-` and `-wrongv-` commands which handle expressions as complicated as those handled by `-store-` but without the sentence capabilities of `-answer-` and `-wrong-`. There are also `-ansu-` and `-wrongu-` commands which are similar to `-ansv-` and `-wrongv-` but treat scientific units on a dimensional basis.)

If the student says “37 women and 5 men,” the incorrect number 37 will have xx under it, whereas the response “6.5 women and 5 men” will have the 6.5 underlined since it is nearly correct (similar to a misspelling of a word). Normally `-answer-` and `-wrong-` consider numbers off by less than 10% to be “misspelled.” You can alter these specifications by preceding the list of `-answer-` and `-wrong-` commands with a `-specs-` command:

```
unit      trial
arrow    1815
specs    toler,nodiff
answer   7 women <and> 5 men
```

The `-specs-` command is a *judging* command which affects the operation of other judging commands which follow it. Here it has been used to specify that a “tolerance” of 1% is permitted and that “no difference will be allowed for underlining” (normally 10%). Having specified both “toler” and “nodiff,” any expressions within 1% of 7 and 5 will be accepted, but expressions with larger discrepancies will not be underlined.

Note carefully that since `-specs-` is a judging command, it terminates the processing of regular commands. Among other things, this means that a `-long-` command must *precede* the `-specs-`, not follow it. If `-long-` comes after `-specs-`, TUTOR could not prevent the student from entering a longer response (since it could not see the `-long-` command before it paused for the student’s response).

Here are some other useful applications of `-specs-`:

```
specs    okcap,okspell
answer   the antidisestablishmentarianism doctrine
```

This allows the student to capitalize words, and specifies that misspellings are to be considered ok. Note that if the `-answer-` tag contains capitalized words, the student must also capitalize those words. The “okcap” makes capitalization optional only for those words you have not capitalized. You can use `-specs-` to ignore extra words:

## The TUTOR Language

```
specs okextra
answer Washington
```

This states that it is ok to have extra words, so that “It was George Washington” will be an acceptable response. The following is another example of -specs- capabilities:

```
specs noorder
answer apples pears and peaches
```


This specifies that no particular word order is required. Note the absence of commas in the -answer- tag. (Such punctuation marks are not allowed there, but all punctuation marks are ignored in the student’s response, so he or she may use commas). Also, note that “answer apples, pears and peaches” would represent two synonymous answers, and the student could respond either with “apples” or with “pears and peaches”. There exists a much less powerful -exact- command (as well as other techniques) for judging particular punctuation when that is necessary. For example, it is possible to use the -change- command to redefine the comma to be a “word” rather than a punctuation mark. In that case, some otherwise unused character must be defined to take the place of the comma in specifying synonyms.

```
specs nookno
ok
```

Here we specify that no “ok” or “no” be displayed beside the student’s response, contrary to the normal situation. (As an alternative, the -okword- and -noword- commands can be used to change the words TUTOR uses from “ok” and “no” to something else.)

(For other -specs- capabilities see reference material described in Appendix A.)

Another important feature of -specs- (in addition to its use in specifying various options) is that it marks a place to return to *after* judging. Consider the following unit. You do *not* define the system variable “spell”.

```
unit      presi
at        1212
write     Name one of the first three U.S. presidents.
arrow     1513
specs     bumpshift      $$ delete shift codes
          at             2508
          writec        spell, No misspellings!,

          Underlining indicates a misspelled word.
```



```

answer  washington
write   Good old George.
answer  adams
answer  jefferson

```

Suppose the student types "WASHINGTON". TUTOR starts judging just after the -arrow- and encounters -specs-, a judging command. The tag ("bumpshift") tells TUTOR to change the response to "washington" for judging purposes. (Incidentally, this operation changes "jcount", the character count of the judging copy of the student's response, from 20 to 10 because the "shift" characters are knocked out.) Moreover, TUTOR makes a note that it encountered a -specs- command as the fourth command in unit "presi", and this marker will be used in a moment. TUTOR skips the following -at- and -writec- because regular commands are skipped in the judging state.

Next, TUTOR encounters "answer washington" which matches the student's (altered) response, and this terminates judging. The succeeding regular commands are processed as usual. In this case, there is only a "write Good old George" before we run into another judging command ("answer adams") which stops the processing.

Actually, processing has not completely stopped. It is at this point that TUTOR asks one last question: "Did I pass a -specs- command in processing this response?" The answer is yes (at the fourth command in unit "presi"). *TUTOR now processes any regular commands following that -specs- marker.* In this case, TUTOR does an "at 2508" and a -writec- before finally being stopped (*really* stopped this time) by the first -answer- command.

The -writec- refers to the system variable "spell" which is true (-1) if the spelling is correct, and false (0) if a misspelling has been detected. The variable "spell" is -1 if there are no underlined words, but there may be X'ed words (words that are completely different).

The usefulness of the marker property of -specs- is that you can specify a central place to put messages and calculations, which should be done no matter which judging command is matched. We will see additional applications of this useful feature of -specs-. Notice that a later -specs- command will override an earlier -specs- marker in a manner analogous to the way a later -help- command overrides an earlier setting of the "help" marker. Note, too, that if no regular commands follow the -specs-, TUTOR finds nothing to do when it comes there after being nearly stopped as described above. This was the situation in our previous examples such as:

```

specs  nookno
ok

```

## The TUTOR Language

In this example, there are no regular commands between the `-specs-` and the `-ok-`.

Let us return for a moment to the `-answer-` command. We had examples involving synonyms such as `(right,rt)` or `(Va, Virginia)`. A convenient way to specify synonym lists which occur frequently in a lesson is to define a `-list-`:

```
list affirm,yes,ok,yep,yeah,sure,certainly
```

Here “`affirm`” is the title of a list of synonyms (“`affirm`” is not itself a member of that list). With this definition, which should be placed at the very beginning of your lesson along with your `-define-` statement, you can write:

```
answer ((affirm))
wrong  maybe ((affirm))
```

These are equivalent to:

```
answer (yes,ok,yep,yeah,sure,certainly)
wrong  maybe (yes,ok,yep,yeah,sure,certainly)
```

Note that “`answer we affirm`” does not imply this list of synonyms, just as a single important word by itself does not refer to a list. You can use the list equally well to specify optional words, as in:

```
answer <<affirm>> it is
```

Here `<<affirm>>` is equivalent to `<yes,ok,yep,yeah,sure,certainly>`. Note that `<affirm>` merely refers to the single word “`affirm`”. Double marks are needed to refer to the list whose title is “`affirm`”. You can combine references to synonym lists with individual words. For example:

```
wrong  usually (definite, (affirm))
answer often <definite, <affirm>>
```

The following list might also be particularly useful:

```
list negate,no,nope,not,never,huhuh
```

This covers the main capabilities of the `-answer-` and `-wrong-` commands and their associated `-list-` definitions. The `-specs-` command may be used to modify how `-answer-` works and also serves as a useful marker. The marker function of `-specs-` is not limited to `-answer-` but holds for any judging commands which follow it, including `-ok-` and `-no-`.

The -answer- (or -wrong-) command can nicely handle responses which involve a relatively small vocabulary of words. It is, therefore, adequate when the context limits the diversity of student responses (such as foreign language translation drills where there are only a few permissible translations of the sentence and each such sentence contains a rather small number of allowable words). The detailed markup of the response provides the student with useful feedback in such a drill.

The -answer- command is not well-suited to a more free dialog with the student where the context is broader and where the vocabulary used by the student may encompass hundreds of words. In the next section we discuss the -concept- command which can cope with more complexity.

### Building Dialogs With -concept- and -vocabs-

An excellent example of a dialog is a lesson on qualitative organic chemistry analysis written by Prof. Stanley Smith of the Department of Chemistry, University of Illinois, Urbana. This lesson helps students practice their deductive skills on PLATO before they identify unknown compounds in a laboratory. Prof. Smith has PLATO randomly choose one of several organic compounds and then invites the student to ask experimentally-oriented questions aimed at identifying the unknown. Typical questions are: "what is the melting point;" "does it dissolve in sulfuric acid;" "show me the infrared spectrum;" "is it soluble in H<sub>2</sub>O." There are over a hundred such concepts important in this simulated laboratory situation, and since each concept has many equivalent forms drawing upon a vocabulary of hundreds of words, the number of possible responses is astronomical. How can this be handled?

Although the context is far broader than that of a language drill, it is, nevertheless, sufficiently limited to be tractable. No attempt is made to recognize arbitrary student responses such as "cook me some apple pie." With this quite reasonable restriction, the situation can be handled by using the -vocabs- command (analogous to -list-) to define a large vocabulary (with appropriate "synonymization") associated with a list of -concept- commands (analogous to -answer-) which express the basic concepts meaningful in the context of this lesson. The following is a fragment of the -vocabs- command:

```

vocabs labtest      $$ vocabulary must have a name
      <is,it,a,does,in,what>  $$ ignorable words
      (color,red,blue,green)  $$ word number 1 and synonyms
      (water,H2O)           $$ word number 2 and synonym
      (dissolve,soluble)     $$ word number 3 and synonym
      .
      .
      .

```

## The TUTOR Language

And here are a couple of the many -concept- commands:

```

.
.
arrow    1213
concept  what color
write    It is red.
concept  soluble in water
write    It's slightly soluble in water.
.
.
.
.
.
.

```

Consider what TUTOR does with “concept soluble in water”. TUTOR knows that -concept- has a tag consisting of words defined by a previous -vocabs-. (As usual with such matters, the -vocabs- should be at the beginning of the lesson.) The first word in the tag is “soluble” which TUTOR finds is the *third* very important word in the vocabulary (discounting the ignorable or optional words “is,it,a,” etc.). TUTOR groups synonyms together so that “dissolve”, too, would be considered a “number 3” vocabulary word. The next word of the tag is “in” which TUTOR throws away because the -vocabs- command says that the word is ignorable. The next word is “water”, which is in the *second* set of important -vocabs- synonyms. The net result is that “concept soluble in water” is converted to the sequence “3 2”.

Now, consider a student in this lesson who types “does it dissolve in H<sub>2</sub>O”. Superficially, this looks quite different from the -concept- tag “soluble in water”. However, TUTOR encounters a -concept- command which, unlike -answer-, indicates that the student’s response should be looked up in the defined vocabulary. (In the case of -answer- there is no one vocabulary set because each -answer- may include various -list- references and particular words specific to that -answer-.) By a process identical to the conversion of the author’s -concept- tag, TUTOR converts the student’s response into “3 2”. This compact form “3 2” does not match the first “concept what color” (which was converted to “1”), so, TUTOR proceeds to the next judging command, which is “concept soluble in water” or rather its converted form “3 2”. This matches, so judging terminates and regular processing begins. The student gets a reply “It’s slightly soluble in water.”

Notice that the first -concept- encountered triggers the transformation of the student's response into the compact form suitable for looking through a very long list of concepts. If the -vocabs- contains an entry such as (five,5,cinco), the student may match this entry with "3+2", just as in an -answer- statement involving numbers.

You will have to experiment a little with this machinery in order to learn how best to manage the synonymization in the vocabulary. This does depend on the context. In an art lesson it would be disastrous to call red and blue synonyms as was done here, but it makes sense in this context (where the only concept related to color has to do with "what color is it", which means essentially the same as "is it red" or "is it blue").

You will find that the use of words not defined by -vocabs- will result in a markup indicating which words are undefined (X's will appear under these words). If your context is such that you need worry only about key words and don't care if the student asks "does it dissolve superbly in water", you might precede the first -concept- with a "specs okextra" which says that extra student words not found in the vocabulary may be ignored, as though they had been so specified in the -vocabs- tag. In that case, you need not define any ignorable words with -vocabs-, but you would write "concept dissolve water", not "concept dissolve in water" since extra *author* words are not tolerated. If you don't use "specs okextra", the student's word "superbly" will be marked (xxxxxxx). If the student misspells a vocabulary word, that word will be underlined such as "saluble in water."

The following is an alternative and more detailed version of the heart of the dialog lesson, which illustrates several points. It is a rather complex example which brings together many aspects of TUTOR. Note particularly that the -concept- statements now are listed one after the other. The variable "unknown" is a number from 1 to 4 (associated with which compound the student is attempting to identify). The *system* variable "anscnt" is set to zero when judging starts (and when a -specs- is encountered) and it counts the number of -answer-, -wrong-, -ok-, -no-, and -concept- commands passed through. If the third such command terminates judging, "anscnt" will have the value 3. If no match is found, "anscnt" is set to -1.

```

.
.
arrow 1213
wrong what is it
write That is for you to determine!

```

(Continued on next page.)

## The TUTOR Language

```

specs          $$ to clear ansent again
goto          ansent>0,unknown,x
writec        vocab,I don't understand your sentence.,
              The xxxx words are not in my vocabulary.
concept       what color
concept       soluble in water
concept       boiling point
.
.
.
.
unit          unknown
goto          unknown-2,reply1,reply2,reply3,reply4
*
unit          reply1
writec        ansent,,It is colorless.,
              It is slightly soluble in water.,
              The boiling point is 245-247° C.,
.
.

```

The statement “wrong what is it” is necessary because a “concept what is it” contains only ignorable words and would, therefore, not distinguish between “what is it” and “does it what”, which also contains only ignorable words. Since -specs- resets “ansent” to zero, “ansent” will have the value 2 if the student’s response matches the second -concept- (“soluble in water”). No regular commands follow this -concept-, so TUTOR goes right to the -specs- marker to execute the regular commands there. Since “ansent” is greater than zero, TUTOR does a -goto- to unit “unknown”, where there is a -goto- to unit “reply1” (assuming we are working on unknown number 1), which writes “It is slightly soluble in water” on the student’s screen.

This structure makes it *very* easy to add a fifth unknown compound to the lesson. The -vocabs- and list of -concept- commands do not have to be changed, since the basic concepts and vocabulary are pertinent to the analysis of *any* compound. All that is necessary is to add “reply5” to the end of the conditional -goto- in unit “unknown” and to write a unit “reply5” patterned after unit “reply1”. The lesson revision is completed!

What happens if the student says “it what does”? This will not match the -wrong- nor any of the -concept- commands, so “ansent” will be -1. Therefore, the -goto- just after the -specs- will fall through to the

following `-writec-`, which gives one of the two messages dependent on the system variable `"vocab"`: *true* if all words are found in vocabulary, *false* if some words are not found (these words would be underscored with `xxxx`). In this case, the student will get the message "I don't understand your sentence", whereas if the student says "what is elephant" he will see the `xxxx`'s under "elephant" and get the message "The `xxxx` words are not in my vocabulary".

That was a fairly complicated example, but the discussion is justified by the general usefulness of many of the techniques employed and by the extraordinary power such a structure yields, both in its sophisticated handling of student responses and in the ease of expansion to additional options.

Suppose the `-arrow-` is in unit "analysis". One way to proceed from one question to the next would be to place a "next analysis" in this unit. There is an efficient way to avoid erasing and recreating the display associated with this unit. Instead of proceeding, let's judge each response "wrong" so that we stay at this `-arrow-`. Replace the `-specs-` command with these two statements:

```

.
.
.
.
.
.
.
specs nookno $$ so "no" doesn't appear
judge wrong
.
.
.
.

```

Despite its name, `-judge-` is a *regular* command, *not* a judging command. It can be used to alter the judgment made by the judging commands. In this case, TUTOR first skips over this regular command to get to the `-concept-` commands. If one of these commands matches the student response, TUTOR makes an "ok" judgment, but upon going to the `-specs-` marker TUTOR finds a "judge wrong" which overrides the earlier judgment. TUTOR keeps going, processing regular commands, and produces a message as we have seen before. The "nookno" specification prevents a "no" from appearing on the screen and the student simply sees

## The TUTOR Language

our message. But the `-arrow-` has not been satisfied, so when the student presses NEXT, TUTOR erases the response and awaits a new response. Each time, the student gets a reply to his or her experimental question, and the “wrong” judgment takes us back to the `-arrow-`.

This is a good way to manage the screen because only a small portion of the display changes (the surrounding text and figures remain untouched). The “next analysis” re-entry to this same main unit would quickly get tiresome because of the repetitious replotting of the surrounding material.

You should now be able to use `-answer-`, `-wrong-`, and `-list-` in situations where the vocabulary is small and `-concept-` and `-vocabs-` where the vocabulary is large. You have seen how to use `-specs-` both to specify various judging options and to mark a place where post-judging actions can be centralized. You have seen one form of the regular `-judge-` command “judge wrong” which overrides an “ok” judgment made by an `-answer-` or `-concept-`.

Another way to get a “wrong” judgment is to use `-miscon-` (“misconception”) commands instead of `-concept-` commands. Just as `-wrong-` is the opposite of `-answer-`, `-miscon-` is the opposite of `-concept-`.

There is a particularly convenient way to make different concepts equivalent, including different word orders:

```
concept  dissolve in water
          water soluble
          drop in water
write    It's soluble in H2O.
```

The “continued” `-concept-` specifies synonymous concepts. If the student’s response matches any of these three concepts the same message will be given. Also, “anscnt” will be the same no matter which of these concepts makes the match.

Use of `-vocabs-` makes possible the underlining of misspelled vocabulary words (or their acceptance with a “specs okspell”), just as with the `-answer-` command. Similarly, “specs noorder” can be used to indicate that no particular word order is required. There is a `-vocab-` command which permits a larger vocabulary (at the price of giving up these spelling and order capabilities). Just as the multi-word phrase “sodium\*chloride” can be used with the `-answer-` command, so can such phrases be specified in a `-vocabs-` vocabulary.

At times you may be interested mainly in root words, no matter what endings are attached. The words “walk”, “walks”, “walked”, “walker”, and “walking” can be added to a `-vocabs-` very simply as “walk/s/ed/er/ing”, which saves you some typing effort. If you want all of these *except*



for “walk” itself to be added to the vocabulary, use a double slash after the root: “walk//s/ed/er/ing”.

An even more compact way to define common endings is with -endings- commands:

```

endings  0,s,ed,ing
endings  9,er,est
...
vocabs   sample
         will/0,full//9
    
```

The use of the “0” and “9” sets of endings causes the vocabulary to contain these words: will, wills, willed, willing, fuller, and fullest (“full” itself is missing, due to the double slash). An -endings- set must be identified by a number from 0 to 9.

### Numbering Vocabulary Words

Suppose the student is encouraged to ask questions such as “What is the capital of Alabama?” or “What is the area of Alaska?” A compact and powerful way to handle all the states is to specify a vocabulary class (“state”) and number the various states. For example:

```

define   st=v1
vocabs   inquiry
         <What,is,the,of>
         (state, Alabama=1, Alaska=2, Arizona=3, . . . . .)
         capital, area
...
concept  capital of state,st<=state
writec   st,,,Montgomery,Juneau,Phoenix
concept  area of state,st<=state
writec   st-2151,6091586,4001113,9091 . . . . .
write    sq. mi.
    
```

If the student asks “What is the capital of Alaska?” the first -concept- is matched (“capital of state”), and variable “st” is assigned the value “2”, since “Alaska” was given the value “2” in the vocabulary. Now “st” can be used in the following -writec- to tell the student the name of the capital (Juneau). Similarly, if the student asks “What is the area of Arizona”, the second -concept- is matched, “st” is assigned the value “3”, and the student is given the reply “113,909 sq. mi.”

## The TUTOR Language

We can go even further. Consider this altered version, in which the two -concept-s are combined:

```

define    st=v1,prop=v2
vocabs    inquiry
          <What,is,the,of>
          (state,Alabama=1,Alaska=2,Arizona=3, . . . . .)
          (property,capital=1,area=2)
. . .
concept   property of state, st←state,prop←property
writec    2(state-1)+(prop-1)↑Montgomery↑51,609
          Juneau↑586,400↑Phoenix↑113,909↑ . . . . .
writec    prop=2↑ sq. mi.↑↑

```

Suppose the student asks about “the area of Alabama”. The -concept- is matched, “st” is assigned the value “1”, and “prop” is assigned the value “2”. The expression “2(state-1)+(prop-1)” reduces to “2(0)+1” or “1”, which picks out “51,609” from the first -writec-. Since “prop” does equal “2”, the second -writec- will write “sq. mi.” on the screen beside the area number. (It would be good practice for you to determine the steps that would be taken if the student asked about “the capital of Arizona.”)

Synonyms, phrases, and endings can be numbered, as in this -vocabs- entry:

```
(verbs, walk=1/ed=2, stroll=1/ed=2, went*past=3)
```

According to this numbering scheme, “walk” and “stroll” are number 1 among the “verbs,” “walked” and “strolled” are number 2, and the phrase “went past” is number 3.

### The -judge- Command


We have encountered the regular command -judge- (*not* a judging command) and have seen how it can be used to “judge wrong” a response that had already received an “ok” judgment. The -judge- command may also be used to “judge ok” a response (disregarding what a previous judging command may have had to say). The following is a conditional form for this type of -judge- command:

```
judge 3a-b,ok,x,wrong
```

This form will either make the judgment “ok”, leave the current judgment as is (the “x” option), or make the judgment “wrong”, depending on the condition “3a-b”.

Here is a useful example:


```

unit    negative
at      1214
write   Give me a
        negative number:
arrow   1516
store   num
write   Cannot evaluate your expression.
ok      $$ terminate judging
 judge num<0,ok,wrong
writec  num<0,Good!,That's positive!
```

We could just as well have written “judge num<0,x,wrong” since the original judgment was a universal “ok”. (Later we will study -ansv- and -wrongv- which are also useful in numerical judging.) Note that “judge ok” and “judge wrong” do *not* cut off the following commands. In the above example, the -writec- is performed, even though it follows the -judge- command. The -judge- command here merely alters the judgment. If you want to cut off the following commands, you can use “judge okquit” or “judge noquit”.

We have been using the -ok- or -no- commands to terminate judging unconditionally, as in the last example. It is sometimes useful to be able to switch in the other direction, from the regular state to the judging state. For example, suppose you want to count the number of attempts the student makes to satisfy the -arrow-:

```

.
.
.
.
calc    attempt<=0
arrow   1518
ok
 calc    attempt<=attempt+1
judge   continue
answer  cat
etc.
```

## The TUTOR Language

Judging starts just after the -arrow-. The -ok- terminates judging to permit executing the regular -calc- which increments the “attempt” counter. Then the regular -judge- command says “continue judging”, which switches TUTOR back into the judging state to examine the -answer- and other judging commands which follow. If the response is finally judged “no”, the student will respond again, and since judging starts each time from the -arrow-, the “attempt” counter will record each try. (Actually, system variable “ntries” *automatically* counts the number of tries, but structures similar to the structure illustrated here are often useful.)

Leaving out the -ok- and “judge continue” (which permit counting each attempt) is a common mistake. If you write:

```

calc      attempt←∅
arrow    1518
calc      attempt←attempt+1
answer   cat

```

then “attempt” will stop at one. TUTOR initializes “attempt” to ∅, then encounters the -arrow- and notes its position in the unit. Then, the following -calc- increments “attempt” to 1, after which the -answer-judging command terminates this regular processing to await the student’s response. The student then enters his or her response and TUTOR starts judging. The first command after the -arrow- is the incrementing -calc-, which is skipped because it is a *regular* command and TUTOR is looking for judging commands. This will happen on each response entry, so “attempt” never gets larger than one. This explains the importance of bracketing the -calc- with -ok- and “judge continue”.

A related option is “judge rejudge” which is similar to “judge continue”. We have seen that “specs bumpshift” alters the “judging copy” of the response by knocking out the shift characters. The judging copy is the version of the response which is examined by the judging commands (such as -answer-). This version may differ from the student’s actual response due to various operations such as “specs bumpshift”. It is also possible to -bump- other characters or to -put- one string of characters in place of another. All such operations affect the judging copy *only* and do not touch the original response, which remains unmodified. The statement “judge rejudge” replaces the judging copy of the response with the original response, thus cancelling the effects of any previous modifications of the judging copy. The statement also initializes the system variables associated with judging, including “anscnt”. It is, therefore, much more drastic than “judge continue”, which merely

switches TUTOR to the judging state without affecting the judging copy or the system variables.

Another exceedingly useful `-judge-` option is “judge ignore” which erases the student’s response from the screen and permits him or her to type another response without first having to use `NEXT` or `ERASE`. Unlike “judge wrong”, “ok”, or “continue”, “judge ignore” stops all processing and waits for new student input. (Even the commands following a `-specs-` won’t be performed.) On the other hand, TUTOR goes on to the following commands after processing `-judge-` with tags “ok”, “wrong”, or “continue”.

The following routine (which permits the student to move a cursor on the screen) is a good example of the heightened interaction made possible through the use of “judge ignore”. We use the typewriter keys `d`, `e`, `w`, `q`, `a`, `z`, `x`, and `c` which are clustered around a 3 key by 3 key square on the keyboard, to indicate the eight compass directions for the cursor to move on the screen. These keys (shown in Fig. 7-2) have small arrows on them to indicate their common use for moving a cursor.



Fig. 7-2.

## The TUTOR Language

```

unit      cursor
calc      x←y←250  $$ initialize cursor position
          dx←dy←10  $$ cursor step size
do        plot    $$ plot cursor on screen
inhibit   arrow   $$ don't show the arrowhead
arrow     3201
long      1
specs
do        move    $$ come here after judging
          -do- is a regular command
answer    d       $$ east:  ansct=1
answer    e       $$ northeast  2
answer    w       $$ north     3
answer    q       $$ northwest  4
answer    a       $$ west      5
answer    z       $$ southwest  6
answer    x       $$ south     7
answer    c       $$ southeast  8
ignore
*
unit      move
*erase   old cursor
mode     erase
do       plot
mode     write
*increment x and y on the basis of "ansct"
calcs    ansct-2,x←x+dx,x+dx,x-x-dx,x-dx,x-dx,
          x,x+dx
calcs    ansct-2,y←y+dy,y+dy,y+dy,y,
          y-dy,y-dy,y-dy
do       plot
judge    ignore
*
unit     plot
at       x,y
write    +          $$ use "+" for cursor

```



This routine permits the student to move the cursor rapidly in any direction on the screen. A letter which matches one of the -answer- statements will cause the -calcs- statements to update  $x$  and  $y$  appropriately to move in one of the eight compass directions. The "long 1" makes it unnecessary to press NEXT to initiate judging, and the "judge ignore" after the replotting of the cursor again leaves TUTOR awaiting a new response. The "judge ignore" greatly simplifies repetitive response

handling such as that which arises in this example. Normally, such a cursor-moving routine would be associated with options to perform some action, such as drawing a line. This would make it possible for the student to draw figures on the screen.

In addition to the `-judge-` options discussed above, there is a “judge exit” which throws away the `NEXT` or `timeup` key that had initiated judging. This leaves the student in a state to type another letter on the end of his or her response. This can be used to achieve special timing and animation effects.

To summarize, the `-judge-` command is a *regular* command used for controlling various judging aspects. The `-ok-`, `-no-`, and `-ignore-` are *judging* commands which somewhat parallel the “judge ok”, “judge no”, and “judge ignore” options. The “judge rejudge” and “judge continue” options make it possible to switch from the regular state to the judging state (with or without reinitializing the judging copy of the student response and the system variables associated with judging). All of these options may appear in a conditional `-judge-` with “x” meaning “do nothing”:

```
judge expr,no,x,ok,continue,wrong,rejudge,x,ignore,ok
```

The subtle difference between “judge wrong” and “judge no” will be discussed in Chapter 12 in the section on “Student Response Data”. Basically, “judge wrong” is used to indicate an anticipated (specific) wrong response, whereas “judge no” indicates an unanticipated student response. Additional `-judge-` options are “quit”, “okquit”, and “noquit”.

### Finding Key Words: The `-match-` and `-storen-` Commands

The `-match-` command, a judging command, makes it easy to look for key words in a student’s response. The `-match-` command will not only find a word in the midst of a sentence, but it will replace the found word in the judging copy with spaces, to facilitate the further use of additional judging commands (including `-match-`) to analyze the remainder of the response. Here is the form of a `-match-` statement:

```
match num,dog,(cat,feline),horse,(pig,hog,swine)
           0      1      2      3
```

Here “num” is a variable which will be set to `-1` if none of the listed words appear in the student’s response, to `0` if “dog” appears, to `1` if “cat” or “feline” is present, `2` if “horse” is in the response, etc. In any case,

The TUTOR Language

-match- terminates judging, with a “no” judgment if num= -1 or an “ok” judgment otherwise. What if more than one of the words appear in the student’s response? Suppose the student says:

“horse and dog”

In this case “num” will be set to 2 because in looking at the first student word we find a match (horse). The judging copy of the response is altered by replacing “horse” with spaces so that it looks like:

“        and dog”

If we were to execute the same -match- again we would get the number 0 corresponding to “dog”, and the judging copy would then look like:

“        and        ”

Note that -match- always terminates judging, so that a “judge continue” is needed before another -match- can be executed. Also note that the key words are pulled out in the order in which they appear in the student’s response, not in the order they appear in the -match- statement.

There are many other ways in which the -match- can be utilized. First, we can improve greatly on our cursor program:

```

.
.
.
.
inhibit  arrow
arrow    3201
long     1
match    num,d,e,w,q,a,z,x,c
do       num,x,move
judge    ignore
.
.
.
.

```



Unit “move” remains unchanged except to replace (in two places) the expression “anscnt-2” by the expression “num-1” (and we can delete the “judge ignore” in unit “move”). We see that -match is useful for converting a word to a number which represents the word’s position in a list.



Another good use of `-match-.is` in an index:

```

unit      table
base
term      index
at        1218
write     Choose a chapter:
          a) Introduction
          b) Nouns
          c) Pronouns
          d) Verbs
arrow     1822
long      1
match     chapter,a,b,c,d
calc      chapter<-chapter+1
jump      chapter,x,x,intro,unoun,pron,verb,x
write     Pick a,b,c, or d.
```

Notice that we must increment “chapter” by one if we want topic “a” to be chapter 1, since `-match-` associates  $\emptyset$  with the first element in its list (`-1` is reserved for the case where no match is found). If no match is found, there is a “no” judgment. (Again, `-base-` could come later in the unit, or at the beginning of the chapters, in which case the `BACK` key would still be active for returning to the place from which the index was accessed.)

These applications barely scratch the surface of `-match-s` capabilities. Here are some other ideas on how to use `-match-`:

- 1) Use `-match-` to pull out negation words such as no, not, never, etc. Then “judge continue” and use `-answer-` or `-concept-` commands to analyze the remainder of the response. You can in this way separate the basic concept from whether it is negated, with the negation information held in the `-match-` variable for easy use in conditional statements.
- 2) Use `-match-` to identify and remove a key-word directive before processing the rest of the information. This comes up in simulating computer compilers, in games (“move” or “capture”), etc.

A related command is `-store-`, which will find a simple numeric expression in a sentence, store it in your specified variable, and replace the expression with spaces. This is particularly useful for pulling out several numbers. The `-store-` command will handle much more complicated expressions including variables as well as numbers, but can get only one number. For example, the student might respond to a question about graph-plotting coordinates with “32.7,38.3”. These two numbers can be acquired by:

## The TUTOR Language

```

.
.
arrow 1215
store x
write You haven't given me numbers.
store y
write You only gave me one number.
answer $$ remainder should be essentially blank
no
write There should just be two numbers.

```

Like `-store-`, `-storen-` will terminate judging on an error condition (in which no number was found). In the example, the first `-storen-` removes and stores one number in “x” and the second `-storen-` looks for a remaining number to store in “y”. The first `-storen-` will terminate judging if there are no numbers. The second `-storen-` will terminate judging if there is no number remaining after one has been removed. The blank `-answer-` will be matched if only punctuation, such as commas, remains after the actions of the two `-storen-s`.

### Numerical and Algebraic Judging: `-ansv-` and `-wrongv-`

We have already had some experience in handling numerical and algebraic responses by using `-store-` to evaluate numerically the student’s expression. The `-ansv-` (for “answer is variable”) and `-wrongv-` judging commands evaluate the student’s expression in the same way as `-store-` and also perform a comparison with a specified value.

The `-ansv-` command is useful in association with `-store-`. If you ask the student for a chapter number or a launch velocity of a moon rocket, it is convenient to use `-ansv-` to check whether his number is within the range you allow. For example:

```

.
arrow 1314
store chapter
write Choose a chapter from 1 to 9.
ansv 5,4 $$ match if in the range 5±4 (1 to 9)
no

```

Another common use is in arithmetic drills:

```

define  b=v1,c=v2
unit    drill      $$ multiplication drill
next    drill
randu   b,10       $$ pick an integer from 1 to 10
randu   c,10       $$ pick another integer
at      1513
write   What is <s,b> times <s,c>?
arrow   1715
ansv    b×c        $$ no tolerance
write   Right!
wrongv  b+c
write   You added.
wrongv  b×c,1      $$ plus or minus 1
write   You are off by 1.
wrongv  b×c,20%    $$ plus or minus 20%
write   You are fairly close.
no
write   You are way off!

```

The drill as written will run forever. It could be modified to stop after 5 straight correct responses, or after some other criterion has been met. Note that the response “bc” or “b×c” is judged “no” (unless you define these variables in the “student” set of defines). Also note that the student need not do any mental multiplication for this drill (since if the student is asked to multiply 7 times 9, he or she could respond with 7×9 which matches the -ansv-).

Let’s make a change to require some multiplication on the part of the student:

```

.
.
ansv    b×c
judge   opcnt=0,ok,wrong
writec  opcnt=0,Right!,Multiply!
wrongv  b+c
.
.

```

## The TUTOR Language

Do not define “opcnt”! It is a system variable which counts the number of operations in the student’s response. If the student says “ $7(5+8+3)/2$ ” then “opcnt” will be 4 because the student’s expression contains:

- 1) an (implied) multiplication (*7 times* a parenthesized expression);
- 2) two additions; and
- 3) a division.

In this drill we want the student to give the result with no operations, so “opcnt” should be zero (“specs noops,novars” can also be used to prevent the student from using operations or variables in his or her response).

Recall that the first -concept- command encountered will trigger the reduction of the student’s response to a compact form, through the use of the -vocabs-. This compact form can be compared rapidly to all succeeding -concept- commands. Similarly, the first -store- or -ansv- or -wrongv- causes TUTOR to “compile” the student’s expression into a form which can be quickly evaluated when another of these commands is encountered. It is during the compilation process that “opcnt” is set. Just as the -vocabs- list tells TUTOR how to interpret the student’s words, so the “define student” set of names tells TUTOR how to treat names encountered in the compilation of a student’s algebraic response. So, there are many parallels between -ansv- and “define student” on the one hand and -concept- and -vocabs- on the other.

Let’s look at an algebraic example, as opposed to the numerical examples we have treated:

```

define student
      x=v1
unit simplify
at 1215
write Simplify the expression
      3x + 7 + 2x - 5
randu x $$ pick a fraction between 0 and 1
calc x<x+1 $$ change to 1 to 2 range
arrow 1418
ansv 5x+2 $$ 0 tolerance
goto varcnt-1,toofew,x,manyvar $$ how many x's
goto opcnt-2,toofew,x,manyop $$ how many operations
wrongv 5x+12
write You should subtract 5, not add it.
no
goto formok,x,tellerr
*
```

unit	toofew
write	Your expression is not sufficiently general.
judge	wrong
*	
unit	manyvar
write	"x" should appear only once.
judge	wrong
*	
unit	manyop
write	Not simplest form.
judge	wrong

Unit "tellerr" would contain a -writec- involving the system variable "formok" to tell the student precisely why his or her expression could not be evaluated. There could be several -wrongv- statements in the example to check for specific errors. The system variable "varcnt" during compilation of the student's expression counts the number of references to variables. For example, " $x+3x+x+2$ " is numerically equivalent to  $(5x+2)$ , so that this response will match the -ansv-, but "varcnt" will be 3 because "x" is mentioned three times. If both x and y were defined, the expression " $2x+y+4x$ " would yield a "varcnt" of 3 (two x's and one y) and an "opcnt" of 4 (two implied multiplications and two additions).

In this way "opcnt" and "varcnt" may be used to distinguish among equivalent algebraic responses which differ only in form. Roughly speaking, what is usually called "simplest algebraic form" often corresponds to the smallest possible values of "opcnt" and "varcnt".

There are some minor technical points in the preceding example. For example, -randu- with only one argument produces a fraction between 0 and 1. If this should happen to be very close to 0 then "x" would be unimportant in the expression  $(5x+2)$ , so it seems better to add one and give "x" a value between 1 and 2, which is comparable to the other quantities in the expression. We could have used the two-argument form (e.g., "randu x,8") to pick an *integer* value for "x". However suppose TUTOR chooses the integer 2 for "x". In this case, a student who happens to give "12" as his or her response will match the -ansv- by accident since  $5x+2 = 5 \times 2 + 2 = 10 + 2 = 12$ . On the other hand, with TUTOR picking a *fraction*, the student would have to type something like "8.93172462173" to accidentally match the -ansv-. This just won't happen. You would have to type different numbers 24 hours a day for hundreds of years to match accidentally. If you want even more security against an accidental match, just change the value of "x" and check again. In skeleton form, here is a way to do it:

The TUTOR Language

```

ansv      5x+2
goto      varcnt-1,toofew,x,manyvar
goto      opcnt-2,toofew,checkup,manyop
wrongv    5x+12
.
.
unit      checkup
randu     x          $$ new value of x
calc      x←x+1
judge     continue
ansv      5x+2      $$ try again
.
.
.

```

A further check is that we require exactly one “x” and exactly two operations.

There is a way to give detailed feedback to the student in case his or her expression is not algebraically equivalent to the desired expression ( $5x+2$ ). Suppose the student’s incorrect expression is “ $6x+2$ ”, and that you have done a -storea- to save the response and a -store- to evaluate it for some *integer* value of x. Then ask the student this question:

```

.
.
.
write    What is the numerical value of
          3(<s,x>)+7+2(<s,x>)-5?

```

If x is 4, this will appear on the screen as:

```

What is the numerical value of
3(4)+7+2(4)-5?

```

Many students can handle a numerical example even if an algebraic example gives them trouble, so this student is likely to reply correctly, either with or without some help, that this expression gives 22. You can then reply to the student with this statement (assuming the student’s alphanumeric response is in “string” and its value is in “result”):

write But your expression,  $\langle a, \text{string}, \text{count} \rangle$ ,  
gives  $\langle s, \text{result} \rangle$  in this case.

If the student's response was " $6x+2$ ", with a value of 26 (if  $x$  is 4), this appears on the screen as:

But your expression,  $6x+2$ ,  
gives 26 in this case.

The student now sees that his or her expression " $6x+2$ " does not give the value 22 which it should in the case where  $x$  is 4. You have fed back the student's own expression, evaluated for a particular case where the student can see there is a conflict. (In other words, anything the student says may be used against him or her.) Here is an opportunity for the student to learn, by example, a useful technique in simplifying complicated expressions: try some numerical cases for which you know the results and see whether they agree with the simplified expression.

It is possible to judge equations as well as expressions. Suppose we ask the student to simplify the equation " $4x+3=x+12y-5$ ". A suitable response might be " $12y=3x+8$ " or " $x=(12y-8)/3$ ". Every time the student enters a response, let TUTOR pick a random value for the independent variable  $x$ , and calculate the corresponding value of the dependent variable  $y$ :  $y \leftarrow (3x+8)/12$ . Consequently, any correct equation will be true (with value  $-1$ ), and an incorrect equation will be false (with value  $\emptyset$ ). Here is a unit embodying these concepts:

```
define student,x=v1,y=v2
unit equate
at 1215
write Simplify the equation
      4x+3=x+12y-5
arrow 1718
ok
randu x          $$ random x on each judging
calc  x←x+1
      y←(3x+8)/12    $$ y depends on x
judge continue
ansv  -1           $$ logical true
do    ident
wrongv  $\emptyset$       $$ logical false
write That is false.
```

(Continued on the next page.)

The TUTOR Language

```

no                $$ anything else
write            Give me an equation!
*
unit            ident
calc            y<=3.72y        $$ change y arbitrarily
judge          continue
wrongv        -1                $$ should not now be true
write          That is an identity!
ok
judge          varcnt>2,wrong,ok
write          varcnt>2,Not simplified.,Fine.

```

If the student writes “3+4”, this expression has the numerical value 7, so the reply is “Give me an equation!”

If the student writes “3=4”, this expression has the numerical value 0, since it is logically false, and the reply is “That is false.”

If the student writes “3<sup>2</sup>+5=17-3”, which is equivalent to 14=14, TUTOR replies “That is an identity!” The student’s response *is* true (14 *does* equal 14), so that this true relationship has the value -1 which matches the -ansv- statement. A “do ident” follows, where the dependent variable y is changed so that y no longer bears the correct relationship to x. If the student’s response had been a correct simplification of the given equation, his or her expression would no longer be true (-1), since y is no longer the correct function of x. In the case of “3<sup>2</sup>+5=17-3”, however, changing y has no effect and the value is still -1, which matches the -wrongv- statement in unit “ident”. The student gets the message “That is an identity!”

Only if the student enters an equation which is not an identity will he or she get an “ok” judgment. Note the check on “varcnt”. There could also be a check on “opcnt”.

To summarize, -ansv- and -wrongv- are extremely powerful commands for algebraic or numeric responses, particularly in association with variables defined in the “define student” set. The system variables “opcnt” and “varcnt” give you additional information about the *form* of the response.

CAUTION: Since TUTOR performs multiplications before divisions (unless parentheses intervene), a student response of “1/2x” is taken to mean “1/(2x)”, whereas the student might have in mind “(1/2)x”. It is important to warn your students of this convention at the beginning of a lesson which uses algebraic judging. Scientific journals and most textbooks follow this same convention, but many students are unaware of this. Usually, printed materials use the forms  $\frac{x}{2}$  or  $\frac{1}{2}x$  or  $\frac{1}{2x}$ . These forms avoid the ambiguities that arise from the slash (/) or quotient sign



(÷) used on a single typewritten line. It is hoped that eventually TUTOR will make it easy for students to type fractions with the horizontal bar rather than with the slash or quotient sign. Until then, it is important to point out this convention to your students.

### Handling Scientific Units: -ansu-, -wrongu-, and -storeu-

Suppose you want to ask the student for the density of mercury. A correct answer would be "13.6 grams/cm<sup>3</sup>", but there are many equivalent ways to write the same thing. For example, the student might write "13.6×10<sup>-3</sup>kg/ (.01 meter)<sup>3</sup>" or "13.6 gm-cm<sup>-3</sup>", and both of these responses are equivalent to "13.6 grams/cm<sup>3</sup>". TUTOR provides a convenient way not only to judge such responses appropriately, but to give the student specific feedback if he or she makes specific errors (such as omitting the units or giving the right units but the wrong number).

The TUTOR scheme is based on the judging performed by human instructors when grading exam questions involving numbers and units. The instructor makes two separate checks, one for the numerical value and the other for the *dimensionality* of the units. The dimensionality of density is (mass)<sup>1</sup> (length)<sup>-3</sup>, and it is the powers (1,-3) that we are interested in as well as the number 13.6. All of the equivalent correct responses listed above have a numerical value of 13.6 (in the gram-cm system of units) and a mass-length dimensionality of (1,-3). The -storeu- command (-store- with units) can be used to get the numerical part and the dimensionality if we define the units appropriately:

```
define student      $$ units will be used by student
  units,gm,cm      $$ can define up to 10 basic units
  gram=gm,grams=gm,kg=1000gm  $$ synonyms
  meter=100cm,cc=cm3
define mine,student  $$ include student define set
  num=v1,dimens(n)=v(1+n)$$ see "Arrays", Chapter 10
unit dense
at 1215
write What is the density of mercury?
  (Include units!)
arrow 1618
storeu num,dimens(1)
write Cannot evaluate.
no
```

(Continued on the next page.)

## The TUTOR Language

```

goto    num≠13.6,badnum,x
goto    dims(1)≠1,badmass,x
goto    dims(2)≠-3,badleng,x
judge   ok
write   Good!

```

We will go to a unit “badnum”, “badmass”, or “badleng” (not shown here) if there is something wrong with number, mass, or length. The `-storeu-` command has two variables in its tag. The first variable will get the numerical part of the student’s response, and the second (`dims(1)` in this case) is the starting point for receiving the dimensional information. Here are some examples of what will end up in `num`, `dims(1)`, and `dims(2)` for various student responses:

student response	num	dims(1)	dims(2)
13.6 grams/cm <sup>3</sup>	13.6	1	-3
13.6	13.6	∅	∅
13.6 cm-gm <sup>2</sup>	13.6	2	1
13.6 kg/1∅cm	136∅	1	-1

Notice (in the third example) that a minus sign preceding a unit name is taken as a dash meaning multiplication, not subtraction. Note in the last example that “kg” brings in a factor of 1000 relative to the basic unit (gm). Note also that, as usual, TUTOR does multiplication before doing division so that the “1∅ cm” is all in the denominator, with the result that we have  $(\text{length})^{-1}$ . Similarly, “1/2 kg” will be taken to mean  $1/(2 \text{ kg})$ , *not*  $(1/2) \text{ kg}$ . As mentioned earlier, it is best to point out this matter to the student at the beginning of the lesson.

Like `-store-`, the `-storeu-` judging command will flip TUTOR to the regular state (with a “no” judgment) if it cannot evaluate the student’s response. The system variable “formok” can be used in a `-writec-` to tell the student *why* his or her response can’t be evaluated. One example characteristic of responses involving units is “5 grams + 3 cm”, which is absurd. You cannot add masses and lengths, and `-storeu-` will give up. On the other hand, the student can say “65 cm + 2 meter” and `-storeu-` will set `num` to 265, `dims(1)` to ∅ (no mass), and `dims(2)` to 1. As another example, “cos(3cm)” is rejected, but “cos(3cm/meter)” is accepted. The argument of most functions must be dimensionless. (Exceptions are “abs” and “sqrt”.)


A related difficulty faces students unless they are specifically warned about “3+6 cm” being rejected by `-storeu-` (although it looks reasonable in context to the human eye). As far as `-storeu-` is concerned, however, the student is trying to add 3 “nothings” to 6 cm, and the units do not have

the same dimensionality. For -storeu- this is as improper as “3 kg + 6 cm”. Unfortunately, until -storeu- and TUTOR become more sophisticated, it will be necessary to give explicit instructions to the students that:

- 1) Multiplications are done before divisions (unless parentheses intervene), so that  $1/2$  kg does *not* mean  $(1/2)$  kg.
- 2) Responses such as “3 + 6cm” must be written rather as “(3+6)cm”.

Note that these rules also apply in scientific journals and almost all textbooks, but your students may not be consciously aware of these standard rules. Given only these standard conventions, -storeu- will correctly handle an enormous variety of student responses.

While -storeu- can be used to get the number and dimensionality, the -ansu- and -wrongu- commands are primarily used to check for specific cases. Let us modify our sample unit to use these commands, which are like -ansv- and -wrongv- except for checking for correct units:

.	
.	
	arrow 1618
	storeu num,dimens(1)
	write Cannot evaluate!
	ansu 13.6 gm/cm <sup>3</sup> ,.1
	write Good!
	wrongu 13.6,.1
	write Right number, but give the units!
	wrongu (num)gm/cm <sup>3</sup> ,.1
	write Right dimensionality, but wrong number!
	wrongv 13.6,.1
	write Right number but wrong dimensionality.
	no
	writec dimens(2)=-3,Length ok.,Length incorrect.

The -ansu- will make a match only if the dimensionality is correct and the -wrongu- checks for  $13.6 (\text{mass})^0 (\text{length})^0$ , that is, no units given at all. The second -wrongu- looks for a number equal to (num), and finds it since it is the number the student gave (as determined by -storeu-). Therefore, this -wrongu- will match if the number is not 13.6 but the dimensionality is correct. The -wrongv-, unlike -wrongu-, is only concerned with the numerical element rather than the dimensionality. It is used here to check for responses such as “13.6 cm”.

## The TUTOR Language

The `-exact-` and `-exactc-` Commands

It is occasionally useful (in special cases) to use a command *less* powerful than `-answer-` to judge a response. Suppose you are teaching the precise format required on some business form, and you want the student to type "A B C" *exactly*, with three spaces between the letters. A match to "answer A B C" would occur no matter how the student separates the letters. One space, four spaces, a comma or a semicolon (any of these punctuations) are permissible separators as far as `-answer-` is concerned. Normally, this flexibility is beneficial to students because it keeps them from getting too hung up on petty details. If, however, it is the details that are important in a particular response, use an `-exact-` command. In the present case, the statement "exact A B C" will be matched only if the student types *exactly* that string of characters: A, space, space, space, B, space, space, space, C.

The `-answer-` command does not permit punctuation marks in its tag, so that a response such as "a:b" must be judged with an `-exact-` command if the colon is important. While punctuation marks cannot appear in the tag of the `-answer-` command, the student can use them in a response. The `-answer-` command will treat all punctuation marks that the student uses as being equivalent to *spaces*. (As an alternative, the `-change-` command can be used to redefine the colon to be considered a "word" and not just as a punctuation mark, in which case the `-answer-` command can be used.)

It should be emphasized that it is easy to misuse the `-exact-` command. The student should normally be given considerable latitude in the form of his or her response, such as is permitted by the `-answer-`, `-concept-`, and `-ansv-` commands. The `-exact-` command should be used sparingly, and only for short responses. It may be important for the student to know the exact format of something that is as long as:

### 3 No. 6 screws/516-213-86xq-4: New Orleans

In this case, it would certainly be preferable to have the student pick this correct form out of a displayed set of samples than to ask him or her to type it exactly. (Then, all the student would need to say is that item number 3 is the correct form.)

There is also a conditional form of the `-exact-` command, `-exactc-`. (The conditional `-answer-` command is called `-answerc-`.) In the case of the conditional form of the `-do-` command, the presence of commas tells TUTOR that the statement is conditional, so a `-doc-` command name is not needed. But `-write-`, `-answer-`, and `-exact-` may have tags which

include commas, so the conditional command names must be different (-writec-, -answerc-, -exactc-).

### The -answerc- Command: A Language Drill

The conditional -answer- command, -answerc-, may be used to create vocabulary or translation drills. Here is a sample unit which will give the student practice with Esperanto numbers:

```

unit      espo
next      espo
at        1812
write     Give the Esperanto for
randu     item,5                $$ pick an integer from 1 to 5
at        2015
writec    item-2,one,two,three,four,five
arrow     2113
answerc   item-2;unu;du;tri;kvar;kvin  $$ note semicolons

```

Each item in the -answerc- can be as complicated as the tag of an -answer- command. For example, “answerc select↓ <it,is,a> (right,rt) triangle, <it,is,a> three\*sided (polygon,figure)↑↑ circle,ring” will accept either “rt triangle” or “three sided polygon” if “select” is -1, will accept nothing if “select” is zero, and will accept “circle” or “ring” if “select” is one or more. Note that items must be separated by a semicolon or by the -writec-delimiter. There is also a conditional -wrong- command, -wrongc-.

You might write yourself a similar unit to drill yourself on historical dates, capitals of nations, etc. The drill just shown has three defects: (1) it never ends; (2) you may see the same item two or three times in a row; and (3) no help is available if you get stuck. Let's revise the sample unit to have the following characteristics: it should present the five items in a random order but without repeating any item; any items missed will then be presented again; the student may press HELP to get the correct answer.

We will be using a random sequence of non-repeating item numbers such as:

4,2,1,5,3.

## The TUTOR Language

This is called a “permutation” of the five integers. The following sequence is another permutation:




2,5,3,1,4.

You can see that there is a large number (120) of different permutations of five integers. Correspondingly, there is a large number of different permutation sequences for presenting the drill to the student. Such sequences of non-repeating integers are quite different from the sequences we get from repeated execution of our “randu item,5”, which produces sequences (with some integers repeating and some not showing up for a long time) such as:

3,2,4,4,1,5,1,2,4,3,5,5,2,etc.

We need some way of asking TUTOR to produce a permutation for us, rather than the kind of sequence produced by -randu-. This is done by telling TUTOR to set up a permutation of 5 integers (“setperm 5”) from which to draw integers (“randp item”) until the sequence is finished (indicated by “item” getting a value of zero). The -setperm- command actually sets up *two* copies of the permutation, and the “remove item” statement can be used to remove an integer from the second copy. (The -randp- draws integers from the *first* copy.) If we -remove- only those integers corresponding to items correctly answered on the first try, the second copy will contain only the difficult items (after completing the first pass over the five items). At this time, we can use -modperm- (which has no tag) to modify the first copy by shoving the second copy into the first copy. Having replenished the first copy with the difficult items we can use -randp- to choose these again.

Here is a form of the drill incorporating these ideas:

	unit setperm jump *	begin 5 choose	\$\$ set up two copies of a permutation
	unit calc randp jump	choose attempt $\leftarrow$ 0 item item $>$ 0,espo,x	\$\$ initialize number of attempts \$\$ pick an integer \$\$ jump if first copy not empty
	modperm randp jump	item item $>$ 0,espo,x	\$\$ use second copy if first copy empty \$\$ jump if second copy not empty

```

at          2115
write      Congratulations!
           You finished the drill.
end        lesson          $$ end the lesson
*
unit       espo
next      choose
help      esphelp
at        1812
write     Give the Esperanto for
at        2015
writec    item-2,one,two,three,four,five
arrow     2113
answerc   item-2;unu;du;tri;kvar;kvin
goto      attempt>0,q,x
remove    item            $$ remove item if correct on first attempt
no
calc      attempt<=attempt+1
*
unit       esphelp
calc      attempt<=attempt+1  $$ count HELP as an attempt
at        1613
writec    item-2,unu,du,tri,kvar,kvin
end

```

We want to remove an item only if the student gets it right on the first try, which means “attempt” should be zero. The “goto attempt>0,q,x” means “goto a fictitious, empty unit ‘q’ if attempt is greater than 0, else fall through.” If we fall through, we remove the item (“remove item”). We increment “attempt” on each try (and also when help is requested) so that if the student has to see the answer, the item is not removed and will be seen again. Note that the student *is* required to type the correct response and cannot see this answer while he or she types, which gives the student additional practice on the difficult items.

## Summary

This chapter has demonstrated an array of techniques for judging various types of student responses. There are -answer- and -wrong- (aided by -list-) for handling sentences composed from a relatively small vocabulary of words. There are -concept- and -miscon- (supported by -vocabs-) to handle dialogs involving a large vocabulary. The -match- and

## The TUTOR Language

-storen- commands can be used to pull out pieces of a student's response. The -storea- and -store- commands allow the student to specify alphanumeric or numeric parameters. There are -ansv-, -wrongv-, -ansu-, and -wrongu-, aided by "define student", for judging numerical and algebraic responses. The -exact- and -exactc- commands can be used when it is important that the response take a particular *precise* form. The -specs- command permits you to exercise various options associated with these commands and also provides a convenient marker of centralized post-judging processing. The *regular* -judge- command offers additional control over the judging process.

The construction of randomized drills using -setperm-, -randp-, -remove-, and -modperm- (and featuring the conditional commands -answerc- and -wrongc-) was also illustrated in this chapter.

It is hoped that you will read over this chapter occasionally in the course of writing curriculum materials. The TUTOR judging capabilities are extremely rich (because of the wide range of student responses that must be handled in order for lesson material to be successful). Reread appropriate sections of this chapter at a later time, when you need the details. For now it is sufficient to know what is available, and roughly in what form. You may find it helpful to think of the judging commands introduced in this chapter as making up two major classes: those used for handling words and sentences (-answer-, -answerc-, -list-, -concept-, -vocabs-, -match-, -storen-, -storea-, and -exact-), and those used for handling numbers and algebraic expressions (-ansv-, -define-, -ansu-, -store-, and -storeu-).



# Conditional Commands

# 6

It is important to be able to specify the sequencing of a lesson *conditionally*. We might like to jump past some material on the condition that the student has demonstrated mastery of the concept and needs no further practice. Or we might like to take the student to a remedial sequence conditionally (the condition being poor performance on the present topic). Or, which help sequence we offer might be conditional on the number of times help has been requested. All of these examples imply a need for *conditional* sequencing or branching statements, where the condition may be specified by calculations involving the status of the student.

The usefulness of conditional branching is not limited to the sequencing of major lesson segments, but extends to many calculational or display situations. For example, we might need to -do- conditionally one of several possible subroutines in the course of presenting a complex display to the student. This chapter will show you how to perform these and similar conditional operations.


Here is an example involving a *conditional -do-* statement:

```

unit      setup
calc     N←-1
jump     home
*
```

(Continued on the next page.)

The TUTOR Language

	unit	home
	next	home
	at	2010
	do	N,neg,uzero,One,utwo
	at	1215
	write	N equals <s,N>.
	calc	N←N+1
	*	
	unit	neg
	write	Unit "neg".
	*	
	unit	uzero
	draw	210,260;2060;2010
	*	
	unit	One
	circleb	50,0,270
	*	
	unit	utwo
	write	Unit "two".

The new element is the *conditional* -do- statement in unit "home". If N is negative, that statement is equivalent to "do neg". If N is zero, the statement is equivalent to "do uzero", and so on. The statement:

do N,neg,uzero,One,utwo

is equivalent to:

do	neg	if N is negative
do	uzero	if N is zero
do	One	if N is 1
do	utwo	if N is 2 or greater

Note that unit "utwo" will come up repeatedly because it is the last unit named in the conditional -do- statement. The list of unit names can be up to 100 long:

do N,neg,uzero,One,utwo,dispone,  
zon,zip,figure,ultima

If N is 7 or greater, this statement is equivalent to "do ultima".

The “conditional expression” (N in this case) can be anything. It can be as complicated as “ $3x - 5 \sqrt{N}$ ” and can even involve assignments as in “ $N \leftarrow 35 - x$ ”. The value of the expression is *rounded to the nearest integer* before choosing a unit from the list of units. If the *rounded* value is negative, the *first* unit in the list is chosen. For example, if the expression is  $-4$ , it rounds to *zero*, in which case the *second* unit in the list is chosen.

In a conditional -do- each unit named may involve the passing of arguments:

```
do 3N-4,circ(25,75),box(45),x,flag,circ(10,30)
    neg      0    1  2    ≥3
```

So far we have encountered the following sequencing commands: -do-, -jump-, -next-, -next1-, -back-, -back1-, -help-, -help1-, -lab-, -lab1-, -nextnow-, -data-, -data1-, and -base-. When the tag of such a command is just a single unit name (e.g., in a statement like “help uhelper”), we say it is “unconditional”. To make a “conditional” statement out of any of these, we follow the same rule: state the conditional expression, followed by a list of unit names. So we might have:

```
data N-5,zonk,q,zap,zing,x
expression
negative zero one two three
or
greater
```

Here, as in unconditional pointer-associated statements, “q” means the “data” pointer is cleared so that the DATA key is disabled. This can be used to cancel the effect of an earlier -data- command in this main unit. (Remember that all the unit pointers are cleared when we start a new main unit.) The unit name “x” has the special meaning “don’t do anything!” In the example shown, if the condition (N-5) is three or greater, this -data- command has no effect at all and we “fall through” to the next statement without affecting the “data” pointer. Similarly, if a unit name in the conditional -do- discussed above is replaced by “x”, no unit will be done for the corresponding condition and we “fall through” to the next statement.

This “x” option is extremely useful. Consider the following situation:

## The TUTOR Language

```
jump correct-5,x,done
(then show the next item)
```

If  $(\text{correct}-5)$  is negative (that is, the student has made fewer than 5 correct answers), we “fall through” to the presentation of the next item. If, however, the student has 5 or more correct, the condition  $(\text{correct}-5)$  will be zero or greater and we jump to unit “done”.

## Logical Expressions

The last example can be written in an alternative form which improves the readability:

```
jump correct<5,x,done
```

This says “fall through if correct is less than 5, otherwise jump to done”. The condition  $(\text{correct}<5)$  we call a “logical expression” because it has only two possible values, “true” (-1) or “false” ( $\emptyset$ ), whereas numerical expressions can have any numerical value. Since a logical expression can have only two values (-1 if true, or  $\emptyset$  if false) it is pointless to list more than two unit names after the condition.

Actually, because of rounding, the form “jump  $N<5,x,done$ ” is more precise than the form “jump  $N-5,x,done$ ”. Suppose that  $N$  is 4.8. Then “ $N<5$ ” is true (-1), which rounds to -1, which implies “x”. But “ $N-5$ ” is  $-0.2$ , which rounds to zero, which implies “done”. Such differences appear whenever you have variables which can have non-integer values.

Here is another example:

```
do c-b,far,near,far
```

The above will do unit “near” if  $c$  and  $b$  differ by no more than 0.5, since (in that case) “ $c-b$ ” will lie between  $-0.5$  and  $+0.5$ , which rounds to zero. On the other hand:

```
do c=b,same,diff
```

will do unit “same” only if  $c$  and  $b$  are *equal*. The condition “ $c=b$ ” is true (-1) only if  $c$  is equal to  $b$ .

There are six basic logical operators:  $=$ ,  $\neq$ ,  $<$ ,  $>$ ,  $\leq$ , and  $\geq$ , which mean equal, not equal, less than, greater than, less than or equal, and

greater than or equal. The statement “do a≠b,diff,same” is equivalent to “do a=b,same,diff”. These comparison operators consider two numbers to be equal if they differ by less than one part in  $10^{11}$  (relative tolerance) or by an absolute difference of  $10^{-9}$ , whichever is larger. This is done to compensate for small roundoff errors, inherent to computers, due to their very high but not infinite precision. One consequence is that all numbers within  $10^{-9}$  of zero are considered equal by these logical operators. If it is necessary to test very small numbers, scale up the numbers:  $1000a < 1000b$  can be used if a and b are larger than  $10^{-12}$  (since multiplying by 1000 brings the quantities up above the  $10^{-9}$  threshold).

You can mix logical expressions with numerical expressions in many effective ways. For example:

calc x←100-25(y>13)

gives “x←125” if y is greater than 13 (“y>13” if true is -1) or it gives “x←100” if y is less than or equal to 13 (“y>13” if false is 0). To clarify this, suppose that y is 18 or y is 4:

y=18	y=4
100-25(y>13)	100-25(y>13)
100-25(18>13)	100-25(4>13)
100-25(-1)	100-25(0)
100+25	100-(0)
125	100

In these applications it would be nice if “true” were +1 rather than -1, but the much more common use of logical expressions in conditional branching commands dictates the choice of -1 (since the first unit listed is chosen if the condition is negative).

You can combine logical expressions. For example:

[(3<b) \$and\$ (b<5)]

is true (-1) only if *both* conditions (3<b) and (b<5) are true. In other words, b must lie between 3 and 5 for this expression to have the value -1. Similarly,

(y>x) \$or\$ (b=2)

will be true if *either* (y>x) is true *or* (b=2) is true (or both are true).

Finally, you can “invert” the truth of an expression:

## The TUTOR Language

$$\text{not}(b=3c)$$

is true if  $(b=3c)$  is *not* true. This complete expression is equivalent to “ $b \neq 3c$ ”.

The combining operations  $\&$ ,  $\&$ , and “not” make sense only when used in association with logical expressions (which are  $-1$  or  $0$ ). For instance,  $[b > c \ \& \ 19]$  is meaningless and will give unpredictable results. (If you have done a great deal of programming, you might wonder about special bit manipulations, but there are separate operators for masking, union, and shift operations, as discussed in Chapter 10.)

## The Conditional -write- Command (-writec-)

A very common situation is that of needing to write one of several possible messages on the screen. For example, you might like to pick one of five congratulatory messages to write after receiving a correct response from the student:

```

unit   congrat
randu  N,5           $$ let TUTOR pick an integer from 1 to 5
at     1215
do     N-2,ok1,ok2,ok3,ok4,ok5
*
unit   ok1
write  Good!
*
unit   ok2
write  Excellent!
*
unit   ok3
write  I'm proud of you.
*
unit   ok4
write  Hurray!
*
unit   ok5
write  Great!


```

The `-randu-` command, “random on a uniform distribution,” tells TUTOR to pick an integer between 1 and 5 and put it in  $N$ . We then use this value of  $N$  to do one of five units to write one of five messages. There is a *much* more compact way of writing this:

```

unit   congrat
randu  N,5
at     1215
writec N-2,Good!,Excellent!,
      I'm proud of you.,
      Hurray!,Great!,

```



The `-writec-` command is similar to that of a conditional branching command, but the listed elements are pieces of text rather than unit names. Because `-write-` can be used to display any kind of text (including commas), it is necessary to use a different command name (`-writec-`) to indicate the conditional form of `-write-`, whereas in branching statements the commas separating the unit names are enough to tell TUTOR that it is a conditional rather than an unconditional form. (In conversation, “writec” is pronounced “write-see.”)

You can write whole paragraphs with nice left margins, just as with the `-write-` command:

```

writec N,,,Good!,Excellent!,
      I'm proud of
      you and so
      is your mother.,
      Hurray!,Great!,

```

The elements of text are set off by commas. If  $N$  is 3, the student will see a three-line paragraph, since there are no commas at the end of “of” and “so”. If  $N$  is  $-1$  or  $\emptyset$ , no text will be displayed, since there is no text between the first few commas. Note that “x” is not the fall-through that it is for a unit name in a conditional branching command. Here, “x” is a legitimate piece of text which can be displayed, so the “,” form is the “fall-through”.

If you want commas to appear in some of your text elements, you have a problem, since the commas delimit elements. Consider this:

```

writec N,Hello!,How are you, Bill?,Hi there!,

```

If  $N$  is zero, we will see “How are you”, not “How are you, Bill?” The solution is to use a special character ( $\dagger$ ):

```

writec N\daggerHello!\daggerHow are you, Bill?\daggerHi there!\dagger

```

Now, if  $N=\emptyset$  we will see “How are you, Bill?” While this special character ( $\dagger$ ) is required if text elements contain commas, you may prefer to use it always, even when there are no commas. This special character is often called “the writec delimiter”.

## The TUTOR Language

The same kinds of embedding of other commands which are permitted by `-write-` are also permitted with `-writec-`:

```
writec 2c=b,I have <s,ap> apples.,
        I will buy <s,peachy> peaches.,
```

The `-writec-` is affected by `-size-` and `-rotate-` commands, just like `-write-`.

## The Conditional `-calc-` Commands: `-calcc-` and `-calcs-`

The effects of `-writec-` can be achieved by a conditional `-do-` and a bunch of units containing the text elements, but we have seen that this is a clumsy way to do it. We would often like to calculate one of several things based on a condition. This, too, could be done with a conditional `-do-` to one of several units containing the calculations, but this is cumbersome. We saw one shortcut already:

```
calc x<=100-25(y>13)
```

This statement is equivalent to “ $x \leq 125$ ” if  $y > 13$ , and to “ $x \leq 100$ ” if  $y \leq 13$ . This can also be written as:

```
calcc y>13,x<=125,x<=100
```

The `-calcc-` (pronounced “calc-see”) is strictly analogous to `-writec-`. It indicates a list of calculations to be performed, dependent on a condition. The elements in the list are calculations rather than pieces of text or unit names.

Very often each of the calculations in the list consists of assigning a value to the *same* variable. In the example above, both calculations assign a value to the variable “ $x$ ”. An even shorter way to write this kind of thing is:

```
calcs N-5y,bin<=37.5.2,y3+2,,2/N
```

The `-calcs-` (pronounced “calc-ess”) will store one of five values in “bin”, depending on the condition “ $N - 5y$ ”. Note that if “ $N - 5y$ ” rounds to two, we do nothing. Two commas in a row (,,) indicate “do nothing” in `-calcs-`, `-calcc-`, and `-writec-`. Just as “ $x$ ” can be a legitimate piece of text to write, so “ $x$ ” might be a defined variable, which is why it cannot be used as the “do-nothing” indicator in these commands.



### The Conditional -mode- Command

For completeness it should be mentioned that the -mode- command can also be made conditional:

```
mode count-3,write,x,rewrite,erase,write
```

Here the list of elements following the condition is similar to the list of unit names in a -help- command. In this case, they are the names of the various possible screen display modes. The “x” option means “do nothing—do not change the present mode.”

### The -goto- Command

The -goto- command is a very mild version of the -jump- command. It does not initiate a new main unit and does not perform the initializations associated with starting a main unit (the screen is not erased, the help and other unit pointers are not cleared, and how deep we are in “do” levels is unaffected). It is most often used in its conditional form so we waited until this chapter to introduce it.

One common use of the -goto- command is to “cutoff” a unit prematurely:

```

unit A
at 1315
write You have now finished the quiz.
goto score<90,fair,x
size 4
at 2205
write Congratulations!
size 0
*

unit B
at 1912
write The next topic is . . . . .
.
.
.

unit fair
at 1815
write Your score was below 90.
*

unit blah
.
.
.
```

The TUTOR Language

In this example, a score of 90 or better will mean that we fall through the -goto- to display the large-size "Congratulations!" A score of less than 90 will take us to unit "fair" to add "Your score was below 90" to the "You have finished the quiz" already on the screen. The -goto- does not erase the screen, nor does it change the fact that the main unit is still "A". When the student presses NEXT, he proceeds to unit "B", the main unit following unit "A". He does *not* proceed to unit "blah".

Like -do-, the -goto- command attaches a unit without changing which unit is "home", whereas -jump- changes the main unit and performs the many initializations associated with entering a new main unit (full-screen erase, clearing the help pointers, forgetting any -do-s, etc.). The main difference between -goto- and -do-, is that the -do- will normally come back upon completion of the attached unit, whereas -goto- does not come back and statements following the -goto- are normally not executed. (Some people like to think of the -goto- coming back to the *end* of the unit, whereas -do- comes back to the next statement.)

The relationships among main units and attached units and among -jump-, -goto-, and -do- may be clearer if you think of a lesson as being made up of a number of nodes or clusters, each consisting of a main unit and its attached units:




Fig. 6-1.

## CONDITIONAL COMMANDS

Movement between main units is made by pressing NEXT (or HELP, BACK, etc.) or by executing a -jump-. These main units may form a normal sequence or a help sequence (see Chapter 5). The -goto- and -do- commands attach auxiliary units to these main units.

Notice that completion of a unit reached by one or more -goto-s will cause TUTOR to “undo” one level, if one or more -do-s had intervened in reaching this unit. The reason this occurs is that whenever TUTOR encounters a -unit- command (which terminates the preceding unit) TUTOR asks “Are we at the main-unit level?” If so, we have completed processing; if not, we must “undo” to the statement immediately following the last -do- encountered. This point deserves an illustration:

```

unit  calcit
do    sum
show  total
.
.
.
unit  sum
calc  total←0  $$ initialize “total”
goto  addup   $$ -goto- used instead of -do-, for
*      $$ purposes of illustration
unit  addup
.
.      $$ a calculation of “total”
.
unit  other

```

In unit “calcit” we -do- “sum”, which initializes “total” and does a -goto- to unit “addup”, where some kind of calculation is performed. When we run out of work (by encountering a -unit- command at the end of unit “addup”), TUTOR asks whether there was a -do-. There *was* a -do-, so control passes to the statement following the last -do-, which is “show total”. All of this is perfectly reasonable and useful, but it should be pointed out that this property of the -goto- (that it preserves the required information to permit “undoing”) has an odd side-effect. The presence of a -goto- in a done unit causes an exception (the *only* exception) to the description of -do- as a text-insertion device. Except for this case, the effect of a -do- is equivalent to inserting all the statements, contained in the done unit, in place of the -do- statement. But suppose we replace our -do- with the statements contained in unit “sum”. We would have:

## The TUTOR Language

```

unit   calcit
calc   total←0 } in place of "do sum"
goto   addup
show   total
*
unit   addup
.
.
unit   other

```

Now the `-goto-` cuts off the rest of unit “calcit”, and the `-show-` will not be performed, in contrast with the case where we used a `-do-`. So, the presence of a `-goto-` in a done unit causes a (useful) exception to the text-insertion nature of `-do-`.

Here is a summary of the basic properties of the `-goto-` command:

- 1) `-goto-` may be used to attach units with none of the initializations associated with `-jump-`;
- 2) statements which follow the `-goto-` will not be executed (like `-jump-` and unlike `-do-`);
- 3) a `-goto-` in a done unit does *not* cut off statements following the original `-do-` statement, which is an exception to the normal text-insertion nature of `-do-`.

Additional aspects of `-goto-` (in relation to judging student responses) are discussed in Chapter 8.

It is often convenient to cut off a unit with a `-goto-` in the form shown in this example:

```

unit   cuts
goto   expression,x,zonk,empty,x,empty
write  We fell through . . .
.
.
.
unit   empty
*
unit   zonk
.
.
.

```


Note that unit “empty” has nothing in it but serves merely to have a place to go to in order to cut off the end of unit “cuts”. This is such a common situation that TUTOR provides an empty unit named “q” (for quit). The previous `-goto-` can be written as:

```
goto expression,x,zonk,q,x,q
```

The statement “goto q” means go to an empty unit. The special meaning of “q” here makes it illegal to have your own unit named “q”, just as it is not possible to name a unit “x”. Since “do empty” can be rendered by the equivalent “do x”, the statement “do q” (or a conditional form) is given the special interpretation of acting like a “goto q”. The use of “q” in a -goto- statement is somewhat different from the use of “q” in a -help- statement. You will recall from Chapter 5 that “help q” means to quit specifying a help unit, by clearing the -help- pointer.

The -goto- can be used in association with the -entry- command to skip over statements:

```

.
.
.
  calc   b<=0
  goto   3f>5,leavit,x
  calc   b<=f/2
         f<=0
 entry leavit
.
.
.

```

If 3f is greater than 5, we skip over intervening statements to entry “leavit”. The -entry- command is equivalent to a special -goto- plus a -unit-:

```

.
.
.
{ special goto leavit }
{ unit         leavit } equivalent to (entry leavit)
.
.
.

```

So, unlike a -unit- command, -entry- does not terminate a unit but merely provides a named place to branch to. Its equivalence to a special hidden -goto- followed by a -unit- command means that an entry is completely equivalent to a unit, except for not terminating the preceding statements. For this reason it is possible to use an entry name with -do-, -jump-, -help-, etc.

The TUTOR Language

The conditional `-goto-` is often used for repetitive operations similar to those carried out with `-do-`. Here are two versions of a subroutine to add the cubes of the first ten integers:

<pre> <u>-do-</u> unit  add calc  total←0 do    add2,i←1,10 * unit  add2 calc  total←total+i<sup>3</sup> </pre>	<pre> <u>-goto-</u> unit  add calc  i←1       total←0 * goto  add2 * unit  add2 calc  total←total+i<sup>3</sup>       i←i+1 goto  i≤10,add2,x </pre>
---	--

The last two statements in the `-goto-` example could be combined as:

```
goto (i←i+1)≤10,add2,x
```

For the simple task of adding ten numbers, the `-do-` form is certainly easier to construct, but situations occasionally arise where it is easier to construct a repetitive loop using a conditional `-goto-`.

Except for not changing how many levels deep in `-do-s` we are, `-goto-` is quite similar to `-do-`. Although the feature is seldom used, it is even possible to pass arguments to a subroutine with a `-goto-`:

```
goto zonk(12,25)
```

Arguments may also be passed in a conditional `-goto-`:

```
goto 3N-4,alpha(2+count),x,beta(15,2N),q
```

The Conditional Iterative `-do-`

The conditional and iterative `-do-` can be combined so that, on each iteration, the conditional expression selects which unit to do this time:

```
do N+3,ua,ub,uc,ud,i←1,12
      ↓   ↓   ↓   ↓
      neg 0  1  ≥2
```

For each value of  $i$  (from 1 to 12), the expression " $N+3$ " is evaluated, which determines which subroutine will be done. For example, if " $N+3$ " is 0, the above statement is equivalent to "do ub, $i \leftarrow 1,12$ ". Usually a conditional iterative -do- is used in situations where the conditional expression (" $N+3$ ") is not changing, but doing one of the subroutines can change  $N$  so that a different subroutine is used on the next iteration. The following is an example of such manipulations:

```
do i-2,ua,ub,uc,ud,i<=1,4
```

In the first case, where  $i$  is equal to 1, the condition " $i-2$ " is  $-1$ , so we do "ua". Then  $i$  is incremented to 2, and we do "ub", etc. This is, therefore, equivalent to the sequence:

```
do ua
do ub
do uc
do ud
```

As usual, the specified units can involve the passing of arguments.

In a conditional non-iterative -do- the unit names "x" and "q" mean "don't do anything" and "goto q" respectively. In a conditional iterative -do-, "x" means "don't do anything on this iteration," and "q" means "quit doing this statement and go on to the next statement." In other words, "x" means "fall through to the next iteration," while "q" means "fall through to the next TUTOR statement." For example:

```
do i-2,ua,x,q,ud,i<=1,4
show i
```

will display the number "3". For  $i$  equal to 1 we do "ua"; for  $i$  equal to 2 we do nothing; for  $i$  equal to 3 we quit and go on to the following -show-statement.

### The -if- and -else- Commands

Suppose you want to do one set of statements if  $x$  is greater than  $y$ , and a different set of statements. One way to do this, as we have seen, is to put the two sets of statements in two different units and write "do  $x > y$ , unita, unitb". Another way to perform these operations is to use -if- and -else- commands:

The TUTOR Language

```

Done if x>y {
    if      x>y
    .      calc  Z<=5y
    .      draw  x,Z;x+100,Z+100
    else
Done if x<=y {
    .      at    x,y
    .      circle 50
endif

```

The statements between the -if- and -else- commands are performed only if x is greater than y, and the statements between the -else- and -endif- commands are performed otherwise. The tag of the -if- command must be a logical expression (one that has values -1 or 0). The tag of the -else- command must be blank. The -endif- command identifies the end of the sequence.

Note that the statements bracketed by -if-, -else-, and -endif- must be indented, with an initial period identifying them as indented statements. (It is possible that the details of this indenting format may change. Consult on-line PLATO aids for up-to-date information.)

When do you use a conditional -do-, and when do you use -if- and -else-? This depends mainly on the number of statements involved. If there are few statements to be performed, -if- and -else- is probably more readable. But, if "unita" and "unitb" are long subroutines, the conditional -do- is the more convenient form.

There doesn't have to be an -else-:

```

if      x>y
.      calc  Z<=5y
.      draw  x,Z;x+100,Z+100
endif

```

This will do the -calc- and -draw- only if x is greater than y.

There is also an -elseif- for specifying an additional condition:

```

Done if x>y {
    if      x>y
    .      calc  Z<=5y
    .      draw  x,Z;x+100,Z+100
elseif

```



```

Done if x>.5y
but x not
greater than
y
Done if
neither of
the above
is valid
elseif x>.5y
  at 1225
  write This paragraph will be
        displayed only if x is
        not greater than y but
        is greater than .5y.
else
  at 1225
  write x is less than .5y!
endif
    
```

It is possible to have additional levels of indented -if- structures:

```

A second
level of
indenting
if a=b $or$ b>3
  calc x<=b+2
  if count<8
    at 2513
    write Two levels!
  else
    do subr
  endif
else
  at 912
  show x
endif
    
```

The text "Two levels!" will appear on the screen if (a=b \$or\$ b>3) and if (count<8).

## Sequencing of Units Within a Lesson

# 5

We have discussed many units which make different kinds of displays. In some cases, the main units had other units attached to them by means of -do-. Upon completion of a main unit, the student can proceed to the next one by pressing NEXT. A greater variety of inter-unit connections is needed to build a complete lesson which includes optional help sequences, branches to remedial sections when the student is having trouble, an index that gives the student some control over the order of presentation, etc. This section will discuss, in more detail, how to build rich interconnections into a lesson. This discussion builds on the introduction to such matters presented in Chapter 1.

It is often desirable to skip over some units, particularly if they are used as subroutines, not as main presentation units. We have seen that this can be done by using a -next- command to name the main unit which is to follow. For example:

```

unit   one
next  two
do     dispone
at     1515
write This is unit one.
*
unit   dispone
calc  radius←(x←y←200)−50

```

(Continued on the next page.)

## The TUTOR Language

```
do    halfcirc
*
unit  two
at    412
write This is unit two.
```




Fig. 5-1a.




Fig. 5-1b.

When TUTOR begins “executing” the statements in unit “one”, it starts out assuming that the next physical unit, unit “dispone”, will be the next main unit. However, TUTOR encounters a “next two” statement which says, “No, make a note that unit ‘two’ will be next, rather than the next physical unit”. The “do dispone” is then executed, which involves drawing a figure. Finally, we write “This is unit one”, which is at the end of unit “one”. Nothing more will happen until the student presses the NEXT key, at which time TUTOR looks at its notes and finds that unit “two” comes next, whereupon it erases the screen and starts executing unit “two”. Had we not inserted the -next- command, TUTOR would have gone on to unit “dispone” by default.

To put it another way, TUTOR has a pointer which tells which main unit should come next. At the beginning of a main unit, TUTOR places zero in this pointer to indicate that the *next physical unit* should be next.

SEQUENCING OF UNITS WITHIN A LESSON

If no -next- command is encountered, we reach the end of the unit with the pointer still zero, and when the student presses NEXT, TUTOR will by default proceed to the next physical unit. On the other hand, if we encounter a -next- command anywhere in the unit, it will alter this pointer so that later, when the student presses NEXT, the pointer is non-zero and is pointing to whatever unit we have specified.

It should be clear from this discussion that the -next- command can be executed anywhere in the unit without changing its effect. Nevertheless, *it is important to place the -next- command near the beginning of the unit.* The advantage is that you can then see at a glance what is the main sequence flow. If the -next- command is buried far down in the unit, you have to hunt for this crucial information. You put such unit information at the beginning of a unit for the same reason that you define appropriate names for the variables you use: you or a colleague may have to read through the lesson months after it was written!

The following is a simple illustration of how the -next- pointer is handled:

```

unit  silly
next  A
next  B
next  C
*
unit  sillier
    
```

Well, what unit *will* be next? Answer: unit "C"! The pointer starts out cleared to zero (which implies the next physical unit), then gets set to "A", then to "B", and finally to "C". Each succeeding -next- command overwrites what had previously been in the pointer.

It is also possible to clear the next pointer yourself by -next- with no tag or "next q" ("q" for "*quit* specifying something"). Either of these forms will clear the next pointer so that the next physical unit will come next. In other words, the sequence:

```

unit  start
next  silly
next  q      $$ or just "next" with no tag
*
unit  again
    
```

will proceed from unit "start" to unit "again" because the "next q" cancels the "next silly".

## The TUTOR Language

Such seemingly meaningless manipulations are mentioned here for completeness and as aids to explaining how TUTOR handles a unit pointer, such as that associated with the `-next-` command. These manipulations will make more sense to you later on in the book. The important thing to remember is that you have complete control over the pointer. You can set it or clear it with an appropriate `-next-` command.

The existence of `"next q"` (and related statements) means that `"unit q"` is not a permitted statement (you are not allowed to name a unit `"q"` because of the possible confusion). For similar reasons you will see later that a unit cannot be named `"x"`.

Another way to utilize pointers is in specifying optional `"help"` sequences which the student can request by pressing the `HELP` key. Such optional sequences are important tools in tailoring the lesson to meet the needs of individual students of diverse backgrounds and abilities. Here is an example. (See Figures 5-2a and 5-2b.)

```

unit dipper
help words $$ specify a help unit
at 1215
write Today we will discuss Ursa Major.
*

unit dippy
help words $$ specify a help unit
at 2213
write Ursa Major is in the northern sky.
.
.
.
unit words
at 1525
write Ursa Major is the Latin name for the
constellation which contains
the "Big Dipper".
(Press NEXT for more help,
or Press BACK.)
*
unit words2
at 1525
write "Ursa" means "bear".
"Major" means "bigger".
end

```

SEQUENCING OF UNITS WITHIN A LESSON




Fig. 5-2a.




Fig. 5-2b.

The -help- command is used to specify a “help” unit, which may be just the first unit in a long help sequence. If you provide help in this way, the student can get it by pressing the HELP key. (Conversely, if there is no -help- command, the HELP key has no effect). When the student enters the help sequence, his or her screen is erased to clear the way for the display generated by the first help unit. The student may at any time press BACK or shift-BACK to return to “home base”, the main unit he or she was in when requesting help. A “base” pointer retains the name of the “base unit” (the unit to return to). In the example, if you press HELP in the base unit “dippy”, TUTOR remembers “dippy” and jumps to “words”, from which the BACK key will take you back to “dippy”. If instead you press NEXT, you advance to “words2”, where you can again press BACK or shift-BACK to return to “dippy”. From “words2” you will also return to “dippy” upon pressing NEXT because the -end- command in unit “words2” signals the end of the help sequence.

It is almost as though the student had two screens to look at! The student starts the lesson in the first unit of a normal, non-help sequence and advances in this sequence until he or she requests help. At this point, the student turns his or her attention to a different, parallel sequence of units, almost as though that student had turned to use another terminal. The student can get back to the original sequence by pressing BACK, as if he or she had turned back to the original terminal. The usefulness of such a parallel sequence is not limited to help sequences but can be used to

## The TUTOR Language

provide review, a desk calculator mode, a dictionary of terms, tables of data, etc., or for any situation in which the student temporarily needs a second terminal “off to the side”.

It is possible to access yet another help sequence when you are already in a help sequence. `BACK`, however, will return you to the original *base* unit, *not* the help unit you were in when you requested the second help sequence. This is due to the fact that there is only one base pointer, which is not changed by the second help request. If there is already a base unit specification, TUTOR does not alter it.

You can alter the base unit pointer yourself with a `-base-` command. If you put a `-base-` command with no tag in unit “words” you will prevent a return to “dipper” or “dippy”. The `-base-` command with no tag or a “base q” statement clears the base pointer so that TUTOR forgets where to return to and thinks that you are not in a help sequence. (You should notice that the `-end-` command in unit “words2” is now ignored. The `-end-` command has no effect in a non-help sequence.) This `-base-` (blank or “q” tag) is used quite often since it is frequently convenient to put the student into a non-help sequence, even though he reached a certain point by pressing `HELP`. Also, TUTOR automatically clears the base pointer whenever and by whatever means the student reaches the corresponding base unit.

You can change the base pointer to point to some unit other than the original one. Imagine that we place the following statement in unit “words”:

```
base dispone
```

This means TUTOR will eventually return to “dispone” rather than “dipper” or “dippy”. This is occasionally a useful technique. For example, you might like to return to a unit just ahead of the original one in order to ease back into the original context. Notice, too, that while `-base-` with no tag (or “q”) can change a help sequence into a non-help sequence, so “base unitname” can change a non-help sequence into a help sequence by naming a unit to return to.

You probably will not need all of the features of `-help-`, `-base-`, and `-end-` described above, but hopefully the discussion has clarified how they do their work. You have also discovered some terms which will be quite useful in later discussions and can now talk about “non-help sequences” of “main units” and “help sequences” of “main units”. It should be pointed out that a base unit may have other (auxiliary) units attached to it by `-do-`; and, of course, you return to the base unit itself, not to one of these attached units, even if the `-help-` command is located in an attached unit. Moreover, a lesson may be thought of as a collection of

main units which have attached units, and the student moves from one main unit to another. The student may enter a help sequence of main units, each of which may -do- attached units. While the student is in the help sequence, TUTOR remembers which main unit is the “base” unit to return to when -end- is encountered, or when BACK or shift-BACK is pressed. The following is a diagram of this structure:




Fig. 5-3.

You may have realized that -help- and -base- are quite similar to -next- in that all three commands set pointers. (The pointers have different uses, however). For example, if we say:

```
unit lotshelp
help a
help b
help c
```

then the last one wins—the help pointer ends up pointing at unit “c”. We saw earlier that -next- works this way. Similarly, “help q” or -help- with no tag will clear the help pointer, thus making the HELP key inoperative.



## The TUTOR Language

You may find it helpful to think of a help sequence as a “slow” subroutine. Whereas a -do- command takes us to a unit and right back again, -help- makes possible an optional jump to a unit or to a sequence of units where the student may study for many minutes before returning to the base unit. Aside from the “slowness” and the necessity of pressing keys to go and return, there is one fundamental difference from a -do- situation. In a help sequence, we return from help to the *beginning* of the base unit and re-execute the statements in the unit in order to restore the original display, whereas the return from a -do- is to the statement following the -do-.

This last point is sufficiently important to warrant an example:

```

unit   initial
at     2513
write  Set "a" to 0.
calc   a←0
*
unit   repeat
help   trivial
at     2715
write  Increment "a" to <(s,a←a+1)>.
*
unit   trivial
at     312
write  Press NEXT or BACK.
end

```

(Of course, “a” must be defined.) If we repeatedly press HELP, then BACK, while we are in unit “repeat” we will repeatedly increment variable “a”. Variable “a” increases by one on every return from the help sequence because the return is to the *beginning* of the base unit, and *all* the statements in unit “repeat” are re-executed. This is necessary to restore (to the screen) the display associated with unit “repeat”, since the entire screen is erased when the HELP and BACK keys are pressed.

This example brings up a fundamental programming point: the question of initialization. We might use a structure like that shown above for counting the number of times the student presses the HELP key (although we would then most likely put the “a←a+1” in the help unit). In order to count something (requests for help, number of wrong answers, etc.), it is necessary to “initialize” the counting variable to zero before starting the process, and this initialization must *precede* (and be outside) the process itself. This can perhaps best be seen by moving the statement “calc a←0” from unit “initial” to the beginning of unit “repeat”:

```

.
.
.
unit   repeat
help   trivial
calc   a←0
at     2715
write  Increment "a" to <s,a←a+1>.
.
.
.

```

Imagine pressing HELP (and BACK) repeatedly. There would never be a change in the displayed value of "a", because on each return from the help unit, "a" is again reset to zero (whereas that was previously done *only* within unit "initial").

The question of initialization will be encountered again and again in various guises. These matters were not mentioned earlier partly because the iterative `-do-` command had the initialization built-in. For example:

```
do zonk,i←5,13
```

means "initialize 'i' to 5 and do 'zonk', then repeat by incrementing 'i' by one until it reaches 13".

It should be mentioned here that initialization questions are, of course, not unique to programming. The principal and interest due monthly on your car or house loan depend on the *initial* conditions of the loan. When you make fudge, you start with certain ingredients in the mixing bowl (the *initial* condition) and *then* you beat the mixture 200 times. You would no more restart with new, unmixed ingredients after each beating stroke than you would reinitialize a count of student errors after each attempt. In other words, questions of initialization are mainly questions of common sense, and we will make explicit comments about these matters only where confusion is likely. In the case of a return from a help sequence, you might have thought that TUTOR remembers the entire display originally made by the base unit. However, as you have seen, TUTOR must *re-create* the display by *re-executing* the commands in the base unit (which has side effects related to initialization questions).

Now, let's move the "calc a←0" back to unit "initial" and modify the unit to look like this:

```

unit   initial
calc   a←0
jump   repeat  $$ do not wait for the NEXT key
*

```

## The TUTOR Language

The `-jump-` command acts much like the student pressing NEXT (the screen is erased and we move to a new main unit). The `-jump-` command is particularly useful in association with initializations, as in this example, where it is necessary to separate initializations from a process in a different unit. It would be superfluous to show the student a blank screen and to make the student press NEXT. Indeed, it should be a basic rule to minimize unnecessary keypresses so as not to frustrate the student. Notice that `-jump-` is *immediate* (like `-do-` and unlike the `-next-` or `-help-` commands) and that statements which follow `-jump-` in a unit *will not be executed* (unlike `-do-`, `-next-`, and `-help-`).

The base pointer is not affected by a `-jump-`. The pointer remains zero if we are not in a help sequence, and it retains its base unit specification if we are in a help sequence. The `-jump-` simply takes us from one new main unit to another without having to press NEXT. Since it starts a new main unit, a `-jump-` cancels any `-do-s` which have been encountered (there will be no return from those `-do-s`).

When moving from one main unit to another, by `-jump-` or by pressing NEXT, the entire screen is erased unless the *first* of these two main units contains an “inhibit erase” statement. For example, the sequence:

```
inhibit erase
jump more
```

will leave the old display on the screen, and displays created by unit “more” will be added to the screen.

Since `-jump-` takes the student from one main unit to another without altering the base pointer, it is possible to take a student to a help sequence immediately without pressing HELP:

```
unit model
.
.
.
base model
jump modhelp
.
.
.
```

Initially, the base pointer is zero because we are in a non-help sequence. Then, a `-base-` command is used to set the base pointer to unit “model” (the main unit we are presently in). The `-jump-` takes us to unit “modhelp”.

Now we are in a help sequence because the base pointer has been set. The return from the help sequence will be to the beginning of unit "model". Note the difference between "base model" and "base q" in unit "model": a "base q" statement would clear the already-cleared base pointer, whereas "base model" sets the pointer to "model".

### Summary of Sequencing Commands

You have learned a variety of commands which enable you to control the sequencing of units in a lesson. These include commands which set pointers (-next-, -help-, -base-, etc.) and a couple of immediate branching commands (-do- and -jump-). You have seen how to have two parallel sequences of main units, a non-help sequence and a help sequence, and have used the -end- command to terminate a help sequence. Additional aspects of the connections among units will be discussed in Chapter 6 in the section on the -goto- command. Recall that the LAB, DATA, and BACK keys are activated by -lab-, -data-, and -back- commands, just as the HELP key is activated by the -help- command. The *shifted* HELP, LAB, DATA, NEXT, and BACK keys (abbreviated as HELP1, LAB1, DATA1, NEXT1, and BACK1) are activated by the commands -help1-, -lab1-, -data1-, -next1-, and -back1-. (When in a help sequence, the BACK or BACK1 keys will cause a return to the base unit, unless there are explicit -back- or -back1- commands to alter this.) Here is a unit which uses many of these commands:

unit	central	
help	uhelp	
help1	index	
lab	simulate	
lab1	calc	
data	data	
data1	news	
at	1314	
write	Press	HELP for assistance, shift-HELP for an index, LAB for simulation, shift-LAB for a calculator, DATA for tables of data, shift-DATA for class news.

This is an extreme case, but this unit gives the student *six* choices of help sequences, and which help sequence is entered depends on which key the student presses. In any of these cases, the eventual return will be to this base unit.

## The TUTOR Language

The commands `-next-`, `-next1-`, `-back-`, and `-back1-` are somewhat different in that they do not cause a help sequence to be initiated (pressing the corresponding key does not alter the base pointer, and one simply moves among main units of the help sequence or non-help sequence).

The same conventions apply to all these commands. In particular, a blank tag (or “q”) disables the corresponding key by clearing the associated pointer. A non-help sequence can be changed into a help sequence by specifying a unit to return to with a “base unit” statement. A help sequence becomes a non-help sequence if we encounter a “base q” or “base” statement, since these clear the base pointer.

It is important to point out that *all* the unit pointers, other than “base”, are cleared when we start a new main unit (either by `-jump-` or by pressing a key such as NEXT, BACK, or HELP). Starting a new main unit, therefore, involves a number of important initializations, including erasing the screen to prepare for the new display (unless there was a preceding “inhibit erase”).

Notice that `-jump-` and `-do-` are basically author-controlled branching commands, while `-help-`, `-back-`, `-data-`, etc., permit the student to control the lesson sequence.

There is another way to enter a help sequence, which is particularly useful in offering the student an index to the various parts of the lesson. Suppose the lesson is organized into chapters or topics and you wish to let the student choose his or her own sequence. In particular, the student can skip ahead, go back, or review material. It is desirable that the student be able to go to an index or table of contents at any time. One way to provide access to the index is to put a “data table” statement in every main unit. The student can then hit the DATA key and jump to unit “table” at any time. Unit “table” would contain a list of topics for the student to choose from, and it should contain a “base” statement to insure that the chosen topic be entered as a base sequence. Another way to provide access to this kind of index is by means of a single `-term-` command:

unit	table
base	
term	index
at	1218
write	Choose a chapter:
	a) Introduction
	b) Nouns
	c) Pronouns
	d) Verbs

## SEQUENCING OF UNITS WITHIN A LESSON

```

arrow  1822
answer a
jump   intro
answer b
jump   unoun
answer c
jump   pron
answer d
jump   verb

```

The presence of “term index” in the unit “table” makes it possible for the student to press the TERM key and type “index” in order to reach unit “table” at any time. (The TERM key is the shifted ANS key on the keyboard.) When the student presses TERM, TUTOR responds by asking the student “what term?” at the bottom of the screen, whereupon the student would type “index”. The student then reaches unit “table”, where he or she may choose a chapter. You can see that -term- is complementary to -help-. The -help- command in a main unit specifies where to go if HELP is pressed while in that main unit, whereas the presence of -term- in a unit specifies that the unit can be entered from *anywhere* in the lesson. An error is made if another -term- command (with the same tag) is placed in a different unit. In this case, TUTOR would not know which unit to enter.

While the -base- command can be put at the beginning of the unit, there is some advantage to moving it later on in the unit. With -base- commands just before the -jump- commands, the student retains the option of pressing BACK to return to where he or she came from (if he or she doesn't like the available choices). This option is lost if the -base- command has already cleared the base pointer.

The name -term- stems from an early use of this kind of facility to provide a dictionary of “terms”, whereby the student has access to the special vocabulary used in a lesson. In such an application, there are as many help units as there are terms to be defined and each unit has an appropriate -term- command:

```

unit  cardinfo
term  cardiac
at    1907
write "cardiac" means "pertaining to the heart".
end

```

When the student types TERM-cardiac, the screen is erased and the definition of “cardiac” is displayed by unit “cardinfo”. Immediately

## The TUTOR Language

upon pressing NEXT or BACK, the screen is again erased and the student is sent back to the beginning of the base unit. A better procedure in this case would be to change the statement “term cardiac” to “termop cardiac”. The -termop- command refers to “term on page” and permits the display given by unit “cardinfo” to be added to the original display without any erasing.

Except for such dictionary applications, it is strongly recommended that you limit yourself to having only one unit with a -term- in it, and its tag should be “index”. This greatly simplifies the instructions to the student on how to use the lesson and minimizes what he must remember in order to move around in the lesson. In the index unit you describe the various options that are available. Even for providing a dictionary of terms, this scheme is probably preferable (one of the options could be “dictionary of terms”, which in turn would show a list of the words whose definitions are available).

It is possible to have additional -term- commands in the unit to provide synonyms:

```

unit   table
base
term   index
term   contents
term   choice
at     1218
write  Choose a topic . . .

```

These additions insure that the student will reach this unit by TERM-index, or TERM-contents, or TERM-choice.

### The -helpop- Command: “Help on Page”

Often the help to be provided when the student presses the HELP key is a brief statement or small drawing which will fit easily on the “page” or screen display which the student is viewing. When this is the case, such help can be added to the screen by means of a -help- command if an “inhibit erase” is used to prevent the current display from disappearing.

A better way is to use a -helpop- command. The statement “helpop hint” specifies that unit “hint” should be done when the student presses the HELP key, without erasing the screen. After going through unit “hint”, TUTOR returns to the point in the lesson where you were waiting for the student to press a key. This could be a -pause- statement, the end of a unit (where you were waiting for the student to press NEXT to

proceed to the next main unit), or an `-arrow-` command where the student was entering a response. The fact that TUTOR returns to the waiting point is an additional advantage of `-helpop-` over the `-help-` command, since return from an ordinary help sequence goes all the way to the *beginning* of the base unit, rather than to the waiting point. (Since the original display is still on the screen when `-helpop-` is used, there is no need to redo the base unit to restore the screen display.) No `-end-` command is needed in a `-helpop-` unit, unlike a `-help-` unit.


The set of on-page commands includes `-helpop-`, `-helplop-` (associated with the `HELP1` or `shift-HELP` key), `-dataop-`, `-datalop-` (for the `DATA1` key), `-labop-`, and `-lablop-` (`LAB1` key). The `-termop-` command mentioned earlier permits `TERM`-associated displays “on the page”.

For moving among main units there are the commands `-nextop-`, `-nextlop-`, `-backop-`, and `-backlop-`. These are just like `-next-`, `-next1-`, `-back-`, and `-back1-`, except that the screen is not erased when proceeding to the named unit. These features can be mixed in one unit. If a unit contains a `-nextop-` command and a `-back-` command, the screen will not be erased when `NEXT` is pressed, but it will be erased if `BACK` is pressed.

### The `-imain-` Command

An alternative to “`TERM-index`” is to tell the student to press a key such as `shift-DATA` to reach an index page. If this index is in unit “`table`”, you must then put the statement “`data1 table`” in every main unit, since all unit pointers are cleared when a new main unit is entered. A better way to do this is to use an `-imain-` command which specifies a unit to be done initially in every main unit:

<code>imain setit</code>		<code>...</code>
<code>...</code>		<code>...</code>
<code>unit a</code>		<code>unit a</code>
<code>...</code>		<code>do setit</code>
<code>...</code>		<code>...</code>
<code>unit b</code>		<code>unit b</code>
<code>...</code>		<code>do setit</code>
<code>...</code>		<code>...</code>
<code>unit c</code>		<code>unit c</code>
<code>...</code>		<code>do setit</code>
<code>...</code>		<code>...</code>
<code>unit setit</code>		<code>unit setit</code>
<code>data1 table</code>		<code>data1 table</code>
<code>...</code>		<code>...</code>



IS  
EQUIVALENT  
TO



## The TUTOR Language

The `-imain-` command names unit “setit” to be done at the beginning of every main unit. This saves you the trouble of placing the statement “do setit” at the beginning of each main unit.

You can specify all kinds of initializations to be performed in each main unit. For example, you might advertise the shift-DATA key with this display at the bottom of the screen:

Press shift-DATA for an index

In this case you would write something like:

```

imain  setit
...
unit   setit
data1  table
at     3218
write  Press shift-DATA for an index
box    3217;3148

```

Now the display will appear with each main unit, and the shift-DATA key will be activated. (Incidentally, if you have blank `-pause-` commands in your units, pressing shift-DATA will merely take the student past the pause, not to the table of contents. Similarly, pressing the TERM key at a blank `-pause-` will not offer TERM capabilities but will merely take the student past the pause. Rather than use a blank `-pause-`, use a statement such as “pause keys=next,data1,term”, as discussed in Chapter 8. With this kind of pause, pressing shift-DATA will take the student to the index, and pressing TERM will give normal TERM features, while pressing NEXT will take the student past the pause. Other keys are ignored.)

The `-imain-` command sets a pointer, just as the `-help-` and `-base-` commands do. You can change the associated unit by executing another `-imain-` command:

```

imain  setit
...
imain  other

```

Notice that the new “imain” unit will *not* be done immediately, but only when a new main unit is entered. You must add the statement “do other” if you want unit “other” to be done immediately. You can stop having an imain-associated unit done by using “imain q”, or “imain” (blank tag), to clear the `-imain-` pointer.

SEQUENCING OF UNITS WITHIN A LESSON

While any key may be used to access an index, many authors have agreed to use shift-DATA in order to provide some uniformity from one lesson to another. This procedure reduces the number of new conventions a student must learn when studying new material.

There is a similar -iarrow- command which can be used to specify a unit to be performed every time a student enters a response. If the -iarrow- command is itself located in the -imain- unit, *all* -arrow-s will be affected.

# Doing Calculations in TUTOR

## 4

You can make TUTOR calculate things for you. For example:

```

.
.
.
at      1201
write   Who is buried
        in Grant's tomb?
arrow   1201+308
.
.
.

```

The `-arrow-` statement, as written, is completely equivalent to `"arrow 1509"`. Or consider this:

```
circle (412+72.62)1/2
```

The radius of the circle will be taken to be the square root of the sum of 41 squared and 72.6 squared.

Just about any expression that would have made sense to your high school algebra teacher will be understood and correctly evaluated. For example:

The TUTOR Language

<u>Expression</u>	<u>TUTOR Evaluation</u>
$3.4+5(2^3-3)/2$	15.9
$2\times 3+8$	14 (NOT 22)
$\sin(30^\circ)$	0.5 (See Appendix C for other functions.)
$49^{1/2}$	7
$(4+7)(3+6)$	99
$6/5\times 10^{-3}$	1200 (NOT $1.2\times 10^{-3}$ )

If your high school algebra is rusty, we remind you that “ $2\times 5+3$ ” means “ $(2\times 5)+3$ ” which is 13, *not* “ $2\times(5+3)$ ” which is 16. The rule is that multiplication is “more important” than addition or subtraction and gets done *first*. If you are unsure at some point, you may use parentheses around several portions of your expression to make the meaning unambiguous.

A similar point holds true for division, which is considered “more important” than addition or subtraction. “ $8+6/2$ ” means “ $8+(6/2)$ ” which is 11, *not* “ $(8+6)/2$ ” which would be 7. The only ticklish point is whether multiplication is more or less “important” than division. TUTOR agrees with most mathematical books and journals that multiplication is more important than division, so that “ $6\times 4/3\times 2$ ” means “ $(6\times 4)/(3\times 2)$ ” which is 4. Note that this means that TUTOR considers “ $1/2(6+4)$ ” to be “ $1/(2(6+4))$ ” which is 0.05, *not* “ $(1/2)(6+4)$ ” which would be 5. Again, when in doubt use parentheses. You could write “.5(6+4)” if you wish, which is unambiguous.

Experience has shown that students tend to write algebraic responses according to these rules, and making TUTOR conform to these rules facilitates the correct judging of student algebraic and numerical responses.

Having seen how expressions are handled, we can introduce “student variables” which may be used to hold numerical values obtained by evaluating expressions. These stored results can be used later in the lesson. As an example, a “variable” might hold the student’s score on a diagnostic quiz, and this score could be used later to determine how much drill to give the student. The storage place is called a “variable” because what it holds may *vary* at different times in the lesson. Another variable might count the number of times the student has requested help, in which case the number which it holds would vary from 0 to 1 to 2, etc.

There are 150 “student variables” which can be used for storing up to 150 numerical values. These “student variables” are unimaginatively called:

v1, v2, v3, . . . v148, v149, v150.

Later in this section we will learn how to give variables names (such as "radius," "wrongs," "tries," "speed," etc.) which are appropriate to their particular usage in a specific lesson. But first, we will look at variables using their primitive names: v1 through v150.

These variables are called *student* variables because each of the many students who may simultaneously be studying your lesson has his or her own private set of 150 variables. You might use variable v23 to count the number of correct responses on a certain topic, which will be different for each student. If there are forty students working on your lesson, TUTOR is keeping track of forty different "v23's", each one different. This is done automatically for you, so that you can write the lesson with one individual student in mind, and v23 may be considered simply as containing that individual student's count of correct responses. Thus, one student might be sent to a remedial unit because the contents of his variable number 23 show that he did poorly on this topic. Another student might be pushed ahead because the contents of her variable 23 indicate an excellent grasp of the material. It is through manipulation of the student variables that a lesson can be highly individualized for each student.

Variables are useful in building certain kinds of displays. Let's see how to build a subroutine which can draw a half-circle in various sizes, depending on variables which we set up. In order to specify the size of the figure and its location on the screen, we must specify a center (x and y) and a radius. We let variables v1 and v2 hold the horizontal x and vertical y positions of the center, and we let variable v3 hold the value for the radius.




Fig. 4-1.

We can draw such a figure with the following unit:

## The TUTOR Language

```

unit   halfcirc
at     v1,v2
circle v3,0,180      $$ 180 degree arc
draw  v1-v3,v2;v1+v3,v2  $$ horizontal line

```

In order to use this subroutine we might write:

```

unit   vary
calc  v1←150      $$ x center at 150
calc  v2←300      $$ y center at 300
calc  v3←100      $$ radius 100
do    halfcirc
calc  v1←v1+v3    $$ increment x center
do    halfcirc    $$ y and radius unchanged

```

The statement “calc v2←300” means “perform a calculation to put the number 300 in variable v2”. The statement “calc v1←v1+v3” means “calculate the sum of the numbers presently held in variables v1 and v3, and put the result in variable v1”. In the present case, this operation will store the number 250 (150+100) in variable v1 for use in the second “do halfcirc”. Note that the second “do halfcirc” will use the original values of v2 and v3, which have not been changed. This unit will produce this picture:




Fig. 4-2.

The  $\leftarrow$  symbol is called the “assignment” symbol, because it assigns a numerical value to the variable on its left. This numerical value is obtained by evaluating the expression to the right of the assignment symbol.

A slightly more complicated example of a -calc- statement is:

```
calc v3←5v2+v1
```

This statement means “multiply by 5 the number currently held in v2, add this to the number held in v1, and store the result in v3.” In conversation you might read this as “calc v3 assigned five v2 plus v1” or “calc v3 becomes five v2 plus v1”. Notice that it is common practice to refer simply to “v2” when we really mean “the number currently held in variable v2”.

The simplest possible -calc- statement merely assigns a number to a variable, as in “calc v2←150”. It is permissible to make more than one assignment in a -calc- statement:


```
calc v3←v7←18.62
```

This will assign the value 18.6<sup>2</sup> to both variables v3 and v7.

### Giving Names to Variables: -define-

Your programming can be made much more readable by “defining” suitable names for the student variables which you use. For example, in the units just discussed, the quantities of interest were the center (x and y) and radius of the circular arc. We should *precede* such units with a -define- statement:

```

 define x=v1,y=v2
        radius=v3    $$ names may be 7 characters long
unit vary
calc x←150
        y←300        $$ The command name -calc- may be
        radius←100   $$ omitted on successive lines
do halfcirc
calc x←x+radius
do halfcirc
*
unit halfcirc
at x,y
circle radius,0,180
draw x-radius,y;x+radius,y

```

## The TUTOR Language

The `-define-` statement tells TUTOR how to interpret the defined names when they are encountered later in expressions. The units are now much more readable than they were when we used `v1`, `v2`, and `v3`.

Giving meaningful names to the variables you use is very important. After an absence of several months, you would have difficulty in remembering what you are keeping in, say, variable `v26`, whereas the name “tries” would remind you immediately that this variable holds a count of the number of times the student has tried to answer the question. The importance of readability is even more vital if a colleague is working with you on the material. Your associate would find it extremely frustrating to try to figure out what you are keeping in `v26`. So, use `-define-`!

*There should not be any `v3`'s or `v26`'s anywhere in your lesson except in the `-define-` statement itself.* Put all your definitions at the very beginning of the lesson where you will have ready reference to the variables you are using.

The only reason we started out using the primitive `v`-names was to establish a more concrete feeling for the meaning of a student variable. From here on we will use defined variable names. A preceding `-define-` statement is assumed.

**WARNING:** Normal algebraic notation permits expressions such as “ $r\cos\theta$ ”, but in TUTOR you must write “ $r\times\cos(\theta)$ ” or “ $r(\cos(\theta))$ ”. That is, you *must* use an explicit multiplication sign between names (either your defined names such as “`r`” or TUTOR-defined names such as “`cos`”), and you *must* place parentheses around the arguments of functions. For example, the “ $\theta$ ” in  $\cos(\theta)$  must be enclosed in parentheses.

The reason for this is that TUTOR cannot cope with the ambiguities of trying to decide whether an expression such as “`abc`” means “ $a\times bc$ ” (if there is a name “`bc`”), or “ $ab\times c$ ” (if there is a name “`ab`”), etc. Later, when we discuss the important topic of judging student responses, we will see that TUTOR can make reasonable guesses when treating a *student's* algebraic response and can permit the student the luxury of leaving out multiplication signs and omitting parentheses around function arguments. You, the author, are required to be more explicit, however, in separating one name from another. Notice that “`17angle`” is fine and TUTOR will recognize this as meaning “ $17\times\text{angle}$ ”. But “`rangle`” can't be pulled apart into “ $(r)(\text{angle})$ ” because you *might* have meant “ $(\text{ran})(\text{gle})$ ”.



### Repeated Operations: The Iterative -do-

With very little effort we can make a variety of designs out of our unit "halfcirc". For example:




Fig. 4-3.

The effect of the -do- statement is to set y to 100 and do unit "halfcirc", then set y to 170 (the starting value of 100 plus an increment of 70) and do halfcirc again, and repeat the process until y reaches the final value of 380. The format of the extremely useful iterative -do- statement is:

```
do unitname,index<-start,end,increment
```

In the above example, the index "y" starts at 100 and goes to 380 in increments of 70. If no increment is specified, an increment of one is assumed. For example, "do halfcirc,radius<-101,105" will make an arc five dots wide, as in the following figure:



Fig. 4-4.

```
unit stack
calc x<-256
radius<-70
do halfcirc,y<-100,380,70
at 312
write We used an
iterative -do-.
```

## The TUTOR Language

The iterative -do- statement also helps in making animations. The following statements will cause the half-circle to move horizontally across the screen. (See Figures 4-5a and 4-5b.)

```

unit    march
at      3120
write   Move figure left to right.
calc    y←280
         radius←75
do      anim,x←100,350,50
do      halfcirc    $$ draw final figure
at      3220
write   All done.
*

unit    anim
do      halfcirc    $$ draw figure
catchup
pause   1           $$ pause an additional second
mode    erase
do      halfcirc    $$ erase the figure
mode    write
    
```




Fig. 4-5a.




Fig. 4-5b.

We simply -do- unit "anim" repeatedly for different values of x (the horizontal position of the figure on the screen). Unit "anim" does unit "halfcirc" twice, once to draw and once to erase the figure interrupted by a one-second pause. The -catchup- command insures that a second will elapse from the end of drawing the figure on the screen until the beginning of erasing it.

Now that you have studied -define-, -calc-, and -do-, you have learned the basic techniques of how to tell PLATO what calculations you want performed. We have applied these tools to a variety of display generation problems, and we will later use calculations for controlling sequencing in a lesson and for judging responses. Hopefully, you have gained added insight into the value of a subroutine. Notice how many different ways we have used the single unit "halfcirc"!

### Showing the Value of a Variable

We have learned how to calculate and how to store results in variables. How do we show these results on the screen? Suppose we perform this calculation:

```
calc y←5sqrt(37) $$ or, y←5×371/2; "sqrt" means square root
```

How do we later show the value of y? Assume we have defined y. Perhaps we could use this:

```
write y
```

No, that won't work; that will just put the letter "y" on the screen. The -write- command is basically a device for displaying non-varying text, not for showing the value contained in a variable. We need another command:

```
show y
```

This will show the value of y in an appropriate format (-show- picks an appropriate number of significant figures and will use a scientific format such as  $6.7 \times 10^{13}$ , if the number is large enough to require it). By using -show- instead of -write-, you tell TUTOR that you want the stored value to be shown rather than just the characters in the tag.

The TUTOR Language

The `-show-` command will normally choose 4 significant figures, so that a typical display might be “-23.47”. You can specify a different value by giving a second “argument” (arguments are the individual pieces of the tag of a statement):

```
show y,8 $$ 8 significant figures
```

The arguments of the `-show-` command can, of course, be complicated expressions:

```
show 10+30cos(2angle),format+2
```

In fact, it is a general rule that you can use complicated expressions anywhere in TUTOR statements. For example, “draw 5rad +225,34L;123-L<sup>2</sup>,28L”!

Here is a short program which uses `-show-` to display a table (see Figure 4-6) of square roots of the integers from 1 to 15:

```
define N=v1
unit roots
at 310
write N
at 325
write N1/2
do root,N<=1,15
*
unit root
at 410+100N
show N
at 425+100N
show sqrt(N)
```

\$\$ write titles for the two columns

N	N <sup>1/2</sup>
1	1
2	1.414
3	1.732
4	2
5	2.236
6	2.449
7	2.646
8	2.828
9	3
10	3.162
11	3.317
12	3.464
13	3.606
14	3.742
15	3.873

Fig. 4-6.

The last statement could also be written as “show  $N\frac{1}{2}$ ”. This technique of making tables, including the use of the -do- index (N) to position the displays (as in “at 425+100N”) is an important and powerful tool.

There are other commands for displaying variables: -showe- (exponential), -showt- (tabular), -showa- (alphanumeric), -showo- (octal), and -showz- (show trailing zeroes). These are described in detail in the reference material mentioned in Appendix A.

Although -write- is basically designed for non-variable text, combinations of text and variables occur so often that TUTOR makes it easy to “embed” a -show- command within a -write-:

```
write The area was <(s,13.7w,6)> square miles.
```

The embedded “s” indicates a -show- command and the remainder “13.7w,6” is its tag. Other permissible abbreviations include “o” (showo), “a” (showa), “e” (showe), “t” (showt) and “z” (showz). The above -write- statement is equivalent to:

```
write The area was
show 13.7w,6
write square miles.
```

## Passing Arguments to Subroutines

When you write “show 13.7w,6”, you are passing two pieces of information to the -show- command. You are giving two numerical “arguments” (13.7w and 6) to the TUTOR machinery that performs the -show- operations. Similarly, we created a half-circular arc with “circle radius,0,180” in which we passed three arguments to the TUTOR circle-making machinery. Sometimes certain arguments are optional. For example, “show 13.7w” will use a default second argument of 4 (significant figures), and omitting the last two arguments in a -circle- command (“circle radius”) will cause a full circle to be drawn rather than an arc. When we pass one argument to the -at- command (“at 1215”), we mean coarse grid; when we pass two arguments (“at 125,375”), we mean fine grid.

This notion of passing arguments to TUTOR commands, with some arguments optional, also applies to your own subroutines, such as unit “halfcirc”. The “halfcirc” subroutine needs three arguments (x, y, and radius) to do its job. We passed these arguments by assigning values to variables and letting “halfcirc” pick up those values and use them:


The TUTOR Language

```

define  x=v1,y=v2,radius=v3
unit    vary
calc    x←150
        y←300
        radius←100
do      halfcirc
calc    radius←50
do      halfcirc
*
unit    halfcirc
at      x,y
circle  radius,0,180
draw    x-radius,y;x+radius,y
    
```

Notice that the second -do- will use the original “x” and “y”, since these variables have not been changed. It is as though we passed only one argument (“radius”) to the subroutine.

TUTOR permits another way of writing this sequence which looks similar to the way one passes arguments to the “built-in subroutines” (-show-, -circle-, -at-, etc.):




```

define  x=v1,y=v2,radius=v3
unit    vary
do      halfcirc(150,300,100)
do      halfcirc(50)
*
unit    halfcirc(x,y,radius)
at      x,y
circle  radius,0,180
draw    x-radius,y;x+radius,y
    
```

The statement “unit halfcirc(x,y,radius)” tells TUTOR that when this unit is done as a subroutine, arguments are to be passed to it. The statement “do halfcirc(150,300,100)” tells TUTOR to pass the listed arguments to the “halfcirc” subroutine for its use. The arguments are passed in the order listed:

```

do      halfcirc(150,300,100)
.
.
.
unit    halfcirc(x,y,radius)
    
```


 (Pass 3 Arguments)

These variables are now set for use in the subroutine. It is precisely as though we had assigned values to “x”, “y”, and “radius” by using -calc-. If some arguments are omitted, these variables are not transferred:

```
do   halfcirc(235,,85)
.
.
.
unit halfcirc(x,y,radius)
```

1                  2                  (Pass 2 Arguments)

In this case the variable “y” has not been assigned a new value, so it retains the value it had, which was 300. (The value of “y” could have changed if “halfcirc” itself altered it. For example, if we append “calc y←75” to the end of unit “halfcirc”, “y” would now be 75, although it was originally passed the value of 300 by the first -do- statement during the making of the first display.)

Arguments to be passed need not be simple numbers. Each argument can be a complicated expression. The expressions are evaluated, then passed in order:

```
do   halfcirc(3.4radius-25,radius+25y,200+y)
.
.
.
unit halfcirc(x,y,radius)
```

1                  2                  3

It is as though we had written:

```
calc arg1←3.4radius-25
      arg2←radius+25y
      arg3←200+y
      x←arg1
      y←arg2
      radius←arg3
```


Just as the -at- command handles its arguments differently depending on the number of arguments (one for coarse grid and two for fine grid), so it is possible for your subroutines to do such things. There is a TUTOR-defined “system variable” named “args” which always contains the number of arguments passed the last time a subroutine was done. By “system variable” we mean a variable separate from the student variables

The TUTOR Language


(v1 through v150) whose contents are assigned by TUTOR rather than by you. You do not define system variables; they are already defined for you. (Indeed, if you say “define args=v3”, you will override TUTOR’s definition of the meaning of “args”, so that “args” will mean “v3” rather than “the number of arguments passed to a subroutine”.) In Chapter 6 (Conditional Commands) you will see how you could do a variety of things in a subroutine (conditional on the value of “args”) which are similar to the kinds of things the -at- command does.

Our subroutine “halfcirc” uses three student variables: v1, v2, and v3, defined as “x”, “y”, and “radius”. Another subroutine could use the same variables for carrying out its work, but it must be kept in mind that -do-ing this subroutine will affect v1, v2, and v3, since arguments will be passed.

Suppose one subroutine uses another, with “nested” -do-s like this:



Variable v11 ends up with the value 50. It is advisable to use different variables in the two subroutines. Here unit A uses v11 and unit B uses v25. It can lead to confusion or even logical errors if B also uses v11 to do its work, since -do-ing B will affect the value of v11 used by A. Here is the structure to be avoided:





```

do    B(3+v11)
calc  v11←10v11      $$ v11←80
.
.
.
unit  B(v11)         $$ v11←8

```

PASS

Now variable `v11` ends up with the value `80` rather than `50`. This is due to the effect on `v11` of the “do `B(3+v11)`” statement, which assigns the value of `8` to `v11` by passing the argument to unit “`B`”.

This concludes our discussion of calculations for now. We can calculate, save results, use them to make displays, and show the values. In the next section, we will use calculations in association with guiding the sequencing of a lesson.

# Building Your Own Tools: The-do-Command

## 3

You now know enough about presenting material to the student to be able to make attractive displays. You will be able to do even more when you learn how to tell PLATO to calculate complicated displays for you. Before discussing how to do calculations we will pause to introduce an extremely important concept, the “subroutine”, which is fundamental to all aspects of authoring. We will start off by applying the concept of a subroutine to certain display problems.

To introduce the use of subroutines, consider the problem of placing some standard message on several of your main lesson pages. For example, in the many units where you make help available to the student (if he presses the HELP key) you might like to advertise this fact by placing this display at the bottom of the page:

HELP is available

The corresponding TUTOR statements might be:

```
at      3123
write  HELP is available
box    3022;3141
```

It would be tedious to copy these statements into every unit where they were required. Moreover, if you decided later to move this to the upper right corner of the screen, you would have to find all occurrences of this and change all of them. There is a way around these difficulties, and in

## The TUTOR Language

later work we will find further important advantages to the method. Suppose we write a “subroutine” (a unit to be used many times as needed):

```
unit  helper
at    3123
write HELP is available
box   3022;3141
```

Where we need to show this message, we need only write the statement:

```
do  helper
```

This statement attaches unit “helper” to the present unit. It is as though we had inserted the contents of unit “helper” at the point where we say “do helper”. Now, instead of a dozen copies of the display statements we have only one, plus a dozen -do- commands. The -do- command may appear anywhere in a unit. The location of each -do- command will determine when the associated display appears on the screen in relation to your existing display material. All these displays may be changed by simply changing the subroutine unit! You do not need to change the -do- statements, instead change unit “helper” which they all use.

The use of -do- improves the readability of a TUTOR lesson. When you see “do helper” anywhere in your lesson you recognize at a glance what it is for. The contents of unit “helper” might contain a large number of statements which would clutter up your other units, and decrease readability, if these statements appeared directly in each unit.

Let’s consider another use. Suppose we wish to draw a “Cheshire cat” which fades to a smile as Alice watches. We want to draw a cat face made up of the smile plus all the rest of the face, then erase everything but the smile. Here is an effective way of doing it. (See Figures 3-1a and 3-1b.)

```
unit  Alice
at    512
write Watch the Cheshire cat!
do    cat
catchup    $$ wait for cat to be drawn
pause 4    $$ then pause 4 seconds
mode  erase
do    face
mode  write
at    3012
write See the smile?
```

## BUILDING YOUR OWN TOOLS: THE-DO-COMMAND




Fig. 3-1a.




Fig. 3-1b.

We will need some units to use as the subroutines:

```

unit  cat
do    face
do    smile
*
unit  face
at    250,250
circle 120    $$ outline
at    200,280
circle 30     $$ eye
at    300,280
circle 30     $$ eye
*
unit  smile
at    250,250
circle 80,225,315  $$ smile—arc goes from 225° to 315°

```

Note that unit “Alice” does unit “cat”, which in turn does units “face” and “smile”. TUTOR permits you to go ten levels deep in -do-s. Here we have gone only two levels deep. Note that unit “smile”, on its own, is a useful subroutine and might be done whenever just the smile is desired.

## The TUTOR Language

To summarize, we can build useful tools by constructing “subroutines” (units which may be done from many places in the lesson). The liberal use of -do- improves readability, reduces typing, and facilitates revising the lesson. This last point is particularly important when there is a “bug” (unknown error) in the lesson. Debugging becomes much simpler because of the modular nature of subroutines, and because a lesson which uses -do- extensively has its critical control points well localized.

# Miscellany

12

This chapter will acquaint you with additional features of TUTOR and PLATO. See Appendix A for sources of additional information.

## Other Terminal Capabilities

We have emphasized the keyboard and plasma display panel as the main input and output devices used in communicating with the student. Other devices which may be used include a projector of color photographs, a touch panel, a random-access audio playback device, and other specialized input-output devices. All of these terminal-associated devices are easily managed by TUTOR.

The plasma display panel is flat and transparent, which makes it possible to project photographs on the back, superimposing color photographs with plasma-panel text and line drawings. There exists a microfiche projector for the PLATO terminal which will project any of 256 color photos, with fractional-second access time to any of these 256 pictures. (A "microfiche" is a sheet of film carrying many tiny pictures.) Microfiches can be made from a set of ordinary 35mm slides. Students or teachers can insert the appropriate microfiche in the terminal for the subject to be studied. The `-slide-` command selects any of the 256 photos: `"slide 173"` will project the 173rd photo. Additional options on the

## The TUTOR Language

-slide- command permit the independent control of a shutter in the projector.

The touch panel is a device which puts a grid of 16 vertical and 16 horizontal infrared light beams just in front of the plasma panel. When a student points at the panel, he breaks a horizontal and vertical beam. The information as to which beams were broken is sent to the computer as a "key" and the lesson can use this information to move a cursor, choose a topic pointed at, etc.

We discussed in Chapter 8 how to know where the student touched the screen. Another way is to use the information in the system variable "key", which contains the last "key" input from the student, whether it came from the keyboard, the touch panel, or some other external input device. Here is a unit which will analyze the inputs:

```

unit    getkey
next    getkey
enable
pause
goto    (key $ars$ 8),x,keyset,touch,extin,x
write   Impossible!
unit    keyset
write   You pressed a key
        on the keyboard.
*
unit    touch
calc    x<=(key $ars$ 4) $mask$ o17
        y<=(key $mask$ o17)
write   You touched location
        x=<s,x>,y=<s,y>.
*
unit    extin
write   The external input
        was <s,key $mask$ o377>

```

The -enable- command permits touch inputs as well as inputs from any device connected to the external input connector at the back of the PLATO terminal. (The external input device might be a temperature sensor, an analog-to-digital converter, etc.) Without an -enable- command these inputs are ignored. A -disable- command will also cause inputs to be ignored. The system variable "key" contains a 10-bit integer (see the section on bit manipulations in Chapter 9): the most significant or left-most two bits identify the source of the key (0 for keyset, 1 for touch panel, 2 for external input), and the least significant or right-most eight

bits contain the actual data (which keyset button, which touch panel beams, what external data). In the case of the touch panel, the eight data bits contain four bits of x and four bits of y to specify a position. A succession of external inputs can also be retrieved with a single -collect-command.

If an -enable- command is placed just after an -arrow-, touch inputs can be accepted. There is a -touch- judging command whose tag specifies a screen location (and optionally a spatial tolerance). The -or- command is particularly useful here:

```

.
.
arrow 2513
enable
touch 1215
or
answer book
write Yes, "libro" means book.

```

The student will get the same message whether he or she types "book" or points at a picture of a book displayed at location 1215. (The -or-command can be used to make synonymous any judging commands. The system variable "anscnt" will be the same for all judging commands linked by -or-.)

There is a random-access audio device which stores twenty minutes of speech, music, or other sounds. Segments as short as one-third second can be accessed in a fraction of a second, no matter where the segment is located on the twenty-minute magnetic disk. As with microfiche, students can change the disks themselves. There is a -play- command to choose a section of the disk to play music or talk to the student.

Other devices can be connected to the external output connector at the back of the PLATO terminal and controlled with the -ext- command. The -ext- command can send up to sixty 16-bit quantities per second to a device. Among the interesting devices using this capability is a "music box" that plays four-part harmony.

## Student Response Data

A crucial aspect of TUTOR on the PLATO system is that student response data can be collected easily to aid authors in improving lessons. Detailed information can be collected: unanticipated "wrong" responses (which may have been correct but inadequately judged), requests for



## The TUTOR Language

help, words not found in a -vocabs-, etc. Summary information can also be collected: amount of time spent in an area of a lesson, number of errors made, number of help requests, etc. These detailed and summary data provide an objective basis for revising lessons.

A -dataon- command in a lesson turns on the automatic data collection machinery. Students registered in courses with associated response data files will have their responses logged in their data files. When registering students in a course, specific data collection options can be chosen. For example, one might collect only responses judged “no” (unanticipated incorrect responses). Anticipated correct responses (judged “ok”) and anticipated incorrect responses (judged “wrong”) would not be logged. This is often done because the anticipated responses are precisely those for which the lesson is already replying in a detailed, appropriate manner to the student. Here we see the difference between judge “no” (unanticipated) and judge “wrong” (anticipated). In this connection, -wrong-, -wrongv-, and -wrongu- make a “wrong” judgment, whereas the -no- command makes a “no” judgment.

The -area- command is used to divide a lesson into sections, each of which will produce an area summary in the data file. Each time the student encounters another -area- command, a summary of the previous area is placed in the data file. The area summary includes student name, area name, amount of time spent in the area, number of -arrow-s, number of ok/wrong/no responses, number of helps requested and found, etc. This summary data makes possible a statistical treatment of lesson data which can pinpoint weak areas.

The -output- and -outputl- commands permit you to write your own information and messages into the datafile. This supplements the automatic data logging invoked with -dataon- and -area-.

While PLATO provides a standard mechanism for looking through data files (including sorting the data), you can also read back this information and process it yourself. For example, the -readd- command will read area summaries or -outputl- information from a datafile previously specified by a -readset- command.

## Additional Tools for Teaching Foreign Languages

Usually in a lesson on a language such as Russian, which uses a special alphabet, the student will answer some questions in English and some questions in the foreign alphabet. The responses in the foreign alphabet require a “force font”, or a “force font,left” for leftward-going languages such as Arabic, Hebrew, and Persian. Sometimes a

“force micro” option will also be required in order to re-order the keyboard. Since there may be several things different about the two kinds of -arrow-s, it is convenient to have an alternate -arrow- command, which is named -arrowa-.

The -arrowa- command can cue the student differently, because you can alter the arrowhead displayed by -arrowa- by using the -arheada- command. The -arheada- command is similar to the -okword- and -noword- commands (the tag is what will be shown). Just as an -iarrow- unit is associated with the -arrow- command, so the -iarrowa- command can be used to specify a unit associated with the -arrowa- command. Here is a typical setup:

```

arheada +
iarrow english      ] (in an -imain- unit)
iarrowa persian    ]
...
unit ask1
draw 510;1510;1540;510
at 1812
write What is this figure?
arrow 2015
answer triangle
*
unit ask2
draw 510;1510;1540;510
at 1833
write این شکل چیست؟
arrowa 2030
answer مثلش
*
unit english
force clear
okword ok
noword no
*
unit persian
force clear,font,left,micro
okword نه
noword نه

```

Fig. 12-1.

## The TUTOR Language

Unit “ask2” has an `-arrowa-` command, which is associated with unit “persian”, the unit named in the earlier `-iarrowa-` statement. Unit “persian” clears out any existing `-force-` options and sets up the appropriate typing conditions for the student. Unit “persian” also redefines the words to be shown for correct and incorrect responses. The `-answer-` command in unit “ask2” has the Persian for triangle. The student will see a “←” instead of a “≥” as a cue to give a response, thanks to the `-arheada-` command. On the other hand, the standard `-arrow-` command in unit “ask1” has associated with it the `-iarrow-` unit “english”, which clears the `-force-` options and sets the “ok” and “no” words to English words.

While this machinery is particularly valuable in language lessons, it is also useful whenever your `-arrow-s` fall into two rather different categories. An example might be a physics lesson in which some `-arrow-s` require sentence responses and other `-arrow-s` require algebraic or numerical responses.

Some additional TUTOR commands which are particularly helpful in foreign language lessons include `-change-`, `-getword-`, `-getloc-`, `-getmark-`, and `-compare-`. As an example of the `-change-` command, the statement “change symbol comma to word” (which must be placed in the initial entry unit) will change the normal meaning of a comma as an ignorable punctuation mark, so that the comma will be treated as a separate word. This is useful when teaching punctuation, where you want to check specifically for commas. The `-getword-` command is similar to `-storen-` and is used to pull apart the student’s response into separate words. The `-getloc-` command will tell you where a particular student word is on the screen, so that you could draw a box around that word. The `-getmark-` command gives you information on how TUTOR marked up the student’s response, including whether a word was incorrect, misspelled, or out of word order. The `-compare-` command permits you to check a student’s word against a stored list of words (in a common, for example), including spelling aspects.

Routers and `-jumpout-`

A lesson can be designated as a “router” which routes students through the many lessons making up a complete course. A router is associated with a course. Students registered in a course which uses a router will upon sign-in be sent first to the router, not to the lesson specified by the restart information. A typical router might ask the student, “Do you want to resume studying the lesson you last worked on?” If the student says yes, the router executes a “jumpout resume”, which means “jumpout” of this lesson into the lesson mentioned in the

tag, with “resume” having the special meaning “resume at the restart point”. If the student does not want to resume, the router might offer the student an index of available lessons. Suppose the student chooses a lesson on the list whose name is “espnum”. Then the router does a “jumpout espnum” to take the student to that lesson. (The -jumpout-command can be conditional.) Upon completion of lesson “espnum”, (by “end lesson”) the student is brought back into the router. If the lesson executed a -score- command, the router can use the corresponding value of system variable “lscore” to help decide how to route the student. The router might ask the student what he or she wants to do next, or the router might immediately take the student to an appropriate lesson.

Generally speaking, -jumpout- commands should be placed only in routers, not in instructional lessons. Following this practice insures that lessons can be plugged into routers on a modular basis. An exception exists in the case where one instructional package is spread over two or three physical lessons, in which case -jumpout- is used to connect the lessons.

A router can use up to fifty “router variables” (vr1 through vr50) which are not affected by the instructional lessons. These can be used to keep track of which lessons have been completed, how many times they have been reviewed, how much time was spent in each lesson, etc.

## Instructor Mode

Authors write and test lessons, and students study these lessons. Instructors choose lessons from the library of available lessons to make up a course for their students. Instructors also register students, monitor their progress, leave messages for the class or for individual students, etc. There is an “instructor mode” which makes it easy for instructors to do these things without knowing the TUTOR language. The instructor mode is based on a router coupled with a mechanism for setting up a roster of students. The options available through this router are sufficiently flexible to make it unnecessary in most cases to write specialized routers.

## Special “terms”

Authors have a number of special “terms” to help them in curriculum development. If you press TERM and type “step”, you can step through your lesson one command at a time. (A continued -calc- counts as one command.) This is extremely helpful in tracking down logical errors in a lesson. After each step, you can check the present value of student

## The TUTOR Language

variables. There is also a *-step- command* which will throw the lesson into the step mode. The step features are operative only for authors testing their own lessons.

“TERM-cursor” provides you with a cursor which you can move around the screen using the “arrow” keys. Press “f” for fine grid or “g” for gross (coarse) grid. Also press “f” or “g” to update the display of the current cursor location. This facility is useful for deciding what changes to make in the positioning of displays on the screen.

“TERM-consult” notifies PLATO consultants of your request for help. When a consultant becomes available, he or she will talk to you by typing at the bottom of your screen. The consultant’s screen has the same display you have on your screen. It is as though the consultant were looking over your shoulder as you demonstrate the problem. You can talk to the consultant by typing sentences at *-arrow-s* or by hitting TERM and typing. (If you press NEXT, and you have typed eight or fewer characters, your sentence will be taken as a *-term-* to look for in the lesson. Otherwise your line is erased so that you can type some more.) The consultants not only know TUTOR well but they have also had a great deal of experience in helping authors.

“TERM-talk” asks you for the name of the person you want to talk to, then pages that person if the person is presently working at a PLATO terminal. The person called accepts the call by hitting TERM and typing “talk”. The two of you can then talk to each other at the bottom of the screen, but neither of you can see what is on the rest of the other person’s screen. If you want the other person to see all of your screen, press shift-LAB, which puts you into a mode similar to TERM-consult.

“TERM-calc” provides a convenient one-line desk calculator at the bottom of the screen. Authors get normal, octal, and alphanumeric results. To avoid confusion, students who use TERM-calc are not shown the octal and alphanumeric displays.

# Appendices

Appendix A. Where to get further information

Appendix B. List of TUTOR commands

Appendix C. List of built-in -calc- functions

# Appendix A

## Where to Get Further Information

The document “Summary of TUTOR Commands and System Variables” by Elaine Avner lists each TUTOR command, gives the basic form of the tag, and notes any restrictions such as maximum number of arguments or maximum length of names. Lesson “aids” available on PLATO provides detailed interactive descriptions of each command, as well as a wealth of other information useful to authors.

Lesson “notes” on PLATO provides a forum for discussing user problems. You can write notes to ask questions or to suggest new features that would be helpful in your work. You can read notes written by other users, including replies to your notes. Replies to programming questions generally appear within one day. (For faster service, use TERM-consult.) An extremely important section of “notes” is the list of announcements of new TUTOR features. Check this section regularly for announcements of new TUTOR capabilities. The announcements are followed within a few days by detailed descriptions in “aids”.

Sometimes “notes” will announce a *change* in the TUTOR language involving an automatic conversion of existing lessons. For example, at one time there were several different commands (-line-, -liner-, -figure-, and -figuref-) which did what -draw- now does. When -draw- was implemented, all existing PLATO lessons were run through an automatic conversion routine to change the old commands into appropriate -draw- commands. It is probable that other such refinements will be made in the future. Therefore, be sure to read notes and aids regularly.

# Appendix B

## List of TUTOR Commands

Display	Calculations	Sequencing	Student Responses	Other	
at,atnm	gorigin	calc	unit	arrow,endarrow	pause
write	axes	calcc	entry	iarrow	catchup
writec	bounds	calcs	nextnow	arrowa	time
erase	scalex	define	next,next1	iarrowa	step
eraseu	scaley	do	back,back1	long	keytype
size	lscalex	exit	help,help1	jkey	group
rotate	lscaley	doto	data,data1	copy,edit	collect
mode	labelx	goto	lab,lab1	force	inhibit
charset	labely	branch	term	answer,wrong	enable
lineset	markx	transfr	base	answerc,wrongc	disable
micro	marky	zero	end	concept,miscon	dataon
char	gat,gatnm	set	restart	vocabs,vocab	area
plot	graph	randu	imain	list,endings	output
show	hbar	setperm	finish	ansv,wrongv	outputl
showa	vbar	randp	do	ansu,wrongu	readset
showe	gdraw	remove	join	exact,exactc	readd
showo	gbox	modperm	exit	touch,touchw	dataset
showt	gvector	pack,packc	goto	ok,no,ignore	datain
showz	gcircle	move	jump	ans	dataout
draw	gdot	search	jumpout	match	
box	polar	compute	eraseu	specs	
vector	delta	itoa	nextop,next1op	or	
circle	funcu	clock	backop,back1op	storea	
circleb	slide	name	helpop,help1op	storen	
dot	play	course	dataop,data1op	store	
window	ext	date	labop,lab1op	storeu	
rorigin		day	termop	judge	
rat,ratnm		find		join	
rdraw		findall		bump	
rbox		common		put,putd,putv	
rvector		comload		loada	
rcircle		storage		okword,noword	
rdot		stoload		eraseu	
		initial		getword	
		reserve		getloc	
		release		getmark	
		sort		compare	
		sorta		change	



## The TUTOR Language

### Additional TUTOR Commands Not Discussed in This Book

abort	abort normal updating of common or student record
add1	add one to a variable
allow	allow an instructional lesson to use router common
altfont	use alternate font for all writing
backnd	run lesson at lower priority
chartst	check whether charset already loaded
close	like -loada- but takes one character per variable
dataoff	turn off student response data collection
delay	timed blank output for precise display timing
exactv	character string match to student response
foregnd	run lesson at normal (non-background) priority
iferror	specify unit to go to if -calc- error
lesson	sets "ldone" to inform router about lesson completion
open	like -storea- but stores one character per variable
press	presses a key for the student
readr	read a student record for data processing
record	record a message on audio device
route	specify router units for end of instructional lessons
routvar	set up router variables
stop	like -back- but for the STOP key
sub1	subtract one from a variable
tabset	set up tabs for TAB key
time1	set a time within a lesson
timer	router sets a time for a lesson to finish
use	use sections of another lesson to prepare this lesson

# Appendix C

## List of Built-in-Calc-Functions

sin(x)	sine
cos(x)	cosine
arctan(x)	arctangent

Angles are measured in radians. For example, sin(45) means sine of 45 radians, but sin (45°) means sine of 45 degrees (0.707). The degree sign (MICRO-o) converts to radians. Similarly, arctan(1) is .785 radians, which can be converted to degrees by dividing by 1°, the number of radians in one degree; arctan(1)/1° is 45. Using the degree sign after a number is equivalent to multiplying the number by (2π/360). π (MICRO-p) is 3.14159. . . .

sqrt(x)	square root; can also be written $x^{1/2}$ or $x^{.5}$
log(x)	logarithm, base 10
ln(x)	natural logarithm, base e
exp(x)	$e^x$

abs(x)	absolute value; abs(-7) is 7
round(x)	round to nearest integer; round(8.6) is 9
int(x)	integer part; int(8.6) is 8
frac(x)	fractional part; frac(8.6) is 0.6
sign(x)	+1 if $x > 0$ , 0 if $x = 0$ , -1 if $x < 0$

=, ≠, <, >, ≤, ≥	produce logical values (true=-1, false=0)
not(x)	inverts logical values (true↔false)
x \$and\$ y	true if both x and y are true
x \$or\$ y	true if either x or y is true (or both)

x \$cls\$ y	circular left shift x by y bit positions
x \$ars\$ y	arithmetic right shift x by y bit positions
x \$mask\$ y	sets bits where both x and y have bits set
x \$union\$ y	sets bits where either x or y has bits set (or both)
x \$diff\$ y	sets bits where x and y differ (exclusive union)
bitcnt(x)	counts bits

The logical operators (=, ≠, <, >, ≤ and ≥) consider two quantities to be equal if they differ by less than one part in  $10^{11}$  (relative tolerance) or by

The TUTOR Language

an absolute difference of  $10^{-9}$ . One consequence is that all numbers within  $10^{-9}$  of zero are considered equal. Similarly, “int” and “frac” round their arguments by  $10^{-9}$  so that int(3.999999999) is 4, not 3, and frac(3.999999999) is 0, not 1. This is done because a value of 3.999999999 is usually due to roundoff errors made by the computer in attempting to calculate a result of 4.

System Variables

DISCUSSED IN THIS BOOK	NOT DISCUSSED IN THIS BOOK	
anscnt	baseu	aarea
args	capital	aarrows
clock	dataon	ahelp
formok	entire	ahelpn
jcount	error	aok
key	errtype	aokist
opcnt	extra	asno
spell	judged	aterm
station	ldone	atermn
varcnt	lscore	atime
vocab	lstatus	auno
where	mainu	
wherex	mode	
wherey	nhelpop	
	ntries	
	order	
	phrase	
	size	
	user	
	wcount	
	zreturn	

The third column consists of counters associated with the -area- command.

There are some additional system variables available for special purposes. See the on-line PLATO aids for information.

# Index

- abort- Appendix B
- absolute graphics commands 189, 190
- accent marks 10
- ACCESS key 175, 182
- active lesson 239
- add1 - Appendix B
- aids Appendix A
- algebraic and numerical judging 126
  - algebraic 128
    - judging equations 131
    - warning about  $(1/2 \times)$  132, 135
  - with scientific units 133
    - warning about  $(3+6\text{cm})$  with -storeu- 135
- allow- Appendix B
- alphanumeric information
  - storea- 104
  - showa- 53, 104
  - 10 characters per variable 105, 156, 162, 220
  - difference from numeric 105, 220
  - alphanumeric to numeric conversion 231, 235
- alternate font 175
  - unaffected by -size- and -rotate- 179
  - using -char- and -plot- 199
- altfont- Appendix B
- and ( $\$and\%$ ) logical operator 81
- And(array) 216
- Anderson, B. 4
- animations 28
  - use of iterative -do- 49
  - smooth animations 178
- ans- 154
- anscnt system variable 113
  - zeroed when judging starts and by -specs- 113
  - zeroed by judge rejudge 120
  - not changed for synonomous -concept-s 116
  - cursor moving 122
  - with -or- 251
- ansu- 135
  - warning about  $(3+6\text{cm})$  with -storeu- 135
- ansv- 126
  - wrongv- 126
  - in arithmetic drill 127
  - with opcnt 127
  - specs noops, novars 128
  - concept/vocabs similar to ansv/define 128
- algebraic judging 128

## Index

- ansv- (*Cont.*)
  - warning about (1/2×) 132
  - affected by -bump-, -put-, and judge re-  
judge 232
- ansva- Appendix B
- answer- 16
  - markup of errors in student response 17
  - with numbers 106
    - Limitations 106
    - notoler, nodiff 107
  - with phrase (Santa\*Maria) 17
  - specs 107
  - caps in tag with specs okcap 107
  - no punctuation marks in tag 108
    - punctuation ignored in student re-  
sponse 108
  - with -list- 110
  - answer- useful in limited context 111
    - see -concept- 111
  - interaction with -concept- 114
  - with negation 125
  - with blank tag 126
  - exact- compared with -answer- 136
  - conditional -answer- (-answerc-) 137
    - using -put- to find pieces of words 159
- answerc- 137
- Arabic 177, 253
- area-252
- args system variable 55
- arguments
  - passing arguments to TUTOR com-  
mands 53
  - passing arguments to subroutines 53
    - args system variable 55
    - warning to use different variables in  
different subroutines 56
  - omitted arguments 55
  - order of passing 54
  - passing arguments in conditional -do-  
79
    - passing arguments in -goto- 90
    - can be complicated expressions 55
- arc of a circle 26
- arheada- 253
- arithmetic drill 127
- arithmetic right shift \$ars\$ 224
- with negative numbers 228
- arrays 214 (also see indexed variables 204)
  - array operations 216
    - matrix multiplication (dot product),  
vector product, sum, Prod, Min,  
Max, And, Or, Rev, Transp
- arrays (*Cont.*)
  - offset arrays 217
  - vertically segmented arrays 231
- arraysegv 231
- arrow- 15, 96 (also see -arrowa- 253)
  - multiple -arrow-s in a unit 21, 98
  - displays arrowhead on screen 16, 141
  - inhibit arrow 122
  - location in unit remembered 96, 141
  - restarting at -arrow- for each response  
97, 141
  - satisfy all -arrow-s before leaving main  
unit 97, 142
  - search for additional -arrow-s 97, 99, 142
    - goto- skipped 147
  - delimits preceding -arrow- 97, 99, 142
  - changes search state to regular state 99,  
142
  - sets default long 104
  - summary of processing stages 141
    - interactions with other commands 149
    - sets default long, jkey, copy 150
  - rules for attaching units containing  
-arrow- 100, 148
  - merely collect response 164
  - sets left margin 172
  - with response erasing 192
  - enable- for touch input 250
- arrowa- 253
  - different arrowhead from -arheada- 253
  - associated -iarrowa- 253
- assignment of values in a -calc- 46
  - multiple assignments 47
  - implicitly defined 203
  - in -store-/compute- 235
  - specs okassign 235
- assignment symbol 46, 235
- asterisk for comments 20
- attached unit 40, 64, 86
  - by -do- 40
  - by -goto- 86
- attempts (counting student attempts) 119
- audio device 251
- automated display generation 35
- automatic response-associated erasing 193
- automatic scaling with graphing com-  
mands 182
  - auxiliary unit (see attached unit)
- Avner, E. Appendix A
- at- 14, 24 (also see -atnum- 172) (-gat- 183,  
-rat- 189)
  - default -at- after response 97

- at- (*Cont.*)
  - one or two arguments 24, 53, 55
  - sets left margin 171
  - atnm- does not set a margin 172
  - where system variable 173
  - wherex and wherey system variables 174
- atnm- (-at- with no margin) 172 (-gatnm- 189, -ratnm- 189)
- axes- 183 (also see -bounds- 184)
- back- 18 (also see -backup- 73)
- back1- 69 (also see -backlop- 73)
- backnd- Appendix B
- BACK and BACK1 return from help sequence 62
- backup- 73 (also see -back- 18)
  - alternative to "inhibit erase" 73
- backlop- 73 (also see -back1- 69)
  - alternative to "inhibit erase" 73
- backspace 9, 174
- base- 63
  - base pointer and base unit 63
    - q or blank to clear 64
    - automatically cleared when base unit reached 64
    - set base pointer 63, 64, 198
- base unit 19, 63
- basic TUTOR 13
- binary notation 209, 218
- bit manipulation 217
  - \$cls\$ circular left shift 223, 224
  - \$ars\$ arithmetic right shift 224
    - with negative numbers 228
  - \$mask\$ 225
    - constructing masks in octal 226
  - \$union\$ 229
  - \$diff\$ 229
- bitent function 229
  - packing data 225
  - octal numbers 226
  - complementing bits 228
  - byte manipulation 229, 230
- bitent function 229
- bounds- 184
- box- 25 (-gbox- 185, -rbox- 189)
- branch- 212 (also see -goto- 85, -doto- 213)
  - statement labels 212
  - must not have duplicate labels 212
  - cannot branch past -entry- 212
  - speed advantage compared with -goto- 212
- branching 59
  - conditional 77
  - within a unit, see -branch- 212
- broken or dashed circle -circleb- 26
- bump- 120, 156
  - combinations of -put- and -bump- 158
  - with shift characters 158
  - affects -store-/-ansv- 232
- bumpshift specs option 109
- byte manipulation 229, 230 (also see bit manipulation)
- calc special term 256
- calc- 46, 201
  - conditional -calc- (-calcc- and -calcs-) 84
    - × is not the fall-through option 84
  - summary 201
  - statement label equivalent to -calc- 212
  - with integer variables 222
  - functions Appendix C
- calcc- 84 (see -calc-)
- calcs- 84 (see -calc-)
- calculations 43
- carriage returns and left margins 171
- catchup- 32
- central memory 241
- change- 108, 136, 254
- changes in TUTOR Appendix A
- character grid
  - coarse 14
  - fine 23
- character set 176
- character strings 159
  - see -bump- and -put- for student character strings
  - see -pack-, -move-, and -search- for other strings
  - single quote marks ('dog') 160, 223
  - double quote marks ("dog") 165, 223
  - 6-bit character codes 220
    - and -calc- 221
    - and -compute- 231
- characters
  - character grid (coarse 14, fine 23)
  - character size (8×16) 34
  - 10 per variable 105, 220
  - special characters 175
- char- 199
- charset- 176, 181
- chartst- Appendix B
- charts (see graphing commands)
- Cheshire cat 40

## Index

- Chinese characters with `-rdraw-` 188
- `-circle-` 26 (`-gcircle-` 185, `-rcircle-` 189)
  - ellipses 185, 189
- `-circleb-` 26
- circular left shift `$(s)` 222, 224
- clear (force option) 253
- `-clock-` command 163
- clock system variable 163
- `-close-` Appendix B
- coarse grid 14, 23
- command 13
  - list of commands Appendix B
- comments (\*) 20, (\$) 26
- `-comload-` 243, 247
- `-common-` 237
  - temporary common 237
    - uses of temporary common 239
  - `-common-` not executed 238
  - permanent common 240
    - splitting among many students 248
    - and the swapping process 240
    - reserving common 246
- common variables 237 (also see `-common-`)
- `-compare-` 254
- compile 231
- complementing bits 228
- `-compute-` 231 (see `-itoa-` 235)
- conditional commands 77 (also see `-if-` 91)
  - condition can be complicated expression 79
  - condition rounded to nearest integer 79, 80
    - with logical expressions 80
      - more precise due to rounding 80
- consult special term 256
- continued `-write-` statement 171
- conversions
  - between octal and decimal 226
  - between alphanumeric and numeric 231, 235
- `-course-` 163
  - course registration 199
- `-concept-` 111 (see `-vocabs-` 111)
  - with numbers 113
  - with numbered vocabulary words 117
  - synonyms 111
  - with phrases (Santa\**Maria*) 116, 118
  - with endings 116, 118
  - markup of student response 113
    - missing words 113
    - misspellings 113
    - specs okextra 113
- `-concept-` (*Cont.*)
  - interaction with `-answer-` or `-wrong-` 114
  - with judge wrong 115
  - synonymous `-concept-s` 116
    - anscnt unchanged 116
  - with negation 125
  - `concept/vocabs` similar to `ansv/define` 128
- `-copy-` 150, 10
  - copy key disabled by `-arrow-` 150
  - copy compared with `edit` 150
- cross product (vector product) 216
- cursor moving routine 122
  - with `-match-` 124
  - with `-keytype-` 166
- cursor special term 256
- Curtin, C. 5
- Cyrillic characters 176
- dashed or broken circle (`-circleb-`) 26
- data from student responses 251
- `-data-` 69 (also see `-dataop-` 73)
- `-dataoff-` Appendix B
- `-dataon-` 252
- `-dataop-` 73 (also see `-data-` 69)
- data bases 237
- data files 252
- `-datain-` 248
- `-dataout-` 248
- dataset operations 248
- `-dataset-` 248
- `-data1-` 69 (also see `-data1op-` 73)
  - `-data1op-` 73 (also see `-data1-` 69)
- `-date-` 163
- Davis, C. 2
- `-day-` 163
- debugging facilitated by `-do-` 42
- decimal and octal conversions 226
- `-define-` 47, 202, 235
  - use `-define-`, avoid primitives 48
  - `-define-` must precede related `-calc-` 47
  - explicit multiplication required 48, 235
  - overriding system variable definitions 56, 235
  - student define set 103, 128, 231, 235
    - with algebraic judging 128
    - with scientific units 133
    - with indexed variables 205
    - in graft 234
  - defining functions 202
  - warning about defining `v`, `n`, `vc`, or `nc` 205

- define- (*Cont.*)
  - defining arrays 214
  - defining indexed variables 205
  - defining segmented variables 207
- delay- Appendix B
- delta- 185
- desk calculator 101
- dialog (with -concept- and -vocabs-) 111
- dictionary using -term- 71
- \$diff\$ 229
- dimensionality of scientific units in
  - storeu- 133
- disable- 250
- disk permanent storage 241, 242
- do- 40 (also see -doto- 213, -if- 91)
  - iterative 49
    - compared with conditional -goto- 85
    - caution about slowness of segmented variables 211
  - conditional 78 (also see -if- 91)
  - conditional iterative -do- 90
    - special meaning of q and x 91
  - undo when -unit- command encountered 87
  - do q like goto q 89
    - like -join-, except regular only 98, 142
  - skipped during judging and search 98
  - do-ing -arrow-s 100
  - text-insertion nature 101
    - goto- causes exception 87, 145
  - judging command prevents un-do-ing 142, 145
  - do level saved at -arrow- 149
  - nested -do-s 206
  - exit- from -do-s 236
- dollar signs for comments 26
- dot- 200
- doto- 213
- dot product (matrix multiplication) 216
- dots on screen 24
  - dot- 200
- display screen 3
- displays 23
  - automated display generation 35
- draw- coarse grid 14, fine grid 25
  - automated display generation 35
  - example with complicated expressions 52
  - from current position 174, 186
  - skip option 185
  - updating of where, wherex, wherey 185
- draw- coarse grid (*Cont.*)
  - large number of points 186
  - comparison with -gdraw- 189, 190
  - comparison with -rdraw- 189, 190
  - see -window- 190
  - erasing associated with response 195
  - making dots 200
- drills
  - arithmetic 127
  - vocabulary 137, 138, 196
- edit- 150, 9
  - edit compared with copy 150
- ellipses (-gcircle- 185, -rcircle- 189)
- else- 91
- elseif- 92
- embedded show commands in -write- statement 53
  - s for -show-, a for -showa-, t for -showt-, e for -showe-, and z for -showz- 53
  - in -writec- 84
  - in -pack- and -packc- 162
- enable- 250
- end- 19, 63, 64, 198
  - ignored in non-help sequence 64
  - end lesson 139, 255
  - no -end- with -helpop- 73
- endarrow- 21, 99, 100
  - delimits preceding -arrow- 21, 100, 142
  - changes search state to regular state 100, 142, 148
  - pause between -arrow-s 100
  - at end of unit 100
  - required if -arrow- done or joined 100, 148
- endif- 92
- endings- 116, 117, 118
- entry- 89 (also see -unit-)
  - use in vocabulary drill 197
- equality rounding
  - in logical expressions 81
- equations in algebraic judging 131
- erase- 28, 33
  - automatic full-screen erase for new main unit 22, 60
  - inhibit erase 151, 197
  - explicit -erase- 198
  - nextop- alternative to "inhibit erase" 73
- erase mode 33
  - used in erasing responses 196
- eraseu- 195



## Index

- erasing student responses 192
- exact- 136
  - handles punctuation marks 136
  - blank -answer- not blank -exact- 126
- exactc- 136 (conditional -exact-)
- exclusive union (see \$diff\$ 229)
- exit- 236
- exponential show command, -showe- 53
- exponents in floating-point numbers 219, 229
- expressions (mathematical) 43
  - usable everywhere 52
  - logical expressions 80
    - mixing logical and numerical expressions 81
  - student expressions 101, 102
- ext- 251
- external input 250
- external output 251
  
- false (in logical expressions) 80
- find- 235 (also see -search- 161)
- findall- 236
- fine grid 23
- finish- 239
- flags using segmented variables 210
- floating-point numbers 219, 229
- font 11, 175
  - force font 177
    - unaffected by -size- and -rotate- 179
    - using -char- and -plot- 199
- force- 104
  - force clear 254
  - force long 104
  - force font 177
  - force left 177, 253
  - force micro 253
- foregnd- Appendix B
- foreign languages 137, 196, 252
- formok system variable 102, 129, 134, 233
- funct- 185 (also see 233)
- function keys 9
- functions 48, 202, Appendix C
  - parentheses around function arguments 48, 102
  - dimensionless arguments for -storeu- 134
  - defining your own functions 202
  - int (integer part) 204, 222
  - sin (sine) 48
  - sqrt (square root) 52
  - modulo 204
  - bitcnt (bit count) 229
- functions (*Cont.*)
  - plotting functions 185, 233
  - array functions 216
  
- gat- 183 (also see -at-)
- gbox- 185
- gcircle- 185
- gdraw- 183 (also see -draw- and -rdraw-)
  - comparison with -draw- 189, 190
  - comparison with -rdraw- 189, 190
- getloc- 254
- getmark- 254
- getword- 254 (also see -storen- 125)
- Ghesquiere, J. 2
- gorigin- 183
  - comparison with -rorigin- 189
- goto- 85 (also see -branch- 212, -doto- 213)
  - mild form of -jump- 85
  - cut off a unit 85
  - does not change main unit 85
  - relation to -do- 85, 86, 87
  - exception to text-insertion nature of -do- 87, 146
  - summary of basic properties 88
  - goto q 88, 139
  - with -entry- 89
  - compared with iterative -do- 90
  - passing arguments with -goto- 90
  - a regular command 98, 146
  - skipped during judging and search 98
  - must not use in attached -arrow- unit 100, 148
- graft language 234
- graph- 183
- graphics 23
  - automated display generation 35
  - comparison of absolute, relative, and graphing 189, 190
- graphing commands 182 -gorigin-, -axes-, -bounds-, -scalex-, -scaley-, -labelx-, -labeley-, -lscalex-, -lscaley-, -markx-, -marky-, -gat-, -gatnm-, -graph-, -hbar-, -vbar-, -gdraw-, -gbox-, -gvector-, -gdot-, -polar-, -delta-, -funct-
- grid 23
- group- 167 (see -keytype- 166)
  - with touch panel 168
- gvector- 184
  
- halfcirc subroutine example 46
- hbar- 183 (also see -vbar- 183)

- Hebrew 177, 253
- help- 18, 62 (also see -helpop- 72)
  - later -help- overrides earlier help 21
- helpop- 72
  - return to waiting point, not start of unit 73
  - no -end- command 73
- help1- 69
- helpop- 73
- help sequence 62 (also see -helpop- 72)
  - help sequence is a slow subroutine 66
  - return is to beginning of base unit 66
  - converting between help and non-help sequences 64
  - use of -jkey- to give help 152
  - importance of enabling HELP key 154
  - with inhibit erase 198
- iarrow- 75, 155, 253
- iarrowa- 253
- ieu (see initial entry unit)
- if- 91
- iferror- Appendix B
- ignore- judging command 122
- imain- 73
- inactive lesson 239, 241
- indenting with -if-/-else- 92
- index for students to use
  - with -term- 70
  - with -imain- 73
  - setting and clearing -imain- 74
  - with -store-/-ok- 103
  - with -match- 125
  - with -ansv- 126
- indexed variables 204
  - with -storeu- dimensionality 133
  - warning about defining v, n, vc, or nc 205
- indexed common variables 238
- inhibit-
  - inhibit arrow 122
  - inhibit erase 151, 197
    - interaction with -restart- 199
    - nextop- alternative to "inhibit erase" 73
- initial entry unit (ieu) 177
  - with compute pointers 234
  - relation to -restart- 178, 199
- initializations
  - general questions of initialization 66, 67
  - unit pointers cleared when new main unit entered 70
  - use of -imain- 74
- initializations (*Cont.*)
  - zeroing variables 207
  - zeroing compute pointers 232
    - in ieu 234
  - window- not initialized by main unit 191
  - size- and -rotate- not initialized by main unit 190
  - with -restart- 199
    - initializing variables 199
  - zeroing temporary common 238
  - zeroing -storage- 247
- insertion of subroutine (by -do-) 40
- instructor mode 255
- int function for integer part 204, 222
- integer variables 221
  - common integer variables 238
- interactions of -arrow- with other commands 149
- Introduction to TUTOR, Chesquiere, Davis, Thompson 2
- iterative -do- 49, 67
- itoa- 235
- jcount system variable 105
  - affected by specs bumpshift 109
  - and -bump- 156
  - and -put- 157, 159
- jkey- 150, 151
  - default set by -arrow- 150
  - with response erasing 192
- join- 98 (also see -do-)
  - universally executed (regular, judging, search) 98, 142
  - like -do- except universal 98, 142, 144, 155
  - join-ing -arrow-s 99
  - text-insertion nature 101, 145
    - goto- causes exception 87, 145
  - judging command prevents un-do-ing 142, 145
  - repeated execution in regular, judging, search states 142
- judge- 115, 118
  - judge- is a regular command 115, 118
  - judge wrong used to stay at -arrow- 115, 197
    - does not stop processing 119
    - judge noquit does stop processing 119
    - in student data 252
  - judge ok 118
    - does not stop processing 119

## Index

- judge- (*Cont.*)
  - judge okquit does stop processing 119
  - judge continue 119, 153
    - in algebraic judging 131
  - judge rejudge 120, 156
    - affects -store-/-ansv- 232
  - judge ignore 121
    - stops processing 121
  - judge exit 123
  - judge no 123
    - does not stop processing 119
    - judge noquit does stop processing 119
    - in student data 252
  - judge quit, okquit, noquit 123
    - conditional form of -judge- 118
- judging commands 95
  - (see -arrow-, -answer-, -wrong-, -answerc-, -wrongc-, -concept-, -miscon-, -match-, -ansv-, -wrongv-, -ansu-, -wrongu-, -store-, -storea-, -storen-, -exact-, -exactc-, -ignore-, -ans-, -bump-, -put-, -putd-, -specs-, -endarrow-)
  - summary 139
  - stop processing in regular state 97, 144
    - may terminate judging state 97
    - ok and no judgments 97
      - default no 97
    - require an -arrow- command 96
    - skipped in search state 97, 98
    - delimit regular commands 98, 142
    - accessed by -join- 98
    - switching from regular to judging state 119
  - judging copy of student response 120
    - affected by -bump- 156
  - judging keys 150 (see -jkey-)
  - judging student responses 95
  - jump- 68
    - initializations 68
      - base pointer not affected 68
      - cancels previous -do-s 68
      - screen erased 68
    - used with -base- to initiate help sequence 68
    - compared with -goto- 85
  - jumpout- 254
  - key system variable 152, 165
    - key names 152, 165
    - catching every key 164
    - key codes 165
    - timeup 166
      - with touch and external input 250
  - keyset or keyboard 8
  - keytype- 166 (see -group- 167)
    - with touch panel 168
  - keyword judging 123
  - kstop- Appendix B
  - lab- 69 (also see -labop- 73)
  - lab1- 69 (also see -lab1op- 73)
  - labeling graphs 183
  - labels on statements for -branch- 212, for -doto- 213
    - must not have duplicate labels 212
  - labelx- 183 (see -markx- 184)
  - labeledy- 183 (see -marky- 184)
  - labop- 73 (also see -lab- 69)
  - lab1op- 73 (also see -lab1- 69)
  - languages 137, 196, 252
  - large-size writing 26
  - left shift (see circular left shift 222, 224)
  - leftward writing 177, 252
  - lesson samples 4-6
  - lesson space 181, 240
  - lesson not swapped 244
  - levels of -do- (10 permitted) 41
  - line drawings (see -draw-)
  - line-drawn characters (see -size- and -rotate-) 179, 188
  - lineset- 179, 181, 188
  - list- 110
    - in -answer- and -wrong- 110
  - loada- 159, 160
  - locking common 246
  - logical expressions 80
    - in conditional commands 80
    - mixed with numerical expressions 81
    - logical operators =, ≠, <, >, ≤, ≥ 80
      - roundoff on equality 81
    - logical operators \$and\$, \$or\$, (not) 81, 82
  - long- 103
    - force long 104, 150
    - follows -arrow-, precedes judging commands 104
    - modifies -arrow- 104

- long- (*Cont.*)
  - must precede -specs- 107
  - long 1 with judge ignore 122
  - default set by -arrow- 150
  - edit- for long greater than 150 characters 150
- lscalex- 184 (see -scalex- 183)
- lscaley- 184 (see -scaley- 183)
- lscore (associated with -score-) 255
  
- main unit 59, 64, 85
  - not affected by -goto- 85
- margin set by -at- and -arrow- 171
- marker
  - arrow- marker 96, 97
  - specs- marker 109, 114
- markup of response 97
- markx- 184 (see -labelx- 183)
- marky- 184 (see -labeledy- 183)
- masking in bit manipulations (\$mask\$) 225
- match- 123
  - also see -storen- 125
  - in graft language 234
- mathematical expressions 43
- matrix multiplication 216
- matrix operations 214 (also see arrays)
- Max(array) 216
- merge (see \$union\$ 229)
- micro- 181
  - force micro 253
- microfiche 249
- micro-key options 10
- micro table 181
- Min(array) 216
- mode- (erase, write, rewrite) 33, 174, 179
  - conditional form 85
- modperm- 138 (also see permutations)
- modulo function 204
- move- 160
- multiple -arrow-s 21, 99
- multiplication
  - explicit between defined names 48 (except for students 103)
  - takes precedence over division 44
- music 251
  
- name- 163
- naming variables (-define-) 47
- nc1-nc1500 common variables 243
  
- negative words 110, 125
- next- 18, 59 (also see -nextop- 73)
  - put near beginning of unit 61
  - successive -next- commands override 61
  - “next ” or “next q” to clear pointer 61
- NEXT key 9, 60
  - always a judging key 150
  - ignoring extra NEXT keys 155
- next physical unit 60
- next1- 69, 70 (also see -nextlop- 73)
- nextnow- 18, 20
- nextop- 73 (also see -next- 59)
  - alternative to “inhibit erase” 73
- nextlop- 73 (also see -next1- 69)
  - alternative to “inhibit erase” 73
- no- 103, 123
  - in arithmetic drill 127
- nodiff specs option 107
- non-help sequence 64
  - converting between help and non-help sequences 64
- non-numerical parameters specified by student 104
- nookno specs option 108
- noops specs option 128
- noorder specs option 18, 108, 116
- noquit (judge option) 119, 123
- not (logical function) 82
- notes Appendix A
- notoler specs option 107
- novars specs option 128
- noword- 197, 253
- nr1-nr50 router variables 255
- numbering vocabulary words 117
- numeric information different from alpha-numeric 105
  - range of numerical values 217
- numerical parameters specified by student 101, 126
  - checking for negative 119
- numerical and algebraic judging 126
  - algebraic 128
- n1-n150 student variables 221
  
- octal numbers for masks 226
- octal show command, -showo- 53, 227
- offset arrays 217
- ok- 101, 119, 123
- okassign specs option 235

## Index

- okcap specs option 107
- okextra specs option 18, 108, 113
- okquit (judge option) 119 123
- okspell specs option 107, 116
- okword- 197, 253
- opcnt system variable 128, 129
- open- Appendix B
- operations (see precedence)
- optional words
  - in -answer-/wrong- 16
  - in -vocabs- 111
- or- judging command 251
- or (\$or\$) logical operator 81
- Or(array) 216
- output- 252
- outputl- 252
  
- pack- 162
- packc- 162
- parentheses around function arguments
  - 48, 102
- partial circle 26
- passing arguments 53 (see arguments)
- pause- 28, 164
  - between -arrow-s, with -endarrow- 100
  - catching every key 164
  - no key display 167
  - no help at blank -pause- 167
  - pause keys=a,b,etc. 168
    - help, term, etc. possible 168
    - NEXT key special 168
  - with touch panel 168
- permanent common 240 (also see -common-)
- permanent storage area 240
- permutations 138
  - randp- 138
  - setperm- 138
  - remove- 139
  - modperm- 138
  - vocabulary drill 137
- Persian 177, 253
- photographic projection 249
- phrase (such as Santa\*Maria) 17, 116, 118
- physical next unit 60
- place notation 224
- plasma display panel 3
- play- 251
- plot- 199
- plotting functions 233 (also see -funct- 185)
  
- pointers (next, help, base, etc.) 60
  - q or blank to clear pointer 61, 65
  - successive commands override earlier settings 61, 65
  - cleared when new main unit entered 70
- compute pointer 232
  - zeroing in ieu 234
- pointing at touch panel 168, 250
- polar- 184
- positioning 23
- powers in floating-point numbers 219, 229
- precedence (of mathematical operations)
  - 44, 132
- preparing lesson for active use 239
- press- Appendix B
- primitive variable names (v1-v150) 44, 48, 235
- Prod(array) 216
- punctuation in responses 108, 126, 136, 254
- put- 120, 157
  - affects jcount 159
  - terminates judging if string too long 157
  - combinations of -put- and -bump- 158
  - affects -store-/ansv- 232
- putd- 158 (also see -put-)
- putv- 158 (also see -put-)
  
- q (special unit name) 61, 65
  - clears unit pointers 70, 79
  - goto q 88, 139
  - in conditional iterative -do- 91
- quit (judge option) 123
- quote marks for character strings
  - single ("dog") 160, 223
  - double ("dog") 165, 223
  
- random numbers (see -randu- and permutations)
- randp- 138 (also see permutations)
- randu- 82
  - arithmetic drill 137
  - algebraic judging 128, 129
  - compared with -randp- 138
- range of numerical values 217
- rat- 189
- ratnm- 189
- rcircle- 189
- rdraw- 187
  - affected by -size- and -rotate- 188
  - compared with -gdraw- 189

- readability with subroutines 40
- readd- 252
- readset- Appendix B
- records in datasets 248
- record- Appendix B
- registration records 199, 242
  - storage- not saved 247
- regular commands 96
  - skipped in judging state 96, 120, 141, 146
  - skipped in search state 97, 98, 142
  - do- and -goto- are regular commands 98
  - switching from regular to judging state 119
  - judging command stops and prevents un-do-ing 142, 144
- relative graphics commands 189, 190
- release- 246, 248
- remove- 139 (also see permutations)
- reserve- (common 246, dataset 248)
- reserving common 246
- reserving dataset records 248
- responses (see judging)
- response data 251
- restart- 199 (also see initial entry unit 177)
  - storage- not saved 247
- restarting a lesson 178 (-restart- command 199)
- resume (in -jumpout-) 254
- return from help sequence 63, 66
- Rev(array) 216
- rewrite mode 34, 174, 179
- right shift (see arithmetic right shift 224, 228)
- rotate- 26
  - interaction with -arrow- 149
  - affects -writec- 84
  - does not affect alternate font 179
  - affects -rdraw- even in size zero 188
  - not initialized by main unit 190
- rorigin- 187
  - compared with -gorigin- 189
- rounding
  - of condition in conditional commands 79, 80
  - in equality operation 81
  - in indexed variables 205
  - in segmented variables 211
  - with integer variables 222
- route- Appendix B
- routers 254, 255
- router variables (vr1-vr50) 255
- routvar- Appendix B
- Russian alphabet 176
- rvector- 189
- scalex-/scaley- 183 (also see -lscalex-/lscaley- 184)
  - comparison with -size- 189
- scaling in graphing commands 182
- scientific units 133 (see -ansu-)
- score- 255
- search- (character string command) 161
- search state (looking for additional -arrow-s) 97, 142
  - skips regular and judging commands 97
- segmented variables 207, 230
  - table of ranges and space 209
  - signed segments 208
  - fractional numbers 210
  - slowness 211
  - equivalent bit manipulations 225
  - byte manipulations 229
  - vertical segments 230
- segmentv 230
- selective erase (text 28, graphics 33)
- sequencing 59
  - summary of sequencing commands 69
  - author-controlled and student-controlled 70
  - within a unit, see -branch- 212
- set- (fill array elements) 217
- setperm- 138 (also see permutations)
- Sherwood, B. 5
- Sherwood, J. 7
- shift character 104, 156, 157, 158, 162
- shift operators (\$cls\$ 222, 224) (\$ars\$ 224, 228)
- skip in -draw- 185
- show- 51
  - significant figures 52
  - showa- (alphanumeric) 53, 105
  - default length 105
  - uses 6-bit character codes 220
  - ignores null characters 222
  - with v or n variables 222
  - showe- (exponential) 53
  - showt- (tabular) 53
  - showo- (octal) 53, 228
  - showz- (show trailing zeroes) 53
  - automatic erasing 194

## Index

- sign-in/sign-out 199, 242
- simulation of judging and search 98
- sin (sine function) 48
- size- 26
  - interaction with -arrow- 149
  - affects -writec- 84
  - does not affect alternate font 179
  - affects -rdraw- 188
  - comparison with -scalex- 189
  - not initialized by main unit 190
- skipping over main units 59
- slide- 249
- Smith, S. 4, 111
- smooth animations 178
- sort- 248
- sorta- 248
- sorting lists 248
- special characters 175
- specifying parameters
  - numerical
    - store- 101
    - with -show- 106
  - non-numerical
    - storea- 104
    - with -showa- 106
- specs- 17, 18, 107
  - notoler, nodiff 107
  - bumpshift 109
  - okcap 107
  - okspell 107
    - with -concept- 116
  - okextra 18, 108, 113
  - noorder 18, 108
    - with -concept- 116
  - nookno 108, 115
  - noops, novars 128
  - okassign 235
- specs- is a judging command 107
- specs- sets a marker 109, 141
  - later -specs- overrides earlier marker 109
  - clears ansent 114
- speech 251
- spell system variable 109
- spelling and -compare- 254
- square root function, sqrt(expression) 52
- statement has command and tag 13
- statement label with -branch- 212, with -doto- 213
  - must not have duplicate labels 212
- status bank 242
- step- command 256
- step special term 255
- stoload- 247
- storage- 246 (also see -common-)
  - not saved on sign-out 247
  - zeroed on sign-in 247
- store- 101
  - a judging command 102
  - judges no if cannot evaluate 102
  - with -show- 106
  - compared with -storen- 125
  - with -ansv- 126
  - concept/vocabs similar to ansv/define 128
  - warning about (1/2×) 132
  - affected by -bump-, -put-, and judge rejudge 232
  - no primitive variable names 235
  - no assignments without specs okassign 235
- store values into variables 44
- storea- 104
  - with -showa- 106
  - with character string manipulations 159
  - opposite of -loada- 159
  - compare with -pack- 162
  - merely collect response 164
  - uses 6-bit character codes 220
  - with v or n variables 222
- storen- 126
  - also see -match- 123 and -store- 101 and -getword- 254
- storeu- 133
  - terminates judging if error 134
  - warning about (3+6cm) with -storeu- 135
- strings 159 (see character strings)
- student define set 103 (also see -define-)
- student responses 95
  - storing responses (see specifying parameters)
  - judging responses (see judging commands)
  - student response data 251
- student specification of parameters (see specifying parameters)
- student variables (v1-v150) 44
  - in displays 45
  - compared with common variables 238
  - augment with -storage- 246
- subl- Appendix B
- Sum(array) 216
- superimposing writing 34, 174

- superscripts and subscripts 10, 174
- system variable 55
  - ansent 113
  - args 55
  - clock 163
  - formok 102, 129, 134, 233
  - jcoun 105
    - affected by specs bumpshift 109
    - and -bump- 156
    - and -put- 159
  - key 152, 165
  - opent 128, 129
  - spell 109
  - varcnt 129, 132
  - vocab 115
  - where 173
    - updating in -draw- 185
  - wherex 174
  - wherey 174
- subroutines 39
- superscripts and subscripts 174
- swapping process 240, 243
  - swapping memory 241
  - and common variables 243
- synonyms
  - in -answer- 16, 95 (also see -list- 110 )
  - in -concept- 113 (also see -vocabs- 111)
  - in numbered vocabulary words 118
  
- table of square roots 52
- tabset- Appendix B
- tabular show command, -showt- 53
- tag 13
- talk special term 256
- temporary common 238 (also see -common-)
- Tenczar, P. 6
- term- 70 (also see -termop- 72)
  - complementary to -help- 71
  - dictionary use 71
  - duplicate terms an error 71
  - synonyms 72
  - step, cursor, consult, talk, calc 255
- terminal capabilities 3, 249
- termop- 72
- text (see -write-, -size-, -rotate-)
- text insertion of subroutine (by -do-) 40
  - arrow- in subroutine 100, 148
- Thompson, C. 2
- tick marks on graphs 184
- time- 31
  
- time-slice 245, 246
- timeup key 166
- tolerance
  - with -answer-/-wrong- 107
  - with -ansv-/-wrongv- 126
  - with -ansu-/-wrongu- 135
  - on equality operations 81
- touch- 251 (also see 168)
- touch panel 168, 250
- transfr- 207
  - not with segmented variables 208
  - with -common- or -storage- 247
- Transp(array) 216
- tries (counting student attempts) 119
- true (in logical expressions) 80
  
- unconditional commands 79 (also see conditional commands)
- \$union\$ 229 (also see \$diff\$ exclusive union 229)
- unit- 14
  - terminates preceding unit 87
  - see -entry- (which does not terminate) 89
  - must not have duplicate -unit- names 212
- unit pointers (see pointers) 60
- units (scientific units) 133
- universal execution of -join- 98, 142
- use- Appendix B
  
- varcnt system variable 129
- variables
  - student variables 44
    - with -restart- 199
    - with -storage- 246
  - indexed variables 204
    - with -storeu- dimensionality 133
  - common variables 237
  - segmented variables 207
  - range of numeric values 217
  - router variables 255
- vbar- 183 (also see -hbar- 183)
- vector- 25 (-gvector- 184, -rvector- 189)
- vertical segments 230
- vocab system variable 115
- vocab- 116
- vocabs- 111, 252 (see -concept- 111)
  - numbering vocabulary words 117
- vocabulary drill 137
- vc1-vc1500 common variables 243
- vr1-vr50 router variables 255



## Index

- v1-v150 student variables 44
- where system variable 173
  - updating in `-draw-` 185
- wherex system variable 174
- wherey system variable 174
- `-window-` 190
- `-write-` coarse grid 14, fine grid 24
  - with embedded show commands 53
    - s for `-show-`, a for `-showa-`,
    - t for `-showt-`, e for `-showe-`, and z for `-showz-` 53
  - conditional `-write-` (`-writec-`) 82
  - with left margins 171
  - continued `-write-` statement 171
  - successive `-write-` statements 172
  - also see `-size-` and `-rotate-`
  - size 1 versus size 0 188
  - automatic erasing 192
  - alternate font
    - with `charset` 176
    - using `-char-` and `-plot-` 199
- write mode 34
- `-writec-` 82 (also see `-write-`)
  - x is not the fall-through option 83
  - special character when using commas 83
  - with embedded show commands 84
  - affected by `-size-` and `-rotate-` 84
  - automatic erasing 192
- `-wrong-` 16 (also see `-answer-`)
- `-wrongu-` 135 (also see `-ansu-`)
- `-wrongv-` 126 (also see `-ansv-`)
  - with scientific units 135
- x (special unit name) 62, 79
- `-zero-` 207
  - not with segmented variables 208
- `$$` (permits comments to follow tag) 26
- `<>` embedding `-show-` in `-write-` 53